*Prefazione
Perl Moderno è uno dei modi per descrivere come lavorano i migliori programmatori Perl 5: usano gli idiomi del linguaggio; sfruttano CPAN; hanno un senso estetico e un'arte che gli permettono di scrivere codice potente, manutenibile, scalabile, conciso ed efficace. Anche voi potete imparare a lavorare così!
Perl ha fatto la sua comparsa nel 1987 come semplice strumento per gli amministratori di sistema. Sebbene agli inizi si proponesse di occupare una nicchia tra lo scripting shell e la programmmazione C, col tempo è diventato un potente linguaggio general-purpose. Perl 5 ha una solida storia di pragmatismo e un brillante futuro di ulteriori raffinamenti e migliorie (footnote: Perl 6 è una reinvenzione della programmazione basata sui solidi principi base del Perl, ma è un argomento per un altro libro.).
Durante la ormai lunga storia di Perl--specialmente i 17 anni di Perl 5--la nostra percezione di che cosa sia l'eccellenza per un programma Perl è cambiata. Se da una parte potete scrivere programmi efficaci senza sfruttare tutto ciò che il linguaggio vi offre, dall'altra la comunità Perl mondiale ha inventato, preso a prestito, migliorato e raffinato molte idee e le ha rese disponibili a chiunque le voglia conoscere.
Usare Perl Moderno
Il modulo Modern::Perl, disponibile su CPAN (CPAN(cpan)), richiede a Perl di segnalare l'uso di costrutti ambigui e errori di battitura, e attiva nuove funzionalità introdotte nelle versioni recenti di Perl 5. A meno di indicazioni contrarie, i segmenti di codice presentati suppongono sempre un programma basato sul seguente schema basilare:
#!/usr/bin/env perl
use Modern::Perl 2011;
use autodie;
...che equivale a:
#!/usr/bin/env perl
use 5.012; # implica "use strict;"
use warnings;
use autodie;
Alcuni esempi utilizzano funzioni di test come ok(), like() e is() (Testing(testing)). Lo schema di questi programmi è il seguente:
#!/usr/bin/env perl
use Modern::Perl;
use Test::More;
# il codice dell'esempio va qui
done_testing();
Al momento della scrittura di questo libro, la famiglia di versioni stabili di Perl 5 è Perl 5.14. Gli esempi in questo libro funzionano al meglio con Perl 5.12.0 o successivi. Molti esempi funzionano, con piccole modifiche, anche con versioni di Perl 5 più datate, ma incontrerete difficoltà maggiori con le versioni precedenti alla 5.10.0.
Se non avete un'installazione di Perl 5 (o ne avete una vecchia versione), potete installare voi stessi una nuova release. Gli utenti Windows possono scaricare Strawberry Perl da http://www.strawberryperl.com/ oppure ActivePerl da http://www.activestate.com/activeperl. Gli utenti di altri sistemi operativi che hanno già Perl 5 installato (insieme a un compilatore C e agli altri tool di sviluppo), possono iniziare con l'installare il modulo CPAN App::perlbrew (footnote: Si veda http://search.cpan.org/perldoc?App::perlbrew per le istruzioni di installazione.).
perlbrew vi consente di installare e gestire più versioni di Perl 5. In questo modo, potete passare da una versione di Perl 5 all'altra e, inoltre, potete installare Perl 5 e i moduli CPAN nella vostra home directory senza interferire con la versione di Perl di sistema. Se avete mai dovuto pregare il vostro amministratore di sistema per avere il permesso di installare del software, vi dovrebbe essere chiaro come questo vi possa semplificare, e di molto, la vita.

La Filosofia Perl
Perl risolve problemi--è flessibile, accomodante e malleabile. Programmatori in gamba lo usano ogni giorno per i compiti più diversi, dai programmi di automazione ad hoc lunghi una riga ai progetti che coinvolgono molti programmatori e durano anni.
Perl è pragmatico, e il controllo è in mano al programmatore. Siete voi a scegliere come risolvere un problema, e Perl asseconderà le vostre intenzioni senza darvi fastidio e senza tante cerimonie.
Perl è un linguaggio che cresce di pari passo con le vostra conoscenze. Nel giro di un'ora imparerete abbastanza per poter scrivere dei programmi utili a risolvere problemi reali--e imparerete come funziona il linguaggio e perché funziona in quel modo. A partire da questa conoscenza e dalla esperienza della comunità Perl mondiale, Perl Moderno ci aiuterà a scrivere codice che funziona ed è facilmente manutenibile.
Prima di tutto, avete bisogno di sapere come imparare di più.
Perldoc
Una buona documentazione fa parte della cultura Perl. L'utility perldoc fa parte di qualunque installazione completa di Perl 5 (footnote: Il vostro sistema Unix-like potrebbe richiedere l'installazione di un pacchetto aggiuntivo come perl-doc su Debian o Ubuntu GNU/Linux.). perldoc visualizza la documentazione di qualsiasi modulo Perl installato sul sistema--sia che sia integrato nella distribuzione sia che sia stato installato dal Comprehensive Perl Archive Network (CPAN)--oltre alle migliaia di pagine della ampia documentazione di base di Perl.
Le versioni recenti della documentazione Perl sono disponibili su http://perldoc.perl.org/. Gli indici http://search.cpan.org/ e http://metacpan.org/ forniscono la documentazione di tutti i moduli CPAN. Altre distribuzioni di Perl 5, come ActivePerl e Strawberry Perl, forniscono documentazione in formato HTML sul sistema locale.
Usate perldoc per leggere la documentazione di un modulo o una parte della documentazione di base:
$ perldoc List::Util
$ perldoc perltoc
$ perldoc Moose::Manual
Il primo esempio mostra la documentazione presente all'interno nel modulo List::Util. Il secondo esempio mostra un file di sola documentazione, in questo caso l'indice della documentazione di base. Il terzo esempio mostra un file di sola documentazione incluso come parte di una distribuzione su CPAN (Moose(moose)). perldoc permette di ignorare questi dettagli; non c'è infatti differenza tra leggere la documentazione di una libreria integrata, come Data::Dumper, o di una installata da CPAN.
La struttura standard per la documentazione include una descrizione del modulo, mostra dei semplici esempi di uso, e prosegue con una spiegazione dettagliata del modulo e della sua interfaccia. Mentre il livello di dettaglio della documentazione varia da un autore all'altro, il suo formato presenta una notevole uniformità.
perldoc perltoc mostra l'indice della documentazione di base, e perldoc perlfaq mostra l'indice delle domande frequenti (Frequently Asked Questions) su Perl 5. perldoc perlop e perldoc perlsyn documentano gli operatori e i costrutti sintattici di Perl. perldoc perldiag spiega il significato dei messaggi di avvertenza. perldoc perlvar elenca tutte le variabili simboliche. Dare un'occhiata a questi file è utilissimo per farvi un'idea generale di Perl 5.
L'utility perldoc ha molte altre funzionalità (vedete perldoc perldoc). L'opzione -q cerca i termini che le vengono forniti nelle FAQ del Perl. Per esempio, perldoc -q sort restituisce tre domande frequenti: How do I sort an array by (anything)?, How do I sort a hash (optionally by value instead of key)? e How can I always keep my hash sorted?.
L'opzione -f mostra la documentazione di una funzione predefinita di Perl, ad esempio perldoc -f sort spiega il funzionamento dell'operatore sort. Se non sapete il nome della funzione che vi serve, scorrete la lista di funzioni predefinite in perldoc perlfunc.
L'opzione -v cerca la documentazione su una variabile predefinita. Per esempio, perldoc -v $PID mostra la documentazione sulla variabile che contiene l'id del processo del programma in esecuzione. A seconda della shell che usate, potreste dover quotare opportunamente il nome della variabile.
Con l'opzione -l, perldoc mostra il percorso del file di documentazione anziché il suo contenuto (footnote: Tenete presente che un modulo potrebbe avere un file .pod separato dal file .pm.).
L'opzione -m mostra l'intero contenuto di un modulo, codice incluso, senza effettuare alcuna formattazione particolare.
Il sistema di documentazione di Perl 5 è POD (Plain Old Documentation). perldoc perlpod descrive come funziona POD. Altri tool relativi a POD sono podchecker, che controlla la validità di un documento POD, e Pod::Webserver, che mostra i POD locali in formato HTML con un web server minimale.
Espressività
Gli studi di linguistica e lingue umane di Larry Wall hanno molto influenzato la progettazione di Perl. Come conseguenza, il linguaggio vi dà una enorme libertà nel modo di risolvere i problemi, a seconda dello stile del vostro gruppo di lavoro, del tempo a vostra disposizione, quanto a lungo il programma dovrà continuare a funzionare, e persino di quanto vi sentite creativi. Potete scrivere un pezzo di codice semplice e lineare oppure partecipare alla scrittura di programmi di grandi dimensioni, basati su un progetto ben definito. Potete scegliere tra diversi paradigmi di programmazione, e decidere di evitare o di adottare le funzionalità più avanzate.
Laddove altri linguaggi impongono un singolo modo "corretto" di scrivere ciascun pezzo di codice, Perl lascia a voi decidere che cosa è più leggibile, efficace o semplicemente divertente.
I programmatori Perl hanno uno slogan per questo: TIMTOWTDI, pronunciato "Tim Toady", ovvero "There's more than one way to do it!" ("C'è più di un modo per farlo!").
Questa espressività permette ai maestri del linguaggio di creare programmi superlativi, ma permette anche ai neofiti e ai programmatori poco accorti di fare pasticci. L'esperienza e il gusto estetico vi guideranno a scrivere codice eccellente. La scelta è vostra--ma prestate attenzione alla leggibilità e alla manutenibilità, specialmente per chi verrà dopo di voi.
I principianti di Perl spesso trovano che alcuni dei suoi costrutti siano poco comprensibili. Molti di questi idiomi (Idiomi(idioms)) sono estremamente potenti, anche se in modi sottili e non sempre ovvi. Può essere una buona idea evitarli fino a che non arrivate a padroneggiarli.
Imparare Perl è come imparare una nuova lingua. Imparerete alcune parole, metterete insieme delle frasi, e presto potrete sostenere semplici conversazioni. La padronanza del linguaggio si ottiene esercitandosi sia nella lettura sia nella scrittura. Non dovete conoscere ogni dettaglio di Perl per usarlo in modo produttivo, ma i principi esposti in questo capitolo sono fondamentali per la vostra crescita come programmatori.
Un altro degli obiettivi nel progettare Perl è quello di evitare di sorprendere i programmatori (Perl) esperti. Per esempio, sommare due variabili ($first_num + $second_num) è ovviamente una operazione numerica (Operatori Numerici(numeric_operators)); l'operatore di addizione tratta entrambe le variabili come valori numerici e produce un risultato numerico. Indipendentemente da quale sia il contenuto di $first_num e $second_num, Perl li converte in valori numerici (Coercizione Numerica(numeric_coercion)). In altre parole, se avete espresso l'intenzione di trattarli come numeri usando un operatore numerico Perl sarà felice di accontentarvi.
Gli adepti di Perl spesso chiamano questo principo DWIM, ovvero do what I mean (fai ciò che intendo). Un altro modo per dirlo è che Perl segue il principio di minima sorpresa. Data una conoscenza anche solo superficiale di Perl (e specialmente del contesto; Contesto(context_philosophy)), dovreste riuscire a comprendere lo scopo anche di espressioni Perl che non vi sono familiari. Svilupperete man mano questa capacità.
L'espressività del Perl permette anche ai principianti di scrivere programmi utili senza dover conoscere l'intero linguaggio. Il codice che ne risulta è spesso chiamato baby Perl, nel senso che quasi tutti desiderano aiutare un bambino a imparare a comunicare bene. Tutti noi iniziamo come principianti. Con la pratica e gli insegnamenti dei programmatori più esperti, inizierete a capire e ad adottare gli idiomi e le tecniche più potenti.
Per esempio, un programmatore Perl esperto potrebbe triplicare tutti i numeri di una lista con:
my @triplicati = map { $_ * 3 } @numeri;
... e un adepto potrebbe scrivere:
my @triplicati;
for my $num (@numeri)
{
push @triplicati, $num * 3;
}
... mentre un principiante potrebbe provare con:
my @triplicati;
my $cont = @numeri;
for (my $i = 0; $i < $cont; $i++)
{
$triplicati[$i] = $numeri[$i] * 3;
}
Tutti e tre gli approcci portano allo stesso risultato, ma ciascuno usa Perl in modo diverso.
L'esperienza con la programmazione in Perl vi aiuterà a focalizzarvi sul che cosa volete ottenere piuttosto che sul come ottenerlo. In ogni caso, Perl esegue di buon grado anche i programmi più semplici. Potete progettare e raffinare i vostri programmi per aumentarne la chiarezza, l'espressività, il riutilizzo e la manutenibilità, in parte o completamente. Sfruttate la flessibilità e il pragmatismo: è molto meglio raggiungere il vostro scopo adesso in modo efficace che scrivere un programma di superba purezza e bellezza concettuale l'anno prossimo.
Contesto
In un linguaggio parlato, il significato di una parola o frase può dipendere dal modo in cui viene usata; il contesto aiuta a chiarirne il senso. Per esempio, l'errato uso del plurale in "Vorrei un panini, per favore!" (footnote: Il plurale del nome è in disaccordo con la quantità.) suona subito sbagliato, proprio come l'errore di genere in "la gatto" (footnote: L'articolo è femminile, mentre il nome è maschile.). Considerate anche il pronome "che" o il nome "città" che possono essere singolari o plurali a seconda del contesto.
Il contesto in Perl è qualcosa di simile. Esso determina quanti e quali dati usare. Perl fa del suo meglio per darvi esattamente quello che chiedete--a patto che lo chiediate utilizzando il contesto appropriato.
Alcune operazioni in Perl hanno un diverso comportamento a seconda che vogliate ottenere da loro zero, uno o più di un risultato. Un costrutto Perl può fare due cose diverse se scrivete "Fai questo, ma non mi importa del risultato" piuttosto che "Fai questo, e mi aspetto più di un risultato". Altre operazioni vi permettono di specificare se intendete lavorare con dati numerici, testuali, oppure valori logici vero e falso.
Il contesto può causare confusione se tentate di scrivere o leggere del codice Perl come una sequenza di singole espressioni avulse da ciò che le circonda. Potrà succedervi di volervi dare una botta in testa dopo una lunga sessione di debugging quando finalmente scoprite che l'errore era nelle vostre assunzioni sul contesto. Se invece tenete sempre a mente il contesto, il vostro codice sarà più corretto--oltre che pulito, flessibile e conciso.
Contesti Void, Scalare e Lista
Il contesto di quantità determina quanti elementi vi aspettate da un'operazione. È qualcosa di simile alla concordanza di numero tra soggetto e verbo in italiano. Anche senza conoscere formalmente questo principio, non avete probabilmente problemi a trovare l'errore nella frase "Perl sono un linguaggio divertente". In Perl, il numero di elementi che vi aspettate determina quanti ne otterrete.
Supponete di avere una funzione (Dichiarazione di Funzioni(functions)) con nome cerca_faccende() che ordina la vostra lista di faccende di casa da sbrigare in base alla loro priorità. Il modo in cui chiamate questa funzione determina cosa essa produrrà. Potreste non avere tempo di occuparvi di nulla, e in questo caso chiamare la funzione serve solo a far finta di lavorare. Potreste avere abbastanza tempo per fare un lavoro, oppure avere molte energie in un weekend libero e desiderare di sbrigare più faccende possibile.
Se chiamate la funzione senza chiedere un valore di ritorno, la chiamata viene fatta nel contesto void:
cerca_faccende();
Se assegnate il valore restituito della funzione a un singolo elemento (Scalari(scalars)), la funzione viene valutata in contesto scalare:
my $risultato_singolo = cerca_faccende();
Se assegnate il risultato della chiamata ad un array (Array(arrays)) o a una lista, o lo utilizzate in una lista, la funzione viene valutata in contesto lista:
my @tutti_i_risultati = cerca_faccende();
my ($elemento_singolo, @altri) = cerca_faccende();
elabora_lista_di_risultati( cerca_faccende() );
Le parentesi nella seconda linea dell'esempio precedente raggruppano le due dichiarazioni di variabli (Scope Lessicale(lexical_scope)), e l'assegnamento si comporta come vi aspettate. Se si volesse ignorare @altri, potreste scrivere:
my ($elemento_singolo) = cerca_faccende();
.... nel qual caso le parentesi suggeriscono al parser di Perl 5 che intendete fare l'assegnamento in contesto lista anche se vi interessa assegnarne solo un elemento. Tutto questo non è forse così ovvio, ma ora che lo sapete, la differenza di quantità tra queste due istruzioni dovrebbe esservi evidente:
my $contesto_scalare = cerca_faccende();
my ($contesto_lista) = cerca_faccende();
Valutare una funzione o un'espressione in contesto lista--tranne che negli assegnamenti--può creare qualche confusione. Le liste propagano il contesto lista alle espressioni al loro interno, quindi entrambe queste chiamate a cerca_faccende() sono valutate in contesto lista:
elabora_lista_di_risultati( cerca_faccende() );
my %risultati =
(
operazione_poco_costosa => $risultati_poco_costosi,
operazione_costosa => cerca_faccende(), # OPS!
);
Quest'ultimo esempio sorprende spesso i programmatori principianti, dato che inizializzare un hash (Hash(hashes)) con una lista di valori impone un contesto lista alla chiamata a cerca_faccende. Per imporre esplicitamente un contesto scalare, usate l'operatore scalar:
my %risultati =
(
operazione_poco_costosa => $risultati_poco_costosi,
operazione_costosa => scalar cerca_faccende(),
);
Perché è importante il contesto? Una funzione può esaminare il contesto in cui è stata chiamata e decidere di conseguenza quanto lavoro deve fare. In contesto void, cerca_faccende() potrebbe non fare assolutamente nulla. In contesto scalare, deve trovare soltanto il compito più importante. In contesto lista, deve ordinare e restituire l'intera lista di faccende da sbrigare.
Contesti Numerico, Stringa e Booleano
L'altro contesto di Perl--il contesto di valore--determina come Perl interpreta un dato. Probabilmente avrete già notato la flessibilità del Perl nell'indovinare se un dato è un numero o una stringa, e nel convertire tra i due a seconda delle vostre intenzioni. In cambio della possibilità di non dover dichiarare (o quanto meno tener traccia) esplicitamente del tipo di un dato contenuto in una variable o restituito da una funzione, i contesti di valore di Perl forniscono importanti suggerimenti al compilatore su come gestire tali dati.
Perl converte i valori al tipo specifico appropriato (Coercizione(coercion)), in base agli operatori che usate. Per esempio, l'operatore eq determina se due stringhe contengono la stessa informazione come stringhe:
say "Errore catastrofico di crittografia!" if $alice eq $bob;
Forse vi è capitato di essere spiacevolmente sorpresi in casi in cui siete sicuri che le stringhe siano diverse, ma confrontandole risultano uguali:
my $alice = 'alice';
say "Errore catastrofico di crittografia!" if $alice == 'Bob';
L'operatore eq tratta i suoi operandi come stringhe forzando il contesto stringa su di essi. Invece, l'operatore == impone il contesto numerico. In contesto numerico, entrambe le stringhe vengono valutate come 0 (Coercizione Numerica(numeric_coercion)). Fate attenzione ad usare l'operatore appropriato al tipo di contesto desiderato.
Il contesto Booleano si ha quando usate un valore in una istruzione condizionale. Negli esempi precedenti, if valutava il risultato degli operatori eq e == in contesto booleano.
In rare occasioni, potrebbe essere necessario forzare esplicitamente un contesto se non è possibile trovare un operatore del tipo appropriato. Per forzare un contesto numerico, aggiungete zero ad una variabile. Per forzare un contesto stringa, concatenate la variable con la stringa vuota. Per forzare un contesto booleano, usate due volte l'operatore di negazione:
my $x_numerico = 0 + $x; # forza il contesto numerico
my $x_stringa = '' . $x; # forza il contesto stringa
my $x_booleano = !!$x; # forza il contesto booleano
I contesti di tipo sono più facili da identificare rispetto ai contesti di quantità. Una volta che conoscete i contesti forniti dai vari operatori (Tipi di Operatori(operator_types)), è raro che possiate sbagliarvi.
Idee Implicite
Il contesto è solo una delle abbreviazioni linguistiche in Perl. I programmatori che padroneggiano queste abbreviazioni, dando appena un'occhiata al codice, possono capire immediatamente le sue caratteristiche più importanti. Un'altra importante caratteristica linguistica è l'equivalente Perl dei pronomi.
La Variabile Scalare di Default
La variabile scalare di default (detta anche variabile argomento o topic), $_, si nota principalmente per la sua assenza: molte delle operazioni predefinite di Perl lavorano sul contenuto di $_ in assenza di una variabile esplicita. Potete usare $_ come qualunque altra variabile, ma spesso questo risulta superfluo.
Molti degli operatori scalari di Perl (inclusi chr, ord, lc, length, reverse e uc) usano la variabile scalare di default se non gli fornite un'alternativa. Per esempio, la funzione predefinita chomp rimuove una sequenza di "a capo" dalla fine del suo operando (footnote: Vedete perldoc -f chomp e $/ per maggiori dettagli sul suo comportamento.):
my $zio = "Roberto\n";
chomp $zio;
say "'$zio'";
$_ ha lo stesso ruolo in Perl dei pronomi ciò, quello o del suffisso -lo in italiano. Se non passate esplicitamente una variabile, chomp rimuove la sequenza di "a capo" dalla fine di $_. Perl capisce a che cosa vi riferite quando dite "chomp", e farà sempre "chomp" su "quel" valore. Pertanto, queste due linee di codice sono equivalenti:
chomp $_;
chomp;
Analogamente, say e print operano su $_ in assenza di altri argomenti:
print; # stampa $_ sul filehandle corrente
say; # stampa "$_\n" sul filehandle corrente
Le funzionalità Perl per le espressioni regolari (Espressioni Regolari e Matching(regex)) per default usano $_ per i match, le sostituzioni e le traslitterazioni:
$_ = 'Mi chiamo Paquito';
say if /Mi chiamo/;
s/Paquito/Paquita/;
tr/A-Z/a-z/;
say;
Le direttive di ciclo del Perl (Direttive di Ciclo(looping_directives)) per default usano $_ come variabile su cui iterare. Considerate un ciclo for che itera su una lista:
say "#$_" for 1 .. 10;
for (1 .. 10)
{
say "#$_";
}
... oppure il while:
while (<STDIN>)
{
chomp;
say scalar reverse;
}
... oppure la trasformazione di una lista con map:
my @quadrati = map { $_ * $_ } 1 .. 10;
say for @quadrati;
... o il filtraggio di una lista con grep:
say 'Colazione!'
if grep { /fette biscottate/ } @dispensa;
Come una frase in italiano diventa confusa quando usate troppi pronomi e antecedenti, anche in Perl dovete fare attenzione a mescolare usi impliciti o espliciti di $_. Un incauto uso simultaneo di $_ può fare sì che un pezzo di codice sovrascriva tacitamente un valore scritto da un altro pezzo di codice. Se scrivete una funzione che usa $_, potreste interferire con l'uso di $_ da parte del codice che chiama la funzione.
A partire dal Perl 5.10, potete dichiarare $_ come variabile lessicale (Scope Lessicale(lexical_scope)) per evitare queste interferenze:
while (<STDIN>)
{
chomp;
# ESEMPIO SBAGLIATO
my $modificato = calcola_valore( $_ );
say "Valore originale: $_";
say "Valore modificato: $modificato";
}
Se calcola_valore() o qualche altra funzione modifica $_, tale modifica persiste per l'intera iterazione del ciclo. Aggiungendo una dichiarazione my si evita di sporcare un'altra istanza di $_:
while (my $_ = <STDIN>)
{
...
}
Naturalmente, usare una variabile lessicale con un nome appropriato può essere altrettanto chiaro, se non di più:
while (my $linea = <STDIN>)
{
...
}
Usate $_ come usereste "ciò" o "quello" in un documento formale: occasionalmente, in contesti limitati e ben definiti.
Perl 5.12 ha introdotto l'operatore triplo-punto (...) come segnaposto per del codice che volete scrivere successivamente. In fase di parsing, Perl lo tratta come un'istruzione completa, ma solleva un'eccezione al vostro tentativo di eseguire del codice non implementato se tentate di eseguirlo. Vedete perldoc perlop per maggiori dettagli.
Le Variabili Array di Default
Perl fornisce anche due variabili array implicite. Anzitutto, Perl passa gli argomenti alle funzioni (Dichiarazione di Funzioni(functions)) in un array di nome @_. Le operazioni di manipolazione di array (Array(arrays)) all'interno delle funzioni per default lavorano su questo array, per cui i due seguenti frammenti di codice sono equivalenti:
sub pippo
{
my $arg = shift;
...
}
sub pippo_con_argomenti_espliciti
{
my $arg = shift @_;
...
}
Come $_ corrisponde al pronome quello, @_ corrisponde al pronome quelli. Contrariamente a $_, Perl gestisce automaticamente diverse versioni locali di @_ quando chiamate altre funzioni. Gli operatori nativi shift e pop operano su @_ se non fornite altri operandi.
Nel codice esterno alle funzioni, la variabile array di default @ARGV contiene gli argomenti passati al programma dalla linea di comando. Esternamente alle funzioni, le operazioni sugli array (incluse shift e pop) operano implicitamente su @ARGV. Non potete usare @_ se vi serve @ARGV.
L'operatore Perl <$fh> è equivalente alla funzione predefinita readline. readline $fh ha lo stesso effetto di <$fh>. A partire da Perl 5.10, la chiamata readline senza argomenti è equivalente a <>, quindi ora potete usare readline dappertutto. Per ragioni storiche, l'uso di <> è ancora il più diffuso, ma siete invitati a prendere in considerazione l'uso di readline per la sua maggiore leggibilità. Probabilmente preferite glob '*.html' a <*.html>, giusto? Ecco, l'idea è la stessa.
C'è un caso speciale che riguarda ARGV. Se leggete dal filehandle nullo <>, Perl tratterà ciascun elemento di @ARGV come se fosse il nome di un file da aprire in lettura. (Se @ARGV è vuoto, Perl leggerà da standard input). Questo uso implicito di @ARGV è utile per scrivere programmi brevi, come questo filtro a linea di comando che rovescia il suo input:
while (<>)
{
chomp;
say scalar reverse;
}
Perché quello scalar? say impone un contesto lista ai suoi operandi. reverse propaga il proprio contesto ai suoi operandi, trattandoli come lista in contesto lista e come una stringa concatenata in contesto scalare. Se il comportamento di reverse vi suona confuso, non avete torto. Perl 5 avrebbe probabilmente dovuto separare l'inversione di una stringa dall'inversione di una lista.
Se eseguite il codice sopra passando una lista di file:
$ perl rovescia_linee.pl cifrato/*.txt
... il risultato sarà un lungo testo in output. Se non passate argomenti, potete fornire il vostro standard input con una pipe da un altro programma o scrivendo direttamente sulla tastiera. Ricordatevi comunque che Perl può fare molto di più che scrivere piccoli programmi a linea di comando....
Perl e la Sua Comunità
Il più grande successo di Perl 5 è l'enorme quantità di librerie riusabili che sono state sviluppate per arricchirlo. Mentre di Perl 4 esistevano molte versioni speciali, ad esempio per connettersi a database come Oracle e Sybase, Perl 5 mette a disposizione un vero e proprio meccanismo per le estensioni. Con esso, Larry ha voluto permettere alle persone di creare e manutenere le proprie estensioni senza che questo significasse frammentare Perl in centinaia di dialetti incompatibili.
La realizzazione di questo meccanismo è stata quasi importante quanto la crescita di una comunità intorno al Perl 5. Le persone scrivono librerie. Le persone utilizzano il lavoro di altre persone per svolgere meglio il proprio. Le persone formano una comunità di cui vale la pena fare parte, e che vale la pena preservare ed espandere.
La comunità Perl è solida ed in ottima salute. Accoglie nuovi partecipanti a tutti i livelli, dai neofiti a chi può contribuire allo sviluppo del linguaggio stesso. Approfittate della conoscenza e dell'esperienza di tantissimi altri programmatori Perl per migliorare voi stessi come programmatori.
CPAN
Perl 5 è già di per se stesso un linguaggio pragmatico, ma la sua comunità, ancora più pragmatica, lo ha notevolmente esteso, rendendo poi disponibile a tutti i risultati di questo lavoro. Se avete un problema, è probabile che qualcuno abbia già scritto--e condiviso--del codice Perl che lo risolve.
La programmazione in Perl Moderno fa un uso estensivo di CPAN (http://www.cpan.org/). La Comprehensive Perl Archive Network è un sistema di uploading e mirroring di codice Perl redistribuibile e riusabile. È certamente uno dei più grandi archivi di librerie di codice del mondo, se non il più grande. CPAN offre librerie per qualunque cosa: dall'accesso ai database agli strumenti di profiling, dai protocolli per quasi ogni apparecchiatura di rete che sia mai stata prodotta alle librerie per la grafica e il suono, ai wrapper Perl di librerie esterne disponibili per il vostro sistema.
Perl Moderno senza CPAN è solo un linguaggio. Perl Moderno con CPAN è un linguaggio incredibile.
CPAN contiene distribuzioni, ovvero collezioni di codice Perl riusabile. Una distribuzione può contenere una o più librerie autosufficienti di codice Perl dette moduli. Ogni distribuzione ha un proprio namespace ed è corredata da metadati.
Centinaia di collaboratori registrati e migliaia di moduli indicizzati in centinaia di distribuzioni vengono aggiunti a CPAN ogni mese, senza considerare gli aggiornamenti. Alla fine di Novembre 2011, il sito search.cpan.org riportava 9359 collaboratori, 101656 moduli e 23808 distribuzioni (con incrementi dell'11.5%, 19.4% e 14.3% rispetto agli stessi dati nell'edizione precedente di questo libro).
Il sistema CPAN di per sé è soltanto un servizio di mirroring. Gli autori fanno l'upload delle distribuzioni e CPAN le invia ai mirror, da cui gli utenti e i programmi client possono scaricarli e configurarli, farne il build, testarli e installarli. Il successo di questo sistema è dovuto sia alla sua semplicità che al contributo delle migliaia di volontari che hanno costruito nuovi strumenti su queste basi. In particolare, si sono sempre più affermati nella comunità certi attributi e caratteristiche standard che una distribuzione CPAN deve avere. Tra questi:
Altri servizi di CPAN sono quelli di testing automatico e generazione dei relativi report, che aiutano a verificare la qualità del pacchetto contenente la distribuzione e il suo funzionamento su diverse piattaforme e con diverse versioni di Perl. Ogni distribuzione CPAN ha inoltre una sua coda di ticket http://rt.cpan.org/ per segnalare bug e collaborare con gli autori alla loro soluzione. I siti CPAN forniscono inoltre link alle versioni precedenti delle distribuzioni, valutazioni dei moduli scritte dagli utenti, note alla documentazione e altro ancora. Tutto questo è disponibile partendo da http://search.cpan.org/.
Le installazioni di Perl Moderno contengono due client per connettersi a CPAN e cercare, scaricare, fare il build, testare e installare le distribuzioni: CPAN.pm e CPANPLUS. I due client sono essenzialmente equivalenti per effettuare installazioni di base. In questo libro vi consigliamo l'uso di CPAN.pm semplicemente per la sua onnipresenza. Se usate una versione recente (al momento di questa stesura, l'ultima release stabile è la 1.9800), l'installazione di moduli è ragionevolmente facile. Lanciate il client con:
$ cpan
Per installare una distribuzione da dentro al client, scrivete:
$ cpan
cpan[1]> install Modern::Perl
... oppure, per installarla direttamente da linea di comando, scrivete:
$ cpan Modern::Perl
Il tutorial scritto da Eric Wilhelm sulla configurazione di CPAN.pm (footnote: http://learnperl.scratchcomputing.com/tutorials/configuration/) contiene un'eccellente sezione sulla risoluzione dei problemi.
Anche se il client CPAN è ormai integrato nella distribuzione di Perl 5, potreste dovere installare alcuni strumenti standard di sviluppo, come l'utility make e un compilatore C. Se siete utenti Windows, guardate Strawberry Perl (http://strawberryperl.com/) e Strawberry Perl Professional. Se siete utenti Mac OS X, installate XCode. Se siete utenti Unix o Unix-like dovreste già avere questi strumenti installati sul vostro sistema (per gli utenti Debian e Ubuntu: installate build-essential).
Strumenti per la Gestione di CPAN
Se il vostro sistema operativo possiede una sua installazione di Perl 5, questa potrebbe essere obsoleta o dipendere da specifiche versioni di alcune distribuzioni CPAN. Gli sviluppatori Perl seri spesso creano delle barriere virtuali tra il Perl di sistema e le loro installazioni Perl di sviluppo. Ci sono diversi progetti che possono aiutare a far ciò.
App::cpanminus è un client CPAN relativamente recente, i cui obiettivi principali sono la velocità, la semplicità e l'assenza di costi di configurazione. Installatelo con cpan App::cpanminus, oppure con:
$ curl -LO http://xrl.us/cpanm
$ chmod +x cpanm
App::perlbrew è un sistema che vi consente di installare e gestire diverse versioni e configurazioni di Perl, e di passare da una all'altra. Si installa semplicemente scrivendo:
$ curl -LO http://xrl.us/perlbrew
$ chmod +x perlbrew
$ ./perlbrew install
$ perldoc App::perlbrew
La distribuzione CPAN local::lib vi permette di installare e gestire distribuzioni in una vostra directory personale, anziché nell'intero sistema. È un modo molto efficace per poter lavorare con le vostre distribuzioni CPAN senza interferire con il lavoro degli altri utenti. L'installazione è un po' più complicata di quella delle due distribuzioni descritte sopra, ma App::local::lib::helper può renderla più semplice. Guardate http://search.cpan.org/perldoc?local::lib e http://search.cpan.org/perldoc?App::local::lib::helper per maggiori dettagli.
Tutti e tre i progetti suppongono che lavoriate in ambiente Unix-like (come una distribuzione GNU/Linux, ma anche Mac OS X). Se siete utenti Windows, scaricatevi il pacchetto Padre (http://padre.perlide.org/download.html).
Siti della Comunità
La homepage Perl all'URL http://www.perl.org/ contiene link alla documentazione e al codice sorgente di Perl, oltre che a tutorial, mailing list e a molti progetti importanti della comunità. Se siete nuovi utenti del Perl, la mailing list Perl beginners è un luogo accogliente dove potete porre domande tipiche per un principiante e ottenere risposte utili ed accurate. Guardate http://learn.perl.org/faq/beginners.html.
La homepage per lo sviluppo di Perl è http://dev.perl.org/, e contiene link alle risorse utili per lo sviluppo del core di Perl 5 e Perl 6 (footnote: Vedete anche http://www.perl6.org/).
Perl.com pubblica articoli e tutorial sul Perl e la sua cultura. I suoi archivi contengono materiale la cui datazione va indietro fino al ventesimo secolo. Vedete http://www.perl.com/.
Il sito centrale di CPAN (CPAN(cpan)) è http://www.cpan.org/, anche se gli utenti esperti usano più frequentemente http://search.cpan.org/. Questo centro di distribuzione di codice Perl libero e riusabile è una componente essenziale della comunità Perl. MetaCPAN (https://metacpan.org/) è un'interfaccia alternativa a CPAN apparsa di recente.
Il sito PerlMonks, che si trova all'URL http://perlmonks.org/, è dedicato alle discussioni sulla programmazione in Perl. Essendo attivo da undici anni, esso costituisce uno dei siti più venerandi di domande e risposte su un linguaggio di programmazione.
Diversi siti della comunità riportano e commentano le novità. http://blogs.perl.org/ è una piattaforma gratuita di blog aperta a tutti i membri della comunità Perl.
Altri siti raccolgono le elucubrazioni dei programmatori Perl; tra questi: http://perlsphere.net/, http://planet.perl.org/ e http://ironman.enlightenedperl.org/. Quest'ultimo è parte di una iniziativa della Enlightened Perl Organization (http://enlightenedperl.org/) per incrementare il numero e migliorare la qualità delle pubblicazioni sul Web relative al Perl.
Perl Buzz (http://perlbuzz.com/) raccoglie e ripubblica periodicamente alcune delle novità più utili e interessanti. Perl Weekly (http://perlweekly.com/) offre una panoramica settimanale delle notizie dal mondo Perl.
Perl.it (http://perl.it/) si propone invece come "il punto di riferimento del linguaggio Perl in Italia". L'Associazione di Promozione Sociale Perl.it, che lo gestisce, ha finanziato la traduzione in italiano del presente libro ed organizza annualmente l'Italian Perl Workshop (http://perl.it/workshop/), conferenza nazionale sul linguaggio Perl e tecnologie correlate. Se volete leggere materiale in italiano ed entrare in contatto con altri programmatori Perl in Italia, questo è decisamente il sito che fa per voi.
Siti sullo Sviluppo
Best Practical Solutions (http://bestpractical.com/) gestisce una installazione del loro popolare sistema di tracking, RT, sia per gli autori CPAN che per lo sviluppo del core di Perl 5 e di Perl 6. Ciascuna distribuzione CPAN ha la propria coda RT, raggiungibile da search.cpan.org e accessibile su http://rt.cpan.org/. Perl 5 e Perl 6 hanno code RT separate su http://rt.perl.org/.
La mailing list dei Perl 5 Porters (p5p) è il punto di riferimento degli sviluppatori di Perl 5. Vedete http://lists.cpan.org/showlist.cgi?name=perl5-porters.
La Perl Foundation (http://www.perlfoundation.org/) ospita un wiki per tutto ciò che riguarda Perl 5. Vedete http://www.perlfoundation.org/perl5.
I progetti di molti programmatori Perl sono ospitati su Github (http://github.com/) (footnote: ... inclusi i sorgenti di questo libro all'URL http://github.com/chromatic/modern_perl_book/). Guardate in special modo Gitpan (http://github.com/gitpan/), che ospita i repository Git che tangono traccia della storia completa di ogni distribuzione su CPAN.
GitPAN è aggiornato poco frequentemente. Come alternativa, considerate l'uso del formidabile modulo Git::CPAN::Patch di Yanick Champoux.
Eventi
La comunità Perl tiene innumerevoli conferenze, workshop, seminari e incontri. In particolare, YAPC--Yet Another Perl Conference--è un modello di conferenze locali e a basso costo, adottato con successo in diversi continenti. Vedete http://yapc.org/.
Sul wiki della Perl Foundation potete trovare l'elenco degli altri eventi, all'URL http://www.perlfoundation.org/perl5/index.cgi?perl_events.
Esiste inoltre il già menzionato Italian Perl Workshop, sul quale trovate informazioni all'URL http://perl.it/workshop/.
Centinaia di gruppi Perl Mongers locali si incontrano frequentemente per discussioni tecniche e per socializzare. Vedete http://www.pm.org/.
IRC
Quando i Perl monger non riescono ad incontrarsi di persona, molti di loro collaborano e discutono online attraverso il sistema testuale di chat IRC. Molti dei progetti Perl più seguiti e utilizzati hanno il proprio canale IRC; ad esempio, #moose e #catalyst.
Il server principale per la comunità Perl è irc://irc.perl.org/. Prendete nota dei canali #perl-help, per assistenza generica sulla programmazione Perl, e #perl-qa, dedicato al testing e ad altre problematiche legate alla qualità del codice. Sappiate che invece il canale #perl ha uno scopo molto generale, ed è usato dai suoi partecipanti per discutere di qualunque argomento gli interessi in quel momento (footnote: ... e, come tale, non è principalmente una sorgente di aiuto.).
La comunità italiana ha anche un proprio canale IRC, dove ovviamente potete scrivere (e ottenere risposte!) in italiano. Il server al quale connettersi è irc://irc.freenode.net, e il nome del canale #perl.it.
Il Linguaggio Perl
Come un linguaggio parlato, il Perl è una combinazione di molte piccole parti legate l'una con l'altra. Al contrario del linguaggio parlato, dove sfumature, tono di voce ed intuizione consentono alle persone di comunicare nonostante piccole incompresioni e concetti fumosi, i computer ed il codice richiedono precisione. Potete scrivere codice Perl efficace anche senza conoscere tutti i dettagli di ogni caratteristica del linguaggio, ma dovete comprendere come esse interagiscono per scrivere bene del codice Perl.
Nomi
I nomi (o identificatori) sono dappertutto nei programmi Perl: variabili, funzioni, package, classi e perfino filehandle. Tutti questi nomi iniziano con una lettera o un underscore e possono includere opzionalmente qualunque combinazione di lettere, numberi e underscore. Quando la direttiva utf8 (Stringhe e Unicode(unicode)) è attivata, potete usare negli identificatori qualunque carattere ammesso da UTF-8 in una parola. Questi sono tutti identificatori validi in Perl:
my $nome;
my @_nomi_privati;
my %Da_Nomi_a_Indirizzi;
sub unNomeBuffo3;
# con use utf8; attivata
package Ingy::Döt::Net;
Questi invece sono identificatori non validi:
my $nome scorretto;
my @3;
my %~flag;
package un-nome-in-stile-lisp;
I nomi esistono principalmente a beneficio del programmatore. Queste regole si applicano solo ai nomi che compaiono esplicitamente nel vostro codice sorgente, come sub prendi_torta oppure my $tostapane. Infatti, è il parser del Perl a controllare che le regole sui nomi siano rispettate.
La natura dinamica del Perl vi permette anche di riferirvi a delle entità attraverso nomi generati a runtime o dati in input al programma. Queste ricerche simboliche vi danno maggiore flessibilità al costo di una minore sicurezza. In particolare, invocando indirettamente delle funzioni e dei metodi o effettuando la ricerca dei simboli di un namespace, potete bypassare il parser del Perl.
Tutto questo può portare a del codice confuso. Come raccomanda molto efficacemente Mark Jason Dominus (footnote: http://perl.plover.com/varvarname.html), usate piuttosto un hash (Hash(hashes)) oppure una struttura dati annidata (Strutture Dati Annidate(nested_data_structures)).
Nomi di Variabili e Sigilli
I nomi di variabile iniziano sempre con un sigillo (o simbolo) che indica il tipo del valore della variabile. Le variabili scalari (Scalari(scalars)) usano il simbolo del dollaro ($). Le variabili array (Array(arrays)) usano il simbolo chiocciola (@). Le variabili hash (Hash(hashes)) usano il simbolo della percentuale (%):
my $scalare;
my @array;
my %hash;
I sigilli costituiscono una sorta di namespace visuale per i nomi di variabili. È possibile dichiarare più variabili con lo stesso nome e tipi diversi, ma crea confusione:
my ($cattiva_scelta, @cattiva_scelta, %cattiva_scelta);
Questo codice non confonde il Perl, ma confonde le persone che lo leggono.
I sigilli di Perl 5 sono varianti. Così come il contesto determina quanti elementi vi aspettate da un'operazione o quali tipi di dati vi aspettate di ottenere, i sigilli determinano il modo in cui manipolate i dati in una variabile. Per esempio, per accedere a un singolo elemento di un array o di un hash, dovete usare il sigillo scalare ($):
my $elemento_di_hash = $hash{ $chiave };
my $elemento_di_array = $array[ $indice ]
$hash{ $chiave } = 'valore';
$array[ $indice ] = 'altro valore';
L'analogia con il contesto di numero è importante. Usare un elemento scalare di un aggregato come lvalue (parte di un assegnamento a sinistra del carattere =) impone un contesto scalare (Contesto(context_philosophy)) all'rvalue (parte a destra del carattere =).
In modo simile, un accesso a più elementi di un hash o di un array--una operazione nota come slicing--usa il simbolo chiocciola (@) e impone un contesto lista (footnote: ... anche se la lista stessa ha zero o un solo elemento):
my @elementi_di_hash = @hash{ @chiavi };
my @elementi_di_array = @array[ @indici ];
my %hash;
@hash{ @chiavi } = @valori;
Il modo più affidabile di determinare il tipo di una variabile--scalare, array o hash--è di considerare le operazioni che le vengono applicate. Gli scalari supportano tutte le operazioni di base, come le manipolazioni di stringhe, numeriche e booleane. Gli array supportano accessi indicizzati che usano le parentesi quadre. Gli hash supportano accessi per chiave che usano le parentesi graffe.
Namespace
Perl fornisce un meccanismo per raggruppare funzioni e variabili correlate in un proprio spazio dei nomi univoco--i namespace (Package(packages)). Un namespace è una collezione di simboli con un suo nome. Perl permette di definire namespace multi-livello, i cui nomi sono composti con doppi due punti (::); per esempio, Dolci::Gelato si riferisce a una collezione di variabili e funzioni logicamente correlate, come cono() e versa_cioccolata_calda().
All'interno di un namespace, potete usare dei nomi abbreviati per i suoi membri. Al di fuori del namespace, vi riferite invece a un membro usando il suo nome qualificato. Ovvero, all'interno di Dolci::Gelato, aggiungi_praline() si riferisce alla stessa funzione di Dolci::Gelato::aggiungi_praline() al di fuori del namespace.
Oltre alle regole standard per i nomi, che si applicano anche ai nomi di package, per convenzione i nomi di package definiti dagli utenti iniziano con lettere maiuscole. Il core del Perl riserva infatti i nomi di package tutti minuscoli per le direttive (Direttive(pragmas)) come strict e warnings. Questa politica è imposta essenzialmente attraverso il rispetto delle linee guida della comunità.
Tutti i namespace in Perl 5 sono visibili a livello globale. Quando Perl deve fare la ricerca di un simbolo in Dolci::Gelato::Frigo, inizia col cercare nella symbol table main:: un simbolo che rappresenti il namespace Dolci::, quindi cerca al suo interno il namespace Gelato::, e così via. Il Frigo:: è quindi visibile dall'esterno del namespace Gelato::. Il fatto che i due namespace siano annidati è solo un meccanismo di memorizzazione, e non ha alcuna implicazione su ulteriori relazioni tra package padre e figlio o tra package fratelli. Sta al programmatore usare questo meccanismo per rendere evidenti le relazioni logiche tra entità--attraverso una buona scelta e organizzazione dei nomi.
Variabili
Una variabile in Perl è un'area per la memorizzazione di un valore (Valori(values)). Mentre un programma triviale può manipolare direttamente dei valori, quasi tutti i programmi usano le variabili per semplificare la logica del codice. Una variabile rappresenta valori; è più semplice spiegare il teorema di Pitagora in termini di variabili a, b e c che dover intuire il suo principio da una lunga lista di valori validi. Questo concetto potrà sembrarvi banale, ma per programmare in modo efficace dovete padroneggiare l'arte di bilanciare tra il generico e riusabile e lo specifico.
Scope delle Variabili
Le variabili sono visibili alle diverse parti del vostro programma in base al loro scope (Scope(scope)). Molte delle variabili che incontrerete hanno uno scope lessicale (Scope Lessicale(lexical_scope)). I file sorgente forniscono uno scope lessicale, mentre una dichiarazione di package non crea di per sé un nuovo scope:
package Negozio::Giocattoli;
my $sconto = 0.10;
package Negozio::Musica;
# $sconto ancora visibile
say "Il nostro sconto è attualmente $sconto!";
A partire da Perl 5.14, potete associare un blocco a una dichiarazione package. Questa sintassi fornisce effettivamente uno scope lessicale:
package Negozio::Giocattoli
{
my $sconto = 0.10;
}
package Negozio::Musica
{
# $sconto non disponibile
}
package Negozio::GiochiDaTavolo;
# $sconto sempre non disponibile
Sigilli delle Variabili
Il sigillo della variabile in una dichiarazione determina il tipo della variabile: scalare, array o hash. Il sigillo usato quando si accede alla variabile varia invece in base a ciò che volete fare con la variabile. Per esempio, dichiarate un array con @valori. Accedete al suo primo elemento--un singolo valore--con $valori[0]. E accedete a una lista di valori dentro l'array con @valori[ @indici ].
Variabili Anonime
In Perl non è richiesto che le variabili abbiano un nome. I nomi servono ad aiutare voi, i programmatori, a tener traccia di una $mela, delle @scatole o dei %ristoranti_economici. Le variabili create senza un nome dal vostro codice sorgente sono dette variabili anonime. L'unico modo per accedere a una variabile anonima è per riferimento (Riferimenti(references)).
Variabili, Tipi e Coercizione
Una variabile in Perl 5 rappresenta sia un valore (un costo in euro, dei tipi di pizza disponibili, dei negozi di chitarre con i loro numeri di telefono) che il contenitore che memorizza tale valore. Il sistema dei tipi di Perl gestisce sia i tipi valore che i tipi contenitore. Mentre il tipo contenitore di una variabile--scalare, array o hash--non può cambiare, Perl è flessibile riguardo al tipo valore. Potete memorizzare una stringa in una variabile in una data linea di codice, assegnare un numero alla stessa variabile nella linea successiva e quindi riassegnarle il riferimento a una funzione (Riferimenti a Funzioni(function_references)).
Eseguire su una variabile una operazione che impone uno specifico tipo valore può dar luogo a una coercizione (Coercizione(coercion)) dal precedente tipo valore della variabile stessa.
Per esempio, il modo ufficialmente documentato di determinare il numero di elementi in un array è di valutare l'array in contesto scalare (Contesto(context_philosophy)). Dato che una variabile scalare può sempre e solo contenere uno scalare, assegnare un array a uno scalare impone il contesto scalare all'operazione, e un array valutato in contesto scalare restituisce il numero di elementi nell'array:
my $conteggio = @elementi;
Queste relazioni tra tipi delle variabili, sigilli e contesto sono di importanza fondamentale.
Valori
La struttura di un proramma è fortemente influenzata dal modo in cui modellate i vostri dati con le variabili appropriate.
Mentre le variabili permettono la manipolazione astratta dei dati, i valori che esse contengono rendono i programmi utili e concreti. Più questi valori sono accurati, migliori sono i vostri programmi. I valori rappresentano dati--il nome e l'indirizzo di vostra zia, la distanza tra il vostro ufficio e un campo da golf sulla Luna, o il peso di tutti i biscotti che avete mangiato lo scorso anno. All'interno del vostro programma, le regole sul formato di tali dati sono spesso rigide. Dei programmi efficaci necessitano di modi efficaci (ovvero semplici, veloci, compatti, efficienti) per rappresentare i propri dati.
Stringhe
Una stringa è un dato testuale o binario senza particolari restrizioni di formato o contenuto. Potrebbe essere il vostro nome, il contenuto di un file immagine o lo stesso vostro programma. Una stringa ha un significato nel programma solo quando gliene attribuite uno.
Per rappresentare una stringa nel vostro programma, racchiudetela tra una coppia di caratteri di quotatura. I delimitatori di stringa più comuni sono i singoli e doppi apici:
my $nome = 'Donner Odinson, Portatore di Disperazione';
my $indirizzo = "Stanza 539, Bilskirnir, Valhalla";
I caratteri in una stringa a singola quotatura rappresentano se stessi letteralmente, con due eccezioni. Per inserire un apice singolo in una stringa a singola quotatura fatelo precedere da un carattere backslash di escape:
my $ricorda = 'L\'apice va preceduto '
. 'da un carattere di escape!';
Dovete precedere con l'escape anche un backslash alla fine della stringa per evitare che esso faccia da escape all'apice finale producendo un errore di sintassi:
my $eccezione = 'Questa stringa termina con un '
. 'escape, non con un apice: \\';
Tutti gli altri backslash vengono interpretati letteralmente, a meno che ve ne siano due in sequenza, nel qual caso il primo fa da escape al secondo:
is('Modern \ Perl', 'Modern \\ Perl',
'escape del backslash tra apici singoli');
Una stringa a doppia quotatura ha molti più caratteri speciali a disposizione. Per esempio, potete codificare nella stringa dei caratteri di spaziatura altrimenti invisibili:
my $tab = "\t";
my $newline = "\n";
my $carriage = "\r";
my $formfeed = "\f";
my $backspace = "\b";
Questo esempio mostra un principio utile: la sintassi usata per dichiarare una stringa può variare. Potete rappresentare un tab all'interno di una stringa sia con l'escape \t che digitando direttamente un tab. Per ciò che riguarda il Perl, le due stringhe funzionano allo stesso modo, anche se le rappresentazioni specifiche della stringa differiscono nel codice sorgente.
La dichiarazione di una stringa può contenere diverse linee di testo delimitate da newline; queste due dichiarazioni sono equivalenti:
my $con_escape = "due\nlinee";
my $letterale = "due
linee";
is $con_escape, $letterale, 'equivalenza \n e newline';
Le sequenze di escape sono spesso più facili da leggere della spaziatura equivalente.
Le stringhe Perl hanno lunghezza variabile. Quando le manipolate e modificate, il Perl cambia la loro dimensione in modo appropriato. Per esempio, potete combinare diverse stringhe in una stringa più grande con l'operatore . di concatenazione:
my $gattino = 'Micio' . ' ' . 'Miao';
A tutti gli effetti, è come se aveste inizializzato la stringa tutta insieme.
In una stringa a doppia quotatura potete anche interpolare il valore di una variabile scalare o i valori di un array, in modo tale che il contenuto attuale della variabile diventi parte della stringa come se le aveste concatenate:
my $fatto = "$nome vive a $indirizzo!";
# equivalente a
my $fatto = $nome . ' vive a ' . $indirizzo . '!';
Per inserire un doppio apice in una stringa a doppia quotatura fatelo precedere da un carattere di escape (ovvero, un backslash):
my $citazione = "\"Ahia,\", gridò. \"Fa male!\"";
Quando troppi backslash creano confusione, usate un operatore di quotatura alternativo con cui potete scegliere un diverso delimitatore di stringa. L'operatore q indica singola quotatura, mentre l'operatore qq si comporta come una doppia quotatura. Il carattere che segue l'operatore determina quali sono i caratteri che delimitano la stringa. Se tale carattere è il primo di una coppia di caratteri bilanciati--come aperta e chiusa graffa--il corrispondente carattere di chiusura sarà il delimitatore di fine stringa. Altrimenti, lo stesso carattere fungerà da delimitatore sia di inizio che di fine stringa.
my $citazione = qq{"Ahia", disse. "Fa male!"};
my $ricorda = q^L'apice singolo non va quotato!^;
my $reclamo = q{È un po' troppo presto per svegliarsi.};
Quando dichiarare una stringa complessa con molti caratteri di escape diventa noioso, usate la sintassi heredoc per assegnare una o più linee a una stringa:
my $estratto =<<'FINE_ESTRATTO';
Egli alzò lo sguardo. "Il tempo non è mai dalla nostra parte, figlio mio.
Non ne vedi l'ironia? Tutto ciò che essi conoscono è il cambiamento.
Il cambiamento è la sola costante su cui tutti loro concordano.
Noi, invece, nati fuori dal tempo, restiamo perfetti e
perfettamente auto-coscienti. E possiamo accettare il cambiamento solo
quando noi stessi lo perseguiamo. È contro la nostra natura: ci ribelliamo
contro il cambiamento. Dobbiamo forse considerarli
superiori per questo?"
FINE_ESTRATTO
La sintassi di <<'FINE_ESTRATTO' ha tre parti. Le doppie parentesi angolari introducono l'heredoc. Gli apici determinamo se l'heredoc si comporta come una stringa a singola o doppia quotatura. Il comportamento di default è quello della doppia quotatura con la relativa interpolazione. FINE_ESTRATTO è un identificatore arbitrario che il parser Perl 5 usa come delimitatore finale.
Fate attenzione; indipendentemente dall'indentazione della dichiarazione dell'heredoc, il delimitatore finale deve essere all'inizio della linea:
sub una_funzione {
my $ingredienti =<<'FINE_INGREDIENTI';
Due uova
Una tazza di farina
Due etti di burro
Un quarto di cucchiaino di sale
Una tazza di latte
Un pizzico di vaniglia
Condimento a piacere
FINE_INGREDIENTI
}
Se l'identificatore inizia con una spaziatura, la stessa spaziatura dev'essere presente prima del delimitatore finale. Tuttavia, se indentate l'identificatore, Perl 5 non rimuoverà una spaziatura equivalente dall'inizio di ogni linea dell'heredoc.
Usare una stringa in un contesto diverso da quello stringa causa una coercizione (Coercizione(coercion)).
Stringhe e Unicode
Unicode è un sistema per rappresentare tutti i caratteri delle lingue scritte del mondo. Mentre gran parte dei testi in inglese usano un insieme di soli 127 caratteri (che richiedono sette bit di memoria e stanno comodamente negli otto bit di un byte), sarebbe naïve pensare che non vi possa servire un giorno o l'altro una dieresi.
Le stringhe in Perl 5 possono rappresentare due tipi di dati distinti anche se correlati:
Ogni carattere ha un suo codepoint, un numero univoco che lo identifica nell'insieme di caratteri Unicode.
I dati binari sono sequenze di ottetti--numeri di 8 bit, ognuno dei quali può rappresentare un numero tra 0 e 255.
Perché ottetti e non byte? L'assunzione che un carattere possa stare dentro un byte può essere causa di problemi senza fine con Unicode. Separate quindi l'idea di spazio in memoria da quella della rappresentazione dei caratteri.
Le stringhe Unicode e binarie sono simili. Ciascuna ha una length(). Ciascuna supporta le operazioni standard sulle stringhe come la concatenazione, lo splicing e l'applicazione di espressioni regolari. Ogni stringa che non contenga dati binari puri contiene dati testuali, e dovrebbe quindi essere una sequenza di caratteri Unicode.
Tuttavia, dato che il vostro sistema operativo reppresenta i dati su disco (o in input dall'utente o dalla rete) come sequenze di ottetti, Perl non può sapere se i dati che leggete costituiscono un file immagine o un documento di testo o qualunque altra cosa. Per default, Perl tratta tutti i dati in input come sequenze di ottetti. Tocca a voi assegnare un significato specifico a tali dati.
Codifica dei Caratteri
Una stringa Unicode è una sequenza di ottetti che rappresenta una sequenza di caratteri. Una codifica Unicode mappa sequenze di ottetti in sequenze di caratteri. Alcune codifiche, come UTF-8, possono codificare tutti i caratteri dell'insieme di caratteri Unicode. Altre codifiche rappresentano un sottoinsieme dei caratteri Unicode. Per esempio, l'ASCII codifica un testo in inglese comune senza caratteri accentati, mentre Latin-1 può rappresentare testo nella maggior parte delle lingue che usano l'alfabeto latino.
Per evitare problemi con Unicode, decodificate e codificate sempre dalla/alla codifica appropriata durante l'input e l'output del vostro programma.
Perl 5.12 supporta lo standard Unicode 5.2, mentre Perl 5.14 supporta Unicode 6.0. Se avete bisogno di tenere conto delle differenze tra le versioni di Unicode, probabilmente conoscete già il sito http://unicode.org/versions/.
Unicode nei Vostri Filehandle
Quando comunicate al Perl che uno specifico filehandle (File(files)) lavora con testo codificato, Perl convertirà automaticamente gli ottetti in input in stringhe Unicode. Per ottenere questo, aggiungete un layer di IO al modo della funzione nativa open. Un layer di IO si interpone tra il programma e l'input o l'output convertendo i dati. Nel seguente esempio, il layer :utf8 decodifica dei dati UTF-8:
use autodie;
open my $fh, '<:utf8', $file_di_testo;
my $stringa_unicode = <$fh>;
Con binmode, potete anche modificare un filehandle esistente, sia per l'input che per l'output:
binmode $fh, ':utf8';
my $stringa_unicode = <$fh>;
binmode STDOUT, ':utf8';
say $stringa_unicode;
Senza il modo utf8, stampare delle stringhe Unicode su un filehandle provoca un warning (Wide character in %s), perché il file contiene ottetti, e non caratteri Unicode.
Unicode nei Vostri Dati
Il modulo core Encode fornisce una funzione di nome decode() per convertire uno scalare contenente dei dati in una stringa Unicode. La corrispondente funzione encode() converte dalla codifica interna del Perl alla codifica di output desiderata:
my $da_utf8 = decode('utf8', $dati);
my $a_latin1 = encode('iso-8859-1', $stringa);
Unicode nei Vostri Programmi
Potete includere caratteri Unicode nei vostri programmi in tre modi. Il modo più semplice è quello di usare la direttiva utf8 (Direttive(pragmas)), che indica al parser Perl di interpretare il resto del file di codice sorgente con la codifica UTF-8. Ciò vi permette di usare caratteri Unicode nelle stringhe e negli identificatori:
use utf8;
sub da_£_a_¥ { ... }
my $yen = da_£_a_¥('1000£');
Per poter scrivere questo codice, il vostro editor di testo deve supportare UTF-8 e dovete salvare il file con la codifica appropriata.
All'interno di stringhe a doppia quotatura, potete usare la sequenza di escape Unicode per rappresentare la codifica dei caratteri. La sintassi \x{} rappresenta un carattere singolo; mettete il codice Unicode del carattere in esadecimale dentro alle parentesi graffe:
my $thorn_con_escape = "\x{00FE}";
Alcuni caratteri Unicode hanno dei nomi, che spesso sono più facili da leggere dei codici Unicode. Usate la direttiva charnames per attivarli e la sequenza di escape \N{} per riferirvi ad essi:
use charnames ':full';
use Test::More tests => 1;
my $thorn_con_escape = "\x{00FE}";
my $thorn_per_nome = "\N{LATIN SMALL LETTER THORN}";
is $thorn_con_escape, $thorn_per_nome,
'controllo di equivalenza carattere thorn';
Potete usare \x{} e \N{} all'interno di espressioni regolari e in ogni altro punto dove potete legittimamente inserire una stringa o un carattere.
Conversione Implicita
Molti dei problemi con Unicode in Perl nascono dal fatto che una stringa può essere sia una sequenza di ottetti che una sequenza di caratteri. Perl vi permette di combinare questi due tipi attraverso l'uso delle conversioni implicite. Quando queste conversioni sono sbagliate, raramente lo sono in modo ovvio.
Quando Perl concatena una sequenza di ottetti con una sequenza di caratteri Unicode, decodifica implicitamente la sequenza di ottetti usando la codifica Latin-1. La stringa risultante conterrà caratteri Unicode. Quando stampate dei caratteri Unicode, Perl codifica la stringa usando UTF-8, perché Latin-1 non può rappresentare l'intero insieme di caratteri Unicode--Latin-1 è un sottoinsieme di UTF-8.
Questa asimmetria può dar luogo a stringhe Unicode codificate con UTF-8 in output e decodificate con Latin-1 in input.
Peggio ancora, quando il testo contiene solo caratteri dell'inglese senza accento, il bug rimane nascosto--dato che entrambe le codifiche hanno la stessa rappresentazione di ogni carattere.
my $ciao = "Ciao, ";
my $saluto = $ciao . $nome;
Se $nome contiene un nome in alfabeto inglese come Alice non si riscontra nessun problema, dato che la rappresentazione Latin-1 è uguale alla rappresentazione UTF-8. Se invece $name contiene un nome come José, ci sono diverse possibilità:
Ci sono diversi scenari anche per il letterale stringa:
my $ciao = "Ciao, ";
my $ciao = "¡Hola, ";
use utf8;
my $ciao = "Kuirabá, ";
Se sia $ciao che $nome sono stringhe Unicode, la loro concatenazione produce un'altra stringa Unicode.
Se entrambe le stringhe sono sequenze di ottetti, Perl le concatenerà in una nuova sequenza di ottetti. Se entrambi i valori contengono ottetti di una stessa codifica--per esempio, Latin-1-- la concatenazione funzionerà correttamente. Se invece gli ottetti appartengono a diverse codifiche, per esempio una concatenazione di dati UTF-8 a dati Latin-1, la sequenza di ottetti risultante è priva di senso per entrambe le codifiche. Questo potrebbe succedere se l'utente avesse immesso un nome come stringa UTF-8 e il saluto fosse una costante stringa in Latin-1, ma il programma non avesse decodificato nessuna delle due.
Se solo uno dei valori è una stringa Unicode, Perl decodificherà l'altro come un dato Latin-1. Se questa non è la codifica corretta, i caratteri Unicode risultanti saranno errati. Per esempio, se l'input dell'utente fosse un dato UTF-8 e la costante stringa fosse Unicode, il nome verrebbe decodificato erroneamente come i cinque caratteri Unicode José (sic) invece di José perché i dati UTF-8 hanno un altro significato quando sono decodificati come dati Latin-1.
Vedete perldoc perluniintro per una spiegazione molto più dettagliata di Unicode, delle codifiche e di come gestire i dati in ingresso e in uscita in un mondo Unicode (footnote: Per avere molti più dettagli su una gestione efficace di Unicode in ogni parte dei vostri programmi, vedete la risposta di Tom Christiansen a "Why does Modern Perl avoid UTF-8 by default?" http://stackoverflow.com/questions/6162484/why-does-modern-perl-avoid-utf-8-by-default/6163129#6163129).
Perl 5.12 ha aggiunto una funzionalità, unicode_strings, che attiva la semantica Unicode per tutte le operazioni su stringa all'interno del suo scope. Perl 5.14 ha ulteriormente migliorato questa funzionalità; se lavorate con Unicode in Perl, vale la pena di fare un aggiornamento almeno a Perl 5.14.
Numeri
Perl supporta i numeri sia come valori interi che a virgola mobile. Potete rappresentarli anche con la notazione scientifica e in forma binaria, ottale ed esadecimale:
my $intero = 42;
my $float = 0.007;
my $float_sci = 1.02e14;
my $binario = 0b101010;
my $ottale = 052;
my $esadecimale = 0x20;
I caratteri in grassetto sono i prefissi numerici rispettivamente per le notazioni binaria, ottale ed esadecimale. Ricordate che uno zero iniziale in un intero indica sempre la forma ottale.
Anche se in Perl 5 potete scrivere esplicitamente valori a virgola mobile perfettamente accurati, Perl 5 li memorizza internamente in forma binaria. Questa rappresentazione a volte è causa di specifiche imprecisioni; consultate perldoc perlnumber per maggiori dettagli.
Non potete usare delle virgole per separare le migliaia nei letterali numerici, dato che il parser interpreta le virgole come operatori. Usate invece gli underscore. Il parser li tratterà come caratteri invisibili; per chi legge il codice invece non lo sono. Queste istruzioni sono equivalenti:
my $miliardo = 1000000000;
my $miliardo = 1_000_000_000;
my $miliardo = 10_0_00_00_0_0_0;
Considerate qual è l'alternativa più leggibile.
Grazie alla coercizione (Coercizione(coercion)), raramente i programmatori Perl devono preoccuparsi di convertire in numeri un testo letto dall'esterno del programma. Perl tratta tutto ciò che sembra un numero come un numero in contesto numerico. Nelle rare occasioni in cui volete sapere se qualcosa sembra un numero al Perl, usate la funzione looks_like_number del modulo core Scalar::Util. Questa funzione restituisce il valore vero se Perl considera come numerico l'argomento passato.
Il modulo CPAN Regexp::Common fornisce molte espressioni regolari ben testate per identificare tipi validi più specifici di valori numerici (numero naturale, intero, a virgola mobile).
Undef
Il valore undef del Perl 5 rappresenta un valore non assegnato, indefinito e sconosciuto. Le variabili scalari dichiarate ma non definite contengono undef:
my $nome = undef; # assegnamento superfluo
my $rango; # contiene anche undef
In contesto booleano, undef viene valutato come falso. Valutare undef in contesto stringa --ad esempio interpolarlo in una stringa--causa un warning di uninitialized value:
my $non_definito;
my $definito = $non_definito . '... e così via';
... produce:
Use of uninitialized value $non_definito in
concatenation (.) or string...
La funzione nativa defined restituisce vero se il suo operando viene valutato come valore definito (ovvero qualunque cosa tranne undef):
my $stato = 'ho il raffreddore';
say defined $stato; # 1, che rappresenta il valore vero
say defined undef; # stringa vuota; un valore falso
La Lista Vuota
Quando compare a destra in un assegnamento, il costrutto () rappresenta una lista vuota. In contesto scalare, viene valutato come undef. In contesto lista, è semplicemente una lista vuota. Quando invece compare a sinistra in un assegnamento, il costrutto () impone il contesto lista. Per contare il numero di elementi restituiti da una espressione in contesto lista senza usare una variabile temporanea, usate l'idioma (Idiomi(idioms)):
my $conteggio = () = tutti_i_miei_cappelli();
Dato che l'operatore di assegnamento è associativo a destra (Associatività (associativity)), Perl valuta per primo il secondo assegnamento chiamando tutti_i_miei_cappelli() in contesto lista. Questa chiamata restituisce una lista.
L'assegnamento alla lista vuota fa perdere tutti i valori della lista, ma tale assegnamento ha luogo in contesto scalare, che valuta il numero di elementi nella parte destra dell'assegnamento. Di conseguenza, $conteggio contiene il numero di elementi nella lista restituita da tutti_i_miei_cappelli().
Se per il momento trovate questo concetto un po' confuso, non preoccupatevi. Man mano che capirete come le funzionalità di progetto fondamentali del Perl si incastrano tra loro nella pratica, tutto diventerà più chiaro.
Liste
Una lista è un gruppo di una o più espressioni separate dalla virgola. Le liste possono comparire letteralmente nel codice sorgente come valori:
my @primi_fib = (1, 1, 2, 3, 5, 8, 13, 21);
... come target di un assegnamento:
my ($package, $nomefile, $linea) = chiamante();
... o come liste di espressioni:
say nome(), ' => ', anni();
Non sono le parentesi tonde che creano una lista. È la virgola che la crea. Dove sono presenti in questi esempi, le parentesi raggruppano espressioni per modificarne la precedenza (Precedenza(precedence)).
Usate l'operatore di intervallo per creare liste di letterali in forma compatta:
my @caratteri = 'a' .. 'z';
my @contatore = 13 .. 27;
Usate l'operatore qw() per spezzare un letterale stringa con i caratteri di spaziatura e produrre una lista di stringhe:
my @stooges = qw( Larry Curly Moe Shemp Joey Kenny );
Perl genera un warning se qw() contiene una virgola o il carattere di commento (#), perché non solo tali caratteri sono rari in un qw(), ma la loro presenza indica normalmente una svista.
Le liste possono occorrere (e spesso lo fanno) come risultati di espressioni, ma queste liste non compaiono letteralmente nel codice sorgente.
Le liste e gli array non sono interscambiabili in Perl. Le liste sono valori. Gli array sono contenitori. Potete memorizzare una lista in un array e potete forzare la conversione di un array in una lista, ma essi restano entità separate. Per esempio, l'indicizzazione di una lista occorre sempre in contesto lista. L'indicizzazione di un array può occorrere sia in contesto scalare (per un singolo elemento) che in contesto lista (per una slice):
# non preoccupatevi dei dettagli per ora
sub contesto
{
my $contesto = wantarray();
say defined $contesto
? $contesto
? 'lista'
: 'scalare'
: 'void';
return 0;
}
my @slice_di_lista = (1, 2, 3)[contesto()];
my @slice_di_array = @slice_di_lista[contesto()];
my $indice_di_array = $slice_di_array[contesto()];
say contesto(); # contesto lista
contesto(); # contesto void
Controllo di Flusso
Il controllo di flusso di base del Perl è molto semplice. L'esecuzione del programma comincia dall'inizio (la prima linea del file da eseguire) e prosegue fino alla fine:
say 'All'inizio';
say 'In mezzo';
say 'Alla fine';
Le direttive di controllo di flusso del Perl modificano l'ordine di esecuzione--ovvero che cosa avviene al prossimo passo nel programma--in base ai valori delle loro espressioni.
Direttive di Scelta
La direttiva if esegue l'azione associata solo quando la sua espressione condizionale ha valore vero:
say 'Ciao, Roberto!' if $nome eq 'Roberto';
La forma postfissa è adeguata in caso di espressioni semplici. La forma a blocco raggruppa diverse espressioni in una singola unità:
if ($nome eq 'Roberto')
{
say 'Ciao, Roberto!';
trovato_roberto();
}
Mentre la forma a blocco richiede che la condizione sia tra parentesi, la forma postfissa non lo richiede.
L'espressione condizionale può consistere di diverse sotto-espressioni, purché esse compongano un'unica espressione principale:
if ($nome eq 'Roberto' && not salutato_roberto())
{
say 'Ciao, Roberto!';
trovato_roberto();
}
Nella forma postfissa, aggiungere delle parentesi può aiutare a chiarire il significato del codice al costo di una minor eleganza visiva:
saluta_roberto() if ($nome eq 'Roberto' && not salutato_roberto());
La direttiva unless è una forma negata di if. Perl esegue l'azione quando l'espressione condizionale è falsa:
say "Tu non sei Roberto!" unless $nome eq 'Roberto';
Come if, anche unless ha una forma a blocco, anche se molti programmatori la evitano poiché diventa rapidamente difficile da leggere in presenza di condizionali complessi:
unless (anno_bisestile() and luna_piena())
{
spassatela();
fai_capriole();
}
unless è particolarmente adatta ad essere usata con condizionali postfissi, specialmente per la validazione dei parametri delle funzioni (Validazione Postfissa dei Parametri(postfix_parameter_validation)):
sub spassatela
{
return unless @_;
for my $canzone (@_) { ... }
}
Entrambe le forme a blocco di if e unless accettano la direttiva else, che specifica il codice da eseguire quando l'espressione condizionale non ha valore vero (per if) o falso (per unless):
if ($nome eq 'Roberto')
{
say 'Ciao, Roberto!';
accogli_utente();
}
else
{
say "Non ti conosco.";
ignora_utente();
}
I blocchi else vi permettono di riscrivere i condizionali if e unless l'uno nei termini dell'altro:
unless ($nome eq 'Roberto')
{
say "Non ti conosco.";
ignora_utente();
}
else
{
say 'Ciao, Roberto!';
accogli_utente();
}
Tuttavia, la doppia negazione implicita nell'uso di unless con un blocco else può creare confusione. Questo esempio potrebbe essere l'unico posto dove vi capiterà di vederlo.
Così come vi fornisce sia if che unless per permettervi di esprimere i vostri condizionali nel modo più leggibile, il Perl vi permette anche di scegliere tra operatori condizionali positivi e negativi:
if ($nome ne 'Roberto')
{
say "Non ti conosco.";
ignora_utente();
}
else
{
say 'Ciao, Roberto!';
accogli_utente();
}
... anche se la doppia negazione implicita nella presenza del blocco else suggerirebbe di invertire il condizionale.
Una o più direttive elsif possono seguire un blocco if e precedere un singolo else:
if ($nome eq 'Roberto')
{
say 'Ciao, Roberto!';
accogli_utente();
}
elsif ($nome eq 'Giacomo')
{
say 'Ciao, Giacomo!';
accogli_utente();
}
else
{
say "Non sei mio zio.";
ignora_utente();
}
Anche un blocco unless può essere seguito da un blocco elsif (footnote: Buona fortuna se volete capirci qualcosa!). Invece non esiste elseunless.
Scrivere else if costituisce un errore di sintassi (footnote: Larry preferisce elsif per ragioni estetiche, come nell'antica arte del linguaggio Ada.):
if ($nome eq 'Riccardo')
{
say 'Ciao, cugino!';
}
# attenzione; errore di sintassi
else if ($nome eq 'Cristina')
{
say 'Ciao, cugina!';
}
L'Operatore Condizionale Ternario
L'operatore condizionale ternario valuta un'espressione condizionale e sceglie tra due alternative:
my $suffisso_temporale = dopo_pranzo($tempo)
? 'pomeriggio'
: 'mattino';
L'espressione condizionale precede il carattere punto interrogativo (?) e il carattere due punti (:) separa le due alternative. Le alternative sono espressioni di complessità arbitraria--incluse altre espressioni condizionali ternarie.
Un idioma interessante, anche se un po' oscuro, è l'uso del condizionale ternario per scegliere tra variabili alternative, anziché valori:
push @{ rand() > 0.5 ? \@squadra_rossa : \@squadra_blu },
Giocatore->new;
Come al solito, valutate i benefici della chiarezza rispetto ai benefici della concisione.
Cortocircuiti
Il Perl può cortocircuitare parti di espressioni condizionali complesse. Quando Perl è in grado di determinare che un'espressione complessa è vera o falsa senza aver ancora valutato tutte le sue sottoespressioni, le sottoespressioni rimanenti non vengono più valutate. Un esempio dovrebbe aiutare a chiarire:
say "Entrambi veri!" if ok( 1, 'sottoespressione uno' )
&& ok( 1, 'sottoespressione due' );
done_testing();
Il valore di ritorno di ok() (Testing(testing)) è il valore booleano ottenuto valutando il suo primo argomento, quindi questo codice stampa:
ok 1 - sottoespressione uno
ok 2 - sottoespressione due
Entrambi veri!
Quando la prima sottoespressione--la prima chiamata a ok--ha valore vero, Perl deve valutare la seconda sottoespressione. Se invece la prima sottoespressione avesse valore falso, non sarebbe necessario valutare le sottoespressioni successive, dato che l'intera espressione non potrà comunque essere vera:
say "Entrambi veri!" if ok( 0, 'sottoespressione uno' )
&& ok( 1, 'sottoespressione due' );
Questo esempio stampa:
not ok 1 - sottoespressione uno
Sebbene la seconda sottoespressione sia ovviamente vera, Perl non la valuta. Lo stesso tipo di cortocircuito si applica alle operazioni di or logico:
say "Almeno uno vero!" if ok( 1, 'sottoespressione uno' )
|| ok( 1, 'sottoespressione due' );
Questo esempio stampa:
ok 1 - sottoespressione uno
Almeno uno vero!
Dato il successo della prima sottoespressione, Perl può evitare di valutare la seconda sottoespressione. Se la prima sottoespressione fosse stata falsa, il risultato della valutazione della seconda sottoespressione avrebbe determinato il risultato della valutazione dell'intera espressione.
Oltre a permettervi di evitare computazioni potenzialmente costose, i cortocircuiti possono aiutarvi a evitare errori e warning, come nel caso in cui l'uso di un valore non definito potrebbe causare un warning:
my $barbecue;
if (defined $barbecue and $barbecue eq 'salsiccia') { ... }
Contesto per le Direttive Condizionali
Le direttive condizionali--if, unless e l'operatore condizionale ternario--valutano tutte un'espressione in contesto booleano (Contesto(context_philosophy)). Così come gli operatori eq, ==, ne e != producono tutti un risultato booleano quando vengono valutati, Perl converte i risultati di altre espressioni--inclusi variabili e valori--in forma booleana.
Perl 5 non ha un unico valore vero, né un unico valore falso. Tutti i numeri che hanno valore 0 rappresentano il valore falso. Tra questi: 0, 0.0, 0e0, 0x0, ecc. La stringa vuota ('') e la stringa '0' hanno valore falso, ma non le stringhe '0.0', '0e0', ecc. L'idioma '0 ma vero' valuta a 0 in contesto numerico e a vero in contesto booleano, grazie al suo contenuto come stringa.
Sia la lista vuota che undef hanno valore falso. Gli array e gli hash vuoti restituiscono il numero 0 in contesto scalare, e quindi hanno valore falso in contesto booleano. Un array che contiene un unico elemento--anche se fosse undef--valuta a vero in contesto booleano. Un hash che contiene qualche elemento --anche una chiave con valore associato undef--valuta a vero in contesto booleano.
Il modulo CPAN Want vi permette di individuare un contesto booleano all'interno delle vostre funzioni. La direttiva overloading del core (Overloading(overloading)) vi permette di specificare che cosa devono produrre i vostri tipi di dati quando sono valutati in diversi contesti.
Direttive di Ciclo
Perl fornisce diverse direttive di ciclo e iterazione. Il ciclo in stile foreach valuta un'espressione che produce una lista e esegue un'istruzione o un blocco fino a quando ha consumato tale lista:
foreach (1 .. 10)
{
say "$_ * $_ = ", $_ * $_;
}
Questo esempio usa l'operatore di intervallo per produrre una lista di interi da uno a dieci inclusi. La direttiva foreach cicla su questi numeri, settando la variabile topic $_ (La Variabile Scalare di Default(default_scalar_variable)) a ciascuno di essi a turno. Perl esegue il blocco per ogni intero e stampa il suo quadrato.
Molti programmatori Perl si riferiscono alle iterazioni come cicli foreach, ma Perl tratta i nomi foreach e for in modo interscambiabile. È il codice che li accompagna a determinare il tipo e comportamento del ciclo.
Analogamente a if e unless, questo ciclo ha forma postfissa:
say "$_ * $_ = ", $_ * $_ for 1 .. 10;
Un ciclo for può specificare il nome di una variabile anziché usare la variabile topic:
for my $i (1 .. 10)
{
say "$i * $i = ", $i * $i;
}
Quando un ciclo for usa una variabile iteratore, lo scope della variabile è l'interno del loop. Perl setta questa variabile lessicale al valore di ciascun elemento su cui opera l'iterazione. Perl non modifica la variabile topic ($_). Se avete dichiarato una variabile lessicale $i nello scope più esterno, il suo valore rimane intatto al di fuori del ciclo:
my $i = 'mucca';
for my $i (1 .. 10)
{
say "$i * $i = ", $i * $i;
}
is( $i, 'mucca', 'Valore intatto nello scope esterno' );
Questa localizzazione avviene anche se non ridichiarate la variabile iteratore come lessicale (footnote: ... ma siete caldamente invitati a dichiarare le vostre variabili iteratore come lessicali per ridurne lo scope.):
my $i = 'cavallo';
for $i (1 .. 10)
{
say "$i * $i = ", $i * $i;
}
is( $i, 'cavallo', 'Valore intatto nello scope esterno' );
Iterazione e Alias
Il ciclo for effettua un aliasing della variabile iteratore ai valori su cui opera l'iterazione in modo che ogni modifica al valore dell'iteratore modifica anche il valore corrente dell'iterazione:
my @numeri = 1 .. 10;
$_ **= 2 for @numeri;
is( $numeri[0], 1, '1 * 1 vale 1' );
is( $numeri[1], 4, '2 * 2 vale 4' );
...
is( $numeri[9], 100, '10 * 10 vale 100' );
L'aliasing funziona anche con la forma a blocco del ciclo for:
for my $num (@numeri)
{
$num **= 2;
}
... e con le iterazioni che usano la variabile topic:
for (@numeri)
{
$_ **= 2;
}
Tuttavia, non potete usare l'aliasing per modificare valori costanti:
for (qw( Qui Quo Qua ))
{
$_++;
say;
}
In questo caso, il Perl solleverà un'eccezione relativa al tentativo di modifica di valori read-only.
Potrà succedervi occasionalmente di vedere l'uso del for con una singola variabile scalare, allo scopo di effettuare un aliasing di $_ a tale variabile:
for ($input_utente)
{
s/\A\s+/; # scarta spaziatura iniziale
s/\s+\z/; # scarta spaziatura finale
$_ = quotemeta; # aggiungi escape a caratteri speciali
}
Iterazione e Scope
Lo scope della variabile topic usata come iteratore è una causa comune di confusione. Considerate una funzione scombina_topic() il cui scopo è di modificare $_. Se il codice che itera su una lista chiama scombina_topic() senza proteggere $_, ci sarà poi da divertirsi col debugging:
for (@valori)
{
scombina_topic();
}
sub scombina_topic
{
s/pippo/pluto/;
}
Se proprio dovete usare $_ anziché un'altra variabile, rendete la variabile topic lessicale con my $_:
sub scombina_topic
{
# era $_ = shift;
my $_ = shift;
s/pippo/pluto/;
s/paperino/topolino/;
return $_;
}
Anche l'uso di una variabile iteratore con un nome previene gli aliasing indesiderati di $_.
Il Ciclo For in Stile C
Il ciclo for in stile C vi richiede di gestire le condizioni dell'iterazione:
for (my $i = 0; $i <= 10; $i += 2)
{
say "$i * $i = ", $i * $i;
}
In questo costrutto dovete assegnare esplicitamente a una variabile iteratore, dato che esso non effettua né aliasing né assegnamenti alla variabile topic. Mentre una variabile dichiarata all'interno del costrutto di ciclo ha come scope il blocco lessicale del ciclo, una variabile dichiarata all'esterno del costrutto di ciclo non viene lessicalizzata:
my $i = 'maiale';
for ($i = 0; $i <= 10; $i += 2)
{
say "$i * $i = ", $i * $i;
}
isnt( $i, 'maiale', '$i sovrascritto con un numero' );
Il costrutto di ciclo può avere tre sottoespressioni. La prima sottoespressione--la sezione di inizializzazione--viene eseguita solo una volta, prima dell'esecuzione del corpo del ciclo. Perl valuta la seconda sottoespressione--il confronto condizionale--prima di ciascuna iterazione del corpo del ciclo. Quando essa ha un valore vero, l'iterazione continua. Quando ha un valore falso, l'iterazione termina. L'ultima sottoespressione viene eseguita dopo ogni iterazione del corpo del ciclo.
for (
# sottoespressione di inizializzazione del ciclo
say 'Inizializzazione', my $i = 0;
# sottoespressione di confronto condizionale
say "Iterazione: $i" and $i < 10;
# sottoespressione che fa terminare (prima o poi) l'iterazione
say 'Incremento ' . $i++
)
{
say "$i * $i = ", $i * $i;
}
Notate l'assenza del punto e virgola dopo l'ultima sottoespressione, e l'uso dell'operatore virgola e dell'operatore a bassa precedenza and; questa sintassi è piuttosto complicata. Quando è possibile, preferite i cicli in stile foreach al ciclo for.
Tutte e tre le sottoespressioni sono opzionali. Un ciclo for infinito si ottiene con:
for (;;) { ... }
While e Until
Un ciclo while continua fino a che l'espressione condizionale del ciclo assume un valore booleano falso. Un modo idiomatico di scrivere un ciclo infinito è:
while (1) { ... }
Al contrario dell'iterazione con cicli in stile foreach, la condizione di ciclo del while non ha di per sé alcun effetto collaterale. Ovvero, se @valori contiene uno o più elementi, anche questo codice è un ciclo infinito:
while (@valori)
{
say $valori[0];
}
Per prevenire i cicli while infiniti, fate un aggiornamento distruttivo dell'array @valori modificandolo ad ogni iterazione del ciclo:
while (@valori)
{
my $valore = shift @valori;
say $valore;
}
È anche possibile modificare @valori all'interno della condizione del while, ma ci sono alcune sottigliezze legate al valore di verità di ciascun elemento.
while (my $valore = shift @valori)
{
say $valore;
}
Questo ciclo termina non appena incontra un elemento che ha un valore falso, e quindi non esaurisce necessariamente tutto l'array. Questo potrebbe essere il comportamento desiderato, ma spesso è invece una sorpresa per i principianti.
Il ciclo until inverte il senso del test del ciclo while. L'iterazione continua finchè l'espressione condizionale del ciclo valuta a falso:
until ($fine_corsa)
{
...
}
L'uso canonico del ciclo while è di iterare sull'input da un filehandle:
use autodie;
open my $fh, '<', $file;
while (<$fh>)
{
...
}
Perl 5 interpreta questo ciclo while come se aveste scritto:
while (defined($_ = <$fh>))
{
...
}
Senza l'uso implicito di defined, ogni linea letta dal filehandle che valuta a falso in contesto scalare--una linea vuota o una linea che contiene il solo carattere 0--farebbe terminare il ciclo. L'operatore readline (<>) restituisce un valore non definito solo quando ha raggiunto la fine del file.
Usate la funzione nativa chomp per rimuovere i caratteri di fine linea da ogni linea. Molti principianti se ne dimenticano.
Sia while che until hanno forme postfisse, come il ciclo infinito 1 while 1;. Il while e l'until postfissi si possono applicare a qualunque espressione singola, incluso il classico esempio "Ciao, mondo!" dei computer a 8-bit dei primi anni '80:
print "Ciao, mondo! " while 1;
I cicli infiniti sono più utili di quanto potrebbero sembrare, specialmente per gli event loop nei programmi con GUI, per gli interpreti di programmi o per i server di rete:
$server->invia_risultati() until $esegui_shutdown;
Usate un blocco do per raggruppare diverse espressioni in una singola unità:
do
{
say 'Come ti chiami?';
my $nome = <>;
chomp $nome;
say "Ciao, $nome!" if $nome;
} until (eof);
Un blocco do viene interpretato come una singola espressione che può contenere diverse espressioni. Al contrario della forma a blocco del ciclo while, il blocco do con un postfisso while o until esegue il proprio corpo almeno una volta. Questo costrutto è meno comune delle altre forme di ciclo, ma non meno potente.
Cicli dentro altri Cicli
Potete annidare i cicli dentro altri cicli:
for my $colore (@colori)
{
for my $valore (@valori_delle_carte) { ... }
}
Quando lo fate, ricordatevi di dichiarare le vostre varibili di iterazione! In caso contrario, la potenziale confusione dovuta all'uso della variabile topic e al suo scope sarebbe ingestibile.
Un errore comune quando si annida un ciclo while in un ciclo foreach è che è facile esaurire un filehandle con un ciclo while:
use autodie;
open my $fh, '<', $un_file;
for my $prefisso (@prefissi)
{
# NON USARE; probabilmente contiene un bug
while (<$fh>)
{
say $prefisso, $_;
}
}
Aprendo il filehandle al di fuori del ciclo for, la posizione del file rimane invariata tra due iterazioni del ciclo for. Alla seconda iterazione, il ciclo while non avrà più nulla da leggere e non verrà eseguito. Per ovviare a questo problema, ri-aprite il file all'interno del ciclo for (semplice, ma non sempre ottimale per l'uso delle risorse di sistema), leggete l'intero file in memoria (potrebbe non funzionare se il file è molto grande) oppure riportate il filehandle all'inizio del file ad ogni iterazione con un seek (un'opzione spesso trascurata):
use autodie;
open my $fh, '<', $un_file;
for my $prefisso (@prefissi)
{
while (<$fh>)
{
say $prefisso, $_;
}
seek $fh, 0, 0;
}
Controllo di Ciclo
Talvolta, dovete uscire da un ciclo prima di avere esaurito le condizioni di iterazione. Potete usare i meccanismi standard di controllo del Perl 5--eccezioni e return--oppure le istruzioni di controllo di ciclo.
L'istruzione next fa ripartire il ciclo alla prossima iterazione. Usatelo quando avete già fatto tutto ciò che dovevate fare nell'iterazione corrente. Per scorrere le linee di un file saltando quelle che iniziano col carattere di commento #, scrivete:
while (<$fh>)
{
next if /\A#/;
...
}
Confrontate l'uso di next con l'alternativa: racchiudere il resto del corpo del blocco in un if. E ora considerate cosa succede se ci sono diverse condizioni che possono farvi saltare una linea. I modificatori di controllo di ciclo insieme ai condizionali postfissi possono rendere il vostro codice molto più leggibile.
L'istruzione last fa uscire immediatamente dal ciclo. Per finire di processare un file dopo avere incontrato un marcatore di terminazione, scrivete:
while (<$fh>)
{
next if /\A#/;
last if /\A__END__/
...
}
L'istruzione redo fa ripartire l'iterazione corrente senza valutare di nuovo la condizione. Questo può essere utile in quei pochi casi in cui volete modificare subito la linea che avete letto, e quindi far ripartire l'elaborazione dalla linea modificata senza leggere un'altra linea. Un esempio un po' stupido è il seguente parser di file che unisce tra loro le linee che terminano con un backslash:
while (my $linea = <$fh>)
{
chomp $linea;
# fai il match di backslash alla fine della linea
if ($linea =~ s{\\$}{})
{
$linea .= <$fh>;
chomp $linea;
redo;
}
...
}
Usare le istruzioni di controllo di ciclo nei cicli annidati può creare confusione. Se non potete evitare di annidare i cicli--trasformando i cicli interni in funzioni--usate una etichetta di ciclo per essere più chiari:
LINEA:
while (<$fh>)
{
chomp;
PREFISSO:
for my $prefisso (@prefissi)
{
next LINEA unless $prefisso;
say "$prefisso: $_";
# qui c'è un next PREFISSO implicito
}
}
Continue
Il costrutto continue si comporta come la terza sottoespressione di un ciclo for; Perl esegue il suo blocco prima di passare alla prossima iterazione del ciclo, sia nel caso di una ripetizione normale che nel caso di una re-iterazione prematura dovuta a next (footnote: L'equivalente Perl del continue del C è next.). Potete usarlo con un ciclo while, until, when e for. Gli esempi di uso di continue sono rari, ma è utile ogni volta che volete garantire che qualcosa sia eseguito ad ogni iterazione del ciclo indipendentemente da come si conclude tale iterazione:
while ($i < 10 )
{
next unless $i % 2;
say $i;
}
continue
{
say 'Continuo...';
$i++;
}
Ricordate che un blocco continue not viene eseguito quando il controllo di flusso abbandona un ciclo a causa di un last o di un redo.
Given/When
Il costrutto given è una nuova funzionalità di Perl 5.10. Esso assegna il valore di un'espressione alla variabile topic ed è seguito da un blocco:
given ($nome) { ... }
Al contrario di for, non itera su un aggregato. Valuta la propria espressione in contesto scalare, e assegna sempre alla variabile topic:
given (my $nome_utente = trova_utente())
{
is( $nome_utente, $_, 'assegnamento automatico a variabile topic' );
}
given lessicalizza la variabile topic:
given ('topo')
{
say;
da_topo_a_uomo( $_ );
say;
}
sub da_topo_a_uomo { s/topo/uomo/ }
given è particolarmente utile quando è usato in combinazione con when (Matching Intelligente(smart_match)). given topicalizza un valore dentro un blocco in modo che diverse istruzioni when possano fare il match del topic con delle espressioni usando una semantica di smart-match. Il seguente codice implementa il gioco di Sasso, Carta e Forbici:
my @opzioni = ( \&sasso, \&carta, \&forbici );
my $confuso = "Non capisco la tua mossa.";
do
{
say "Sasso, Carta e Forbici! Scegline uno: ";
chomp( my $mossa_utente = <STDIN> );
my $mossa_computer = $opzioni[ rand @opzioni ];
$mossa_computer->( lc( $mossa_utente ) );
} until (eof);
sub sasso
{
print "Ho scelto il sasso. ";
given (shift)
{
when (/carta/) { say 'Vinci tu!' };
when (/sasso/) { say 'Pareggio!' };
when (/forbici/) { say 'Vinco io!' };
default { say $confuso };
}
}
sub carta
{
print "Ho scelto la carta. ";
given (shift)
{
when (/carta/) { say 'Pareggio!' };
when (/sasso/) { say 'Vinco io!' };
when (/forbici/) { say 'Vinci tu!' };
default { say $confuso };
}
}
sub forbici
{
print "Ho scelto le forbici. ";
given (shift)
{
when (/carta/) { say 'Vinco io!' };
when (/sasso/) { say 'Vinci tu!' };
when (/forbici/) { say 'Pareggio!' };
default { say $confuso };
}
}
Perl esegue la regola di default quando il match con tutte le altre condizioni fallisce.
Il modulo CPAN MooseX::MultiMethods fornisce un'altra tecnica per semplificare questo codice.
Chiamate di coda
Una chiamata di coda ha luogo quando l'ultima espressione in una funzione è una chiamata a un'altra funzione--il valore di ritorno della funzione chiamante è il valore di ritorno della funzione chiamata:
sub traccia_e_saluta_persona
{
my $nome = shift;
log( "Sto salutando $nome" );
return saluta_persona( $nome );
}
Ritornare da saluta_persona() direttamente al chiamante di traccia_e_saluta_persona() è più efficiente che ritornare a traccia_e_saluta_persona() e poi ritornare immediatamente da traccia_e_saluta_persona(). Ritornare direttamente da saluta_persona() al chiamante di traccia_e_saluta_persona() è una ottimizzazione di chiamata di coda.
Il codice fortemente ricorsivo (Ricorsione(recursion)), specialmente quando ha luogo una mutua ricorsione, può consumare molta memoria. Le chiamate di coda riducono la quantità di memoria necessaria per il tracciamento interno del controllo di flusso e possono rendere trattabili degli algoritmi molto costosi. Purtroppo, Perl 5 non applica questa ottimizzazione automaticamente; dovete applicarla voi stessi quando lo ritenete necessario.
L'operatore nativo goto ha una forma per cui chiama una funzione come se la funzione corrente non fosse mai stata chiamata, cancellando in sostanza il tracciamento della chiamata alla nuova funzione. La sintassi poco elegante confonde le persone che hanno sentito dire di "Non usare mai goto", ma funziona:
sub traccia_e_saluta_persona
{
my ($nome) = @_;
log( "Sto salutando $nome" );
goto &saluta_persona;
}
Questo esempio mostra due aspetti importanti. Prima di tutto, goto &nome_funzione e goto &$riferimento_a_funzione richiedono di usare il sigillo di funzione (&) in modo che il parser sappia che deve effettuare una chiamata di coda anziché saltare semplicemente a un'etichetta. In secondo luogo, questa forma di chiamata di funzione passa implicitamente il contenuto di @_ alla funzione chiamata. Potete modificare @_ per cambiare gli argomenti passati.
Questa tecnica è usata piuttosto di rado; è particolarmente utile quando volete assumere voi stessi il controllo di flusso in modo da non disturbare altre funzioni che stanno ispezionando il chiamante (come quando state implementando speciali funzioni di logging o qualche sorta di funzionalità di debugging), o quando siete alle prese con un algoritmo che richiede molta ricorsione.
Scalari
Il tipo di dati fondamentale in Perl 5 è lo scalare, un valore discreto singolo. Tale valore può essere una stringa, un intero, un numero a virgola mobile, un filehandle o un riferimento--ma è sempre e comunque un singolo valore. Gli scalari possono essere variabili lessicali, di package o globali (Variabili Globali(globals)). Potete dichiarare soltanto le variabili lessicali o di package. I nomi delle variabili scalari devono conformarsi alle linee guida standard per i nomi di variabile (Nomi(names)). Le variabili scalari usano sempre, come sigillo (Sigilli delle Variabili(sigils)), il carattere del dollaro ($).
I valori scalari e il contesto scalare sono fortemente legati; assegnare a uno scalare fornisce un contesto scalare. Usare il sigillo scalare con una variabile aggregato impone il contesto scalare per accedere a un singolo elemento dell'hash o dell'array.
Scalari e Tipi
Una variabile scalare può contenere qualunque tipo di valore scalare senza richiedere conversioni o cast speciali, e il tipo del valore memorizzato in una variabile può cambiare:
my $valore;
$valore = 123.456;
$valore = 77;
$valore = "Sono l'alluce di Max.";
$valore = Negozio::Gelato->new();
Anche se questo codice è legale, cambiare il tipo di dato memorizzato in uno scalare è segno di confusione.
La flessibilità di tipo spesso causa una coercizione del valore (Coercizione(coercion)). Per esempio, potreste trattare il contenuto di uno scalare come se fosse una stringa, anche se non gli avete assegnato esplicitamente una stringa:
my $codice_postale = 10100;
my $citta_regione_cap = 'Torino, Piemonte' . ' ' . $codice_postale;
Potete anche applicare operazioni matematiche a delle stringhe:
my $codice_radio = 'KBMIU';
# aggiorna codice e restituisci il nuovo valore
my $prossimo_codice = ++$codice_radio;
# restituisci il vecchio valore, e quindi aggiorna il codice
my $codice_attuale = $codice_radio++;
# così non funziona:
my $nuovo_codice = $codice_radio + 1;
Questo incremento magico di stringa non ha un corrispondente decremento. Non potete ottenere il valore stringa precedente scrivendo $codice_radio--.
L'operazione di incremento di stringa fa diventare la a una b e la z una aa, rispettando l'insieme di caratteri e la distinzione maiuscole/minuscole. Mentre ZZ9 diventa AAA0, ZZ09 diventa ZZ10--i numeri danno luogo a riporto fino a che ci sono cifre più significative da incrementare, come sul contachilometri di un'automobile.
Valutare un riferimento (Riferimenti(references)) in contesto stringa produce una stringa. Valutare un riferimento in contesto numerico produce un numero. Nessuna delle due operazioni modifica il riferimento, ma non potete riottenerlo dai loro risultati:
my $autori = [qw( Pratchett Vinge Conway )];
my $riferimento_come_stringa = '' . $autori;
my $riferimento_come_numero = 0 + $autori;
$autori è ancora un riferimento valido, ma $riferimento_come_stringa è una stringa che non ha connessioni con il riferimento e $riferimento_come_numero è un numero che non ha connessioni con il riferimento.
Per permettere di effettuare coercizioni senza perdere dei dati, gli scalari del Perl 5 possono contenere componenti sia numeriche che stringa. In Perl 5, la struttura dati interna per rappresentare uno scalare ha sia uno slot numerico che uno slot stringa. Accedere a una stringa in contesto numerico produce uno scalare con valori sia numerico che stringa. La funzione dualvar() del modulo core Scalar::Util ci permette di manipolare entrambi i valori direttamente in un singolo scalare.
Gli scalari non contengono uno slot separato per i valori booleani. In contesto booleano, la stringa vuota ('') e la stringa '0' hanno valore falso. Tutte le altre stringhe hanno valore vero. In contesto booleano, i numeri che hanno valore zero (0, 0.0, e 0e0) rappresentano il valore booleano falso. Tutti gli altri numeri rappresentano il valore vero.
Attenzione al fatto che le stringhe '0.0' e '0e0' hanno valore vero; questo è uno dei casi in cui Perl 5 fa una distinzione fra ciò che sembra un numero e ciò che è veramente un numero.
C'è un altro valore che è sempre falso: undef. È il valore delle variabili non inizializzate ma è anche un valore di per sé.
Array
Gli array del Perl 5 sono una struttura dati di prima classe--il linguaggio li supporta come tipo di dati nativo--che memorizza zero o più scalari. Potete accedere a singoli membri dell'array con degli indici interi, e potete aggiungere o rimuovere elementi a piacere. Il sigillo @ denota un array. Per dichiarare un array, scrivete:
my @elementi;
Elementi di un Array
Accedere a un singolo elemento di un array in Perl 5 richiede il sigillo scalare. $gatti[0] è un uso non ambiguo dell'array @gatti, dato che le parentesi quadre ([]) postfisse (Posizione(fixity)) indicano sempre un accesso indicizzato a un array.
Il primo elemento di un array ha indice zero:
# @gatti contiene una lista di oggetti Gatto
my $primo_gatto = $gatti[0];
L'ultimo indice di un array dipende dal numero di elementi nell'array. In contesto scalare (indotto da un assegnamento scalare, una concatenzaione di stringhe, un'addizione o un contesto booleano), un array valuta al numero di elementi nell'array:
# assegnamento scalare
my $numero_gatti = @gatti;
# concatenazione di stringhe
say 'Ho ' . @gatti . ' gatti!';
# addizione
my $num_animali = @gatti + @cani + @pesci;
# contesto booleano
say 'Esatto, sono un padrone di gatti!' if @gatti;
Per ottenere l'indice dell'ultimo elemento di un array, sottraete uno dal numero di elementi dell'array (ricordate che gli indici dell'array iniziano da 0) oppure usate la meno elegante sintassi $#cats:
my $primo_indice = 0;
my $ultimo_indice = @gatti - 1;
# oppure
# my $ultimo_indice = $#gatti;
say "Il mio primo gatto ha indice $primo_indice, "
. "e il mio ultimo gatto ha indice $ultimo_indice."
Quando l'indice è meno importante della posizione di un elemento, usate degli indici di array negativi. L'ultimo elemento di un array è accessibile all'indice -1. Il penultimo elemento dell'array è accessibile all'indice -2, e così via:
my $ultimo_gatto = $gatti[-1];
my $penultimo_gatto = $gatti[-2];
$# ha un ulteriore uso: assegnandogli un valore si può ridimensionare un array. Ricordate che gli array in Perl 5 sono mutabili. Si espandono o contraggono secondo la necessità. Quando restringete un array, Perl scarterà i valori per cui non c'è posto nell'array ridimensionato. Quando espandete un array, Perl riempirà le posizioni espanse di undef.
Assegnamento ad Array
Potete fare assegnamenti a singole posizioni in un array usando gli indici:
my @gatti;
$gatti[3] = 'Jack';
$gatti[2] = 'Tuxedo';
$gatti[0] = 'Daisy';
$gatti[1] = 'Petunia';
$gatti[4] = 'Brad';
$gatti[5] = 'Choco';
Se assegnate un valore a un indice oltre i limiti attuali dell'array, Perl estende l'array alla dimensione giusta e riempie tutte le posizioni intermedie con undef. Dopo il primo assegnamento nell'esempio qui sopra, l'array conterrà undef nelle posizioni 0, 1 e 2 e Jack nella posizione 3.
Come abbreviazione, potete inizializzare un array con una lista:
my @gatti = ( 'Daisy', 'Petunia', 'Tuxedo', ... );
... ma ricordate che non sono le parentesi a creare una lista. Senza parentesi, questa istruzione assegnerebbe Daisy al primo e unico elemento dell'array, a causa delle precedenze tra operatori (Precedenza(precedence)).
Un'espressione qualunque che produce una lista in contesto lista può essere assegnata ad un array:
my @gatti = elenca_i_gatti();
my @info_tempo = localtime();
my @numeri = 1 .. 10;
Assegnare a un elemento scalare di un array impone il contesto scalare, mentre assegnare all'intero array impone il contesto lista.
Per svuotare un array, assegnategli la lista vuota:
my @date = ( 1969, 2001, 2010, 2051, 1787 );
...
@date = ();
my @elementi = (); è una versione più lunga e verbosa di my @elementi, dato che gli array appena dichiarati sono vuoti.
Operazioni sugli Array
Qualche volta è più conveniente vedere un array come una collezione ordinata e mutabile di elementi piuttosto che come una mappatura tra indici e valori. Perl 5 fornisce diverse operazioni per manipolare gli elementi di un array senza usare gli indici.
Gli operatori push e pop aggiungono e rimuovono, rispettivamente, degli elementi dalla coda di un array:
my @piatti;
# cosa c'è da mangiare?
push @piatti, qw( hamburger pizza lasagne rape );
# ... ma vostro nipote odia la verdura
pop @piatti;
Potete fare il push di una lista di valori in un array, ma potete solo fare il pop di un elemento per volta. push restituisce il nuovo numero di elementi dell'array. pop restituisce l'elemento rimosso.
Dato che push opera su una lista, potete facilmente appendere gli elementi di uno o più array tra di loro:
push @piatti, @colazione, @pranzo, @cena;
Analogamente, unshift e shift aggiungono e rimuovono, rispettivamente, degli elementi dalla testa di un array:
# espandiamo i nostri orizzonti culinari
unshift @piatti, qw( tofu spanakopita taquitos );
# un ripensamento su quell'idea della soia
shift @piatti;
unshift antepone una lista di elementi alla testa di un array e restituisce il nuovo numero di elementi dell'array. shift rimuove e restituisce il primo elemento dell'array.
Pochi programmi usano i valori di ritorno di push e unshift.
L'operatore splice rimuove e sostituisce elementi di un array dato un offset, la lunghezza di una slice di lista e i sostituti. Sia la sostituzione che la rimozione sono opzionali; potete omettere sia l'uno che l'altro. La descrizione di splice in perlfunc dimostra la sua equivalenza con push, pop, shift e unshift. Un uso particolarmente efficace è la rimozione di due elementi da un array:
my ($vincitore, $secondo) = splice @finalisti, 0, 2;
# oppure
my $vincitore = shift @finalisti;
my $secondo = shift @finalisti;
Prima del Perl 5.12, iterare su un array per indice richiedeva un ciclo in stile C. A partire da Perl 5.12, each può iterare su un array per indice e valore:
while (my ($indice, $valore) = each @libri)
{
say "#$indice: $valore";
...
}
Slice di Array
Il costrutto di array slice vi permette di accedere agli elementi di un array in contesto lista. A differenza dell'accesso scalare a un elemento dell'array, questa operazione di indicizzazione riceve una lista di zero o più indici e usa il sigillo array (@):
my @gatti_giovani = @gatti[-1, -2];
my @gatti_anziani = @gatti[0 .. 2];
my @selezione_di_gatti = @gatti[ @indici ];
Le slice di array possono essere utili negli assegnamenti:
@utenti[ @indici_sostituzioni ] = @sostituisci_utenti;
Una slice può contenere zero o più elementi--o anche uno solo:
# slice di array con un singolo elemento; contesto lista
@gatti[-1] = dammi_altri_gatti();
# accesso a un singolo elemento dell'array; contesto scalare
$gatti[-1] = dammi_altri_gatti();
L'unica differenza sintattica tra una slice di array con un solo elemento e l'accesso scalare a un elemento dell'array è il sigillo usato. La differenza semantica è maggiore: una slice di array impone sempre il contesto lista. Una slice di array valutata in contesto scalare produce un warning:
Scalar value @gatti[1] better written as $gatti[1] at...
(NdT: tradotto sarebbe "È meglio scrivere il valore scalare @gatti[1] come $gatti[1]...").
Una slice di array impone il contesto lista anche all'espressione che usa come indice:
# la chiamata di funzione è in contesto lista
my @gatti_affamati = @gatti[ indici_di_gatti() ];
Array e Contesto
In contesto lista, gli array diventano liste. Se passate diversi array a una normale funzione Perl 5, essi si appiattiscono in un'unica lista:
my @gatti = qw( Daisy Petunia Tuxedo Brad Jack );
my @cani = qw( Rodney Lucky );
porta_dal_veterinario( @gatti, @cani );
sub porta_dal_veterinario
{
# SBAGLIATO: non usatelo!
my (@gatti, @cani) = @_;
...
}
Nella funzione, @_ conterrà sette elementi, non due, poiché l'assegnamento di una lista ad un array è greedy. Un array consuma più elementi possibile dalla lista. Dopo l'assegnamento, @gatti conterrà tutti gli argomenti passati alla funzione. @cani invece sarà vuoto.
Questo appiattimento può confondere i principianti che tentano di creare array annidati in Perl 5:
# crea un singolo array, non un array di array
my @numeri = ( 1 .. 10,
( 11 .. 20,
( 21 .. 30 ) ) );
... ma questo codice è a tutti gli effetti equivalente a questo:
# crea un singolo array, non un array di array
my @numeri = 1 .. 30;
... perché le parentesi raggruppano semplicemente le espressioni. Non creano liste, in questo caso. Per ovviare a questo appiattimento, usate i riferimenti ad array (Riferimenti ad Array(array_references)).
Interpolazione di Array
Gli array vengono interpolati in stringhe come liste della conversione in stringa di ciascun elemento separate dal valore corrente della variabile globale predefinita $". Il valore di default di tale variabile è lo spazio singolo. Il suo mnemonico in English.pm è $LIST_SEPARATOR. Quindi:
my @alfabeto = 'a' .. 'z';
say "[@alfabeto]";
[a b c d e f g h i j k l m
n o p q r s t u v w x y z]
Localizzate $" con un delimitatore per facilitare il debugging (footnote: Il credito per questa tecnica va a Mark Jason Dominus.):
# cosa c'è già in questo array?
local $" = ')(';
say "(@dolciumi)";
(crostata)(torta)(ciambelle)(biscotti)(pane dolce)
Hash
Un hash è una struttura dati nativa del Perl che associa delle chiavi stringa a dei valori scalari. Allo stesso modo in cui una variabile corrisponde a un'area di memoria, una chiave di un hash si riferisce a un valore. Pensate a un hash come se fosse una guida telefonica, con cui potete usare i nomi dei vostri amici per trovare i loro numeri di telefono. In altri linguaggi gli hash sono detti tablelle, array associativi, dizionari o mappe.
Gli hash hanno due proprietà importanti: memorizzano uno scalare per ogni chiave univoca e non impongono alcun ordine specifico sulle chiavi.
Dichiarazione di Hash
Gli hash usano il sigillo %. Dichiarate un hash lessicale con:
my %gusti_preferiti;
Un hash appena dichiarato è vuoto. Potreste anche scrivere my %gusti_preferiti = ();, ma sarebbe superfluo.
Gli hash usano il sigillo $ quando accedete a singoli elementi e le parentesi graffe { } per specificare la chiave:
my %gusti_preferiti;
$gusti_preferiti{Gabri} = 'Cioccolato';
$gusti_preferiti{Anna} = 'Vaniglia';
È possibile assegnare una lista di chiavi e valori ad un hash in un'unica espressione:
my %gusti_preferiti = (
'Gabri', 'Cioccolato',
'Anna', 'Vaniglia',
);
Se assegnate un numero dispari di elementi ad un hash, riceverete un warning. Un idioma del Perl suggerisce di usare l'operatore fat comma (=>) per associare chiavi e valori, poiché rende più esplicite tali associazioni:
my %gusti_preferiti = (
Gabri => 'Menta e cioccolato',
Anna => 'Vaniglia',
);
L'operatore fat comma funziona come la virgola, ma quota anche la bareword (Bareword(barewords)) che lo precede. La direttiva strict non genererà un warning per la presenza di tale bareword--e se avete una funzione con lo stesso nome di una chiave dell'hash, l'operatore fat comma non chiamerà la funzione:
sub nome { 'Leonardo' }
my %indirizzo =
(
nome => 'Via Fibonacci 1123',
);
La chiave in questo hash sarà nome e non Leonardo. Se volete chiamare la funzione, dovete farlo esplicitamente:
my %indirizzo =
(
nome() => 'Via Fibonacci 1123',
);
Per svuotare un hash, assegnategli la lista vuota (footnote: Potreste vedere occasionalmente anche undef %hash.):
%gusti_preferiti = ();
Indicizzazione di Hash
Per accedere ai singoli valori di un hash dovete eseguire un'operazione di indicizzazione. Usate una chiave (operazione di accesso per chiave) per estrarre un valore da un hash:
my $indirizzo = $indirizzi{$nome};
In questo esempio, $nome contiene una stringa che è anche una chiave dell'hash. Come per l'accesso a un singolo elemento di un array, il sigillo dell'hash è cambiato da % a $ per indicare l'accesso per chiave a un valore scalare.
Potete anche usare letterali stringa come chiavi di hash. Perl quoterà le bareword automaticamente seguendo le stesse regole del fat comma:
# quotato automaticamente
my $indirizzo = $indirizzi{Vittorio};
# quotatura necessaria; non è una bareword valida
my $indirizzo = $indirizzi{'Maria Luisa'};
# una chiamata di funzione deve essere disambiguata
my $indirizzo = $indirizzi{dammi_un_nome()};
I principianti tendono a quotare i letterali stringa che usano come chiavi di hash, mentre i programmatori esperti evitano la quotatura quando è possibile. La quotatura delle chiavi di hash viene usata per segnalare l'intenzione di fare qualcos'altro.
Anche le funzioni native del Perl 5 sono soggette alla quotatura automatica:
my %indirizzi =
(
Leonardo => 'Via Fibonacci 1123',
Utako => 'Hotel Cantor, Stanza 1',
);
sub da_nome_a_indirizzo
{
return $indirizzi{+shift};
}
L'operatore più unario (Coercizioni Unarie(unary_coercions)) converte quella che sarebbe una bareword (shift) soggetta alla quotatura automatica in un'espressione. Da questo esempio si vede che potete usare una espressione arbitraria--non solo una chiamata di funzione--come chiave di un hash:
# non fatelo davvero, per favore
my $indirizzo = $indirizzi{reverse 'odranoeL'};
# l'interpolazione funziona
my $indirizzo = $indirizzi{"$nome $cognome"};
# e anche le chiamate di metodi
my $indirizzo = $indirizzi{ $utente->nome() };
Le chiavi di un hash possono solo essere stringhe. Qualunque cosa valuti ad una stringa è una chiave di hash ammissibile. Perl si spinge fino a fare la coercizione (Coercizione(coercion)) di altri tipi nel tipo stringa, in modo che se usate un oggetto come chiave di hash, ottenete la versione stringa dell'oggetto anziché l'oggetto stesso:
for my $isbn (@lista_isbn)
{
my $libro = Libro->per_isbn( $isbn );
# difficile che sia quello che volevate fare
$libri{$libro} = $libro->prezzo;
}
Esistenza di Chiavi di Hash
L'operatore exists restituisce un valore booleano che indica se un hash contiene una data chiave:
my %indirizzi =
(
Leonardo => 'Via Fibonacci 1123',
Utako => 'Hotel Cantor, Stanza 1',
);
say "Ho l'indirizzo di Leonardo"
if exists $indirizzi{Leonardo};
say "Ho l'indirizzo di Galileo"
if exists $indirizzi{Galileo};
Usare exists anziché tentare di accedere direttamente a una chiave di hash evita due problemi. Prima di tutto, il controllo non viene effettuato sul valore associato alla chiave; infatti una chiave di hash potrebbe esistere e avere un valore anche se tale valore corrisponde a falso in contesto booleano (per esempio undef):
my %valore_chiave_falso = ( 0 => '' );
ok( %valore_chiave_falso,
'un hash che contiene falso sia come chiave che come valore
dovrebbe valutare a vero' );
Inoltre, exists evita l'autovivificazione (Autovivificazione(autovivification)) nelle strutture dati annidate (Strutture Dati Annidate(nested_data_structures)).
Se una chiave di hash esiste, il suo valore potrebbe essere undef. Potete controllarlo con defined:
$indirizzi{Leibniz} = undef;
say "Gottfried vive a $indirizzi{Leibniz}"
if exists $indirizzi{Leibniz}
&& defined $indirizzi{Leibniz};
Accedere a Chiavi e Valori degli Hash
Gli hash sono variabili aggregate, ma la loro struttura a coppie offre diverse possibilità di iterazione: sulle chiavi dell'hash, sui valori dell'hash o sulle coppie chiave/valore. L'operatore keys produce una lista di chiavi di un hash:
for my $destinatario (keys %indirizzi)
{
say "Trovato l'indirizzo di $destinatario!";
}
L'operatore values produce una lista di valori di un hash:
for my $indirizzo (values %indirizzi)
{
say "Qualcuno vive in $indirizzo";
}
L'operatore each produce una lista di liste di due elementi contenenti una chiave e il suo valore:
while (my ($destinatario, $indirizzo) = each %indirizzi)
{
say "$destinatario vive in $indirizzo";
}
Al contrario degli array, non c'è un ordinamento ovvio per queste liste. L'ordinamento dipende dall'implementazione interna dell'hash, la versione di Perl che state usando, la dimensione dell'hash e un ulteriore fattore random. Ciò nonostante, l'ordine degli elementi imposto da keys, values e each è consistente. Modificando l'hash l'ordine potrebbe cambiare, ma se l'hash rimane immutato potete fare affidamento su tale ordine.
Ogni hash ha un unico iteratore per l'operatore each. Non potete iterare su un hash con più di un each senza incorrere in problemi; se iniziate una nuova iterazione mentre un'altra è in corso, la prima terminerà prematuramente e la seconda inizierà da qualche parte in mezzo all'hash. Durante una iterazione each, fate attenzione a non chiamare funzioni che possano a loro volta tentare di iterare sull'hash con each.
In pratica questo succede raramente, ma quando vi dovesse servire resettate l'iteratore di un hash con keys o values in contesto void:
# resetta l'iteratore dell'hash
keys %indirizzo;
while (my ($destinatario, $indirizzo) = each %indirizzi)
{
...
}
Slice di Hash
Una slice di hash è una lista di chiavi o valori di un hash indicizzata in una singola operazione. Per inizializzare diversi elementi di un hash in una sola istruzione:
# %gatti contiene già altri elementi
@gatti{qw( Jack Brad Mars Grumpy )} = (1) x 4;
Questo codice è equivalente all'inizializzazione:
my %gatti = map { $_ => 1 }
qw( Jack Brad Mars Grumpy );
...eccetto che l'inizializzazione di uno slice di hash non sostituisce i contenuti di un hash che esistono già.
Le slice di hash vi permettono anche di estrarre diversi valori da un hash in un'unica operazione. Come per le slice di array, il sigillo dell'hash cambia per indicare il contesto lista. L'uso delle parentesi graffe indica accesso per chiave rendendo non ambigua la presenza di un hash:
my @indirizzi_compratori = @indirizzi{ @compratori };
Le slice di hash rendono facile anche fondere due hash insieme:
my %indirizzi = ( ... );
my %indirizzi_canadesi = ( ... );
@indirizzi{ keys %indirizzi_canadesi }
= values %indirizzi_canadesi;
Questo codice è equivalente a ciclare manualmente sui contenuti di %indirizzi_canadesi, ma è molto più compatto.
Che cosa succede se la stessa chiave compare in entrambi gli hash? L'uso di uno slice di hash sovrascrive sempre le coppie chiave/valore in %indirizzi. Se desiderate un comportamento diverso, è più appropriato ciclare manualmente.
L'Hash Vuoto
Un hash vuoto non contiene né chiavi né valori. In contesto booleano ha valore falso. Un hash che contiene almeno una coppia chiave/valore ha invece valore vero in contesto booleano anche se tutte le chiavi o tutti i valori o entrambi avessero valore falso in contesto booleano.
use Test::More;
my %vuoto;
ok( ! %vuoto, 'un hash vuoto dovrebbe avere valore falso' );
my %chiave_falsa = ( 0 => 'valore vero' );
ok( %chiave_falsa, 'un hash contenente falso come chiave
dovrebbe avere valore vero' );
my %valore_falso = ( 'chiave vera' => 0 );
ok( %valore_falso, 'un hash contenente falso come valore
dovrebbe avere valore vero' );
done_testing();
In contesto scalare, un hash valuta a una stringa che rappresenta la frazione di posizioni occupate nell'hash--ma potete tranquillamente ignorare questi dettagli dell'implementazione interna dell'hash.
In contesto lista, un hash valuta a una lista di coppie chiave/valore simile a quella restituita dall'operatore each. Tuttavia, non potete iterare su tale lista allo stesso modo in cui potete iterare sulla lista prodotta da each, poiché il ciclo non terminerebbe mai:
# ciclo infinito per hash non vuoti
while (my ($chiave, $valore) = %hash)
{
...
}
È possibile ciclare su questa lista di chiavi e valori con un ciclo for, ma la variabile iteratore conterrà una chiave ad una iterazione e il suo valore in quella successiva, poiché Perl appiattisce l'hash in un'unica lista di chiavi e valori alternati.
Idiomi degli Hash
Poiché ogni chiave può comparire una sola volta in un hash, assegnare la stessa chiave ad un hash più di una volta memorizza solo il valore più recente. Potete sfruttare questo fatto per trovare gli elementi distinti di una lista:
my %unic;
undef @unic{ @elementi };
my @unici = keys %unic;
L'uso di undef con uno slice di hash setta i valori dell'hash a undef. Questo idioma rappresenta il modo più efficiente di effettuare operazioni insiemistiche con un hash.
Gli hash sono utili anche per contare degli elementi, come gli indirizzi IP in un file di log:
my %indirizzi_ip;
while (my $linea = <$file_di_log>)
{
my ($ip, $risorsa) = analizza_linea( $linea );
$indirizzi_ip{$ip}++;
...
}
Inizialmente ogni valore di un hash è undef. L'operatore di postincremento (++) lo tratta come se fosse zero. In questo esempio, se il valore della chiave esiste già viene rimpiazzato con il suo incremento. Se invece non esiste alcun valore per la chiave, Perl crea un valore (undef) e lo incrementa immediatamente a uno, dato che il valore numerico di undef è 0.
La seguente strategia costituisce un utile meccanismo di caching per memorizzare il risultato di un'operazione costosa con un piccolo overhead:
{
my %cache_utenti;
sub restituisci_utente
{
my $id = shift;
$cache_utenti{$id} //= crea_utente($id);
return $cache_utenti{$id};
}
}
Questa manovra orchescese (footnote: NdT: in inglese è un gioco di parole tra orcish (manovra da orchi) e or-cache.) restituisce il valore dell'hash, se esso esiste. Altrimenti calcola, mette in cache e restituisce il valore. L'operatore di assegnamento defined-or (//=) valuta il suo operando di sinistra. Se tale operando non è definito, l'operatore gli assegna il valore del suo operando di destra. In altre parole, se non c'è un valore nell'hash per una data chiave, questa funzione chiama crea_utente() passandole la chiave e poi aggiorna l'hash.
Perl 5.10 ha introdotto gli operatori di defined-or e di assegnamento defined-or. Prima del 5.10, gran parte del codice usava l'operatore di assegnamento boolean-or (||=) per lo stesso scopo. Purtroppo, alcuni valori validi valutano a falso in contesto booleano, quindi valutare la definitezza dei valori è quasi sempre più accurato. Essendo pigra, la manovra orchescese testa la definitezza del valore in cache, non la sua verità.
Se la vostra funzione accetta diversi argomenti, sfruttate lo slurping degli argomenti (Slurping(parameter_slurping)) per raccogliere delle coppie chiave/valore in un unico hash trattandole come argomenti passati per nome alla funzione:
sub prepara_coppa
{
my %parametri = @_;
...
}
prepara_coppa( gusto => 'Limone',
glassa => 'biscotti macinati' );
Questo approccio vi permette anche di settare dei valori di default, in questo modo:
sub prepara_coppa
{
my %parametri = @_;
$parametri{gusto} //= 'Vaniglia';
$parametri{glassa} //= 'cioccolato';
$parametri{praline} //= 100;
...
}
... oppure includendoli direttamente nell'inizializzazione dell'hash, dato che un assegnamento in una certa posizione sovrascrive gli assegnamenti che lo precedono:
sub prepara_coppa
{
my %parametri =
(
gusto => 'Vaniglia',
glassa => 'cioccolato',
praline => 100,
@_,
);
...
}
Bloccare gli Hash
Dato che le chiavi di hash sono delle bareword, offrono poca protezione contro gli errori di scrittura rispetto alla protezione offerta dalla direttiva strict sui nomi di funzione e di variabile. Il poco conosciuto modulo core Hash::Util fornisce dei meccanismi per migliorare la situazione.
Per prevenire l'inserimento accidentale di una chiave di hash indesiderata (a causa di una svista o di input utente inaffidabile), usate la funzione lock_keys() per congelare l'hash al suo insieme corrente di chiavi. Ogni tentativo di aggiungere una nuova chiave all'hash solleverà un'eccezione. Questa è ovviamente una sicurezza debole, adatta solo a prevenire errori accidentali; chiunque può usare la funzione unlock_keys() per rimuovere questa protezione.
In modo simile potete bloccare o sbloccare il valore corrente di una data chiave dell'hash (lock_value() e unlock_value()) e impostare o rilasciare il modo read-only per l'intero hash con lock_hash() e unlock_hash().
Coercizione
Una variabile Perl può contenere diversi tipi di valori in momenti diversi--stringhe, interi, numeri razionali, ecc.. Anziché associare un'informazione di tipo alle variabili, Perl sfrutta il contesto fornito dagli operatori (Contesti Numerico, Stringa e Booleano(value_contexts)) per decidere come deve interpretare i loro valori. Per scelta di progetto, Perl tenta di fare quello che intendete (footnote: Abbreviato come DWIM, che sta per do what I mean (fai quello che intendo).), quindi dovete essere sufficientemente specifici sulle vostre intenzioni. Se trattate una variabile che contiene un numero come se fosse una stringa, Perl farà del suo meglio per forzare quel numero a una stringa con una coercizione.
Coercizione Booleana
La coercizione booleana ha luogo quando testate la verità di un valore, per esempio nella condizione di un if o di un while. Il numero 0, undef, la stringa vuota e la stringa '0' valutano tutti a falso. Tutti gli altri valori--incluse le stringhe che sarebbero numericamente uguali a zero come '0.0', '0e' e '0 ma vero'--valutano a vero.
Qaundo uno scalare ha sia una componente stringa che una componente numerica (Dualvar(dualvars)), Perl 5 preferisce testare la verità booleana della componente stringa. '0 ma vero' valuta a zero come numero, ma non è una stringa vuota, quindi in contesto booleano valuta a vero.
Coercizione a Stringa
La coercizione a stringa ha luogo quando usate degli operatori stringa come il confronto (eq e cmp), la concatenazione, split, substr e le espressioni regolari, e quando usate un valore come chiave di hash. Il valore undef diventa la stringa vuota e produce un warning di "use of uninitialized value" ("uso di valore non inizializzato"). I numeri diventano stringhe contenenti i loro valori, per esempio il valore 10 diventa la stringa '10'. Potete fare lo split di un numero in singole cifre con:
my @cifre = split '', 1234567890;
Coercizione Numerica
La coercizione numerica ha luogo quando usate operatori di confronto numerico (come == e <=>), quando effettuate operazioni matematiche e quando usate un valore come indice di un array o di una lista. Il valore undef diventa il numero zero e produce un warning di "Use of uninitialized value". Anche le stringhe che non iniziano con una parte numerica diventano il numero zero e producono un warning di "Argument isn't numeric" ("L'argomento non è numerico"). Le stringhe che iniziano con caratteri permessi nei letterali numerici diventano dei numeri corrispondenti a tali caratteri e non producono warning; per esempio 10 salti di leptoni diventa 10 e 6.022e23 scorrerie di talpe diventa 6.022e23.
Il modulo core Scalar::Util contiene una funzione looks_like_number() che usa le stesse regole di parsing della grammatica del Perl 5 per estrarre un numero da una stringa.
Le stringhe Inf e Infinity rappresentano il valore infinito e si comportano come numeri. La stringa NaN rappresenta il concetto "not a number" ("non è un numero"). In contesto numerico non danno luogo al warning "Argument isn't numeric".
Coercizione di Riferimento
Usare un'operazione di dereferenziazione su un valore che non sia un riferimento trasforma quel valore in un riferimento. Questo processo di autovivificazione (Autovivificazione(autovivification)) è utile quando manipolate strutture dati annidate (Strutture Dati Annidate(nested_data_structures)):
my %utenti;
$utenti{Paolo}{id} = 228;
$utenti{Giacomo}{id} = 229;
Anche se l'hash non conteneva valori per Paolo e Giacomo, Perl si è preso il disturbo di creare dei riferimenti ad hash in queste chiavi e di assegnare una coppia chiave/valore con chiave id a ciascuno di essi.
Cache di Coercizioni
La rappresentazione interna dei valori in Perl 5 memorizza sia un valore stringa che un valore numerico. La coercizione di un valore numerico a stringa non rimpiazza il valore numerico. Invece, attacca un nuovo valore stringa, in modo che la rappresentazione contiene entrambi i componenti. Analogamente, la coercizione di una stringa a numero popola la componente numerica lasciando la componente stringa inalterata.
Certe operazioni del Perl preferiscono usare una componente di un valore piuttosto che l'altra--per esempio, i test booleani preferiscono la stringa. Se un valore si è memorizzato in cache la rappresentazione in una forma che non vi aspettate, affidarvi a una conversione implicita può dare delle sorprese. Non avrete quasi mai la necessità di esplicitare che cosa vi aspettate (footnote: Il vostro autore ricorda di averlo dovuto fare solo due volte in oltre dieci anni di programmazione in Perl 5), ma sapere che questo caching ha luogo potrebbe aiutarvi una volta o l'altra a fare la diagnosi di qualche strana situazione.
Dualvar
La natura multi-componente dei valori Perl è disponibile agli utenti in forma di variabili doppie. Il modulo core Scalar::Util fornisce una funzione dualvar() che vi permette di bypassare la coercizione del Perl e manipolare separatamente le componenti stringa e numerica di un valore:
use Scalar::Util 'dualvar';
my $nome_falso = dualvar 0, 'Gioie & Dolori';
say 'Vero come booleano!' if !! $nome_falso;
say 'Falso come numero!' unless 0 + $nome_falso;
say 'Vero come stringa!' if '' . $nome_falso;
Package
Un namespace in Perl associa e incapsula varie entità con nome all'interno di una categoria con un proprio nome, come potrebbe essere il vostro cognome o il nome di una marca. Al contrario di un nome nel mondo reale, un namespace non implica alcuna relazione diretta tra le entità. È possibile, ma non necessario, che vi siano tali relazioni.
In Perl 5 un package è una collezione di codice contenuto in un singolo namespace. La distinzione è sottile: il package rappresenta il codice sorgente e il namespace rappresenta l'entità creata quando Perl effettua il parsing di tale codice.
La funzione nativa package dichiara un package e un namespace:
package MioCodice;
our @scatole;
sub aggiungi_scatola { ... }
Tutte le variabili e funzioni globali che dichiarate o a cui fate riferimento dopo la dichiarazione del package si riferiscono a simboli del namespace MioCodice. Potete riferirvi alla variabile @scatole dal namespace main solo col suo nome qualificato @MioCodice::scatole. Un nome qualificato include un nome completo di package, quindi potete chiamare la funzione aggiungi_scatola() soltanto con MioCodice::aggiungi_scatola().
Lo scope di un package continua fino alla successiva dichiarazione package oppure fino alle fine del file, a seconda di ciò che avviene per primo. Perl 5.14 ha potenziato package in modo che potete fornirgli un blocco che delimita esplicitamente lo scope della dichiarazione:
package Giochi::Flipper
{
our $VERSION = 1969;
}
Il package di default è il package main. In assenza di una dichiarazione di package, il package corrente è main. Questa regola si applica tanto ai programmi a linea di comando quanto ai programmi standalone e anche ai file .pm.
Oltre al nome, un package ha una versione e tre metodi impliciti: import() (Importazione(importing)), unimport() e VERSION(). VERSION() restituisce il numero di versione del package. Questo numero è in realtà una serie di numeri contenuti in una variabile globale del package di nome $VERSION. Per convenzione, le versioni tendono a essere una serie di interi separati dal punto, come 1.23 o 1.1.10, dove ogni componente è un intero.
Perl 5.12 ha introdotto una nuova sintassi intesa a semplificare i numeri di versione, come documentato in perldoc version::Internals. Questi numeri di versione con forma più restrittiva devono iniziare con il carattere v e contenere almeno tre componenti intere separate da punti:
package MioCodice v1.2.1;
In Perl 5.14, la forma opzionale a blocco di una dichiarazione di package è:
package Giochi::Flipper v1969.3.7
{
...
}
Nella versione 5.10 e precedenti, il modo più semplice per dichiarare la versione di un package è:
package MioCodice;
our $VERSION = 1.21;
Ogni package eredita un metodo VERSION() dalla classe base UNIVERSAL. Potete fare l'override di VERSION(), anche se non ci sono molti motivi per volerlo fare. Tale metodo restituisce il valore di $VERSION:
my $versione = Qualche::Plugin->VERSION();
Se gli passate un numero di versione come argomento, il metodo solleverà un'eccezione a meno che la versione del modulo sia uguale o successiva all'argomento:
# richiesta almeno versione 2.1
Qualche::Plugin->VERSION( 2.1 );
die "Il tuo plugin $version è troppo vecchio"
unless $version > 2;
Package e Namespace
Ogni dichiarazione package crea, se necessario, un nuovo namespace e impone al parser di inserire tutti i successivi simboli globali del package (variabili globali e funzioni) in tale namespace.
Perl ha dei namespace aperti. Potete aggiungere funzioni e variabili a un namespace in qualunque punto, sia con una dichiarazione di package:
package Pacco
{
sub prima_sub { ... }
}
Pacco::prima_sub();
package Pacco
{
sub seconda_sub { ... }
}
Pacco::seconda_sub();
... che specificando dei nomi di funzione qualificati al momento della dichiarazione:
# implicito
package main;
sub Pacco::terza_sub { ... }
Potete fare aggiunte a un package in ogni punto sia durante la compilazione che a runtime, indipendentemente dal file corrente, sebbene costruire un package con diverse dichiarazioni separate può rendere il codice difficile da esplorare.
I namespace possono avere tanti livelli quanti ne richiede il vostro schema organizzativo, ma i namespace non sono gerarchici. L'unica relazione tra i package è semantica, non tecnica. Molti progetti e aziende creano il proprio namespace di top-level. In questo modo viene ridotta la possibilità di conflitti globali e viene semplificata l'organizzazione del codice su disco. Per esempio:
... e così via.
Riferimenti
Normalmente Perl fa ciò che vi aspettate, anche quando ci sono delle sottigliezze. Considerate ciò che accade quando passate dei valori a una funzione:
sub saluto_rovesciato
{
my $nome = reverse shift;
return "Ciao, $nome!";
}
my $nome = 'Max';
say saluto_rovesciato( $nome );
say $nome;
Al di fuori della funzione, $nome contiene Max, anche se il valore passato alla funzione viene invertito in xaM. Probabilmente è quello che vi aspettavate. Il valore di $nome al di fuori della funzione è separato dal $nome dentro alla funzione. Modificarne uno non ha effetti sull'altro.
Considerate l'alternativa. Se doveste fare delle copie di ogni valore prima che qualcosa possa cambiarli senza il vostro consenso, dovreste scrivere un sacco di codice extra.
Tuttavia alcune volte è utile che i valori possano essere modificati. Se volete passare a una funzione un hash pieno di dati perché vengano modificati, creare e restituire un nuovo hash per ogni modifica è scomodo (per non dire nulla dell'efficienza).
Perl 5 fornisce un meccanismo con cui potete riferirvi a un valore senza farne una copia. Ogni modifica fatta a quel riferimento ne aggiorna il valore in un modo visibile a tutti i riferimenti che lo possono raggiungere. Un riferimento è un tipo di dato scalare di prima classe di Perl 5 che si riferisce ad un altro tipo di dato di prima classe.
Riferimenti a Scalari
L'operatore di riferimento è il backslash (\). In contesto scalare, crea un singolo riferimento ad un altro valore. In contesto lista, crea una lista di riferimenti. Per ottenere un riferimento a $nome:
my $nome = 'Larry';
my $rif_a_nome = \$nome;
Dovete dereferenziare un riferimento per accedere al valore cui si riferisce. La dereferenziazione vi richiede di aggiungere un sigillo extra per ogni livello di indirezione:
sub inverti_sul_posto
{
my $rif_a_nome = shift;
$$rif_a_nome = reverse $$rif_a_nome;
}
my $nome = 'Chiacchierone';
reverse_in_place( \$nome );
say $nome;
Il sigillo scalare doppio ($$) dereferenzia un riferimento a scalare.
Nell'array @_, i parametri si comportano come alias alle variabili del chiamante (footnote: Ricordate che i cicli for producono un comportamento simile.), quindi potete modificarli:
sub inverti_sul_posto
{
$_[0] = reverse $_[0];
}
my $nome = 'ollirdoccoC';
inverti_sul_posto( $name );
say $name;
Normalmente, non desiderate modificare dei valori in questo modo--e i chiamanti raramente se lo aspettano. Assegnare i parametri a dei lessicali nelle vostre funzioni evita questo rischio.
Modificare direttamente un valore o restituire un riferimento a uno scalare può farvi risparmiare memoria. Dato che gli assegnamenti in Perl copiano i valori, potreste ritrovarvi con diverse copie di una stringa molto lunga. Passare dei riferimenti implica che il Perl deve solo copiare tali riferimenti--un'operazione molto meno dispendiosa.
Riferimenti complessi possono richiedere un blocco tra parentesi graffe per disambiguare parti dell'espressione. Potete sempre usare questa sintassi, anche se non sempre rende il codice più chiaro:
sub inverti_sul_posto
{
my $rif_a_nome = shift;
${ $rif_a_nome } = reverse ${ $rif_a_nome };
}
Se vi dimenticate di dereferenziare un riferimento a scalare, probabilmente Perl applicherà una coercizione al riferimento. Il valore stringa sarà nella forma SCALAR(0x93339e8) e il valore numerico sarà la porzione 0x93339e8. Questo valore codifica il tipo di riferimento (in questo caso, SCALAR) e la locazione di memoria del riferimento.
Perl non offre nativamente accesso alle locazioni di memoria. L'indirizzo di un riferimento è un valore usato come identificatore. Diversamente dai puntatori in un linguaggio come il C, non potete modificare questo indirizzo o comunque trattarlo come un indirizzo in memoria. Questi indirizzi sono solo quasi univoci dato che il Perl potrebbe riutilizzare le stesse locazioni di memoria dopo avere reclamato della memoria inutilizzata.
Riferimenti ad Array
I riferimenti ad array sono utili in diverse circostanze:
Usate l'operatore di riferimento per creare un riferimento ad un array dichiarato in precedenza:
my @carte = qw( K Q J 10 9 8 7 6 5 4 3 2 A );
my $rif_a_carte = \@carte;
Ogni modifica fatta attraverso $rif_a_carte modificherà @carte e viceversa. Potete accedere all'intero array con il sigillo @, sia per appiattirlo in una lista che per contarne gli elementi:
my $conteggio_carte = @$rif_a_carte;
my @copia_carte = @$rif_a_carte;
Accedete a singoli elementi dell'array usando una freccia di dereferenziazione (->):
my $prima_carta = $rif_a_carte->[0];
my $ultima_carta = $rif_a_carte->[-1];
La freccia è necessaria per distinguere tra uno scalare di nome $rif_a_carte e un array di nome @rif_a_carte. Notate l'uso del sigillo scalare (Sigilli delle Variabili(sigils)) per accedere a un singolo elemento.
Una sintassi alternativa antepone il sigillo scalare al riferimento ad array. Anche se meno elegante, my $prima_carta = $$rif_a_carte[0]; è più breve da scrivere.
Usate le parentesi graffe per dereferenziare una slice (Slice di Array(array_slices)) di un riferimento ad array:
my @carte_alte = @{ $rif_a_carte }[0 .. 2, -1];
È possibile omettere le parentesi graffe, ma usarle come raggruppamenti spesso migliora la leggibilità.
Per creare un array anonimo--che non è stato dichiarato--racchiudete una lista di valori tra parentesi quadre:
my $rif_a_cose_varie = [qw( Scimmie Robot Dinosauri Formaggi )];
Questo riferimento ad array è identico a un riferimento a un array con nome, eccetto che le parentesi quadre dell'array anonimo creano sempre un nuovo riferimento. Se chiedete diversi riferimenti ad un array con nome essi si riferiscono sempre allo stesso array all'interno di uno stesso scope. Per esempio, se scrivete:
my @piatti = qw( zuppa panini pizza );
my $rif_per_domenica = \@piatti;
my $rif_per_lunedi = \@piatti;
push @piatti, 'coppa di gelato';
...alla fine sia $rif_per_domenica che $rif_per_lunedi conterranno un dessert, mentre se scrivete:
my @piatti = qw( zuppa panini pizza );
my $rif_per_domenica = [ @piatti ];
my $rif_per_lunedi = [ @piatti ];
push @piatti, 'crostata di mirtilli';
...né $rif_per_domenica né $rif_per_lunedì conterranno un dessert. All'interno delle parentesi quadre usate per creare l'array anonimo, il contesto lista appiattisce l'array @piatti in una lista separata da @piatti.
Riferimenti a Hash
Usate l'operatore di riferimento su un hash per creare un riferimento a hash:
my %colori = (
nero => 'negro',
blu => 'azul',
oro => 'dorado',
rosso => 'rojo',
giallo => 'amarillo',
porpora => 'morado',
);
my $rif_a_colori = \%colori;
Per accedere alle chiavi o ai valori dell'hash anteponete il sigillo di hash % al riferimento:
my @colori_italiani = keys %$rif_a_colori;
my @colori_spagnoli = values %$rif_a_colori;
Per accedere a singoli valori dell'hash (ovvero modificarli, cancellarli, controllarne l'esistenza ed estrarli) usate la freccia di dereferenziazione o un sigillo doppio:
sub traduci_in_spagnolo
{
my $colore = shift;
return $rif_a_colori->{$colore};
# oppure return $$rif_a_colori{$colore};
}
Per le slice di riferimenti ad hash usate il sigillo array (@) e le parentesi graffe per disambiguare:
my @colori = qw( rosso blu verde );
my @colores = @{ $rif_a_colori }{@colori};
Per creare hash anonimi usate le parentesi graffe:
my $rif_a_cibo = {
'torta di compleanno' => 'la torta de cumpleaños',
dolci => 'dulces',
biscotto => 'bizcochito',
gelato => 'helado',
};
Come per gli array anonimi, ogni volta vengono creati dei nuovi hash anonimi.
Un errore comune per i principianti è quello di assegnare un hash anonimo ad un hash standard, il che produce un warning relativo ad un numero dispari di elementi nell'hash. Usate le parentesi tonde per assegnare a un hash standard e le parentesi graffe per creare un hash anonimo.
Dereferenziazione Automatica
A partire dalla versione 5.14, il Perl può dereferenziare automaticamente alcuni riferimenti al posto vostro. Dato un riferimento ad array in $rif_a_array, potete scrivere:
push $rif_a_array, qw( lista di valori );
Potete fare la stessa cosa con un'espressione che restituisce un riferimento ad array:
push $case{$luogo}[$armadi], \@scarpe_nuove;
Lo stesso vale per gli operatori su array pop, shift, unshift, splice, keys, values e each e gli operatori su hash keys, values e each.
Se il riferimento usato non è del tipo opportuno--ovvero se non può essere dereferenziato correttamente--Perl solleverà un'eccezione. Questo potrebbe sembrare più pericoloso di dereferenziare esplicitamente i riferimenti, ma è invece la stessa cosa:
my $rif = sub { ... };
# lancia un'eccezione
push $rif, qw( lista di valori );
# anche questo lancia un'eccezione
push @$rif, qw( lista di valori );
Riferimenti a Funzioni
Perl 5 supporta le funzioni di prima classe, ovvero la funzione è un tipo di dati esattamente come lo sono l'array e l'hash. Questo fatto è reso particolarmente evidente dai riferimenti a funzione, e rende possibili molte funzionalità avanzate (Chiusure(closures)). Per creare un riferimento a funzione usate l'operatore di riferimento sul nome della funzione:
sub cuoci_torta { say 'Sto cuocendo un\'ottima torta!' };
my $rif_a_torta = \&cuoci_torta;
Se non usate il sigillo funzione (&), otterrete un riferimento al valore o ai valori di ritorno della funzione.
Per creare delle funzioni anonime usate la keyword sub senza nome:
my $rif_a_dolce = sub { say 'Sto facendo una torta deliziosa!' };
L'uso di sub senza un nome compila la funzione normalmente, ma non la installa nel namespace corrente. L'unico modo per accedere a tale funzione è attraverso il riferimento restituito da sub. Invocate il riferimento a funzione con la freccia di dereferenziazione:
$rif_a_torta->();
$rif_a_dolce->();
Un modo alternativo per invocare riferimenti a funzione è quello di usare il sigillo funzione (&) invece della freccia di dereferenziazione. Evitate questa sintassi, dato che ha alcune insidiose implicazioni per il parsing e per il passaggio degli argomenti.
Pensate alle parentesi vuote come a un'operazione che indica la dereferenziazione di un'invocazione così come le parentesi quadre indicano un accesso per indice ad un array e le parentesi graffe indicano un accesso a un hash. Passate gli argomenti alla funzione dentro le parentesi:
$rif_a_cuoci_qualcosa->( 'biscotti' );
Potete anche usare i riferimenti a funzione come metodi di oggetti (Moose(moose)). Questo può essere utile quando avete dovuto fare una ricerca del metodo stesso (Riflessione(reflection)):
my $pulisci = $colf_robotica->can( 'pulisci' );
$colf_robotica->$pulisci( $cucina );
Riferimenti a Filehandle
Quando usate la forma di open (e opendir) con filehandle lessicale, vi trovate ad avere a che fare con riferimenti a filehandle. Internamente, questi filehandle sono oggetti IO::File. Potete chiamare direttamente dei metodi su di essi. A partire da Perl 5.14, potete farlo semplicemente scrivendo:
open my $out_fh, '>', 'file_di_output.txt';
$out_fh->say( 'Ecco un po\' di testo!' );
Per attivare questa funzionalità, in Perl 5.12 dovete fare la use IO::File; in Perl 5.10 e versioni precedenti dovete fare la use IO::Handle;. Anche il codice meno recente può creare dei riferimenti a typeglob:
local *FH;
open FH, "> $file" or die "Non posso scrivere su '$file': $!";
my $fh = \*FH;
Questo idioma è antecedente alla comparsa dei filehandle lessicali (avvenuta con Perl 5.6.0 nel Marzo del 2000). Potete ancora usare l'operatore di riferimento a typeglob per ottenere riferimenti ai filehandle globali del package come STDIN, STDOUT, STDERR e DATA--ma questi nomi sono comunque globali.
Dovunque è possibile, è preferibile usare i filehandle lessicali. Con tutti i benefici di uno scope esplicito, i filehandle lessicali vi permettono di gestire la durata della vita di un filehandle tramite le funzionalità di gestione dello scope di Perl 5.
Reference Count
Perl 5 usa una tecnica di gestione della memoria nota come reference counting. Ogni valore Perl ha un contatore associato. Perl incrementa tale contatore ogni volta che viene richiesto un riferimento al valore, sia implicito che esplicito. Analogamente, Perl decrementa tale contatore ogni volta che un riferimento scompare. Quando il contatore arriva a zero, Perl può riciclare il valore in modo sicuro.
Come fa il Perl a sapere quando può rilasciare la memoria associata a una variabile senza causare problemi? Come fa il Perl a sapere quando è sicuro chiudere il file aperto nello scope più interno del seguente codice?
say 'file non aperto';
{
open my $fh, '>', 'scope_interno.txt';
$fh->say( 'qui il file e` aperto' );
}
say 'qui il file e` chiuso';
Nel blocco interno dell'esempio, c'è un unico $fh (diverse linee del codice sorgente si riferiscono ad esso, usando però sempre la stessa variabile: $fh). Lo scope di $fh è il blocco interno. Il suo valore non può superare i limiti di tale blocco. Perciò, quando l'esecuzione raggiunge la fine del blocco, Perl ricicla la variabile $fh e decrementa il reference count del filehandle che essa contiene. Il reference count del filehandle arriva a zero e Perl può riciclarlo reclamandone la memoria associata e chiamando close() implicitamente.
Non è necessario che comprendiate tutti i dettagli di questo meccanismo. Dovete però capire che le vostre azioni di creazione di riferimenti e passaggio degli stessi ad altre parti del codice hanno un impatto sulla gestione della memoria del Perl (vedete Riferimenti Circolari(circular_references)).
Riferimenti e Funzioni
Quando usate dei riferimenti come argomenti di funzioni, documentate attentamente le vostre intenzioni. Modificare i valori di un riferimento all'interno di una funzione potrebbe sorprendere il codice chiamante, se esso si aspetta che nessun altro modifichi i suoi dati. Per modificare il contenuto di un riferimento senza effetti per il riferimento stesso, copiate i suoi valori in una nuova variabile:
my @nuovo_array = @{ $array_ref };
my %nuovo_hash = %{ $hash_ref };
Fare questo è necessario solo in qualche caso, ma la duplicazione esplicita aiuta ad evitare spiacevoli sorprese per il chiamante. Se usate strutture dati annidate o altri riferimenti complessi, considerate l'uso del modulo core Storable e la sua funzione dclone (deep cloning).
Strutture Dati Annidate
I dati aggregati del Perl--array e hash--vi permettono di memorizzare scalari e indicizzarli per numero o per chiave. I riferimenti Perl 5 (Riferimenti(references)) vi permettono di accedere ai tipi di dati aggregati attraverso speciali variabili scalari. Le strutture dati annidate del Perl, come un array di array o un hash di hash, sono rese possibili dall'uso dei riferimenti.
Per dichiarare una struttura dati annidata, usate la sintassi della dichiarazione di riferimenti anonimi:
my @terzetti_famosi = (
[qw( emy ely evy )],
[qw( qui quo qua )],
[qw( aldo giovanni giacomo )],
);
my %pasti = (
colazione => { piatto => 'uova',
contorno => 'patatine' },
pranzo => { piatto => 'panini',
contorno => 'mela' },
cena => { piatto => 'bistecca',
contorno => 'insalata di avocado' },
);
Perl permette, ma non richiede, la virgola finale come modo per facilitare l'aggiunta di nuovi elementi alla lista.
Usate la sintassi dei riferimenti Perl per accedere agli elementi delle strutture dati annidate. Il sigillo denota la quntità di dati da estrarre e la freccia di dereferenziazione indica che il valore di una porzione della struttura dati è un riferimento:
my $ultimo_nipote = $terzetti_famosi[1]->[2];
my $parte_del_risveglio = $pasti{colazione}->{contorno};
L'unico modo di annidare una struttura dati multi-livello è attraverso i riferimenti, quindi la freccia è superflua. Potete ometterla per chiarezza, eccetto quando invocate un riferimento a funzione:
my $nipote = $terzetti_famosi[1][2];
my $pasto = $piatti{colezione}{contorno};
$azioni{finanziarie}{compra_cibo}->( $nipote, $pasto );
Usate dei blocchi di disambiguazione per accedere ai componenti delle strutture dati annidate come se fossero array o hash di prima classe:
my $conteggio_nipoti = @{ $terzetti_famosi[1] };
my $per_la_cena = keys %{ $piatti{cena} };
... o per le slice di strutture dati annidate:
my ($piatto, $contorno) = @{ $piatti{colazione} }
{qw( piatto contorno )};
La spaziatura può aiutare, ma non elimina completamente la difficoltà di leggere questo costrutto. Per essere più chiari, usate delle variabili temporanee:
my $rif_a_pasto = $piatti{colazione};
my ($piatto, $contorno) = @$rif_a_pasto{qw( piatto contorno )};
... oppure sfruttare l'aliasing implicito a $_ del ciclo for per evitare l'uso di un riferimento intermedio:
my ($piatto, $contorno) = @{ $_ }{qw( piatto contorno )}
for $pasti{colazione};
perldoc perldsc, il ricettario sulle strutture dati, fornisce numerosi esempi di uso di varie strutture dati in Perl.
Autovivificazione
Quando tentate di scrivere dentro un componente di una struttura dati annidata, Perl creerà il cammino necessario nella struttura dati fino alla destinazione:
my @adadada;
$adadada[0][0][0][0] = 'profondamente annidato';
Dopo la seconda linea di codice, questo array di array di array di array contiene un riferimento ad array in un riferimento ad array in un riferimento ad array in un riferimento ad array. Ogni riferimento ad array contiene un elemento. Analogamente, se trattate un valore non definito come se fosse un riferimento ad hash in una struttura dati annidata, Perl lo inizializzerà di conseguenza:
my %hdh;
$hdh{Computer}{Disco} = 'molto pericoloso';
Questo utile processo è detto autovivificazione. Anche se riduce il codice di inizializzazione delle strutture dati annidate, non distingue tra un vero intento di creare elementi mancanti e gli errori di battitura. La direttiva autovivification di CPAN (Direttive(pragmas)) vi permette di disabilitare l'autovivificazione in uno scope lessicale per specifici tipi di operazioni.
Potreste vedere una contraddizione nell'idea di sfruttare l'autovivificazione attivando al contempo controlli come strict. È un problema di bilanciamento. È più conveniente catturare gli errori che fanno cambiare il comportamento del vostro programma al costo di disabilitare i controlli su alcuni riferimenti simbolici ben identificati? È più conveniente permettere alle strutture dati di crescere piuttosto che dover specificare la loro dimensione e le chiavi permesse?
Le risposte dipendono dal vostro progetto. All'inizio dello sviluppo, concedetevi la libertà di sperimentare. Quando arrivate al test e alla distribuzione, considerate di incrementare le restrizioni per evitare effetti collaterali indesiderati. Grazie allo scope lessicale delle direttive strict e autovivification, potete abilitarle dove e quando lo ritenete necessario.
Naturalmente potreste verificare le vostre aspettative prima di dereferenziare ogni livello di una struttura dati complessa, ma il codice risultante sarebbe lungo e tedioso. La cosa migliore è di evitare strutture dati annidate troppo profondamente rivedendo il vostro modello dei dati e migliorarne l'incapsulamento.
Debugging di Strutture Dati Annidate
La complessità della sintassi di dereferenziazione di Perl 5 combinata con la potenziale confusione dovuta a molti livelli di indirezione possono rendere difficile il debugging delle strutture dati annidate. Fortunatamente esistono due validi strumenti di visualizzazione.
Il modulo core Data::Dumper converte valori di complessità arbitraria in stringhe di codice Perl 5:
use Data::Dumper;
print Dumper( $struttura_complessa );
Può essere utile per vedere che cosa contiene una struttura dati, a che cosa volevate accedere e a che cosa avete invece acceduto. Data::Dumper può fare il dump di oggetti e anche di riferimenti a funzione (se settate $Data::Dumper::Deparse al valore vero).
Anche se Data::Dumper è un modulo core e stampa codice Perl 5, il suo output è verboso. Alcuni sviluppatori preferiscono usare i moduli YAML::XS o JSON per il debugging. Questi moduli non producono codice Perl 5, ma i loro output possono essere molto più chiari da leggere e capire.
Riferimenti Circolari
Il sistema di gestione della memoria di Perl 5 ha un problema con il reference counting (Reference Count(reference_counts)) che diventa visibile al livello del codice utente. Due riferimenti che finiscono indirettamente per puntare l'uno all'altro formano un riferimento circolare che Perl non può neutralizzare automaticamente. Considerate un modello biologico, dove ogni entità ha due genitori e zero o più figli:
my $alice = { madre => '', padre => '' };
my $roberto = { madre => '', padre => '' };
my $carla = { madre => $alice, padre => $roberto };
push @{ $alice->{figli} }, $carla;
push @{ $roberto->{figli} }, $carla;
Sia $alice che $roberto contengono un riferimento ad array che contiene $carla. Dato che $carla è un riferimento a un hash che contiene sia $alice che $roberto, Perl non decrementerà mai il reference count di queste tre persone a zero. Non riconosce l'esistenza di questi riferimenti circolari e non può gestire la durata della vita di queste entità.
Potete spezzare il circolo manualmente (resettando i figli di $alice e $roberto o i genitori di $carla), oppure usare i riferimenti deboli. Un riferimento debole è un riferimento che non incrementa il reference count del suo referente. I riferimenti deboli sono resi disponibili dal modulo core Scalar::Util. La funzione weaken() previene l'incremento di un reference count:
use Scalar::Util 'weaken';
my $alice = { madre => '', padre => '' };
my $roberto = { madre => '', padre => '' };
my $carla = { madre => $alice, padre => $roberto };
push @{ $alice->{figli} }, $carla;
push @{ $roberto->{figli} }, $carla;
weaken( $carla->{madre} );
weaken( $carla->{padre} );
In questo modo $carla mantiene i suoi riferimenti ad $alice e $roberto, ma tali riferimenti non prevengono il garbage collector del Perl dal distruggere le relative strutture dati. Molte strutture dati non necessitano dei riferimenti deboli, ma quando sono necessari hanno un valore inestimabile.
Alternative alle Strutture Dati Annidate
Anche se il Perl è ben contento di processare strutture dati annidate a qualunque profondità possiate immaginare, il costo di comprendere queste strutture dati e le loro relazioni per un lettore umano è alto--per non dire nulla della complessità della sintassi. Oltre i due o tre livelli di annidamento, considerate se modellare i diversi componenti del vostro sistema con classi e oggetti (Moose(moose)) non potrebbe rendere più chiaro il vostro codice.
Operatori
Alcune persone definiscono il Perl come un "linguaggio orientato agli operatori". Per riuscire a capire un programma Perl, dovete sapere come i suoi operatori interagiscono con i propri operandi.
Un operatore Perl è una sequenza di uno o più simboli usati come parte della sintassi di un linguaggio. Ogni operatore opera su zero o più operandi. Pensate ad un operatore come a uno speciale tipo di funzione che il parser riconosce e ai suoi operandi come ad argomenti.
Caratteristiche degli Operatori
Ogni operatore ha molte importanti caratteristiche che ne determinano il comportamento: il numero di operandi su cui opera, le sue relazioni con altri operatori e le sue possibilità sintattiche.
perldoc perlop e perldoc perlsyn contengono molte informazioni sugli operatori Perl, ma la documentazione assume che abbiate già familiarità con alcuni dettagli sul loro funzionamento. I concetti di informatica di base richiesti possono forse sembrare altisonanti a prima vista, ma se andate al di là dei loro nomi risultano semplici. E anche se non li avete mai usati esplicitamente, ne conoscete già il significato.
Precedenza
La precedenza di un operatore determina quando esso viene valutato da Perl in una espressione. L'ordine di valutazione va dalla precedenza più alta a quella più bassa. Poiché la precedenza della moltiplicazione è più alta della precedenza dell'addizione, 7 + 7 * 10 valuta a 77 e non 140.
Per forzare la valutazione di un operatore prima di un altro, raggruppate le sue sottoespressioni tra parentesi. Se scrivete (7 + 7) * 10, raggruppando l'addizione in una singola unità forzate la sua valutazione prima della moltiplicazione. Il risultato diventa 140.
perldoc perlop contiene una tabella delle precedenze. Leggetela, cercate di capirne il contenuto, ma non preoccupatevi di impararla a memoria (quasi nessuno lo fa). Impegnatevi piuttosto a mantenere semplici le vostre espressioni e aggiungete delle parentesi per chiarire le vostre intenzioni.
Quando due operatori hanno la stessa precedenza, altri fattori quali l'associatività (Associatività (associativity)) e la posizione (Posizione(fixity)) aiutano a eliminare l'ambiguità.
Associatività
L'associatività di un operatore determina se esso valuta da sinistra verso destra o da destra verso sinistra. L'addizione è associativa a sinistra, quindi 2 + 3 + 4 valuta prima 2 + 3 e poi aggiunge 4 al risultato. L'elevamento a potenza è associativo a destra, quindi 2 ** 3 ** 4 valuta prima 3 ** 4 e poi eleva 2 all'81esima potenza.
Può valere la pena memorizzare la precedenza e l'associatività degli operatori matematici più comuni, ma come detto sopra la regola aurea è comunque la semplicità. Usate le parentesi per rendere chiare le vostre intenzioni.
Il modulo core B::Deparse è uno strumento indispensabile di debugging. Eseguite perl -MO=Deparse,-p su un pezzo di codice per vedere esattamente come Perl gestisce la precedenza e l'associatività degli operatori. Il flag -p aggiunge parentesi extra di raggruppamento che aiutano a chiarire l'ordine di valutazione.
Ricordate che l'ottimizzatore del Perl semplifica le operazioni matematiche come quelle negli esempi precedenti di questa sezione; per evitarlo usate delle variabili, ad esempio $x ** $y ** $z.
Arità
L'arità di un operatore è il numero di operandi su cui opera. Un operatore nullario opera su zero operandi. Un operatore unario opera su un operando. Un operatore binario opera su due operandi. Un operatore ternario opera su tre operandi. Un operatore poliadico opera su una lista di operandi. La documentazione di un operatore e i relativi esempi dovrebbero chiarire qual è la sua arità.
Per esempio, gli operatori aritmetici sono operatori binari e di solito sono associativi a sinistra. 2 + 3 - 4 valuta prima 2 + 3; infatti l'addizione e la sottrazione hanno la stessa precedenza, ma essendo binari e associativi a sinistra un corretto ordine di valutazione applica l'operatore più a sinistra (+) ai due operandi più a sinistra (2 e 3) e quindi applica l'operatore più a destra (-) al risultato della prima operazione e all'operando più a destra (4).
I principianti Perl sono spesso confusi dall'interazione tra gli operatori poliadici--specialmente le chiamate di funzione--e le espressioni annidate. Se è vero che spesso le parentesi aiutano, attenzione alle complicazioni del parsing di:
# probabilmente codice bacato
say ( 1 + 2 + 3 ) * 4;
... che stampa il valore 6 e (probabilmente) valuta complessivamente a 4 (il valore di ritorno di say moltiplicato per 4). Il parser di Perl interpreta tranquillamente le parentesi come un operatore di post-raggruppamento (Posizione(fixity)) che denota gli argomenti di say, non come un operatore di raggruppamento di un'espressione usato per cambiare le precedenze.
Posizione
La posizione di un operatore determina il suo posizionamento rispetto ai suoi operandi:
Tipi di Operatori
Gli operatori del Perl forniscono contesti di valore (Contesti Numerico, Stringa e Booleano(value_contexts)) ai propri operandi. Per scegliere l'operatore appropriato, dovete tenere conto dei valori degli operandi che gli passate e del valore che vi aspettate di ricevere.
Operatori Numerici
Gli operatori numerici impongono il contesto numerico ai loro operandi. Questi operatori sono: gli operatori aritmetici standard come addizione (+), sottrazione (-), moltiplicazione (*), divisione (/), elevamento a potenza (**) e modulo (%); le loro varianti di assegnamento composto (+=, -=, *=, /=, **= e %=); e l'auto-decremento (--) sia postfisso che prefisso.
L'operatore di auto-incremento ha uno speciale comportamento in contesto stringa (Operatori Speciali(auto_increment_operator)).
Diversi operatori di confronto impongono il contesto numerico ai loro operandi. Questi operatori sono: uguaglianza numerica (==), disuguaglianza numerica (!=), maggiore di (>), minore di (<), maggiore o uguale (>=), minore o uguale (<=) e l'operatore di confronto per il sort (<=>).
Operatori Stringa
Gli operatori stringa impongono il contesto stringa ai loro operandi. Questi operatori sono il binding positivo e negativo a un'espressione regolare (=~ e !~, rispettivamente) e la concatenazione (.).
Diversi operatori di confronto impongono il contesto stringa ai loro operandi. Questi operatori sono: uguaglianza di stringhe (eq), disuguaglianza di stringhe (ne), maggiore di (gt), minore di (lt), maggiore o uguale (ge), minore o uguale (le) e l'operatore di confronto tra stringhe per il sort (cmp).
Operatori Logici
Gli operatori logici impngono il contesto booleano ai loro operandi. Questi operatori sono &&, and, || e or. Sono tutti infissi e tutti possono dar luogo a cortocircuiti (Cortocircuiti(short_circuiting)). Le varianti and e or hanno precedenza più bassa delle corrispondenti forme simboliche.
L'operatore defined-or, //, controlla se il proprio operando è definito. Contrariamente a || che testa la verità del proprio operando, // valuta a vero anche se il suo operando valuta al numero zero o alla stringa vuota. Questo risulta particolarmente utile per impostare i valori di default dei parametri:
sub chiama_il_cane
{
my $nome = shift // 'Fuffy';
...
}
L'operatore condizionale ternario (?:) riceve tre operandi. Valuta il primo in contesto booleano e continua con la valutazione del secondo se il primo è vero e del terzo in caso contrario:
my $verità = $valore ? 'vero' : 'falso';
Gli operatori prefissi ! e not restituiscono l'opposto del valore booleano dei loro operandi. not è la versione di ! con precedenza più bassa.
L'operatore xor è un operatore infisso che valuta all' or-esclusivo dei suoi operandi.
Operatori bit-a-bit
Gli operatori bit-a-bit trattano a livello di bit i numeri rappresentati dai propri operandi. Queste operazioni, piuttosto rare, sono: shift a sinistra (<<), shift a destra (>>), and bit-a-bit (&), or bit-a-bit (|) e xor bit-a-bit (^); e le loro varianti di assegnamento composto (<<=, >>=, &=, |= e ^=).
Operatori Speciali
L'operatore di auto-incremento ha un comportamento speciale. Quando è usato su un valore che ha una componente numerica (Cache di Coercizioni(cached_coercions)), l'operatore incrementa tale componente numerica. Se invece il valore è solamente una stringa (e non ha una componente numerica), l'operatore incrementa tale valore stringa in modo che a diventa b, zz diventa aaa e a9 diventa b0.
my $num = 1;
my $str = 'a';
$num++;
$str++;
is( $num, 2, 'autoincremento numerico' );
is( $str, 'b', 'autoincremento stringa' );
no warnings 'numeric';
$num += $str;
$str++;
is( $num, 2, 'un'addizione numerica con $str' );
is( $str, 1, '...genera una componente numerica per $str' );
L'operatore di ripetizione (x) è un operatore infisso con un comportamento complesso. In contesto lista, se riceve una lista valuta a quella lista ripetuta un numero di volte pari al suo secondo operando. In contesto lista, se riceve uno scalare produce una stringa che consiste nel valore stringa del suo primo operando concatenata con se stessa un numero di volte pari al suo secondo operando.
In contesto scalare, l'operatore produce sempre una stringa ottenuta con il numero appropriato di concatenazioni. Per esempio:
my @sherazad = ('notti') x 1001;
my $calendario = 'notti' x 1001;
my $lung_cal = length $calendario;
is( @sherazad, 1001, 'lista ripetuta' );
is( $lung_cal, 1001 * length 'notti',
'parola ripetuta' );
my @shestringa = 'notti' x 1001;
my $calscalare = ('notti') x 1001;
is( @shestringa, 1, 'contiene una sola stringa' );
is( length $calscalare,
1001 * length 'notti', 'sempre parola ripetuta' );
L'operatore infisso di intervallo (..) produce una lista di elementi in contesto lista:
my @carte = ( 2 .. 10, 'J', 'Q', 'K', 'A' );
Può produrre dei semplici intervalli incrementali (sia di interi che di stringhe), ma non è capace di intuire pattern di intervalli più complessi.
In contesto booleano, l'operatore di intervallo diventa l'operatore flip-flop. Questo operatore produce un valore falso fino a quando il suo operando sinistro non diventa vero. Tale valore rimane vero fino a quando l'operando destro non diventa vero, dopo di che il valore prodotto è di nuovo falso fino a quando l'operatore sinistro non diventa di nuovo vero. Immaginate di effettuare il parsing del testo di una lettera formale con:
while (/Hello, $user/ .. /Sincerely,/)
{
say "> $_";
}
L'operatore virgola (,) è un operatore infisso. In contesto scalare valuta il suo operando sinistro e quindi restituisce il valore prodotto dalla valutazione del suo operando destro. In contesto lista, valuta entrambi gli operatori da destra a sinistra.
L'operatore fat comma (=>) effettua anche la quotatura di una bareword che compaia come suo operando sinistro (Hash(hashes)).
Funzioni
In Perl, una funzione è una unità di codice discreta ed incapsulata. Un programma è una collezione di queste piccole "scatole nere" le cui interazioni governano il controllo di flusso del programma stesso. Una funzione può avere o meno un nome. Può consumare informazioni in input oppure no. Può produrre informazione in output oppure no.
Le funzioni sono un meccanismo primario per l'astrazione, l'incapsulamento ed il riutilizzo del codice in Perl 5.
Dichiarazione di Funzioni
Usate il costrutto nativo sub per dichiarare una funzione:
sub salutami { ... }
Da questo momento salutami() può essere invocata dovunque all'interno del programma.
Non è necessario definire una funzione nello stesso punto in cui la dichiarate. Una dichiarazione anticipata dice al Perl di ricordarsi il nome della funzione anche se essa verrà definita dopo:
sub saluta_dopo;
Invocazione di Funzioni
Usate delle parentesi postfisse (Posizione(fixity)) e il nome di una funzione per invocare la funzione ed eventualmente passarle una lista di argomenti:
salutami( 'Giacomo', 'Paolo' );
salutami( 'Bianca' );
salutami();
In questi esempi le parentesi non sono strettamente necessarie--anche se strict è attivata--ma rendono le cose più chiare sia a un lettore umano che al parser del Perl. In caso di dubbio, è meglio usarle.
Gli argomenti delle funzioni possono essere espressioni arbitrarie, incluse delle semplici variabili:
salutami( $name );
salutami( @authors );
salutami( %editors );
... anche se la gestione di default dei parametri in Perl 5 riserva a volte delle sorprese ai principianti.
Parametri delle Funzioni
Una funzione riceve i suoi parametri in un singolo array, @_ (Le Variabili Array di Default(default_array_variables)). Perl appiattisce tutti i parametri in input in un'unica lista. La funzione può spacchettare i parametri in opportune variabili oppure operare direttamente su @_:
sub saluta_uno
{
my ($nome) = @_;
say "Ciao, $nome!";
}
sub saluta_tutti
{
say "Ciao, $_!" for @_;
}
Anche se gran parte del codice usa shift o lo spacchettamento in una lista, @_ si comporta come un normale array del Perl, quindi potete riferirvi a singoli elementi per indice:
sub saluta_uno_shift
{
my $nome = shift;
say "Ciao, $nome!";
}
sub saluta_due_senza_shift
{
my ($supereroe, $aiutante) = @_;
say "Ma sono proprio $supereroe e $aiutante... Benvenuti!";
}
sub saluta_uno_indice
{
my $nome = $_[0];
say "Ciao, $nome!";
# oppure, in modo meno chiaro
say "Ciao, $_[0]!";
}
... ma anche usare shift, unshift, push, pop, splice e le slice con @_.
Ricordate che le funzioni native che operano sugli array usano @_ come operando di default all'interno delle funzioni. Sfruttate questo idioma.
L'assegnamento di un parametro scalare da @_ richiede uno shift, un accesso indicizzato a @_ oppure la presenza di un contesto lista a sinistra dell'assegnamento stesso determinato dalle parentesi. Negli altri casi, Perl 5 non avrà problemi a valutare per voi @_ in contesto scalare e assegnare il numero di parametri passati:
sub saluta_uno_con_bug
{
my $nome = @_; # bug
say "Ciao, $nome; oggi sembri un numero!"
}
L'assegnamento di diversi parametri a una lista è spesso più chiaro delle corrispondenti linee di codice contenenti degli shift. Confrontate:
sub calcola_valore
{
# shift multipli
my $valore_sinistro = shift;
my $operazione = shift;
my $valore_destro = shift;
...
}
... con:
sub calcola_valore
{
my ($valore_sinistro, $operazione, $valore_destro) = @_;
...
}
Occasionalmente, può essere necessario estrarre alcuni parametri da @_ e passare i restanti ad un'altra funzione:
sub metodo_delegato
{
my $self = shift;
say 'Chiamo metodo_delegato()'
$self->delega->metodo_delegato( @_ );
}
Usate shift quando la vostra funzione necessita di un solo parametro. Usate un assegnamento a lista quando dovete accedere a diversi parametri.
Molte distribuzioni di CPAN estendono la gestione dei parametri di Perl 5 con sintassi e opzioni aggiuntive. signatures e Method::Signatures sono potenti. Method::Signatures::Simple è elementare, ma utile. MooseX::Method::Signatures funziona bene in coppia con Moose (Moose(moose)).
Appiattimento
L'appiattimento dei parametri in @_ avviene nel chiamante di una funzione. Passare un hash come argomento produce una lista di coppie chiave/valore:
my %nome_e_tipo_cuccioli = (
Lucky => 'cane',
Rodney => 'cane',
Tuxedo => 'gatto',
Petunia => 'gatto',
);
mostra_cuccioli( %nome_e_tipo_cuccioli );
sub mostra_cuccioli
{
my %cuccioli = @_;
while (my ($nome, $tipo) = each %cuccioli)
{
say "$nome è un $tipo";
}
}
Quando il Perl appiattisce %nome_e_tipo_cuccioli in una lista, l'ordine delle coppie chiave/valore dell'hash può variare, ma la lista conterrà sempre una chiave seguita immediatamente dal suo valore. L'assegnamento di hash all'interno di mostra_cuccioli() funziona essenzialmente allo stesso modo dell'assegnamento a %nome_e_tipo_cuccioli, anche se quest'ultimo è più esplicito.
L'appiattimento è spesso utile, ma occorre fare attenzione a non mescolare scalari e aggregati appiattiti nelle liste di parametri. Se scrivete una funzione mostra_cuccioli_per_tipo(), dove un parametro è il tipo di cuccioli da mostrare, passate tale tipo come primo parametro (o usate pop per rimuoverlo dalla fine di @_):
sub mostra_cuccioli_per_tipo
{
my ($tipo, %cuccioli) = @_;
while (my ($nome, $specie) = each %cuccioli)
{
next unless $specie eq $tipo;
say "$nome è un $specie";
}
}
my %nome_e_tipo_cuccioli = (
Lucky => 'cane',
Rodney => 'cane',
Tuxedo => 'gatto',
Petunia => 'gatto',
);
mostra_cuccioli_per_tipo( 'cane', %nome_e_tipo_cuccioli );
mostra_cuccioli_per_tipo( 'gatto', %nome_e_tipo_cuccioli );
mostra_cuccioli_per_tipo( 'alce', %nome_e_tipo_cuccioli );
Slurping
Un assegnamento di un aggregato a una lista è sempre greedy, per cui assegnare a %cuccioli fa lo slurp di tutti i valori restanti di @_. Se il parametro $tipo fosse alla fine di @_, Perl produrrebbe un warning relativo all'assegnamento di un numero dispari di elementi ad un hash. Se volete, potete aggirare questo problema:
sub mostra_cuccioli_per_tipo
{
my $tipo = pop;
my %cuccioli = @_;
...
}
... al costo però di una minore chiarezza. Lo stesso principio si applica naturalmente anche all'assegnamento ad un array come parametro. Usate i riferimenti (Riferimenti(references)) per evitare appiattimenti e slurping indesiderati degli aggregati.
Aliasing
C'è una sottigliezza legata all'array @_; esso contiene infatti degli alias agli argomenti della funzione, e questo rende possibile modificarli direttamente. Per esempio:
sub modifica_nome
{
$_[0] = reverse $_[0];
}
my $nome = 'Arancia';
modifica_nome( $nome );
say $nome;
# stampa aicnarA
Quando modificate direttamente un elemento di @_ modificate anche il parametro originale. Fate attenzione a spacchettare @_ con cura.
Funzioni e Namespace
Ogni funzione è contenuta in un namespace (Package(packages)). Le funzioni senza un esplicito namespace--ovvero le funzioni che non sono dichiarate dopo un'istruzione package--appartengono al namespace main. Potete anche dichiarare una funzione in un altro namespace anteponendo un prefisso al suo nome:
sub Estensioni::Matematica::somma {
...
}
Questo codice dichiara la funzione e, se necessario, crea il namespace. Ricordate che i package di Perl 5 sono modificabili in qualunque punto del codice. Potete dichiarare una sola funzione con un determinato nome in un namespace. In caso contrario Perl 5 genera un warning di ridefinizione di una funzione. Potete disabilitare questo warning con no warnings 'redefine'--se siete sicuri che questo comportamento sia ciò che intendevate ottenere.
Per chiamare funzioni in altri namespace usate i loro nomi qualificati:
package main;
Estensioni::Matematica::somma( $scalare, $vettore );
Le funzioni di un namespace sono visibili al di fuori di esso attraverso i loro nomi qualificati. Dentro a un namespace, per chiamare una funzione dichiarata in quel namespace potete usare semplicemente il suo nome. Potete anche importare dei nomi da altri namespace.
Importazione
Quando caricate un modulo con il costrutto nativo use (Moduli(modules)), Perl chiama automaticamente un metodo di nome import() su tale modulo. Un modulo può fornire un proprio import() per rendere alcuni o tutti i simboli che definisce disponibili al package chiamante. Eventuali argumenti che seguono il nome del modulo nell'istruzione use vengono passati al metodo import() del modulo. Quindi:
use strict;
... carica il modulo strict.pm e chiama strict->import() senza argomenti, mentre:
use strict 'refs';
use strict qw( subs vars );
... carica il modulo strict.pm, chiama strict->import( 'refs' ) e infine chiama strict->import( 'subs', 'vars' ).
use chiama implicitamente import() come descritto qui sopra, ma potete chiamare import() anche direttamente. L'esempio fatto con use è equivalente a:
BEGIN
{
require strict;
strict->import( 'refs' );
strict->import( qw( subs vars ) );
}
L'istruzione nativa use aggiunge un blocco BEGIN implicito che racchiude queste linee di codice in modo tale che la chiamata a import() avviene immediatamente dopo che il parser ha compilato l'intera istruzione use. Questo assicura che ogni simbolo importato sia visibile durante la compilazione del resto del programma. In caso contrario, le funzioni importate da altri moduli ma non dichiarate nel file corrente sarebbero interpretate come bareword, e violerebbero la direttiva strict.
Gestione degli Errori
All'interno di una funzione, potete ispezionare il contesto della chiamata alla funzione stessa con la funzione nativa caller. Quando non riceve argomenti, essa restituisce una lista di tre elementi contenente il nome del package chiamante, il nome del file contenente la chiamata e il numero della linea del file dove è avvenuta la chiamata:
package main;
main();
sub main
{
mostra_info_chiamata();
}
sub mostra_info_chiamata
{
my ($package, $file, $linea) = caller();
say "Chiamata da $package in $file:$linea";
}
È possibile ispezionare L'intera catena di chiamate. Passate un singolo argomento intero n a caller() per ispezionare il contesto della chiamata del chiamante n-esimo nella catena. In altre parole, se mostra_info_chiamata() usasse caller(0), riceverebbe informazioni sulla chiamata da main(). Se usasse caller(1), riceverebbe informazioni sulla chiamata dall'inizio del programma.
L'argomento opzionale indica anche a caller di restituire ulteriori valori di ritorno, inclusi il nome della funzione e il contesto di numero della chiamata:
sub mostra_info_chiamata
{
my ($package, $file, $linea, $funz) = caller(0);
say "$funz chiamata da $package in $file:$linea";
}
Il modulo standard Carp usa efficacemente questa tecnica per riportare errori e generare warning nelle funzioni. Quando viene usata al posto di die nel codice di una libreria, croak() solleva un'eccezione dal punto di vista del chiamante di tale codice. Analogamente, carp() genera un warning contenente il nome di file e numero di linea del chiamante (Generare i Warning(producing_warnings)).
Questo comportamento è particolarmente utile quando validate i parametri o le precondizioni di una funzione per indicare che il codice chiamante è in qualche modo errato.
Validazione degli Argomenti
Anche se in genere si adopera per fare ciò che desidera il programmatore, il Perl offre pochi modi nativi per testare la validità degli argomenti passati a una funzione. Potete valutare @_ in contesto scalare per verificare se il numero di parametri passati a una funzione è corretto:
sub somma_numeri
{
croak 'Mi aspettavo due numeri, ne ho ricevuti: ' . @_
unless @_ == 2;
...
}
Il controllo dei tipi è più difficile, a causa delle conversioni di tipo in Perl indotte dagli operatori (Contesto(context_philosophy)). Il modulo CPAN Params::Validate offre la possibilità di controlli più rigorosi.
Argomenti Avanzati sulle Funzioni
Le funzioni sono alla base di molte funzionalità avanzate del Perl.
Conoscenza del Contesto
Le funzioni native di Perl 5 sanno se le avete invocate in contesto void, scalare o lista; la stessa cosa è possibile per le vostre funzioni. La funzione nativa chiamata erroneamente wantarray (footnote: Vedete perldoc -f wantarray per una conferma.) restituisce undef per indicare un contesto void, un valore falso per indicare un contesto scalare e un valore vero per indicare un contesto lista.
sub sensibile_al_contesto
{
my $contesto = wantarray();
return qw( Contesto lista ) if $contesto;
say 'Contesto void' unless defined $contesto;
return 'Contesto scalare' unless $contesto;
}
sensibile_al_contesto();
say my $scalare = sensibile_al_contesto();
say sensibile_al_contesto();
Questo può essere particolarmente utile per evitare che delle funzioni producano dei valori di ritorno computazionalmente costosi in un contesto void. Alcune funzioni idiomatiche restituiscono una lista in contesto lista e il primo elemento della lista o un riferimento ad array in contesto scalare. Ricordate però che non esiste un'unica regola ottimale per l'uso di wantarray. Alcune volte è meglio scrivere funzioni separate per evitare l'ambiguità.
Le distribuzioni CPAN Want di Robin Houston e Contextual::Return di Damian Conway offrono molti modi di scrivere interfacce sensibili al contesto potenti ed usabili.
Ricorsione
Supponete di voler trovare un elemento in un array ordinato. Potreste iterare su ogni singolo elemento dell'array, confrontandolo con ciò che cercate, ma in media dovreste esaminare metà degli elementi dell'array. Un altro approccio consiste nel dividere l'array in due parti, prendere l'elemento di mezzo, confrontarlo e ripetere il processo sulla metà inferiore o superiore dell'array. Divide et impera. Quando non ci sono più elementi da ispezionare oppure avete trovato l'elemento cercato, vi fermate.
Un test automatico di questa tecnica potrebbe essere:
use Test::More;
my @elementi =
(
1, 5, 6, 19, 48, 77, 997, 1025, 7777, 8192, 9999
);
ok esiste_elem( 1, @elementi ),
'trovato primo elemento dell'array';
ok esiste_elem( 9999, @elementi ),
'trovato ultimo elemento dell'array';
ok ! esiste_elem( 998, @elementi ),
'non trovato elemento che non è nell'array';
ok ! esiste_elem( -1, @elementi ),
'non trovato elemento che non è nell'array';
ok ! esiste_elem( 10000, @elementi ),
'non trovato elemento che non è nell'array';
ok esiste_elem( 77, @elementi ),
'trovato elemento in mezzo';
ok esiste_elem( 48, @elementi ),
'trovato ultimo elemento di parte inferiore';
ok esiste_elem( 997, @elementi ),
'trovato primo elemento di parte superiore';
done_testing();
La ricorsione è un concetto che appare ingannevolmente semplice. Ogni chiamata a una funzione in Perl crea un nuovo blocco di attivazione, una struttura dati interna che rappresenta la chiamata stessa, e include l'ambiente lessicale della chiamata corrente alla funzione. Ciò significa che la funzione può richiamare se stessa o ricorrere.
Per poter eseguire con successo il test precedente, dovete scrivere una funzione chiamata esiste_elem() che sappia come chiamare se stessa dividendo ogni volta la lista in due parti:
sub esiste_elem
{
my ($elem, @array) = @_;
# termina ricorsione se non ci sono elementi tra cui cercare
return unless @array;
# arrotonda per difetto se il numero di elementi è dispari
my $punto_in_mezzo = int( (@array / 2) - 0.5 );
my $elem_in_mezzo = $array[ $punto_in_mezzo ];
# restituici vero se elemento trovato
return 1 if $elem == $elem_in_mezzo;
# restituici falso se c'è un solo elemento
return if @array == 1;
# chiama ricorsivamente su parte inferiore
return esiste_elem(
$elem, @array[0 .. $punto_in_mezzo ]
) if $elem < $elem_in_mezzo;
# chiama ricorsivamente su parte superiore
return esiste_elem(
$elem, @array[ $punto_in_mezzo + 1 .. $#array ]
);
}
Anche se è possibile scrivere questo codice in modo procedurale e gestire voi stessi le parti della lista, l'approccio ricorsivo fa sì che sia il Perl a tenere traccia di questi dettagli.
Lessicali
Ogni nuova invocazione di una funzione crea la propria istanza di uno scope lessicale. Sebbene la dichiarazione di esiste_elem() crei un singolo scope per i lessicali $elem, @array, $punto_in_mezzo e $elem_in_mezzo, ogni chiamata a esiste_elem()--anche ricorsiva--memorizza i valori di tali lessicali separatamente.
Non solo esiste_elem() può chiamare se stesso, ma le variabili lessicali di ogni invocazione sono protette e separate:
use Carp 'cluck';
sub esiste_elem
{
my ($elem, @array) = @_;
cluck "[$elem] (@array)";
# segue altro codice
...
}
Chiamate di Coda
Un punto critico della ricorsione è che dovete specificare correttamente le condizioni di terminazione se non volete che la vostra funzione chiami se stessa un numero infinito di volte. La funzione esiste_elem() contiene diverse istruzioni return proprio a questo scopo.
Perl offre un utile warning Deep recursion on subroutine (NdT: ricorsione profonda sulla funzione) quando sospetta la presenza di una ricorsione infinita. Il limite di 100 chiamate ricorsive è arbitrario, ma è spesso utile. Potete disabilitare questo warning con no warnings 'recursion' nello scope della chiamata ricorsiva.
Dato che ogni chiamata a funzione richiede un nuovo blocco di attivazione e dello spazio di memoria per i lessicali, il codice altamente ricorsivo potrebbe usare più memoria del codice iterativo. La eliminazione delle chiamate di coda può essere d'aiuto.
Una chiamata di coda è una chiamata a una funzione che restituisce direttamente il risultato della funzione chiamante. Le seguenti chiamate a esiste_elem():
# chiama ricorsivamente su parte inferiore
return esiste_elem(
$elem, @array[0 .. $punto_in_mezzo ]
) if $elem < $elem_in_mezzo;
# chiama ricorsivamente su parte superiore
return esiste_elem(
$elem, @array[ $punto_in_mezzo + 1 .. $#array ]
);
... sono delle candidate all'eliminazione delle chiamate di coda. Questa ottimizzazione eviterebbe di ritornare alla chiamata corrente e poi alla chiamata precedente, ritornando invece subito alla chiamata precedente.
Purtroppo, Perl 5 non elimina automaticamente le chiamate di coda. Fatelo voi manualmente con una speciale forma dell'istruzione goto. Contrariamente alla forma che spesso produce dello spaghetti code, la goto a funzione sostituisce la chiamata di funzione corrente con quella ad un'altra funzione. Potete indicare una funzione sia per nome che per riferimento. Per passare degli argomenti diversi, impostate direttamente @_:
# chiama ricorsivamente su parte inferiore
if ($elem < $elem_in_mezzo)
{
@_ = ($elem, @array[0 .. $punto_in_mezzo ]);
goto &esiste_elem;
}
# chiama ricorsivamente su parte superiore
else
{
@_ = ($elem, @array[ $punto_in_mezzo + 1 .. $#array ] );
goto &esiste_elem;
}
Qualche volta le ottimizzazioni lasciano a desiderare dal punto di vista estetico.
Trappole e Funzionalità da Evitare
Perl 5 supporta ancora le invocazioni di funzione vecchio stile, ereditate da versioni precedenti del Perl. Mentre ora potete invocare le funzioni Perl per nome, le versioni precedenti di Perl richiedevano di invocarle con una "e commerciale" (&) come primo carattere. Perl 1 richiedeva di usare l'istruzione do:
# stile superato; da evitare
my $risultato = &calcola_risultato( 52 );
# stile Perl 1; da evitare
my $risultato = do calcola_risultato( 42 );
# strano miscuglio; da evitare assolutamente
my $risultato = do &calcola_risultato( 42 );
Se da una parte la sintassi Perl 1 è solo visivamente sgradevole, la forma con la "e commerciale" ha altri effetti sorprendenti. Per prima cosa, disabilita qualunque controllo sul prototipo. Inoltre, passa implicitamente l'esatto contenuto di @_ a meno che specifichiate voi stessi gli argomenti. Entrambe possono causare strani comportamenti del codice.
Un'ultima trappola viene dall'omissione delle parentesi nelle chiamate di funzione. Il parser di Perl 5 usa diverse euristiche per risolvere le bareword ambigue e per determinare il numero di parametri passati a una funzione. Tali euristiche possono sbagliare:
# attenzione; contiene un insidioso errore
ok esiste_elem 1, @elementi, 'trovato primo elemento';
La chiamata a esiste_elem() si ingoia la descrizione del test, intesa invece come il secondo argomento a ok(). Poiché il secondo parametro di esiste_elem() fa slurping, questo problema può rimanere nascosto fino a quando Perl produce dei warning relativi al confronto tra un valore non-numerico (la descrizione del test, che non può essere convertita in un numero) e l'elemento nell'array.
Mentre l'uso indiscriminato di parentesi può peggiorare la leggibilità, un loro uso ponderato può chiarire il significato del codice e rendere improbabili alcuni errori insidiosi.
Scope
In Perl il concetto di scope si riferisce al tempo di vita e alla visibilità delle entità con un nome. Ogni entità Perl con un nome (una variabile, una funzione) ha uno scope. Il meccanismo di scope favorisce l'incapsulamento--che consiste nel mantenere insieme i concetti collegati ed evitare che essi fuoriescano dall'ambito in cui il programmatore li ha confinati.
Scope Lessicale
Lo scope lessicale è uno scope visibile mente leggete un programma. Il compilatore Perl risolve questo tipo di scope durante la compilazione. Un blocco delimitato da parentesi graffe crea un nuovo scope, sia che si tratti di un blocco a se stante, di un blocco di un costrutto di ciclo, del blocco di una dichiarazione sub, di un blocco eval o di qualunque altro blocco purché non sia di quotatura.
Lo scope lessicale determina la visibilità delle variabili dichiarate con my--le variabili dette lessicali. Una variabile lessicale dichiarata in uno scope è visibile in quello scope e negli scope annidati, ma è invisibile agli scope dello stesso livello e a quelli esterni:
# scope lessicale esterno
{
package Robot::Maggiordomo
# scope lessicale interno
my $livello_batteria;
sub pulisci_stanza
{
# scope ancora piu' interno
my $timer;
do {
# scope lessicale ancora piu' interno
my $paletta;
...
} while (@_);
for (@_)
{
# altro scope lessicale ancora piu' interno
my $straccio;
...
}
}
}
... $livello_batteria è visibile in tutti e quattro gli scope. $timer è visibile nel metodo, nel blocco do e nel ciclo for. $paletta è visibile soltanto nel blocco do e $straccio è visibile soltanto nel ciclo for.
Dichiarare un lessicale in uno scope con lo stesso nome di un lessicale in uno scope più esterno nasconde, o oscura, il lessicale esterno nello scope interno. Spesso, è proprio ciò che desiderate:
my $nome = 'Jacob';
{
my $nome = 'Edward';
say $nome;
}
say $nome;
L'oscuramento di lessicali può essere accidentale. Limitate lo scope delle variabili e l'annidamento degli scope per ridurre questo rischio.
Questo programma stampa Edward e quindi Jacob (footnote: Familiari dell'autore, non i noti vampiri.), sebbene ridichiarare una variabile lessicale con lo stesso nome e lo stesso tipo nello stesso scope lessicale produca un messaggio di warning. La possibilità di oscurare un lessicale è una funzionalità dell'incapsulamento.
Ci sono alcune sottigliezze relative alle dichiarazioni di lessicali, per esempio quando una variabile lessicale viene usata come variabile iteratore di un ciclo for. La sua dichiarazione è al di fuori del blocco del ciclo, ma il suo scope è all'interno del blocco stesso:
my $gatto = 'Brad';
for my $gatto (qw( Jack Daisy Petunia Tuxedo Choco ))
{
say "L'iteratore gatto ha il valore $gatto";
}
say "Il gatto esterno ha sempre valore $gatto";
Analogamente, given (Given/When(given_when)) crea un topic lessicale (come my $_) nel proprio blocco:
$_ = 'fuori';
given ('dentro')
{
say;
$_ = 'questo assegnamento e' inutile';
}
say;
... in modo che l'uscita dal blocco ripristina il valore precedente di $_.
Le funzioni--sia con nome che anonime--forniscono uno scope lessicale al proprio corpo. Grazie a questo è possibile definire le closure (Chiusure(closures)).
Lo Scope Our
Dentro a un certo scope, potete dichiarare un alias a una variabile di package con l'istruzione our. Come my, our impone uno scope lessicale all'alias. Il nome qualificato è disponibile ovunque, ma l'alias lessicale è visibile solo all'interno dello scope.
our è particolarmente utile per le variabili di package globali come $VERSION e $AUTOLOAD.
Scope Dinamico
Lo scope dinamico è simile allo scope lessicale per ciò che riguarda le sue regole di visibilità, ma la ricerca dei nomi, invece di procedere dagli scope più interni a quelli più esterni durante la compilazione, procede all'indietro nei contesti di chiamata. Anche se una variabile di package globale è visibile in tutti gli scope, il suo valore cambia a seconda della sua localizzazione e relativo assegnamento:
our $scope;
sub interno
{
say $scope;
}
sub main
{
say $scope;
local $scope = 'scope del main()';
intermedio();
}
sub intermedio
{
say $scope;
interno();
}
$scope = 'scope esterno';
main();
say $scope;
Il programma inizia con la dichiarazione di una variabile our, $scope, e di tre funzioni. Quindi, fa un assegnamento a $scope e chiama la funzione main().
Dentro a main(), il programma stampa il valore corrente di $scope, scope esterno, e poi localizza la variabile. Questo cambia la visibilità del simbolo nello scope lessicale corrente così come in ogni funzione chiamata dallo scope lessicale corrente. Quindi, $scope contiene scope del main() sia nel corpo di intermedio() che di interno(). Quando main() termina perché il controllo di flusso raggiunge la fine del suo blocco, Perl ripristina il valore originale della variabile $scope localizzata. Quindi, l'ultimo say stampa di nuovo scope esterno.
Le variabili di package e quelle lessicali hanno regole di visibilità e meccanismi di memorizzazione diversi nel Perl. Ogni scope che contiene variabili lessicali ha una speciale struttura dati chiamata pad lessicale o lexpad che può memorizzare i valori delle variabili lessicali incluse nello scope. Ogni volta che il controllo di flusso entra in uno di questi scope, Perl crea un nuovo lexpad per i valori delle variabili lessicali in quella particolare chiamata. Questo determina la corretta esecuzione delle funzioni, specialmente nelle chiamate ricorsive (Ricorsione(recursion)).
Ogni package ha un'unica tabella dei simboli che contiene sia le variabili di package che le funzioni con nome. L'importazione (Importazione(importing)) si basa sull'ispezione e la manipolazione di tale tabella dei simboli. Lo stesso vale per local. Potete localizzare solo le variabili globali predefinite e di package--e non le variabili lessicali.
local è particolarmente utile con le variabili globali predefinite. Per esempio, il separatore di record in input $/ determina quanti dati vengono letti da un filehandle da un'operazione readline. $!, la variabile di errore di sistema, contiene il numero di errore della chiamata di sistema più recente. $@, la variabile Perl di errore di eval, contiene l'errore generato dalla operazione eval più recente. $|, la variabile di flush automatico, determina se Perl deve eseguire il flush del filehandle corrente impostato con select dopo ogni operazione di scrittura.
localizzare queste variabili in uno scope il più ristretto possibile limita gli effetti delle vostre modifiche. In questo modo, potete evitare strani comportamenti in altre parti del vostro codice.
Scope di Variabili State
Perl 5.10 ha aggiunto un nuovo scope per supportare l'istruzione nativa state. Lo scope di state è simile allo scope lessicale per quanto riguarda la visibilità, ma impone un'unica initializzazione e la persistenza del valore:
sub contatore
{
state $cont = 1;
return $cont++;
}
say contatore();
say contatore();
say contatore();
Nella prima chiamata a contatore, Perl esegue l'unica inizializzazione di $cont. Nelle chiamate successive, $cont mantiene il suo valore precedente. Questo programma stampa quindi 1, 2 e 3. Se cambiate state in my il programma stamperà 1, 1 e 1.
Potete anche usare un'espressione per impostare il valore iniziale di una variabile state:
sub countatore
{
state $cont = shift;
return $cont++;
}
say contatore(2);
say contatore(4);
say contatore(6);
Sebbene a prima vista si potrebbe pensare che l'output del codice sia 2, 4, e 6, tale output è invece 2, 3, e 4. La prima chiamata alla funzione contatore inizializza la variabile $cont. Nelle chiamate successive la parte di inizializzazione non viene più eseguita.
state può essere utile sia per impostare un valore di default che per predisporre una cache, ma se la usate fate attenzione ad avere compreso il suo comportamento con le inizializzazioni:
sub contatore
{
state $cont = shift;
say 'Il secondo arg ha valore: ', shift;
return $cont++;
}
say contatore(2, 'due');
say contatore(4, 'quattro');
say contatore(6, 'sei');
In questo programma il contatore stampa 2, 3 e 4 come vi aspettate, ma i valori del presunto secondo argomento nelle chiamate a contatore() sono due, 4 e 6--dato che lo shift del primo argumento ha luogo soltanto nella prima chiamata a contatore(). Per evitare questo problema potete cambiare l'interfaccia di contatore(), oppure gestirlo nel modo seguente:
sub contatore
{
my ($valore_iniziale, $testo) = @_;
state $cont = $valore_iniziale;
say "Il secondo arg ha valore: $testo";
return $cont++;
}
say contatore(2, 'due');
say contatore(4, 'quattro');
say contatore(6, 'sei');
Funzioni Anonime
Una funzione anonima è una funzione senza nome. Per il resto si comporta esattamente come una funzione con nome--potete invocarla, passarle degli argomenti, farle restituire dei valori e copiare riferimenti ad essa. Tuttavia, l'unico modo per potervi accedere è attraverso un riferimento (Riferimenti a Funzioni(function_references)).
Un idioma comune in Perl 5 noto come tabella di dispatch usa gli hash per associare i valori di un input a dei comportamenti specifici:
my %dispatch =
(
piu => \&somma_due_numeri,
meno => \&sottrai_due_numeri,
per => \&moltiplica_due_numeri,
);
sub somma_due_numeri { $_[0] + $_[1] }
sub sottrai_due_numeri { $_[0] - $_[1] }
sub moltiplica_due_numeri { $_[0] * $_[1] }
sub dispatch
{
my ($sinistro, $op, $destro) = @_;
return unless exists $dispatch{ $op };
return $dispatch{ $op }->( $sinistro, $destro );
}
La funzione dispatch() riceve argomenti nella forma (2, 'per', 2) e restituisce il risultato dell'operazione.
Dichiarazione di Funzioni Anonime
L'istruzione nativa sub usata senza un nome crea e restituisce il riferimento a una funzione anonima. Potete usare questo riferimento ovunque è ammissibile usare un riferimento a una funzione con nome, per esempio per dichiarare le funzioni di una tabella di dispatch direttamente:
my %dispatch =
(
piu => sub { $_[0] + $_[1] },
meno => sub { $_[0] - $_[1] },
per => sub { $_[0] * $_[1] },
diviso => sub { $_[0] / $_[1] },
elevato => sub { $_[0] ** $_[1] },
);
Questa tabella di dispatch presenta un certo grado di sicurezza; soltanto le funzioni mappate nella tabella stessa sono disponibili per essere chiamate dagli utenti. Se la vostra funzione di dispatch assumesse ciecamente che la stringa passata come nome dell'operatore corrisponde direttamente al nome di una funzione da chiamare, un utente malintenzionato potrebbe chiamare qualunque funzione in qualunque namespace passando 'Funzioni::Interne::funzione_maligna'.
Potreste anche vedere delle funzioni anonime passate come argomenti ad altre funzioni:
sub invoca_funzione__anonima
{
my $funz = shift;
return $funz->( @_ );
}
sub funzione_con_nome
{
say 'Sono una funzione con nome!';
}
invoca_funzione__anonima( \&funzione_con_nome );
invoca_funzione__anonima( sub { say 'Indovina chi sono!' } );
Nomi di Funzioni Anonime
Dato un riferimento a una funzione, è possibile determinare se la funzione ha un nome oppure è anonima usando l'introspezione (footnote: oppure sub_name del modulo CPAN Sub::Identify.):
package MostraChiamante;
sub mostra_chiamante
{
my ($package, $file, $linea, $sub) = caller(1);
say "Chiamata da $sub in $package:$file:$linea";
}
sub main
{
my $sub_anonima = sub { mostra_chiamante() };
mostra_chiamante();
$sub_anonima->();
}
main();
Il risultato potrebbe cogliervi un po' di sorpresa:
Chiamata da MostraChiamante::main
in MostraChiamante:chiamante_anonimo.pl:20
Chiamata da MostraChiamante::__ANON__
in MostraChiamante::chiamante_anonimo.pl:17
La stringa __ANON__ nella seconda linea dell'output mostra che la funzione anonima non ha un nome che Perl possa identificare. Questo può rendere il debugging più complicato. La funzione subname() del modulo CPAN Sub::Name vi permette di associare dei nomi alle funzioni anonime:
use Sub::Name;
use Sub::Identify 'sub_name';
my $anonima = sub {};
say sub_name( $anonima );
my $con_nome = subname( 'pseudo-anonima', $anonima );
say sub_name( $con_nome );
say sub_name( $anonima );
say sub_name( sub {} );
Questo programma produce l'output:
__ANON__
pseudo-anonima
pseudo-anonima
__ANON__
Notate che entrambi i riferimenti sono relativi alla stessa funzione anonima sottostante. Chiamando subname() con un riferimento a una funzione anonima, il nome di tale funzione viene modificato in modo tale che tutti gli altri riferimenti ad essa vedono il nuovo nome.
Funzioni Anonime Implicite
Perl 5 permette la dichiarazione implicita di funzioni anonime attraverso l'uso dei prototipi (Prototipi(prototypes)). Sebbene questa funzionalità ufficialmente sia presente per permettere ai programmatori di definire una propria sintassi come fanno map e eval, un esempio interessante è l'uso di funzioni ritardate che non sembrano funzioni.
Considerate il modulo CPAN Test::Fatal, che riceve una funzione anonima come primo argomento della sua funzione exception:
use Test::More;
use Test::Fatal;
my $muore = exception { die 'Sto morendo!' };
my $vive = exception { 1 + 1 };
like( $muore, qr/Sto morendo/, 'die() causa la morte' );
is( $vive, undef, 'la somma lascia vivere' );
done_testing();
Potete riscrivere questo codice in modo più verboso:
my $muore = exception( sub { die 'Sto morendo!' } );
my $vive = exception( sub { 1 + 1 } );
...o passare riferimenti a funzioni con nome:
sub muore { die 'Sto morendo!' }
sub vive { 1 + 1 }
my $muore = exception \&muore;
my $vive = exception \&vive;
like( $muore, qr/Sto morendo/, 'die() causa la morte' );
is( $vive, undef, 'la somma lascia vivere' );
... ma non potete invece passare tali riferimenti come scalari:
my $rif_a_muore = \&muore;
my $rif_a_vive = \&vive;
# BUG: non funziona
my $muore = exception $rif_a_muore;
my $vive = exception $rif_a_vive;
... perché il prototipo modifica il modo in cui il parser di Perl 5 interpreta questo codice. Non potendo essere sicuro al 100% di che cosa conterranno $muore e $vive, il parser solleva un'eccezione.
Type of arg 1 to Test::Fatal::exception
must be block or sub {} (not private variable)
Ricordate anche che una funzione che riceve una funzione anonima come il primo di diversi argomenti non ammette la virgola dopo il blocco della funzione:
use Test::More;
use Test::Fatal 'dies_ok';
dies_ok { die 'La Forza sia con te!' }
'Non mi piacciono le citazioni dai film';
Questo piccolo neo, dovuto a un'idiosincrasia del parser di Perl 5, può occasionalmente causare confusione nell'uso di una sintassi altrimenti utile. La pulizia sintattica che deriva dalla promozione di blocchi isolati a funzioni anonime può essere vantaggiosa, ma usatela con parsimonia e documentate la vostra API con cura.
Chiusure
Ogni volta che il controllo di flusso entra in una funzione, tale funzione riceve anche un nuovo ambiente che rappresenta lo scope lessicale di quella particolare chiamata (Scope(scope)). Questo vale anche per le funzioni anonime (Funzioni Anonime(anonymous_functions)). Le implicazioni sono importanti. Nell'informatica di base, il termine funzioni di ordine superiore si riferisce a funzioni che manipolano altre funzioni. Le chiusure consentono di sfruttare pienamente le potenzialità di tali funzioni.
Creazione di Chiusure
Una chiusura è una funzione che usa le variabili lessicali di uno scope esterno. Probabilmente avete già creato e utilizzato delle chiusure senza rendervene conto:
package Chiusura::Invisibile;
my $nomefile = shift @ARGV;
sub dammi_il_nomefile { return $nomefile }
Se questo codice vi sembra ovvio, è un buon inizio! Naturalmente la funzione dammi_il_nomefile() può vedere il lessicale $nomefile. È propio il modo in cui funziona lo scope!
Immaginate di voler iterare su una lista di elementi senza dover gestire voi stessi l'iteratore. Potete creare una funzione che restituisce una funzione che, quando viene invocata, restituisce il prossimo elemento dell'iterazione:
sub crea_iteratore
{
my @elementi = @_;
my $cont = 0;
return sub
{
return if $cont == @elementi;
return $elementi[ $cont++ ];
}
}
my $cugini = crea_iteratore(
qw( Riccardo Ale Carlo Enrico Camilla Maria Cristina )
);
say $cugini->() for 1 .. 5;
Sebbene crea_iteratore() sia terminata, la funzione anonima memorizzata in $cugini ha fatto una chiusura sui valori delle due variabili così come essi si presentavano all'interno dell'invocazione di crea_iteratore(). Tali valori, dunque, persistono (Reference Count(reference_counts)).
Poiché ogni invocazione di crea_iteratore() crea un nuovo ambiente lessicale, ogni sub anonima che viene creata e restituita fa una chiusura su un proprio ambiente lessicale unico:
my $zie = crea_iteratore(
qw( Carolina Filomena Viviana Silvia Monica Lara )
);
say $cugini->();
say $zie->();
Dato che crea_iteratore() non restituisce questi lessicali né per valore, né per riferimento, nessun'altra parte del codice Perl può accedervi, eccetto la chiusura. Essi risultano incapsulati allo stesso modo degli altri lessicali, e quindi solo il codice che condivide con essi un ambiente lessicale può accedervi. Questo idioma fornisce un incapsulamento migliore di quello che fornirebbe invece una variabile globale di file o di package:
{
my $variabile_privata;
sub imposta_privata { $variabile_privata = shift }
sub leggi_privata { $variabile_privata }
}
Ricordate che non potete annidare le funzioni con nome. Le funzioni con nome hanno uno scope globale al package. La condivisione di ogni variabile lessicale condivisa tra funzioni annidate viene perduta quando la funzione più esterna distrugge il suo primo ambiente lessicale (footnote: Se questo vi sembra complicato, immaginate come dev'essere stato complicato implementarlo.).
Il modulo CPAN PadWalker vi permette di violare l'incapsulamento lessicale, ma chiunque lo usi introducendo dei bug nel vostro codice si guadagna di diritto l'onore di risolvere tali bug senza il vostro aiuto.
Uso delle Chiusure
Iterare su una lista di dimensione fissata con una chiusura è interessante, ma le chiusure possono fare molto altro, come iterare su una lista che è troppo costosa da calcolare o troppo grande da tenere in memoria tutta in una volta. Considerate una funzione per creare la serie di Fibonacci man mano che avete bisogno dei suoi elementi. Invece di ricalcolare la serie ricorsivamente, usate una cache e siate pigri, creando gli elementi solo quando vi servono:
sub gen_fib
{
my @seriefib = (0, 1);
return sub
{
my $elem = shift;
if ($elem >= @seriefib)
{
for my $calc (@seriefib .. $elem)
{
$seriefib[$calc] = $seriefib[$calc - 2]
+ $seriefib[$calc - 1];
}
}
return $seriefib[$elem];
}
}
Ogni chiamata alla funzione restituita da gen_fib() riceve un argomento, l'indice dell'n-esimo elemento della serie di Fibonacci. La funzione genera tutti i valori precedenti necessari della serie, li memorizza in cache e restituisce l'elemento richiesto--ritardando le computazioni fino a quando non sono assolutamente necessarie. C'è un pattern specifico di caching intrecciato con la serie numerica. Che cosa succede se estraete il codice specifico di caching (inizializzazione della cache, escuzione di codice dedicato al popolamento deglii elementi arbitrari della cache e restituzione di un valore calcolato o estratto dalla cache) in una funzione genera_chiusura_cache()?
sub genera_chiusura_cache
{
my ($calcola_elem, @cache) = @_;
return sub
{
my $elem = shift;
$calcola_elem->($elem, \@cache)
unless $elem < @cache;
return $cache[$elem];
};
}
Le funzioni native map, grep e sort sono anch'esse delle funzioni di ordine superiore.
Ora gen_fib() diventa:
sub gen_fib
{
my @seriefib = (0, 1, 1);
return genera_chiusura_cache(
sub
{
my ($elem, $seriefib) = @_;
for my $calc ((@$seriefib - 1) .. $elem)
{
$seriefib->[$calc] = $seriefib->[$calc - 2]
+ $seriefib->[$calc - 1];
}
},
@seriefib
);
}
Il programma funziona come prima, ma l'uso di riferimenti a funzione e chiusure separa il codice di inizializzazione della cache dal calcolo del prossimo numero della serie di Fibonacci. Personalizzare il comportamento del codice--in questo caso, di genera_chiusura_cache()--passando una funzione vi mette a disposizione una enorme flessibilità.
Chiusure e Applicazione Parziale
Le chiusure possono anche limitare la generalità indesiderata di una funzione. Considerate una funzione che riceve molti parametri:
sub prepara_coppa
{
my %argomenti = @_;
my $gelato = dammi_gelato( $argomenti{gelato} );
my $banana = dammi_banana( $argomenti{banana} );
my $sciroppo = dammi_sciroppo( $argomenti{sciroppo} );
...
}
Avere un'enorme quantità di personalizzazioni possibili può essere giusto per una elegante gelateria in centro città, ma per un venditore ambulante che serve solo gelato alla vaniglia su banane di qualità Cavendish, ogni chiamata a prepara_coppa() passa alcuni argometi che non cambiano mai.
Una tecnica detta applicazione parziale vi permette di specificare alcuni degli argumenti a una funzione e di fornire i restanti in un momento successivo. Racchiudete la chiamata alla funzione in una chiusura e passatele i parametri che volete fissare.
Considerate il venditore ambulante che serve solo gelato alla vaniglia su banane di qualità Cavendish:
my $prepara_coppa_ambulante = sub
{
return prepara_coppa( @_,
gelato => 'Vaniglia',
banana => 'Cavendish',
);
};
Invece di chiamare direttamente prepara_coppa(), invocate il riferimento a funzione $prepara_coppa_ambulante e passate soltanto gli argomenti interessanti, senza dovervi preoccupare di dimenticare o di non passare correttamente quelli fissi (footnote: Potete anche usare il modulo CPAN Sub::Install per importare questa funzione direttamente in un altro namespace.).
Questo è solo un assaggio di quello che potete fare con le funzioni di ordine superiore. Il libro Higher Order Perl di Mark Jason Dominus è il riferimento bibliografico canonico per trattare funzioni e chiusure come oggetti di prima classe. Potete leggerlo online all'URL http://hop.perl.plover.com/.
State e Chiusure
Le chiusure (Chiusure(closures)) sfruttano lo scope lessicale (Scope(scope)) per mediare l'accesso a variabili semi-private. Anche le funzioni con nome possono sfruttare i legami imposti dallo scope lessicale:
{
my $sicurezza = 0;
sub abilita_sicurezza { $sicurezza = 1 }
sub disabilita_sicurezza { $sicurezza = 0 }
sub fai_qualcosa_di_straordinario
{
return if $sicurezza;
...
}
}
L'incapsulamento delle funzioni che (dis)attivano la sicurezza permette alle tre funzioni di condividere uno stato senza esporre la variabile lessicale direttamente al codice esterno. Questo idioma funziona bene nei casi in cui il codice esterno deve poter cambiare lo stato interno, ma è un po' ingombrante quando tale stato deve essere gestito da una singola funzione.
Immaginate che ogni cento clienti della vostra gelateria, uno riceva gratis un'aggiunta di panna:
my $conta_cli = 0;
sub servi_cliente
{
$conta_cli++;
my $ordine = shift;
aggiungi_panna($ordine) if $conta_cli % 100 == 0;
...
}
Questo approccio funziona, ma creare un nuovo scope lessicale per un'unica funzione introduce più complessità di quanto sembri necessario. L'istruzione nativa state vi permette di dichiarare una variabile con scope lessicale il cui valore persiste tra un'invocazione e l'altra:
sub servi_cliente
{
state $conta_cli = 0;
$conta_cli++;
my $ordine = shift;
aggiungi_panna($ordine)
if ($conta_cli % 100 == 0);
...
}
Dovete attivare questa funzionalità usando un modulo come Modern::Perl, la direttiva feature (Direttive(pragmas)) o richiedendo una versione specifica di Perl che sia la 5.10 o successiva (per esempio, con use 5.010; o use 5.012;).
state funziona anche nelle funzioni anonime:
sub crea_contatore
{
return sub
{
state $cont = 0;
return $cont++;
}
}
... anche se questo approccio non presenta molti vantaggi evidenti.
State e Pseudo-State
A partire da Perl 5.10 è stata deprecata una tecnica usata con le versioni precedenti del Perl per emulare il comportamento di state. Una funzione con nome poteva fare una chiusura sul suo scope lessicale precedente abusando di un'idiosincrasia dell'implementazione. L'aggiunta di un condizionale postfisso che valutasse a falso in coda a una dichiarazione my evitava di reinizializzare una variabile lessicale a undef o al suo valore iniziale.
Adesso la presenza di un'espressione condizionale postfissa come modificatore di una dichiarazione di variabile lessicale viene deprecato con un warning. In effetti, con questa tecnica è troppo facile introdurre inavvertitamente dei bug nel codice; usate invece state dove è possibile o in alternativa una vera chiusura. Riscrivete in modo corretto questo idioma quando lo incontrate:
sub cattivo_stato
{
# my $contatore = 1 if 0; # DEPRECATO; non usatelo
state $contatore = 1; # usate invece questo
...
}
Attributi
Le entità Perl con un nome--variabili e funzioni--possono essere associate a dei metadati aggiuntivi in forma di attributi. Gli attributi sono nomi e valori arbitrari usati da determinati tipi di metaprogrammazione (Generazione di Codice(code_generation)).
La sintassi per la dichiarazione degli attributi è poco elegante e usare gli attributi in modo efficace è più un'arte che una scienza. La maggior parte dei programmi evita di utilizzarli, ma quando sono usati bene essi possono offrire vantaggi in termini di chiarezza e manutenibilità.
Uso degli Attributi
Un attributo semplice è un identificatore preceduto dal carattere due punti e associato a una dichiarazione:
my $fortezza :nascosto;
sub eruzione_vulcano :ProgettoScientifico { ... }
Queste dichiarazioni causano l'invocazione degli handler di attributi di nome nascosto e ScienceProject, se esistono per il tipo appropriato (rispettivamente, scalare e funzione). Tali handler possono fare qualunque cosa. Se non esistono degli handler appropriati, Perl solleva un'eccezione durante la compilazione.
Gli attributi possono specificare una lista di parametri. Perl tratta tali parametri come liste composte da sole stringhe costanti. Il modulo CPAN Test::Class fa un buon uso degli attributi parametrici:
sub configura_test :Test(setup) { ... }
sub creazione_test_automatico :Test(10) { ... }
sub termina_test :Test(teardown) { ... }
L'attributo Test identifica i metodi che contengono asserzioni di test e specifica opzionalmente il numero di asserzioni che tali metodi intendono eseguire. Anche se l'introspezione (Riflessione(reflection)) di queste classi, guidata da buone euristiche, potrebbe identificare appropriatamente i metodi di test, l'attributo :Test rende più chiare le vostre intenzioni.
I parametri setup e teardown permettono alle classi di test di definire i propri metodi di supporto senza preoccuparsi dei conflitti con metodi analoghi in altre classi. In questo modo si separa il problema di specificare ciò che una classe deve fare da come funzionano altre classi, e in questo modo si offre molta flessibilità al programmatore.
Anche il framework web Catalyst usa gli attributi per determinare la visibilità e il comportamento di alcuni metodi all'interno di applicazioni web.
Svantaggi degli Attributi
Gli attributi hanno anche degli svantaggi. La direttiva canonica per usare gli attributi, attributes, ha catalogato la propria interfaccia come sperimentale per molti anni. Il modulo Attribute::Handlers di Damian Conways integrato nella distribuzione semplifica la loro implementazione. Attribute::Lexical di Andrew Main è un approccio più recente. Ovunque sia possibile usarli, entrambi sono preferibili a attributes.
La peggiore caratteristica degli attributi è la loro propensione a produrre strani effetti sintattici a distanza. Dato un frammento di codice con attributi, come potete prevedere i loro effetti? Una buona documentazione può essere d'aiuto, ma se la dichiarazione apparentemente innocente di una variabile lessicale memorizza da qualche parte un riferimento a tale variabile, le vostre attese sul suo tempo di vita potrebbero rivelarsi sbagliate. Analogamente, un handler potrebbe racchiudere a vostra insaputa una funzione in un'altra funzione e sostituirla nella tabella dei simboli--pensate a un attributo :memoize che invochi automaticamente il modulo Memoize integrato nella distribuzione.
Gli attributi sono a vostra disposizione per risolvere alcuni problemi ostici. Possono essere molto utili, se usati in modo appropriato--ma la maggior parte dei programmi non ne ha bisogno.
AUTOLOAD
Perl non richiede che dichiariate ciascuna funzione prima di chiamarla e fa del suo meglio per chiamare una funzione anche se non esiste. Considerate questo programma:
use Modern::Perl;
cuoci_crostata( ripieno => 'mele' );
Quando lo eseguite, Perl solleva un'eccezione di chiamata a una funzione cuoci_crostata() non definita.
Ora aggiungete una funzione di nome AUTOLOAD():
sub AUTOLOAD {}
Se rieseguite il programma, non succede nulla di evidente. Perl chiama la funzione AUTOLOAD() di un package--se esiste-- ogni volta che il normale meccanismo di chiamata fallisce. Modificate AUTOLOAD() in modo da stampare un messaggio:
sub AUTOLOAD { say 'Sono in AUTOLOAD()!' }
... per dimostrare che viene chiamata.
Funzionalità Base di AUTOLOAD
La funzione AUTOLOAD() riceve direttamente in @_ gli argomenti che erano stati passati alla funzione non definita, mentre il nome della funzione non definita è memorizzato nella variabile globale di package $AUTOLOAD. Potete manipolare questi argomenti a vostro piacimento:
sub AUTOLOAD
{
our $AUTOLOAD;
# stampa gli argomenti separati da virgola
local $" = ', ';
say "Sono in AUTOLOAD(@_) per colpa di $AUTOLOAD!"
}
Con la dichiarazione our (Lo Scope Our(our)), lo scope della variabile diventa il corpo della funzione. La variabile contiene il nome qualificato della funzione non definita (in questo caso, main::cuoci_crostata). Potete rimuovere il nome del package con un'espressione regolare (Espressioni Regolari e Matching(regex)):
sub AUTOLOAD
{
my ($nome) = our $AUTOLOAD =~ /::(\w+)$/;
# stampa gli argomenti separati da virgola
local $" = ', ';
say "Sono in AUTOLOAD(@_) per colpa di $nome!"
}
La chiamata originale riceve ciò che viene restituito da AUTOLOAD():
say funzione_misteriosa( -1 );
sub AUTOLOAD { return 'mu' }
Gli esempi visti finora si sono limitati ad intercettare le chiamate a funzioni non definite. Naturalmente potete fare di più.
Redirezionare dei Metodi in AUTOLOAD()
Un pattern diffuso nella programmazione OO (Moose(moose)) è di avere un oggetto che delega o fa da proxy per alcuni metodi di un altro oggetto, spesso contenuto o comunque accessibile dal primo. Un proxy di logging può aiutarvi a fare debugging:
package Proxy::Log;
sub new
{
my ($classe, $delegata) = @_;
bless \$delegata, $classe;
}
sub AUTOLOAD
{
my ($nome) = our $AUTOLOAD =~ /::(\w+)$/;
Log::chiamata_metodo( $nome, @_ );
my $self = shift;
return $$self->$nome( @_ );
}
La funzione AUTOLOAD() estrae il nome del metodo non definito e fa il log della sua chiamata. Quindi, ricava l'oggetto delegato dereferenziando un riferimento scalare battezzato e, infine, invoca il metodo sull'oggetto delegato passandogli i parametri ricevuti.
Generazione di Codice in AUTOLOAD()
Questa doppia chiamata è facile da scrivere, ma inefficiente. L'esecuzione di ogni chiamata di metodo sull'oggetto che fa da proxy deve fallire per entrare in AUTOLOAD(). Potete pagare questa penalità una sola volta, installando nuovi metodi nella classe che fa da proxy man mano che il programma li richiede:
sub AUTOLOAD
{
my ($nome) = our $AUTOLOAD =~ /::(\w+)$/;
my $metodo = sub
{
Log::chiamata_metodo( $nome, @_ );
my $self = shift;
return $$self->$nome( @_ );
}
no strict 'refs';
*{ $AUTOLOAD } = $metodo;
return $metodo->( @_ );
}
Il corpo della AUTOLOAD() precedente è diventato una chiusura (Chiusure(closures)) legata con il nome del metodo non definito. Installare tale chiusura nell'appropriata tabella dei simboli permette alle chiamate successive di quel metodo di trovare la chiusura che avete creato (ed evitare AUTOLOAD()). Infine, la chiusura invoca direttamente il metodo e restituisce il risultato.
Sebbene questo approccio sia più pulito e quasi sempre più trasparente rispetto a gestire il comportamento direttamente in AUTOLOAD(), il codice chiamato da AUTOLOAD() ha la possibilità di scoprire che la chiamata è passata attraverso AUTOLOAD(). Infatti, una semplice chiamata a caller() rivela la doppia chiamata di entrambe le tecniche descritte qui sopra. Se da una parte tener conto di questo può essere visto come una violazione dell'incapsulamento, lo stesso si potrebbe dire dell'esporre i dettagli di come un oggetto fornisce un metodo.
Alcuni autori usano una chiamata di coda (Chiamate di coda(tailcalls)) per sostituire la chiamata corrente di AUTOLOAD() con una chiamata al metodo destinatario:
sub AUTOLOAD
{
my ($nome) = our $AUTOLOAD =~ /::(\w+)$/;
my $metodo = sub { ... }
no strict 'refs';
*{ $AUTOLOAD } = $metodo;
goto &$metodo;
}
Questo ha lo stesso effetto di chiamare direttamente $metodo, eccetto che AUTOLOAD() non compare più nella lista di chiamate generate da caller(), e dà quindi l'illusione che il metodo generato sia stato chiamato direttamente.
Svantaggi di AUTOLOAD
AUTOLOAD() può essere uno strumento utile, ma è difficile da usare correttamente. Un approccio naïf alla generazione di metodi a tempo di esecuzione significa tra l'altro che il metodo can() non è in grado di restituire un'informazione corretta sulle capacità degli oggetti e delle classi. La soluzione più semplice consiste nel predichiarare con la direttiva subs tutte le funzioni che pensate di gestire con AUTOLOAD() :
use subs qw( rosso verde blu ocra grigio );
Le dichiarazioni anticipate sono utili solo nei due casi piuttosto rari dell'uso degli attributi e dell'autoload (AUTOLOAD(autoload)).
Questa tecnica ha il vantaggio di documentare le vostre intenzioni ma per contro vi obbliga a mantenere una lista statica di funzioni e metodi. A volte è meglio fare l'override di can():
sub can
{
my ($self, $metodo) = @_;
# usa il risultato del can() della classe padre
my $rif_a_met = $self->SUPER::can( $metodo );
return $rif_a_met if $rif_a_met;
# aggiungi qui qualche filtro
return unless $self->devo_generare( $metodo );
$rif_a_met = sub { ... };
no strict 'refs';
return *{ $metodo } = $rif_a_met;
}
sub AUTOLOAD
{
my ($self) = @_;
my ($nome) = our $AUTOLOAD =~ /::(\w+)$/;>
return unless my $rif_a_met = $self->can( $nome );
goto &$rif_a_met;
}
AUTOLOAD() è come un bulldozer; può catturare funzioni e metodi che non avevate alcuna intenzione di gestire, come DESTROY(), il distruttore degli oggetti. Se scrivete un metodo DESTROY() senza implementazione, Perl non avrà problemi a invocarlo al posto di chiamare AUTOLOAD():
# evita AUTOLOAD()
sub DESTROY {}
I metodi speciali import(), unimport() e VERSION() non passano mai attraverso AUTOLOAD().
Se mescolate funzioni e metodi in un unico namespace che eredita da un altro package, e questo package a sua volta fornisce AUTOLOAD(), potreste vedere questo strano messaggio di errore:
Use of inherited AUTOLOAD for non-method
sbatti_porta() is deprecated
Se vi dovesse succedere, provate a semplificare il vostro codice; avete chiamato una funzione che non esiste in un package che eredita da una classe che contiene un proprio AUTOLOAD(). Ci sono diversi problemi: mescolare funzioni e metodi nello stesso namespace è spesso indice di un errore di progettazione; l'ereditarietà e AUTOLOAD() portano a un aumento molto rapido della complessità; e, infine, ragionare su del codice quando non sapete quali metodi vengono forniti dagli oggetti è difficile.
AUTOLOAD() è utile per programmi piccoli e scritti velocemente, ma è meglio evitarlo per avere del codice robusto.
Espressioni Regolari e Matching
La potenza di Perl nell'elaborazione di testi è dovuta al suo uso della espressioni regolari. Un'espressione regolare (spesso abbreviata in regex o regexp) è un pattern (footnote: NdT: si mantiene la terminologia inglese per i concetti principali, come appunto pattern, matching, ecc.) che descrive le caratteristiche di un pezzo di testo. Un motore di espressioni regolari interpreta i pattern e li applica per fare il match su pezzi di testo e modificarli.
La documentazione base di Perl sulle regex include un tutorial (perldoc perlretut), una guida di riferimento (perldoc perlreref) e un trattato completo (perldoc perlre). Il libro Mastering Regular Expressions di Jeffrey Friedl spiega la teoria e i meccanismi alla base del funzionamento delle espressioni regolari. Sebbene padroneggiare completamente le espressioni regolari sia un'impresa piuttosto ardua, anche una conoscenza limitata vi fornisce degli strumenti estremamente potenti.
Letterali
Le regex più semplici sono dei pattern costituiti da una sottostringa:
my $nome = 'Castellamonte';
say 'Ho trovato una stella!' if $nome =~ /stella/;
L'operatore di match (m//, abbreviato //) identifica una espressione regolare--in questo esempio, stella. Questo pattern non è una parola. Il suo significato è "il carattere s, seguito dal carattere t, seguito dal carattere e, ecc.". Ogni carattere nel pattern è un elemento indivisibile, o atomo. Può combaciare o non combaciare con un carattere in una stringa.
L'operatore di binding a regex (=~) è un operatore infisso (Posizione(fixity)) che applica la regex del suo secondo operando a una stringa fornita dal suo primo operando. Quando viene valutato in contesto scalare, esso restituisce vero se il match ha successo. La forma negata dell'operatore di binding (!~) valuta a vero se il match non ha successo.
Potete anche usare la funzione predefinita index per cercare una sottostringa letterale in una stringa. Usare un motore di espressioni regolari per questo compito è come andare al negozio giù all'angolo a comprare il latte su un elicottero da combattimento--ma Perl vi permette di scegliere la forma che vi sembra più facilmente manutenibile.
L'operatore di sostituzione, s///, è sotto certi aspetti un operatore di raggruppamento (Posizione(fixity)) con due operandi. Il suo primo operando è un'espressione regolare, di cui si vuole fare il match in una stringa quando è presente anche l'operatore di binding di regex. Il secondo operando è una sottostringa usata per sostituire la porzione di stringa riconosciuta col match. Per esempio, per curare una fastidiosa allergia estiva, potete scrivere:
my $stato = 'Sto male.';
$stato =~ s/male/bene/;
say $stato;
L'operatore qr// e la Combinazione di Regex
L'operatore qr// crea delle regex come oggetti di prima classe. Per usarle, interpolatele nell'operatore di match:
my $stella = qr/stella/;
say 'Ho trovato una stella!' if $nome =~ /$stella/;
... o combinate diversi oggetti regex in dei pattern complessi:
my $stella = qr/stella/;
my $monte = qr/monte/;
say 'Ho trovato una stella sopra un monte!'
if $nome =~ /$stella$monte/;
like( $nome, qr/$stella$monte/,
'Ho trovato una stella sopra un monte!' );
La funzione like di Test::More verifica se c'è un match tra il primo argomento e la regex fornita come secondo argumento.
Quantificatori
La potenza delle espressioni regolari è incrementata dall'uso dei quantificatori di regex, che vi permettono di specificare quante volte una componente della regex può comparire in una striga con cui viene fatto il match. Il più semplice è il quantificatore zero o uno, scritto ?:
my $miao_o_mio = qr/mia?o/;
like( 'miao', $miao_o_mio, "il match di 'miao' con /mia?o/ ha successo" );
like( 'mio', $miao_o_mio, "il match di 'mio' con /mia?o/ ha successo" );
La presenza di un atomo in una espressione regolare seguito dal carattere ? significa "fai il match con questo atomo zero o una volta". L'espressione regolare qui sopra trova un match se zero o una occorrenza del carattere a precedono immediatamente un carattere o, quindi il match ha successo con le sottostringhe letterali miao e mio.
Il quantificatore uno o più, scritto +, trova un match solo se c'è almeno una occorrenza dell'atomo quantificato:
my $qualche_a = qr/mia+o/;
like( 'miao', $qualche_a, "il match di 'miao' con /mia+o/ ha successo" );
like( 'miaao', $qualche_a, "anche quello di 'miaao'" );
like( 'miaaao', $qualche_a, "anche quello di 'miaaao'" );
like( 'miaaaao', $qualche_a, "anche quello di 'miaaaao'" );
unlike( 'mio', $qualche_a, "invece quello di 'mio' fallisce" );
In teoria non c'è alcun limite massimo al numero di occorrenze dell'atomo quantificato per cui il match ha successo.
Il quantificatore zero o più, scritto *, trova un match se ci sono zero o più occorrenze dell'atomo quantificato:
my $qualunque_a = qr/mia*o/;
like( 'miao', $qualunque_a, "il match di 'miao' con /mia*o/ ha successo" );
like( 'miaao', $qualunque_a, "anche quello di 'miaao'" );
like( 'miaaao', $qualunque_a, "anche quello di 'miaaao'" );
like( 'miaaaao', $qualunque_a, "anche quello di 'miaaaao'" );
like( 'mio', $qualunque_a, "e anche quello di 'mio'" );
Per quanto ciò possa sembrare curioso, questo operatore vi permette di specificare componenti opzionali di una regex. Ricordatevi di usarlo con parsimonia: è uno strumento costoso e di scarsa precisione. La maggior parte delle espressioni regolari beneficia dell'uso dei quantificatori ? e + molto più che di *. Spesso precisare le vostre intenzioni aumenta la chiarezza.
I quantificatori numerici definiscono un numero specifico di possibili occorrenze di un atomo. {n} significa che il match deve avvenire esattamente n volte.
# equivalente a qr/miao/;
my $solo_una_a = qr/mia{1}o/;
like( 'miao', $solo_una_a, "il match di 'miao' con /mia{1}o/ ha successo" );
{n,} fa il match con un atomo almeno n volte:
# equivalente a qr/mia+o/;
my $qualche_a = qr/mia{1,}o/;
like( 'miao', $qualche_a, "il match di 'miao' con /mia{1,}o/ ha successo" );
like( 'miaao', $qualche_a, "anche quello di 'miaao'" );
like( 'miaaao', $qualche_a, "anche quello di 'miaaao'" );
like( 'miaaaao', $qualche_a, "anche quello di 'miaaaao'" );
{n,m} significa che il match deve avvenire almeno n volte e non più di m volte:
my $poche_a = qr/mia{1,3}o/;
like( 'miao', $poche_a, "il match di 'miao' con /mia{1,3}o/ ha successo" );
like( 'miaao', $poche_a, "anche quello di 'miaao'" );
like( 'miaaao', $poche_a, "anche quello di 'miaaao'" );
unlike( 'miaaaao', $poche_a, "invece quello di 'miaaaao' fallisce" );
Potete esprimere i quantificatori simbolici in termini di quantificatori numerici, ma la maggior parte dei programmi usa i primi molto più spesso dei secondi.
Greediness
I quantificatori + e * sono greedy (NdT avidi, aggressivi), nel senso che tentano di includere nel match il maggior numero possibile di caratteri della stringa in input. Questo punto è particolarmente pericoloso. Considerate un uso naïf del pattern .*, ovvero "zero o più caratteri, a capo escluso":
# una regex mediocre
my $pasto_caldo = qr/pasto.*caldo/;
say 'Ho trovato un pasto caldo!'
if 'Ti offro un pasto caldo' =~ $pasto_caldo;
say 'Ho trovato un pasto caldo!'
if 'l'impasto della torta deve essere caldo' =~ $pasto_caldo;
I quantificatori greedy iniziano con l'includere tutto nel match, e scartano i caratteri uno alla volta, solo quando diventa ovvio che non possono far parte del match.
Potete usare il modificatore di quantificatori ? per rendere parsimoniosi i quantificatori greedy:
my $per_nulla_avido = qr/pasto.*?caldo/;
Quando elabora un quantificatore non-greedy, il motore di espressioni regolari preferisce il più corto dei match potenziali e aggiunge altri caratteri a quelli già identificati dal pattern .*? solo se con quelli correnti il match fallirebbe. Dato che * accetta zero o più occorrenze, potenzialmente il numero minimo di caratteri con cui il pattern d'esempio fa il match è zero:
say 'Ho trovato un pasto caldo'
if 'vorreiunpastocaldo' =~ /$per_nulla_avido/;
Usate +? per fare il match di una o più occorrenze in modo non-greedy:
my $per_nulla_avido_bis = qr/pasto.+?caldo/;
unlike( 'vorreiunpastocaldo', $per_nulla_avido_bis );
like( 'vorrei un pasto caldo', $per_nulla_avido_bis );
Il modificatore di quantificatori ? si applica anche al quantificatore ? (zero o un match) e ai quantificatori numerici. In ogni caso, il suo effetto è quello di fare sì che la regex includa nel match il minor numero possibile di caratteri di input.
I pattern greedy .+ e .* possono essere allettanti, ma sono pericolosi. Un appassionato di cruciverba che deve riempire le quattro caselle del 7 Verticale ("La regina dei fiori") riceverà un sacco di proposte sbagliate con il pattern:
my $sette_verticale = qr/r$solo_lettere*a/;
Dovrebbe scartare Armenia, Bretagna e Brasilia molto prima che il programma suggerisca rosa. Non solo queste parole sono troppo lunghe, ma il match inizia in mezzo ad esse. Una comprensione di prima mano del funzionamento della greediness può aiutare, ma non può sostituire l'esperienza acquisita con un gran numero di test su dati reali.
Ancore di Regex
Le ancore di regex forzano il motore di regex a iniziare o terminare un match in una posizione assoluta. L'ancora di inizio stringa (\A) prescrive che l'inizio del match coincida con l'inizio della stringa:
# riconosce anche "ricarica", "relax" e "ripartire"
my $sette_verticale = qr/\Ar${solo_lettere}{2}a/;
L'ancora di fine linea (\Z) richiede che la fine del match coincida con la fine di una linea nella stringa.
# riconosce anche "rara", ma e` sicuramente un bel passo avanti
my $sette_verticale = qr/\Ar${solo_lettere}{2}a\Z/;
L'ancora di limite parola (\b) trova un match solo al limite tra un carattere di una parola (\w) e un carattere che non appartiene alla parola (\W). Usate una regex con questa ancora per riconoscere rosa e scartare invece carica:
my $sette_verticale = qr/\br${solo_lettere}{2}a\b/;
Metacaratteri
Perl interpreta alcuni caratteri nelle espressioni regolari come metacaratteri, ovvero caratteri che rappresentano qualcosa di diverso dalla loro interpretazione letterale. I metacaratteri danno a chi sa usare le regex una potenza che va ben al di là del match di sottostringhe. Il motore di regex tratta tutti i metacaratteri come atomi.
Il metacarattere . significa "fai il match con qualunque carattere eccetto il carattere di a capo". Ricordate l'avvertimento dato sopra; molti principianti lo dimenticano. Una semplice ricerca con regex--che ignora le migliorie possibili con l'uso delle ancore--per il 7 Vericale potrebbe essere /r..a/. Naturalmente, ci sono sempre diversi modi di ottenere la risposta corretta:
for my $parola (@parole)
{
next unless length( $parola ) == 4;
next unless $parola =~ /r..a/;
say "Potrebbe essere: $parola";
}
Se le potenziali soluzioni in @parole non contengono solo delle semplici parole in italiano, potreste ottenere dei falsi positivi. Infatti, . riconosce anche la punteggiatura, la spaziatura e i numeri. Siate specifici! Il metacarattere \w rappresenta tutti i caratteri alfanumerici (Stringhe e Unicode(unicode)) e il carattere di sottolineatura:
next unless $parola =~ /r\w\wa/;
Il metacarattere \d riconosce le cifre (sempre all'interno di Unicode):
# un riconoscitore di numeri di carta di credito non molto robusto
next unless $numero =~ /\d{4}-\d{4}-\d{4}-\d{4}/;
say "Il tuo numero di carta di credito e`: $numero";
Usate il metacarattere \s per riconoscere la spaziatura, che si tratti di un semplice carattere di spazio, di un carattere di tabulazione, di un carriage return, di un form-feed o di un a capo:
my $due_parole_di_tre_lettere = qr/\w{3}\s\w{3}/;
Questi metacaratteri hanno delle corrispondenti forme negate. Usate \W per riconoscere qualunque carattere eccetto un carattere di parola. Usate \D per riconoscere un carattere che non sia una cifra. Usate \S per riconoscere qualunque cosa tranne un carattere di spaziatura. Usate \B per iniziare il match in qualunque posizione eccetto che ai limiti di una parola.
Classi di Caratteri
Quando nessuno dei metacaratteri descritti sopra è sufficientemente specifico, potete definire la vostra classe di caratteri racchiudendoli tra parentesi quadre:
my $vocali_ascii = qr/[aeiou]/;
my $forse_un_gatto = qr/g${vocali_ascii}tto/;
Senza l'uso delle parentesi graffe, il parser Perl interpreterebbe il nome di variabile come $vocali_asciitto, il che potrebbe causare un errore di compilazione relativo all'uso di una variabile sconosciuta oppure interpolare il contenuto di una variabile esistente $vocali_asciitto nella regex.
Il carattere trattino (-) vi permette di specificare una serie di caratteri contigui in una classe, come in questa regex $solo_lettere_ascii:
my $solo_lettere_ascii = qr/[a-zA-Z]/;
Per includere il trattino in una classe, posizionatelo all'inizio o alla fine:
my $punteggiatura_utile = qr/[-!?]/;
... o fatene l'escape:
my $caratteri_con_lineette = qr/[|=\-_]/;
Usate l'accento circonflesso (^) come primo elemento di una classe di caratteri per specificare "qualunque cosa eccetto questi caratteri":
my $non_una_vocale_ascii = qr/[^aeiou]/;
Usate un accento circonflesso in qualunque posizione eccetto la prima per includerlo in una classe di caratteri. Per includere un trattino in una classe di caratteri negata, posizionatelo subito dopo l'accento circonflesso o alla fine, o fatene l'escape.
Cattura
Le espressioni regolari vi permettono di raggruppare e catturare parti del match per usarle in seguito. Per estrarre un numero telefonico americano nella forma (202) 456-1111 da una stringa:
my $prefisso = qr/\(\d{3}\)/;
my $numero_locale = qr/\d{3}-?\d{4}/;
my $numero_telefonico = qr/$prefisso\s?$numero_locale/;
Notate in particolare l'escape delle parentesi dentro al $prefisso. Le parentesi sono caratteri speciali nelle espressioni regolari di Perl 5. Esse raggruppano degli atomi in unità più grandi e servono anche a catturare delle porzioni della stringa con cui si fa il match. Per riconoscere le parentesi in modo letterale, fatene l'escape con il backslash come è stato fatto in $prefisso.
Catture con Nome
Perl 5.10 ha introdotto le catture con nome, che vi permettono di catturare porzioni di match durante l'applicazione di un'espressione regolare e accedervi in seguito; ad esempio, potete catturare un numero telefonico in una stringa contenente informazioni su un vostro contatto:
if ($info_contatto =~ /(?<tel>$numero_telefonico)/)
{
say "Ho trovato un numero telefonico $+{tel}";
}
Le regex possono sembrare un'accozzaglia di segni di punteggiatura se non ne raggruppate le diverse parti in pezzi separati. La sintassi delle catture con nome è la seguente:
(?<nome cattura> ... )
Le parentesi tonde racchiudono la cattura. Il costrutto ?< nome >, che segue immediatamente la parentesi aperta, assegna un nome alla cattura. Il resto della cattura è un'espressione regolare.
Quando il match di un pattern contenente la cattura ha successo, Perl memorizza la parte di stringa che corrisponde al pattern interno alla cattura nella variabile globale predefinita %+. In tale hash, la chiave è il nome della cattura e il valore è la rispettiva porzione di stringa.
Catture Numerate
Perl supporta le catture numerate dall'alba dei tempi:
if ($info_contatto =~ /($numero_telefonico)/)
{
say "Ho trovato un numero telefonico $1";
}
Questa forma di catture non fornisce un nome identificativo e non usa %+ per la memorizzazione. Invece, le sottostringhe catturate vengono memorizzate da Perl in una serie di variabili globali predefinite. La parte di match della prima cattura trovata finisce in $1, la seconda in $2 e così via. Il conteggio del numero di una cattura avviene quando si incontra la sua parentesi aperta; quindi dopo la prima parentesi aperta inizia la memorizzazione in $1, dopo la seconda in $2 e così via.
Anche se la sintassi per le catture con nome è più verbosa di quella per le catture numerate, in genere risulta più chiara. Contare le parentesi aperte è un lavoro noioso e la complessità di combinare diverse regex ognuna delle quali contiene delle catture numerate diventa proibitiva. Le catture con nome aumentano il grado di manutenibilità delle regex--sebbene ci possano essere collisioni di nome, esse sono relativamente rare. Per minimizzare i rischi, usate le catture con nome solo nelle regex più esterne.
In contesto lista, il match di una regex restituisce una lista di sottostringhe catturate:
if (my ($numero) = $info_contatto =~ /($numero_telefonico)/)
{
say "Ho trovato un numero telefonico $numero";
}
Le catture numerate sono utili anche per semplici sostituzioni in cui le catture con nome risulterebbero troppo verbose:
my $ordine = 'Pasta per vegani!';
$ordine =~ s/(\w+) per vegani/$1 per vegetariani/;
# oppure
$ordine =~ s/(?<cibo>\w+) per vegani/$+{cibo} per vegetariani/;
Raggruppamento e Alternativa
In tutti gli esempi precedenti i quantificatori erano applicati a semplici atomi. In realtà è possibile applicarli a qualunque elemento di una regex:
my $maiale = qr/maiale/;
my $fagioli = qr/fagioli/;
like( 'maiale con fagioli', qr/\A$maiale?.*?$fagioli/,
'forse maiale, sicuramente fagioli' );
Se espandete manualmente la regex, il risultato potrebbe sorprendervi:
my $maiale_e_fagioli = qr/\Amaiale?.*fagioli/;
like( 'maiale con fagioli', qr/$maiale_e_fagioli/,
'forse maiale, sicuramente fagioli' );
like( 'mai fagioli', qr/$maiale_e_fagioli/,
'un momento... non c'e` il fillochinone!' );
A volte essere specifici migliora l'accuratezza dei pattern:
my $maiale = qr/maiale/;
my $con = qr/con/;
my $fagioli = qr/fagioli/;
like( 'maiale con fagioli', qr/\A$maiale? $con? $fagioli/,
'forse maiale, forse con, sicuramente fagioli' );
Alcune regex devono riconoscere una cosa oppure un'altra. Il metacarattere da usare in questi casi è quello di alternativa (|):
my $riso = qr/riso/;
my $fagioli = qr/fagioli/;
like( 'riso', qr/$riso|$fagioli/, 'Ho trovato il riso' );
like( 'fagioli', qr/$riso|$fagioli/, 'Ho trovato i fagioli' );
Il metacarattere di alternativa indica che il match deve essere fatto con l'uno o l'altro dei frammenti adiacenti. Ricordate che l'alternativa ha una precedenza (Precedenza(precedence)) minore delle sequenze di atomi:
like( 'riso', qr/riso|fagioli/, 'Ho trovato il riso' );
like( 'fagioli', qr/riso|fagioli/, 'Ho trovato i fagioli' );
unlike( 'risf', qr/riso|fagioli/, 'Ho trovato un ibrido' );
Anche se viene naturale interpretare riso|fagioli come la stringa ris, seguita dal carattere o oppure f, seguito da agioli, le alternative includono sempre l'intero frammento fino al delimitatore di regex più vicino, che può essere l'inizio o la fine del pattern, una parentesi di raggruppamento, un altro carattere di alternativa oppure una parentesi quadra.
Per ridurre la possibilità di confusione, usate delle variabili per dare dei nomi ai frammenti ($riso|$fagioli) oppure raggruppate le possibili alternative in gruppi non-catturanti:
my $contiene_amido = qr/(?:pasta|patate|riso)/;
La sequenza (?:) raggruppa una serie di atomi senza definire una cattura.
Una espressione regolare convertita in stringa viene automaticamente racchiusa in un raggruppamento non-catturante; qr/riso|fagioli/ diventa la stringa (?:riso|fagioli).
Altre Sequenze di Escape
Per fare il match letterale di un metacarattere, fatene l'escape con un backslash (\). L'avete visto fare negli esempi sopra, dove \( si riferiva a una parentesi tonda aperta e \] si riferiva a una parentesi quadra chiusa. \. si riferisce al carattere letterale punto anziché all'atomo "fai il match con qualunque cosa eccetto un carattere di a capo esplicito".
È probabile che vi succeda di dover fare l'escape del metacarattere di alternativa (|), del metacarattere di fine linea ($) e dei quantificatori (+, ?, *).
I caratteri di disattivazione dei metacaratteri (\Q e \E) disattivano l'interpretazione dei metacaratteri nella parte di regex che si trova tra di essi. Possono essere particolarmente utili quando ricevete l'espressione regolare da una sorgente sulla quale avete il controllo quando scrivete il programma:
my ($testo, $testo_letterale) = @_;
return $testo =~ /\Q$testo_letterale\E/;
Il parametro $testo_letterale potrebbe contenere qualunque cosa--per esempio la stringa ** ALLARME **. Nel frammento delimitato da \Q e \E, Perl interpreta la regex come \*\* ALLARME \*\* e quindi tenta di riconoscere letteralmente dei caratteri asterisco, anziché applicare dei quantificatori greedy.
Siate prudenti quando elaborate delle espressioni regolari provenienti da dell'input utente non attendibile. Un esperto di regex malintenzionato potrebbe provocare un attacco di denial-of-service contro il vostro programma.
Asserzioni
Le ancore di regex come \A, \b, \B e \Z sono una forma di asserzione di regex che richiede alla stringa di soddisfare certe condizioni. Queste asserzioni non fanno il match di singoli caratteri della stringa. Indipendentemente dal contenuto della stringa, il match della regex qr/\A/ avrà sempre successo.
Le asserzioni di lunghezza zero fanno il match di un pattern. La loro peculiarità è che non consumano la porzione della stringa che riconoscono. Per esempio, se vi serve trovare un'auto e nulla di più, potete usare un'asserzione di limite di parola:
my $auto_solitaria = qr/auto\b/;
... ma se volete trovare un'auto senza avere problemi con la polizia, potreste usare un'asserzione di lunghezza zero con negazione in avanti:
my $auto_sicura = qr/auto(?!velox)/;
Il costrutto (?!...) riconosce la stringa auto solo se non è seguita immediatamente dalla stringa velox.
L'asserzione di lunghezza zero in avanti:
my $auto_problematica = qr/auto(?=velox)/;
... riconosce la stringa auto solo se è seguita immediatamente dalla stringa velox. Sebbene una normale espressione regolare possa ottenere lo stesso risultato, considerate una regex per trovare tutte le parole del dizionario che iniziano con auto e vi evitano guai con la polizia:
my $auto_sicura = qr/auto(?!velox)/;
while (<$parole>)
{
chomp;
next unless /\A(?<auto>$auto_sicura.*)\Z/;
say "Trovata un'auto sicura '$+{auto}'";
}
L'asserzione di lunghezza zero non consuma alcun carattere della stringa in esame, lasciando tutto il match al frammento ancorato <.*\Z>. Altrimenti, la parte catturata consisterebbe solo nella porzione auto della stringa in esame.
Per asserire che la vostra auto non occorre all'inizio di una linea, potete usare l'asserzione di lunghezza zero con negazione all'indietro. Le asserzioni di questo tipo devono avere lunghezza prefissata; quindi, non potete usare dei quantificatori:
my $auto_media = qr/(?<!\A)auto/;
Il costrutto (?<!...) contiene un pattern di lunghezza fissa. Potete anche richiedere che l'auto si trovi sempre dopo un carattere di spazio con un'asserzione di lunghezza zero all'indietro:
my $auto_spaziale = qr/(?<=\s)auto/;
Il costrutto (?<=...) contiene un pattern di lunghezza fissa. Questo approccio può esservi utile quando combinate un match globale di regex con il modificatore \G, ma è improbabile che vi troviate spesso ad usare questa funzionalità avanzata.
Una nuova funzionalità delle regex di Perl 5 è l'asserzione keep \K. Questa asserzione di lunghezza zero all'indietro può essere associata a un pattern di lunghezza variabile:
my $auto_spaziale = qr/\s+\Kauto/;
like( 'la mia auto e` andata nello spazio', $auto_spaziale );
like( 'la mia auto e` andata nello spazio doppio',
$auto_spaziale );
\K è particolarmente utile per certe sostituzioni che rimuovono la parte finale di un pattern:
my $esclamazione = 'Ho visto l\'autovelox!';
$esclamazione =~ s/auto\K\w+!/./;
like( $esclamazione, qr/\bauto\./,
"Forse non c'era da preoccuparsi tanto!" );
Modificatori di Regex
Potete utilizzare svariati modificatori per cambiare il comportamento degli operatori delle espressioni regolari. Tali modificatori compaiono alla fine degli operatori di match, di sostituzione e qr//. Per esempio, per attivare l'insensibilità tra maiuscole e minuscole in un match:
my $fiore = 'CaMeLiA';
like( $fiore, qr/Camelia/, 'Che bel fiore!' );
like( $fiore, qr/Camelia/i, 'tasto shift fuori uso' );
Il primo like() fallisce, perché le due stringhe contengono lettere diverse. Invece il secondo like() ha successo, perché il modificatore /i fa sì che la regex ignori la distinzione tra maiuscole e minuscole. Nella seconda regex, M e m sono considerate equivalenti grazie al modificatore.
Potete anche incorporare i modificatori di regex in un pattern:
my $trova_un_gatto = qr/(?<felino>(?i)gatto)/;
La sintassi (?i) attiva l'insensibilità tra maiuscole e minuscole solo nel raggruppamento in cui è racchiuso: in questo caso, una cattura con nome. Potete attivare diversi modificatori in questo modo. Per disattivare degli specifici modificatori fateli precedere dal carattere trattino (-):
my $trova_un_razionale = qr/(?<numero>(?-i)Raz)/;
Il modificatore di multilinea, /m, permette alle ancore \A e \Z di fare un match ad ogni occorrenza del carattere di a capo contenuta nella stringa.
Il modificatore /s tratta la stringa in esame come un'unica linea, quindi il metacarattere . riconosce anche il carattere di a capo. Damian Conway suggerisce un espediente mnemonico per cui /m modifica il comportamento di molti metacaratteri nella regex, mentre /s modifica il comportamento di un singolo metacarattere.
Il modificatore /r fa sì che un'operazione di sostituzione restituisca il risultato della sostituzione, lasciando inalterata la stringa originale. Se la sostituzione ha successo, il risultato è una copia modificata dell'originale. Se la sostituzione fallisce (perché il pattern non viene riconosciuto), il risultato è una copia identica all'originale:
my $stato = 'Sono affamato di crostata.';
my $nuovostato = $stato =~ s/crostata/torta/r;
my $statocopia = $stato
=~ s/fegato e cipolle/wurstel/r;
is( $stato, 'Sono affamato di crostata.',
'la stringa originale dovrebbe essere inalterata' );
like( $nuovostato, qr/torta/, 'voglio la torta' );
unlike( $statocopia, qr/wurstel/, 'niente insaccati' );
Il modificatore /x vi permette di inserire spaziatura e commenti aggiuntivi nei pattern. Quando questo modificatore è attivo, il motore di regex ignora la spaziatura e i commenti. Spesso la regex risultante è molto più leggibile:
my $regex_attrib = qr{
\A # inizio linea
(?:
[;\n\s]* # spaziatura e punto e virgola
(?:/\*.*?\*/)? # commenti C
)*
ATTR
\s+
( U?INTVAL
| FLOATVAL
| STRING\s+\*
)
}x;
Questa regex non è per nulla semplice, ma i commenti e la spaziatura migliorano la sua leggibilità. Anche se componete una regex da frammenti precompilati, il modificatore /x può comunque migliorare il vosto codice.
Il modificatore /g fa il match di una regex globalmente nell'intera stringa. Il suo uso ha senso nelle sostituzioni:
# un po' di pace in Via col Vento
my $contenuto = slurp( $file );
$contenuto =~ s/Rossella O'Hara/Camomilla Calmante/g;
Quando è usato con un match--e non una sostituzione--il metacarattere \G vi permette di elaborare una stringa un pezzo alla volta all'interno di un ciclo. Il match di \G ha successo alla posizione dove è terminato il match precedente più recente. Per memorizzare in modo ordinato i numeri telefonici americani contenuti in un file poco strutturato, potreste scrivere:
while ($contenuto =~ /\G(\w{3})(\w{3})(\w{4})/g)
{
push @numeri, "($1) $2-$3";
}
Ricordate che l'ancora \G riprende all'ultima posizione della stringa riconosciuta dal match dell'iterazione precedente. Se il match precedente terminava con un quantificatore greedy come .*, il match successivo avrà a disposizione una parte più ridotta di stringa. Anche le asserzioni in avanti possono esservi utili.
Il modificatore /e vi permette di scrivere codice Perl 5 arbitrario nella parte destra di un'operazione di sostituzione. Se il match ha successo, il motore di regex esegue il codice e usa il valore restituito come valore per la sostituzione. L'esempio di sostituzione globale visto prima potrebbe essere reso più semplice con del codice come quello che segue:
# un po' di pace in Via col Vento
$sequel =~ s{Rossella( O'Hara)?}
{
'Camomilla' . defined $1
? ' Calmante'
: ''
}ge;
Ogni occorrenza aggiuntiva del modificatore /e provoca un'altra valutazione del risultato dell'espressione, ma solo gli appassionati di golf in Perl si spingono oltre l'uso di /ee.
Matching Intelligente
L'operatore di match intelligente, ~~, confronta due operandi e restituisce un valore vero se corrispondono. L'indeterminatezza della definizione dimostra l'intelligenza dell'operatore: il tipo di confronto dipende dai tipi di entrambi gli operandi. given (Given/When(given_when)) effettua implicitamente un match intelligente.
L'operatore di match intelligente è un operatore infisso:
say 'Concordano (per qualche ragione)' if $operandos ~~ $operandod;
Il tipo di confronto dipende generalmente in primo luogo dal tipo dell'operando destro e in secondo luogo dell'operando sinistro. Per esempio, se l'operando destro è uno scalare con una componente numerica, il confronto sarà basato sull'uguaglianza numerica. Se l'operando destro è una regex, il confronto userà un grep o un match di pattern. Se l'operando destro è un array, il confronto effettuerà un grep o un match intelligente ricorsivo. Se l'operando destro è un hash, il confronto verificherà l'esistenza di una o più chiavi. Una tabella di dimensioni scoraggianti in perldoc perlsyn fornisce molti più dettagli sui confronti che il match intelligente può effettuare.
Una proposta importante per la versione 5.16 suggerisce di semplificare il match intelligente in modo sostanziale. Più i vostri operandi sono complicati, più è probabile che i risultati siano poco chiari. Se evitate di confrontare oggetti e vi limitate a semplici operazioni con scalari o al più tra uno scalare e un aggregato, otterrete risultati migliori.
Ciò detto, il match intelligente può essere utile:
my ($x, $y) = (10, 20);
say 'Numericamente diversi' unless $x ~~ $y;
my $z = '10 piccoli indiani';
say 'Pseudo-numericamente uguali' if $x ~~ $z;
# match di espressione regolare
my $ago = qr/ago/;
say 'Match di pattern' if 'ago' ~~ $ago;
say 'Grep su array' if @pagliaio ~~ $ago;
say 'Grep su chiavi di hash' if %hashiago ~~ $ago;
say 'Grep su array' if $ago ~~ @pagliaio;
say 'Gli elementi dell'array esistono come chiavi di hash'
if %hashiago ~~ @pagliaio;
say 'Match intelligente tra elementi di array' if @mucchio_di_paglia ~~ @pagliaio;
say 'Grep su chiavi di hash' if $ago ~~ %hashiago;
say 'Gli elementi dell'array esistono come chiavi di hash'
if @pagliaio ~~ %hashiago;
say 'Chiavi di hash identiche' if %hashiago ~~ %mappago;
Il match intelligente funziona anche quando uno degli operandi è un riferimento a uno dei tipi di dati elencati:
say 'Chiavi di hash identiche' if %hashiago ~~ \%hashiago;
Oggetti
Programmare è un'attività di gestione. Più il programma è lungo, più dettagli vi trovate a dover gestire. L'unica possibilità di gestire con successo una tale complessità è un opportuno uso dell'astrazione (ossia la capacità di trattare cose simili in modo simile) e dell'incapsulamento (ossia il raggruppamento di dettagli che sono in relazione tra loro).
Il solo uso delle funzioni non è sufficiente per i programmi più grandi. Ci sono diverse tecniche per raggruppare le funzioni in unità di comportamenti correlati. Una di queste tecniche è l'orientamento agli oggetti (OO), o programmazione orientata agli oggetti (OOP), secondo la quale i programmi lavorano su oggetti--entità univoche e discrete con una propria identità.
Moose
Il sistema ad oggetti di default di Perl 5 è flessibile, ma minimale. Anche se potete creare cose eccellenti a partire da esso, tale sistema vi fornisce poca assistenza anche per i compiti basilari. Moose è un sistema ad oggetti completo per Perl 5 (footnote: Vedete perldoc Moose::Manual per maggiori informazioni.). Fornisce dei default più semplici e delle funzionalità avanzate mutuate da linguaggi quali Smalltalk, Common Lisp e Perl 6. Il codice Moose è interoperabile con il sistema ad oggetti di default e rappresenta attualmente il modo migliore di scrivere codice orientato agli oggetti in Perl 5 moderno.
Classi
Un oggetto Moose è un'istanza concreta di una classe, che a sua volta è un modello che descrive i dati e il comportamento specifici dell'oggetto. Le classi usano i package (Package(packages)) per fornire dei namespace:
package Gatto
{
use Moose;
}
Questa classe Gatto sembrerebbe non fare nulla, ma questo è tutto ciò di cui Moose ha bisogno per definire una classe. Potete creare oggetti (o istanze) della classe Gatto con la seguente sintassi:
my $brad = Gatto->new();
my $jack = Gatto->new();
Così come può dereferenziare un riferimento, una freccia può invocare un metodo su un oggetto o su una classe.
Metodi
Un metodo è una funzione associata a una classe. Così come le funzioni appartengono a dei namespace, i metodi appartengono a delle classi, con due differenze. Per prima cosa, un metodo opera sempre su un invocante. Chiamare new() su Gatto invia un messaggio alla classe Gatto. Il nome della classe, Gatto, è l'invocante di new(). Quando chiamate un metodo su un oggetto, l'invocante è l'oggetto stesso:
my $choco = Gatto->new();
$choco->dormi_sulla_tastiera();
In secondo luogo, una chiamata a un metodo coinvolge sempre una strategia di dispatch, per mezzo della quale il sistema ad oggetti sceglie il metodo appropriato. Data la semplicità della classe Gatto, la strategia di dispatch è ovvia, ma gran parte della potenza dell'OO deriva da questa idea.
All'interno di un metodo, il primo argomento è l'invocante. Il nome idiomatico usato in Perl 5 è $self. Considerate che un Gatto (normalmente) miagola():
package Gatto
{
use Moose;
sub miagola
{
my $self = shift;
say 'Miao!';
}
}
Ora tutte le istanze di Gatto possono svegliarvi al mattino perché non hanno ancora mangiato:
my $strana_sveglia = Gatto->new();
$strana_sveglia->miao() for 1 .. 3;
I metodi che accedono ai dati dell'invocante sono metodi d'istanza, perché dipendono dalla presenza dell'invocante appropriato per funzionare correttamente. I metodi (come miagola()) che non accedono ai dati d'istanza sono metodi di classe. Potete invocare i metodi di classe sulle classi e i metodi di classe e di istanza sulle istanze, ma non potete invocare metodi di istanza sulle classi.
I costruttori, che creano le istanze, sono ovviamente dei metodi di classe. Moose vi fornisce un costruttore di default.
I metodi di classe sono a tutti gli effetti delle funzioni globali di namespace. Dato che non accedono ai dati d'istanza, offrono pochi vantaggi rispetto alle funzioni di namespace. Gran parte del codice OO è giustamente basato sui metodi d'istanza, che hanno accesso ai dati d'istanza.
Attributi
Ciascun oggetto Perl 5 è unico. Gli oggetti possono contenere dei dati privati associati a ciascun singolo oggetto--detti attributi, dati d'istanza o stato dell'oggetto. Per definire un attributo, dichiaratelo come parte della classe:
package Gatto
{
use Moose;
has 'nome', is => 'ro', isa => 'Str';
}
Parafrasato in italiano, questo codice dice che che "gli oggetti Gatto hanno un attributo nome. Tale attributo è di sola lettura ed è una stringa".
Moose fornisce la funzione has(), che dichiara un attributo. Il primo argomento, 'nome' nel nostro esempio, è il nome dell'attributo. La coppia di argomenti is => 'ro' specifica che l'attributo è read only, ovvero che non potete modificarlo dopo averlo inizializzato. Infine, la coppia isa => 'Str' specifica che il valore dell'attributo può essere solo una stringa.
In conseguenza all'esecuzione di has, Moose crea un metodo accessore chiamato nome() e vi permette di passare un parametro nome al costruttore di Gatto:
for my $nome (qw( Tuxie Petunia Daisy ))
{
my $gatto = Gatto->new( nome => $nome );
say "Creato un gatto chiamato ", $gatto->nome();
}
La documentazione di Moose usa delle parentesi per separare i nomi degli attributi dalle loro caratteristiche:
has 'nome' => ( is => 'ro', isa => 'Str' );
Tale sintassi è equivalente a:
has( 'nome', 'is', 'ro', 'isa', 'Str' );
L'approccio di Moose è appropriato per dichiarazioni complesse:
has 'nome' => (
is => 'ro',
isa => 'Str',
# opzioni avanzate di Moose; perldoc Moose
init_arg => undef,
lazy_build => 1,
);
... mentre questo libro adotta l'altra punteggiatura vista sopra per semplici dichiarazioni. Scegliete la punteggiatura che vi permette di essere più chiari.
Se passate qualcosa che non sia una stringa, Moose avrà da ridire. Va però ricordato che gli attributi non devono necessariamente avere un tipo. In tal caso, accettano tutto:
package Gatto
{
use Moose;
has 'nome', is => 'ro', isa => 'Str';
has 'eta', is => 'ro';
}
my $sbagliato = Gatto->new( nome => 'micio',
eta => 'bizzarro' );
Specificando un tipo permettete a Moose di effettuare delle validazioni sui dati al posto vostro. Talvolta i benefici di una maggior rigidità sono inestimabili.
Se dichiarate un attributo come leggibile e scrivibile (con is => rw), Moose crea un metodo mutatore che può modificare il valore dell'attributo:
package Gatto
{
use Moose;
has 'nome', is => 'ro', isa => 'Str';
has 'eta', is => 'ro', isa => 'Int';
has 'dieta', is => 'rw';
}
my $grasso = Gatto->new( nome => 'Ciccio',
eta => 8,
dieta => 'Frutti di Mare' );
say $grasso->nome(), ' mangia ', $grasso->dieta();
$grasso->dieta( 'Crocchette a Basso Contenuto di Sodium' );
say $grasso->nome(), ' ora mangia ', $grasso->dieta();
L'uso di un accessore ro come mutatore solleva l'eccezione Cannot assign a value to a read-only accessor at ... (NdT: "Non posso assegnare un valore a un accessore di sola lettura ...").
L'uso di ro oppure di rw è determinato da scelte di progetto, di convenienza e di purezza di stile. Moose non impone alcuna filosofia particolare a questo riguardo. Alcune persone suggeriscono di rendere ro tutti i dati d'istanza, in modo che dobbiate passare tali dati d'istanza al costruttore (Immutabilità (immutability)). Nell'esempio della classe Gatto, eta() potrebbe continuare ad essere un accessore, ma il costruttore potrebbe ricevere l'anno di nascita del gatto e calcolare automaticamente l'età in base all'anno corrente. Questo approccio rafforza il codice di validazione e garantisce che tutti gli oggetti creati abbiano dati validi.
I dati d'istanza sono una prima dimostrazione dei benefici dell'orientamento agli oggetti. Un oggetto contiene dati correlati e può effettuare operazioni con tali dati. Una classe descrive sia i dati che le operazioni associate.
Incapsulamento
Moose vi permette di dichiarare quali sono gli attributi delle istanze di una classe (un gatto ha un nome) e quali sono gli attributi di tali attributi (non potete modificare il nome di un gatto; potete solo leggerlo). Moose stesso decide come memorizzare tali attributi. Se volete potete cambiare questo comportamento, ma permettere a Moose di gestire per voi la memorizzazione incoraggia l'incapsulamento, ovvero la capacità di mantenere i dettagli interni di un oggetto nascosti agli utenti esterni di tale oggetto.
Considerate un cambiamento nel modo un cui un Gatto gestisce la propria età. Invece di passare al costruttore un valore per l'età, dovete passare l'anno di nascita del gatto e calcolare l'età quando serve:
package Gatto
{
use Moose;
has 'nome', is => 'ro', isa => 'Str';
has 'dieta', is => 'rw';
has 'anno_nascita', is => 'ro', isa => 'Int';
sub eta
{
my $self = shift;
my $anno = (localtime)[5] + 1900;
return $anno - $self->anno_nascita();
}
}
Mentre è cambiata la sintassi per creare gli oggetti Gatto, la sintassi per usarli è rimasta invariata. Al di fuori di Gatto, eta() funziona esattamente come ha sempre fatto. Come essa funzioni internamente è un dettaglio interno alla classe Gatto.
Potete preservare la vecchia sintassi per creare oggetti Gatto personalizzando il costruttore generato per Gatto in modo da permettere il passaggio di un parametro eta. Con esso, potete poi calcolare anno_nascita. Vedete perldoc Moose::Manual::Attributes.
Il calcolo dell'età ha un altro vantaggio. Un valore di default dell'attributo può aiutare a fare la cosa giusta quando qualcuno crea un nuovo oggetto Gatto senza passare un anno di nascita:
package Gatto
{
use Moose;
has 'nome', is => 'ro', isa => 'Str';
has 'dieta', is => 'rw', isa => 'Str';
has 'anno_nascita',
is => 'ro',
isa => 'Int',
default => sub { (localtime)[5] + 1900 };
}
La parola chiave default nella dichiarazione di un attributo riceve un riferimento a funzione (footnote: Potete anche usare direttamente un valore semplice come un numero o una stringa, ma usate un riferimento a funzione se avete bisogno di cose più complesse.) che restituisce il valore di default per tale attributo quando viene costruito un nuovo oggetto. Se il codice che crea un oggetto non passa alcun valore per quell'attributo al costruttore, viene usato il valore di default. Se scrivete:
my $gattino = Gatto->new( nome => 'Choco' );
... il nuovo gattino avrà età 0 fino al prossimo anno.
Polimorfismo
L'incapsulamento è utile, ma la vera potenza dell'orientamento agli oggetti è molto più ampia. Un programma OO ben progettato può gestire molti tipi di dati. Quando delle classi ben progettate incapsulano i dettagli specifici degli oggetti in modo appropriato, succede qualcosa di curioso: spesso il codice diventa meno specifico.
Definire i dettagli di ciò che il programma sa di ciascun Gatto (gli attributi) e di ciò che un Gatto può fare (i metodi) all'interno della classe Gatto permette al codice che deve gestire le istanze di Gatto di ignorare come un Gatto fa ciò che fa.
Considerate una funzione che visualizza i dettagli di un oggetto:
sub mostra_stat_vitali
{
my $oggetto = shift;
say 'Il mio nome e` ', $oggetto->nome();
say 'La mia eta` e` ', $oggetto->eta();
say 'Mangio ', $oggetto->dieta();
}
È ovvio che, dato il presente contesto, questa funzione opera correttamente su un oggetto Gatto. In effetti, essa opera correttamente su qualunque oggetto che abbia i tre accessori appropriati, indipendente da come tale oggetto fornisce i tre accessori e dal tipo dell'oggetto: Gatto, Bruco o PesceGatto. La funzione è sufficientemente generica da accettare come parametro valido qualunque oggetto che rispetti questa interfaccia.
La proprietà del polimorfismo significa che potete sostituire un oggetto di una classe con un oggetto di un'altra classe purché forniscano la stessa interfaccia esterna.
Alcuni linguaggi e ambienti richiedono che esista una relazione formale tra due classi per permettere a un programma di sostituire le istanze di una con quelle dell'altra. Perl 5 fornisce dei modi di imporre questi controlli, ma non li richiede. Per il suo sistema ad-hoc di default è sufficiente che due istanze qualunque abbiano metodi con lo stesso nome per poterle trattare come equivalenti. Alcune persone lo definiscono duck typing, sostenendo che qualunque oggetto che può fare qua() è sufficientemente simile a un'anatra perché possiate trattarlo effettivamente come un'anatra.
mostra_stat_vitali() si preoccupa soltanto che un invocante valido supporti tre metodi--nome(), eta() e dieta()--che non ricevono argomenti e restituiscono qualcosa che può essere concatenato in contesto stringa. Potreste avere centinaia di classi diverse nel vostro codice, non legate da relazioni ovvie, che funzionano tutte con questo metodo purché si conformino al comportamento atteso.
Immaginate di dover passare in rassegna tutti gli animali di uno zoo senza poter usare una funzione polimorfa. I benefici della genericità dovrebbero essere evidenti. Inoltre, gli eventuali dettagli che differenziano il calcolo dell'età di un gattopardo e di una piovra possono essere specificati nelle relative classi--dove tali dettagli sono importanti.
Naturalmente, la mera esistenza di un metodo nome() o di un metodo eta() non ha di per sé stessa implicazioni sul comportamento di un oggetto. Un oggetto Cane potrebbe avere un accessore taglia() tramite il quale potete scoprire che $rodney è di taglia piccola e $lucky è di taglia media. Un oggetto Formaggio potrebbe avere un metodo taglia() che vi permette di preparare delle fette di $fontina da mettere in un panino. taglia() può essere un accessore in una classe e non in un'altra:
# di che taglia e` il cane?
my $dimensioni = $zeppie->taglia();
# e` ora di uno spuntino
$formaggio->taglia();
Talvolta è utile sapere che cosa fa un oggetto e qual è il significato di ciò che fa.
Ruoli
Un ruolo è una collezione di comportamenti e variabili di stato con un nome (footnote: Vedete i documenti di progetto di Perl 6 sui ruoli all'URL http://feather.perl6.nl/syn/S14.html e le ricerche sui trait di Smalltalk all'URL http://scg.unibe.ch/research/traits per avere abbondanti dettagli.). Mentre una classe organizza i comportamenti e lo stato in un modello per la creazione di oggetti, un ruolo organizza e assegna un nome a una collezione di comportamenti e variabili di stato. Potete istanziare una classe, ma non un ruolo. Un ruolo rappresenta qualcosa che la classe fa.
Dato un Animale che ha una taglia e un Formaggio che potete tagliare, una differenza potrebbe essere che Animale ha il ruolo di EssereVivente, mentre il Formaggio ha il ruolo di Cibo:
package EssereVivente
{
use Moose::Role;
requires qw( nome eta dieta taglia );
}
Qualunque cosa abbia questo ruolo deve fornire i metodi nome(), eta(), dieta e taglia(). La classe Gatto deve esplicitare il fatto che ha tale ruolo:
package Gatto
{
use Moose;
has 'nome', is => 'ro', isa => 'Str';
has 'dieta', is => 'rw', isa => 'Str';
has 'taglia', is => 'ro', isa => 'Str';
has 'anno_nascita',
is => 'ro',
isa => 'Int',
default => sub { (localtime)[5] + 1900 };
with 'EssereVivente';
sub eta { ... }
}
La linea che inizia con with fa sì che Moose componga il ruolo EssereVivente nella classe Gatto. La composizione garantisce che tutti gli attributi e i metodi del ruolo facciano parte della classe. EssereVivente richiede a ogni classe in cui viene composto di fornire dei metodi nome(), eta(), dieta() e taglia(). Gatto soddisfa questo vincolo. Se invece EssereVivente venisse composto in una classe che non fornisce tali metodi, Moose solleverebbe un'eccezione.
La parola chiave with usata per applicare dei ruoli a una classe deve comparire dopo le dichiarazioni di attributi in modo che la composizione possa identificare i metodi accessori generati.
Adesso tutte le istanze di Gatto restituiranno un valore vero quando domandate loro se forniscono il ruolo EssereVivente. Gli oggetti Formaggio invece dovrebbero restituire falso:
say 'Vivo!' if $fuffy->DOES('EssereVivente');
say 'Ammuffito!' if $formaggio->DOES('EssereVivente');
Questa tecnica di progettazione separa le capacità delle classi e degli oggetti dall'implementazione di tali classi e oggetti. Il calcolo dell'età a partire dall'anno di nascita nella classe Gatto potrebbe essere a sua volta un ruolo:
package CalcolaEta::Da::AnnoNascita
{
use Moose::Role;
has 'anno_nascita',
is => 'ro',
isa => 'Int',
default => sub { (localtime)[5] + 1900 };
sub eta
{
my $self = shift;
my $anno = (localtime)[5] + 1900;
return $anno - $self->anno_nascita();
}
}
L'estrazione di questo ruolo da Gatto rende il suo utile comportamento disponibile ad altre classi. Ora possiamo comporre entrambi i ruoli in Gatto:
package Gatto
{
use Moose;
has 'nome', is => 'ro', isa => 'Str';
has 'dieta', is => 'rw';
has 'taglia', is => 'ro', isa => 'Str';
with 'EssereVivente',
'CalcolaEta::Da::AnnoNascita';
}
Notate che il metodo eta() di CalcolaEta::Da::AnnoNascita soddisfa il vincolo posto dal ruolo EssereVivente. Notate anche che ogni controllo relativo al fatto che Gatto abbia il ruolo di EssereVivente restituisce un valore vero. Estrarre il metodo eta() in un ruolo ha modificato soltanto i dettagli di come Gatto calcola un'età. Il fatto che sia un EssereVivente rimane valido. Gatto può decidere se implementare l'età da sé oppure ottenerla da qualcun altro. Ciò che conta è che esso fornisca un metodo eta() per soddisfare i vincoli di EssereVivente.
Così come il polimorfismo indica che potete gestire nello stesso modo oggetti diversi che hanno uno stesso comportamento, questo allomorfismo indica che un oggetto può implementare uno stesso comportamento in diversi modi.
Un uso pervasivo dell'allomorfismo può ridurre le dimensioni delle vostre classi e incrementare la quantità di codice che condividono. Vi permette anche di dare un nome a una specifica collezione di comportamenti--una cosa molto utile per testare le capacità anziché le implementazioni.
Per confrontare i ruoli con altre tecniche di progettazione quali i mixin, l'ereditarietà multipla e il monkeypatching, vedete http://www.modernperlbooks.com/mt/2009/04/the-why-of-perl-roles.html.
Ruoli e DOES()
Quando componete un ruolo in una classe, tale classe e le sue istanze restituiscono un valore vero se chiamate DOES() su di esse:
say 'Questo e` un Gatto vivo!'
if $micio->DOES( 'EssereVivente' );
Ereditarietà
Il sistema ad oggetti di Perl 5 supporta l'ereditarietà, che stabilisce una relazione tra due classi in base alla quale una delle due classi specializza l'altra. La classe derivata si comporta come la superclasse--ha lo stesso numero e tipo di attributi e può usare gli stessi metodi. Anche se una classe derivata può avere dati e comportamenti aggiuntivi, potete sempre utilizzare una sua istanza dove il codice si aspetta un'istanza della superclasse. In un certo senso, una sottoclasse fornisce il ruolo implicato dall'esistenza della sua superclasse.
È meglio usare i ruoli o l'ereditarietà? I ruoli garantiscono composizioni sicure, un miglior controllo sui tipi, una maggior fattorizzazione del codice e un controllo più granulare sui nomi e i comportamenti, ma l'ereditarietà è più familiare agli sviluppatori con esperienza in altri linguaggi. Usate l'ereditarietà quando una classe è realmente un'estensione di un'altra. Usate invece un ruolo quando una classe necessita di comportamenti aggiuntivi ed è possibile assegnare un nome significativo alla collezione di tali comportamenti.
Considerate una classe SorgenteLuminosa che fornisce due attributi pubblici (attiva e candele) e due metodi (accendi e spegni):
package SorgenteLuminosa
{
use Moose;
has 'candele', is => 'ro',
isa => 'Int',
default => 1;
has 'attiva', is => 'ro',
isa => 'Bool',
default => 0,
writer => '_imposta_attiva';
sub accendi
{
my $self = shift;
$self->_imposta_attiva(1);
}
sub spegni
{
my $self = shift;
$self->_imposta_attiva(0);
}
}
(Notate che l'opzione writer di attiva crea un accessore privato utilizzabile all'interno della classe per impostare il valore di tale attributo).
Ereditarietà e Attributi
Una sottoclasse di SorgenteLuminosa potrebbe definire una maxi-candela che fornisce una quantità di luce cento volte superiore:
package MaxiCandela
{
use Moose;
extends 'SorgenteLuminosa';
has '+candele', default => 100;
}
extends riceve una lista di nomi di classi da usare come superclassi della classe corrente. Se questa fosse l'unica istruzione nella classe qui sopra, gli oggetti MaxiCandela si comporterebbero esattamente come gli oggetti SorgenteLuminosa. In particolare, avrebbero gli attributi candele e attiva e i metodi accendi() e spegni().
Il + all'inizio di un nome di attributo (come candele) indica che la classe corrente tratta quell'attributo in modo speciale. In questo caso maxi-candela sovrascrive il valore di default della sorgente luminosa, in modo che ogni nuova MaxiCandela abbia una potenza di 100 candele.
Quando invocate il metodo accendi() o spegni() su un oggetto MaxiCandela, Perl lo cerca anzitutto nella classe MaxiCandela, e quindi in ciascuna superclasse. Nel nostro esempio, tali metodi si trovano nella classe SorgenteLuminosa.
L'ereditarietà di attributi funziona in modo simile (vedete perldoc Class::MOP).
Ordine di Dispatch dei Metodi
L'ordine di dispatch dei metodi--o ordine di risoluzione dei metodi (MRO dall'inglese method resolution order, NdT)--è banale per le classi con un'unica superclasse. Si cerca nella classe dell'oggetto, quindi nella sua superclasse e così via finchè non si trova il metodo oppure si arriva al termine della catena di superclassi. Le classi che ereditano da diverse superclassi (ereditarietà multipla)--Hovercraft estende sia Barca che Automobile--richiedono un dispatch più sofisticato. Ragionare sull'ereditarietà multipla è complicato; quando è possibile, evitatela.
Perl 5 adotta una strategia di risoluzione in profondità. Cerca nella prima superclasse e, ricorsivamente, in tutte le superclassi di tale superclasse prima di cercare nelle superclassi successive. La direttiva mro (Direttive(pragmas)) fornisce strategie alternative, inclusa la strategia MRO C3 che cerca in tutte le superclassi immediate di una data classe prima di cercare nelle loro superclassi.
Vedete perldoc mro per maggiori dettagli.
Ereditarietà e Metodi
Come per gli attributi, le sottoclassi possono sovrascrivere dei metodi. Immaginate una luce che non potete spegnere:
package Lightstick
{
use Moose;
extends 'SorgenteLuminosa';
sub spegni {}
}
Chiamare spegni() su un lightstick non ha alcun effetto, nonostante il corrispondente metodo in SorgenteLuminosa ne abbia. Il dispatch di metodi trova il metodo della sottoclasse, che non sempre è ciò che desideravate. Se invece questa era davvero la vostra intenzione, usate l'override di Moose per esprimerla chiaramente.
All'interno di un metodo che ne sovrascrive un altro, la funzione super() di Moose vi permette di chiamare il metodo sovrascritto:
package SorgenteLuminosa::Pignola
{
use Carp 'carp';
use Moose;
extends 'SorgenteLuminosa';
override accendi => sub
{
my $self = shift;
carp "Non posso accendere una luce gia` accesa!"
if $self->attiva;
super();
};
override spegni => sub
{
my $self = shift;
carp "Non posso spegnere una luce gia` spenta!"
unless $self->enabled;
super();
};
}
Questa sottoclasse genera un messaggio quando provate ad accendere o spegnere una sorgente di luce che si trova già in quello stato. La funzione super() fa il dispatch all'implementazione del metodo corrente più vicina nella catena delle superclassi, seguendo il normale ordine di risoluzione di Perl 5.
I modificatori di metodo di Moose permettono di fare altre cose simili--e molto di più. Vedete perldoc Moose::Manual::MethodModifiers.
Ereditarietà e isa()
Il metodo Perl isa() restituisce vero se il suo invocante è o estende una classe con un dato nome. Tale invocante può essere il nome di una classe o una sua istanza:
say 'Sembrerebbe una SorgenteLuminosa'
if $applique->isa( 'SorgenteLuminosa' );
say 'Gli ominidi non emettono luce'
unless $gorilla->isa( 'SorgenteLuminosa' );
Moose e il Sistema OO di Perl 5
Moose fornisce molte funzionalità in più del sistema OO di default di Perl 5. Vale sicuramente la pena di usarlo anche se, in linea di principio, potete implementare voi stessi tutto ciò che Moose vi mette a disposizione (Riferimenti Blessed(blessed_references)) oppure assemblarlo alla meno peggio a partire da diverse distribuzioni CPAN. Moose rappresenta un tutto coerente ed è corredato di documentazione di buona qualità. Molti progetti importanti lo usano con successo, e la sua comunità di sviluppatori è attenta e matura.
Moose si occupa dei costruttori, dei distruttori, degli accessori e dell'incapsulamento. A voi spetta il compito di dichiarare ciò che volete, ma ciò che ne ottenete è codice sicuro e facile da usare. Gli oggetti Moose possono estendere e funzionare con oggetti del sistema base di Perl 5.
Moose permette anche la metaprogrammazione--ovvero la manipolazione dei vostri oggetti attraverso lo stesso Moose. Se vi capitasse di chiedervi quali metodi sono disponibili in una classe o in un oggetto, oppure quali attributi sono supportati da un oggetto, queste informazioni sono a vostra disposizione:
my $metaclasse = Pantaloni::Corti->meta();
say 'le istanze di Pantaloni::Corti hanno gli attributi:';
say $_->name for $metaclasse->get_all_attributes;
say 'le istanze di Pantaloni::Corti supportano i metodi:';
say $_->fully_qualified_name
for $metaclasse->get_all_methods;
Potete anche scoprire quali sono le classi derivate da una data classe:
my $metaclasse = Pantaloni->meta();
say 'Pantaloni e` superclasse di:';
say $_ for $metaclasse->subclasses;
Vedete perldoc Class::MOP::Class per maggiori informazioni sulle operazioni delle metaclassi e perldoc Class::MOP per informazioni sulla metaprogrammazione in Moose.
Moose e il suo protocollo di meta-oggetti (MOP dall'inglese meta-object protocol, NdT) offre una sintassi migliore per dichiarare e lavorare con le classi e gli oggetti in Perl 5. Questo codice è valido in Perl 5:
use MooseX::Declare;
role EssereVivente { requires qw( name eta dieta ) }
role CalcolaEta::Da::AnnoNascita
{
has 'anno_nascita',
is => 'ro',
isa => 'Int',
default => sub { (localtime)[5] + 1900 };
method eta
{
return (localtime)[5] + 1900
- $self->anno_nascita();
}
}
class Gatto with EssereVivente
with CalcolaEta::Da::AnnoNascita
{
has 'nome', is => 'ro', isa => 'Str';
has 'dieta', is => 'rw';
}
La distribuzione CPAN MooseX::Declare usa Devel::Declare per aggiungere sintassi specifica per Moose. Le parole chiave class, role e method riducono la quantità di copia e incolla necessario per scrivere del codice orientato agli oggetti di buona qualità in Perl 5. Notate in particolare la natura dichiarativa di questo esempio e l'assenza di my $self = shift; in eta().
Anche se Moose non fa parte dei pacchetti preinstallati con Perl 5, la sua popolarità garantisce che esso è disponibile per molti sistemi operativi. Anche le distribuzioni di Perl 5 Strawberry Perl e ActivePerl lo includono. Anche se Moose è un modulo CPAN anziché una libreria core, la sua purezza di stile e semplicità lo rendono essenziale per la programmazione moderna in Perl.
Moose non è una libreria leggera, ma è molto potente. Il modulo CPAN Any::Moose aiuta a ridurre i costi delle funzionalità che non usate.
Riferimenti Blessed
Il sistema ad oggetti base di Perl 5 è deliberatamente minimale. È basato su tre sole regole:
Potete costruire tutto il resto su queste tre regole, ma esse sono tutto ciò che vi viene offerto come default. Questo minimalismo può essere poco adatto a progetti di grandi dimensioni--in particolare, le possibilità di avere una maggiore astrazione con la metaprogrammazione (Generazione di Codice(code_generation)) sono scomode e limitate. Moose (Moose(moose)) è una scelta migliore per programmi moderni con più di un paio di centinaia di linee di codice, sebbene ci sia moltissimo codice legacy che usa ancora il sistema OO di default di Perl 5.
Il terzo e ultimo pezzo del sistema OO base di 5 sono i riferimenti blessed. L'istruzione predefinita bless associa il nome di una classe ad un riferimento. Tale riferimento diventa così un invocante valido, e Perl esegue il dispatch di metodi su di esso usando la classe associata.
Un costruttore è un metodo che crea un riferimento e ne fa il bless. Per convenzione, il nome dei costruttori è new(), ma non è obbligatorio. Quasi sempre i costruttori sono metodi di classe.
bless riceve due argomenti, un riferimento e un nome di classe, e restituisce il riferimento. Il riferimento può essere vuoto e non è neanche richiesto che la classe esista già. Potete usare bless anche al di fuori di un costruttore o di una classe, ma tutti i programmi eccetto i più semplici dovrebbero usarlo all'interno di un vero costruttore. La forma canonica di un costruttore è simile a questa:
sub new
{
my $classe = shift;
bless {}, $classe;
}
Un costruttore come questo è progettato per ricevere il nome della classe come invocante del metodo. Potreste anche specificare il nome della classe nel codice, ma perdereste in flessibilità. Un costruttore parametrico permette infatti il riutilizzo del codice con l'ereditarietà, la delega e l'esportazione.
Il tipo del riferimento usato è rilevante solo per quanto riguarda il modo in cui l'oggetto memorizza i propri dati di istanza. Non ha altri effetti sull'oggetto risultante. I riferimenti a hash sono i più comuni, ma potete fare il bless di qualunque tipo di riferimento:
my $ogg_array = bless [], $classe;
my $ogg_scalare = bless \$scalare, $classe;
my $ogg_sub = bless \&una_sub, $classe;
Le classi Moose definiscono gli attributi degli oggetti in modo dichiarativo, ma il sistema OO di default di Perl 5 non è altrettanto solerte. Una classe che rappresenta giocatori di basket memorizzandone il numero e la posizione potrebbe usare un costruttore come questo:
package Giocatore
{
sub new
{
my ($classe, %attr) = @_;
bless \%attr, $classe;
}
}
... e creare i giocatori con:
my $joel = Giocatore->new( numero => 10,
posizione => 'centro' );
my $dante = Giocatore->new( numero => 33,
posizione => 'attacco' );
I metodi della classe possono accedere agli attributi dell'oggetto direttamente come elementi dell'hash:
sub formatta
{
my $self = shift;
return '#' . $self->{numero}
. ' gioca in ' . $self->{posizione};
}
... ma la stessa cosa può essere fatta da qualunque pezzo di codice, con la conseguenza che una modifica nella rappresentazione interna dell'oggetto può invalidare del codice esterno alla classe. L'uso di metodi accessori è più sicuro:
sub numero { return shift->{numero} }
sub posizione { return shift->{posizione} }
... e così iniziate a scrivere a mano ciò che Moose vi fornisce gratuitamente. Meglio ancora, nascondendo il codice di generazione degli accessori, Moose incoraggia le persone a usarli al posto degli accessi diretti. E addio tentazioni.
Ricerca di Metodi e Ereditarietà
Dato un riferimento blessed, una chiamata di metodo come questa:
my $numero = $joel->numero();
... cerca anzitutto il nome della classe associata al riferimento blessed $joel--in questo caso, Giocatore. Quindi, Perl cerca una funzione (footnote: Ricordate che Perl 5 non distingue tra funzioni di un namespace e metodi.) di nome numero() in Giocatore. Se tale funzione non esiste e se Giocatore estende una classe, Perl cerca nella superclasse (e così via) finchè non trova una funzione numero(). Quando numero() è stata trovata, viene chiamata come metodo con invocante $joel.
Il modulo CPAN namespace::autoclean può aiutare ad evitare collisioni accidentali tra le funzioni importate e i metodi.
Mentre Moose fornisce l'istruzione extends per specificare le relazioni di ereditarietà, Perl 5 usa una variabile globale di package di nome @ISA. Il dispatcher di metodi scorre l'array @ISA di ciascuna classe per trovare i nomi delle sue superclassi. Se GiocatoreInfortunato estende Giocatore, potreste scrivere:
package GiocatoreInfortunato
{
@GiocatoreInfortunato::ISA = 'Giocatore';
}
È preferibile usare la direttiva parent (Direttive(pragmas)) (footnote: Il vecchio codice potrebbe usare la direttiva base, ma in Perl 5.10 parent ha rimpiazzato base.):
package GiocatoreInfortunato
{
use parent 'Giocatore';
}
Moose ha il proprio metamodello per memorizzare informazioni di ereditarietà estese che offrono funzionalità aggiuntive.
Potete ereditare da diverse superclassi:
package GiocatoreInfortunato;
{
use parent qw( Giocatore Ospedale::Paziente );
}
... sebbene vadano tenuti presenti gli avvertimenti dati prima sulla complessità dell'ereditarietà multipla e del dispatch di metodi. Considerate piuttosto l'uso dei ruoli (Ruoli(roles)) oppure i modificatori di metodo di Moose.
AUTOLOAD
Se non trova alcun metodo appropriato nella classe dell'invocante e nelle sue superclassi, Perl 5 cerca una funzione AUTOLOAD() (AUTOLOAD(autoload)) in ciascuna classe seguendo l'ordine di risoluzione selezionato. Ciascuna AUTOLOAD() trovata viene invocata e può fornire o declinare il metodo desiderato.
AUTOLOAD() rende molto difficile capire l'ereditarietà multipla.
Sovrascrittura di Metodi e SUPER
Come in Moose, anche nel sistema OO base di Perl potete sovrascrivere dei metodi. Al contrario di Moose, però, il sistema base non fornisce meccanismi per indicare la vostra intenzione di sovrascrivere un metodo di una superclasse. Peggio ancora, ogni funzione che predichiarate, dichiarate o importate nella classe derivata potrebbe sovrascrivere accidentalmente un metodo della superclasse semplicemente perché hanno lo stesso nome. Anche quando vi dimenticate di usare il sistema di override di Moose, almeno esso esiste. Nel sistema OO base di Perl 5, invece, tale protezione è del tutto assente.
Per sovrascrivere un metodo in una classe derivata, dichiarate un metodo con lo stesso nome del metodo nella superclasse. All'interno di un metodo che ne sovrascrive un altro, chiamate il metodo sovrascritto usando il suggerimento SUPER:: per farne il dispatch:
sub sovrascritto
{
my $self = shift;
warn 'Chiamata a sovrascritto() nella classe derivata!';
return $self->SUPER::sovrascritto( @_ );
}
Il prefisso SUPER:: al nome del metodo indica al dispatcher di metodi di chiamare un metodo sovrascritto col nome appropriato. Potete passare dei vostri argomenti al metodo sovrascritto, ma gran parte del codice riusa semplicemente @_. In quest'ultimo caso, occorre ricordare di fare lo shift dell'invocante.
SUPER:: ha una funzionalità dannosa: fa il dispatch alla superclasse del package in cui il metodo che sovrascrive è stato compilato. Se avete importato tale metodo da un altro package, Perl non avrà problemi a fare il dispatch alla superclasse sbagliata. Il desiderio di mantenere la retrocompatibilità ha fatto sì che questa funzionalità dannosa fosse mantenuta. Il modulo CPAN SUPER offre una soluzione. La funzione super() di Moose è invece del tutto esente da questo problema.
Strategie per Sopravvivere ai Riferimenti Blessed
Se i riferimenti blessed vi sembrano minimali e complicati e confusi, ebbene: lo sono davvero. Moose rappresenta un enorme miglioramento, quindi usatelo ogni volta che vi è possibile. Se invece vi trovate a fare la manutenzione di codice che usa i riferimenti blessed oppure proprio non siete ancora riusciti a convincere il vostro gruppo ad abbracciare completamente Moose, potete aggirare alcuni dei problemi dei riferimenti blessed con un'opportuna disciplina.
Riflessione
La riflessione (o introspezione) è il processo di interrogare un programma a proposito di sé stesso mentre viene eseguito. Trattando il codice come un dato, è possibile gestirlo nello stesso modo in cui si gestiscono i dati; questo principio è alla base della generazione di codice (Generazione di Codice(code_generation)).
La classe Moose Class::MOP (Class::MOP(class_mop)) semplifica molti casi di utilizzo della riflessione nei sistemi ad oggetti. Se usate Moose, il suo sistema di metaprogrammazione vi sarà di aiuto; altrimenti molti altri idiomi del Perl 5 base vi aiuteranno a ispezionare e manipolare i programmi in esecuzione.
Verificare se un Modulo è stato Caricato
Se conoscete il nome di un modulo, potete verificare se Perl ritiene di averlo caricato ispezionando l'hash %INC. Infatti, quando Perl 5 carica del codice con use o require, memorizza un elemento in %INC la cui chiave è il path del file contenente il modulo da caricare e il cui valore è il path completo su disco di tale modulo. In altre parole, il caricamento di Modern::Perl ha il seguente effetto:
$INC{'Modern/Perl.pm'} =
'.../lib/site_perl/5.12.1/Modern/Perl.pm';
Ovviamente, i dettagli del path completo variano in base alla propria installazione. Per verificare che Perl sia riuscito a caricare un modulo, occorre convertire il nome del modulo in un nome di file in forma canonica e controllare l'esistenza di tale chiave in %INC:
sub modulo_caricato
{
(my $nomemod = shift) =~ s!::!/!g;
return exists $INC{ $nomemod . '.pm' };
}
Come @INC, anche %INC può essere manipolato in qualunque parte del codice. Alcuni moduli (come Test::MockObject o Test::MockModule) manipolano %INC per validi motivi ma, a seconda del vostro livello di paranoia, potreste anche decidere di controllare voi stessi che il path e i contenuti del package siano quelli attesi.
La funzione is_class_loaded() del modulo CPAN Class::Load incapsula questo controllo di %INC.
Verificare l'Esistenza di un Package
Per controllare se un package esiste nel vostro programma -- ovvero se, da qualche parte, del codice ha eseguito una direttiva package con un certo nome -- controllate che il package erediti da UNIVERSAL. Tutto ciò che estende UNIVERSAL deve in qualche modo fornire il metodo can(). Se il package non esiste, Perl solleverà un'eccezione di chiamante non valido, quindi ponete la chiamata all'interno di un blocco eval:
say "$pkg esiste" if eval { $pkg->can( 'can' ) };
In alternativa, potete consultare le tabelle dei simboli del Perl.
Verificare l'Esistenza di una Classe
Dato che Perl 5 non fa una vera distinzione tra package e classi, il meglio che si possa fare senza Moose è controllare se esiste un package con il nome di una data classe. È possibile controllare con can() che il package fornisca new(), ma non vi è alcuna garanzia che gli eventuali new() trovati siano dei metodi e tanto meno dei costruttori.
Verificare il Numero di Versione di un Modulo
I moduli non devono fornire necessariamente un numero di versione, ma ogni package eredita il metodo VERSION() dalla superclasse universale UNIVERSAL (Il Package UNIVERSAL(universal)):
my $ver_mod = $modulo->VERSION();
VERSION() restituisce il numero di versione del modulo, se è definito; altrimenti restituisce undef. Il metodo restituisce undef anche nel caso in cui il modulo non esista.
Verificare l'Esistenza di una Funzione
Per controllare se una funzione esiste in un package, chiamate can() come metodo di classe sul nome del package:
say "$funz() esiste" if $pkg->can( $funz );
Perl solleverà un'eccezione a meno che $pkg sia un invocante valido; se avete dubbi sulla sua validità, ponete la chiamata in un blocco eval. Attenzione che una funzione implementata in termini di AUTOLOAD() (AUTOLOAD(autoload)) potrebbe dare una risposta errata se il suo package non ha predichiarato la funzione oppure non ha sovrascritto can() in modo corretto. In questo caso, significa che tale package ha un bug.
Per determinare se la import() di un modulo ha importato una funzione nel namespace corrente, usate la seguente tecnica:
say "$funz() importata!" if __PACKAGE__->can( $funz );
Come per l'esistenza di un package, potete anche esaminare voi stessi la tabella dei simboli, se avete abbastanza pazienza per farlo.
Verificare l'Esistenza di un Metodo
Non esistono tecniche di riflessione infallibili per distinguere tra una funzione e un metodo.
Consultare le Tabelle dei Simboli
Una tabella dei simboli di Perl 5 è uno speciale tipo di hash, in cui le chiavi sono i nomi dei simboli globali di un package e i valori sono typeglob. Un typeglob è una struttura dati interna che può contenere uno o più dei seguenti valori: uno scalare, un array, un hash, un filehandle e una funzione.
Potete accedere a una tabella dei simboli come hash aggiungendo due volte "due punti" al nome del package. Per esempio, la tabella dei simboli per il package TritaScimmie è accessibile come %TritaScimmie::.
Tramite l'operatore exists è possibile verificare l'esistenza di specifici nomi di simboli all'interno di una tabella dei simboli (o anche manipolare la tabella aggiungendo o rimuovendo simboli). Attenzione però che alcuni cambiamenti interni nel Perl 5 hanno variato i dettagli di che cosa viene memorizzato nei typeglob, quando e perché.
Per maggiori dettagli potete vedere la sezione "Symbol Tables" in perldoc perlmod, ma utilizzate preferibilmente le altre tecniche di riflessione descritte in questa sezione. Se davvero siete costretti a manipolare le tabelle dei simboli e i typeglob, considerate piuttosto di usare il modulo CPAN Package::Stash.
OO Avanzato in Perl
Creare e usare oggetti in Perl 5 con Moose (Moose(moose)) è facile. Invece, progettare dei buoni programmi non lo è affatto. Dovete trovare il giusto compromesso tra l'eccesso e il difetto di progettazione. Soltanto l'esperienza pratica può aiutarvi a comprendere le tecniche fondamentali di progettazione, ma potete farvi guidare da alcuni principi.
Preferite la Composizione all'Ereditarietà
I progetti OO fatti da principianti tendono a utilizzare eccessivamente l'ereditarietà per il riutilizzo del codice e il polimorfismo. Il risultato è spesso una gerarchia di classi troppo profonda, in cui le responsabilità sono sparpagliate nei posti sbagliati. La manutenzione di questo codice risulta molto difficile--come si fa a sapere dove aggiungere o modificare un comportamento? Che cosa succede quando una parte di codice è in conflitto con del codice dichiarato altrove?
L'ereditarietà è solo uno strumento tra tanti. Un'Automobile potrebbe estendere Veicolo::A::Ruote (con una relazione is-a), ma potrebbe più propriamente contenere diversi oggetti Ruota come attributi di istanza (una relazione has-a).
Decomporre classi complesse in entità più piccole e specifiche (che siano classi o ruoli) migliora l'incapsulamento e riduce la possibilità che una singola classe o ruolo cresca troppo. Delle entità più piccole, più semplici e più incapsulate sono più facili da capire, testare e manutenere.
Principio di Singola Responsabilità
Quando progettate il vostro sistema ad oggetti, considerate le responsabilità di ciascuna entità. Per esempio, un oggetto Impiegato potrebbe rappresentare informazioni specifiche sul nome di una persona, gli indirizzi per contattarla e altri dati personali, mentre un oggetto Lavoro potrebbe rappresentare i compiti lavorativi all'interno di un'azienda. Separare queste entità in base alla loro responsabilità permette alla classe Impiegato di considerare solo il problema della gestione di informazioni su chi è la persona e alla classe Lavoro di rappresentare che cosa fa quella persona (per esempio, due oggetti Impiegato potrebbero condividere un Lavoro).
Affidando a ogni classe un'unica responsabilità, si migliora l'incapsulamento dei suoi dati e comportamenti specifici e si riduce la quantità di accoppiamento tra classi.
Non Ripetetevi
La complessità e le duplicazioni rendono più difficili sia lo sviluppo che la manutenzione. Il principio DRY (Don't Repeat Yourself--Non ripetetevi, NdT) è un ammonimento a cercare di eliminare le duplicazioni all'interno del sistema. La duplicazione interessa anche i dati oltre al codice. Invece di ripetere informazioni di configurazione, informazioni sugli utenti e altri dati nel vostro sistema, create un'unica rappresentazione canonica di tali dati da cui potete poi estrarre quelli che vi servono nel formato opportuno.
Questo principio aiuta a ridurre la possibilità che parti importanti del vostro sistema diventino disallineate e vi aiuta a trovare una rappresentazione ottimale del sistema e dei suoi dati.
Principio di Sostituzione di Liskov
Il principio di sostituzione di Liskov suggerisce che dovrebbe essere possibile sostituire una classe o un ruolo con una loro specializzazione senza violare l'API dell'originale. In altre parole, tali specializzazioni dovrebbero essere altrettanto o più generali riguardo a ciò che si aspettano e almeno altrettanto specifiche riguardo a ciò che producono.
Immaginate due classi, Dessert e la sua derivata TortaDiNoci. Se tali classi osservano il principio di sostituzione di Liskov, potete rimpiazzare ciascun uso di un oggetto Dessert con un oggetto TortaDiNoci nel codice di test, e tutti i test dovrebbero passare (footnote: Vedete "IS-STRICTLY-EQUIVALENT-TO-A" di Reg Braithwaite all'URL http://weblog.raganwald.com/2008/04/is-strictly-equivalent-to.html per maggiori dettagli.).
Sottotipi e Coercizioni
Moose vi permette di dichiarare e usare dei tipi e di estenderli con dei sottotipi per dare descrizioni ancora più specifiche di ciò che i vostri dati rappresentano e di come si comportano. Tali annotazioni di tipo sono d'aiuto per verificare che i dati su cui volete operare in funzioni o metodi specifici siano appropriati e vi permettono anche di specificare i meccanismi con cui fare la coercizione dei dati da un tipo ad un altro.
Vedete Moose::Util::TypeConstraints e MooseX::Types per maggiori informazioni.
Immutabilità
I principianti della programmazione OO trattano spesso gli oggetti come se fossero agglomerati di record che utilizzano dei metodi per leggere e scrivere dei valori interni. Questa tecnica semplicistica porta alla sventurata tentazione di spargere le responsabilità dell'oggetto nell'intero sistema.
Un oggetto ben progettato si aspetta che gli diciate che cosa deve fare e non come deve farlo. Come regola pratica, se vi trovate ad accedere ai dati d'istanza dell'oggetto (anche se attraverso i metodi accessori), potreste avere troppa visibilità sui suoi dettagli interni.
Un approccio che aiuta a prevenire questo comportamento è quello di considerare gli oggetti come immutabili. Dopo aver fornito i dati necessari ai loro costruttori vietate ulteriori modifiche di queste informazioni dall'esterno della classe. Non esponete alcun metodo per cambiare i dati d'istanza. Gli oggetti costruiti in questo modo, una volta creati, saranno sempre validi e non potranno divenire inconsistenti a causa di manipolazioni esterne. Seguire questo approccio richiede un'enorme dose di disciplina, ma i sistemi risultanti sono robusti, testabili e manutenibili.
Alcuni stili di progettazione si spingono fino a proibire la modifica dei dati d'istanza dentro alla classe stessa, ma quest'obiettivo è molto più difficile da raggiungere.
Stile ed Efficacia
La qualità è importante.
I programmi hanno dei bug. Necessitano di manutenzione e di estensioni. E coinvolgono molteplici programmatori.
Programmare bene richiede un equilibrio tra la necessità di raggiungere velocemente gli obiettivi e quella di poter continuare a raggiungerli in futuro. Potete bilanciare tra di loro il tempo, le risorse e la qualità. La capacità di arrivare a un buon bilanciamento determina il vostro livello di abilità e di pragmatismo come artigiani.
Conoscere Perl a fondo è importante. Altrettanto importante è sviluppare un senso estetico. L'unico modo di raggiungere questi obiettivi è fare pratica con la manutenzione del codice e con la lettura e scrittura di codice di qualità. Non ci sono scorciatoie, ma almeno potete seguire alcune linee guida.
Scrivere Codice Perl Manutenibile
La manutenibilità è il nebuloso concetto che misura la facilità di comprendere e modificare un programma scritto in precedenza. Mettete via un pezzo di codice per sei mesi, e poi tornate a guardarlo. La manutenibilità misura le difficoltà che dovete affrontare per modificarlo.
La manutenibilità non è né un problema sintattico né una misura di come il vostro codice appare a un non-programmatore. Potete assumere che chi ha a che fare con il codice sia un programmatore competente che ha una buona comprensione della natura del problema che il codice deve risolvere. Quali ostacoli si possono incontrare quando si tenta di risolvere un bug o di fare una miglioria in modo corretto?
La capacità di scrivere codice manutenibile deriva da esperienza guadagnata sul campo con fatica e dalla padronanza degli idiomi, delle tecniche e dello stile dominante del linguaggio. Ciò nonostante, anche i principianti possono migliorare la manutenibilità del loro codice osservando alcuni principi:
I sistemi ben progettati hanno poche duplicazioni. Utilizzano funzioni, moduli, oggetti e ruoli per isolare il codice duplicato in componenti riusabili che modellano in modo accurato il dominio del problema. I migliori progetti vi permettono di aggiungere funzionalità rimuovendo del codice.
Alcuni problemi richiedono soluzioni particolarmente intelligenti. Incapsulate questo codice dietro a semplici interfacce e documentate bene i vostri colpi di genio.
La semplicità non è una scusa per evitare il controllo degli errori o la modularità o la validazione o la sicurezza. Il codice semplice può usare funzionalità avanzate e moduli CPAN in abbondanza; può anche richiedere un certo impegno per essere capito. Ciò che conta è che risolve efficacemente i problemi senza sforzi inutili.
Certe volte vi serve codice robusto e potente. Altre volte ciò di cui avete bisogno è un programma lungo una linea. Semplicità significa conoscere la differenza tra questi casi e ottenere il risultato desiderato.
Scrivere in Perl Idiomatico
Perl prende a prestito liberamente da altri linguaggi e vi permette di scrivere il codice che volete. I programmatori C scrivono spesso codice Perl in C-style, proprio come i programmatori Java scrivono codice Perl in stile Java. I buoni programmatori Perl scrivono codice Perl in stile Perl, adottando gli idiomi del linguaggio.
Gli sviluppatori CPAN, i Perl Monger e i membri delle mailing list hanno esperienze guadagnate sul campo nella soluzione di problemi in innumerevoli modi diversi. Parlategli. Leggete il loro codice. Fate domande. Imparate da loro e lasciate che loro imparino da voi.
Scrivere Codice Perl Efficace
La manutenibilità è, in ultima analisi, un problema di progettazione. A sua volta, una buona capacità di progettazione deriva dalla pratica di alcune buone abitudini:
Se trovate un bug, segnalatelo e, se possibile, risolvetelo. Se trovate un errore di battitura, correggetelo. Chiedete nuove funzionalità. E ricordatevi di dire "Grazie!". Siamo meglio come comunità che come individui separati. E abbiamo tutti più potenza ed efficacia se riusiamo il codice.
Quando avete risolto un nuovo problema e vi sentite pronti, condividetelo. Unitevi a noi. Anche noi risolviamo problemi.
Eccezioni
Programmare bene significa anticipare le situazioni impreviste. File che dovrebbero esistere e non ci sono. Un hard disk enorme che non dovrebbe mai riempirsi ma lo fa. La connessione di rete che dovrebbe sempre essere disponibile ma non lo è. Un database infallibile che fallisce. Le cose eccezionali succedono e un software robusto dovrebbe saperle gestire. Se potete ripristinare il funzionamento, è un'ottima cosa! Ma se non potete, fate almeno il logging delle informazioni rilevanti e riprovate.
Perl 5 gestisce le situazioni eccezionali attraverso le eccezioni: un meccanismo di controllo di flusso con scope dinamico progettato per segnalare e gestire gli errori.
Sollevare Eccezioni
Supponete di voler scrivere su un file di log. Se non riuscite ad aprire tale file, qualcosa sta andando storto. Usate die per sollevare un'eccezione:
sub apri_file_log
{
my $nome = shift;
open my $fh, '>>', $nome
or die "Apertura del file di log '$nome' non riuscita: $!";
return $fh;
}
die() imposta il valore della variabile globale $@ a quello del proprio operando ed esce immediatamente dalla funzione corrente senza restituire nulla. L'eccezione sollevata continua il suo percorso nello stack di chiamate (Esecuzione Controllata(controlled_execution)) finchè incontra del codice che la cattura. Se nessuna chiamata cattura l'eccezione, il programma termina con un errore.
La gestione delle eccezioni adotta lo scope dinamico (Scope Dinamico(dynamic_scope)) come i simboli local.
Catturare Eccezioni
Talvolta è utile che un'eccezione faccia terminare il programma. Un programma eseguito come processo temporizzato potrebbe sollevare un'eccezione quando i log di errore sono pieni, in modo che venga inviato un SMS agli amministratori. Tuttavia non tutte le eccezioni dovrebbero essere fatali. I programmi ben scritti talvolta riescono a ripristinare il funzionamento o almeno a salvare il proprio stato e terminare in modo controllato.
Usate la forma a blocco dell'operatore eval per catturare un'eccezione:
# l'apertura del file di log potrebbe fallire
my $fh = eval { apri_file_log( 'paperopoli.log' ) };
Se l'apertura del file ha successo, $fh conterrà il filehandle. Se fallisce, $fh resta non definito, e il flusso del programma continua.
Il blocco passato come argomento a eval introduce un nuovo scope, sia lessicale che dinamico. Se apri_file_log() chiamasse altre funzioni e una di queste lanciasse un'eccezione, questo eval la catturerebbe.
Un gestore di eccezioni dev'essere uno strumento generico, dato che potrebbe catturare tutte le eccezioni nel suo scope dinamico. Per capire quale eccezione è stata catturata (e anzitutto se c'è stata un'eccezione) controllate il valore di $@. Ricordatevi di localizzare $@ prima di tentare di catturare un'eccezione, dato che $@ è una variabile globale:
local $@;
# l'apertura del file di log potrebbe fallire
my $fh = eval { apri_file_log( 'paperopoli.log' ) };
# eccezione catturata
if (my $eccezione = $@) { ... }
Copiate immediatamente $@ in una variabile lessicale per evitare che il codice eseguito successivamente possa sovrascrivere la variabile globale $@. Non potete sapere se da qualche parte c'è un blocco eval che può reimpostare $@.
Solitamente $@ contiene una stringa che descrive l'eccezione. Ispezionatene il contenuto per capire se potete gestire l'eccezione:
if (my $eccezione = $@)
{
die $eccezione
unless $eccezione =~ /^Il file di log non puo' essere aperto/;
$fh = log_su_syslog();
}
Rilanciate un'eccezione chiamando di nuovo die(). Passatele l'eccezione corrente oppure una nuova eccezione a seconda delle vostre esigenze.
Applicare delle espressioni regolari a eccezioni stringa è poco robusto, dato che i messaggi d'errore possono cambiare nel tempo. Questo vale anche per le eccezioni predefinite sollevate dallo stesso Perl. Per fortuna, potete anche passare a die un riferimento--e perfino un riferimento blessed. Ciò vi permette di fornire molte più informazioni nella vostra eccezione: numeri di linea, file e altre informazioni di debug. Ottenere queste informazioni da un'entità strutturata è molto più facile che estrarle interpretando una stringa. Catturate queste eccezioni come fareste con qualunque altra eccezione.
La distribuzione CPAN Exception::Class facilita la creazione e l'uso di oggetti eccezione:
package Zoo::Eccezioni
{
use Exception::Class
'Zoo::AnimaleFuggito',
'Zoo::GestoreFuggito';
}
sub gabbia_aperta
{
my $self = shift;
Zoo::AnimaleFuggito->throw
unless $self->contiene_animale;
...
}
sub ufficio_aperto
{
my $self = shift;
Zoo::GestoreFuggito->throw
unless $self->contiene_gestore;
...
}
Avvertenze sulle Eccezioni
Mentre sollevare eccezioni è relativamente semplice, catturarle lo è di meno. Usare $@ correttamente vi richiede di scansare svariati rischi piuttosto insidiosi:
Alcuni di questi problemi sono stati risolti in Perl 5.14. Se da una parte essi si presentano molto raramente, dall'altra sono spesso difficili da diagnosticare e risolvere. La distribuzione CPAN Try::Tiny migliora l'affidabilità e la sintassi della gestione di eccezioni (footnote: In effetti, Try::Tiny è stato di ispirazione per le migliorie alla gestione delle eccezioni in Perl 5.14.).
Try::Tiny è facile da usare:
use Try::Tiny;
my $fh = try { apri_file_log( 'paperopoli.log' ) }
catch { log_eccezione( $_ ) };
try sostituisce eval. Il blocco opzionale catch viene eseguito solo quando try cattura un'eccezione. catch riceve l'espressione catturata nella variabile topic $_.
Eccezioni Predefinite
Perl 5 può sollevare svariate eccezioni. perldoc perldiag elenca numerosi "errori fatali intercettabili". Mentre alcuni di essi sono errori di sintassi sollevati durante il processo di compilazione, altri possono essere catturati durante l'esecuzione. Quelli di maggiore interesse sono:
Naturalmente, potete anche catturare le eccezioni prodotte da autodie (La direttiva autodie(autodie)) e i warning lessicali promossi a eccezioni (Registrare i Vostri Warning(registering_warnings)).
Direttive
Molte delle estensioni di Perl 5 sono moduli che forniscono nuove funzioni o definiscono classi (Moose(moose)). Alcuni moduli, invece, come strict e warnings, influenzano il comportamento del linguaggio stesso. Tali moduli sono detti direttive. Per convenzione, le direttive hanno nomi con soli caratteri minuscoli per differenziarli dagli altri moduli.
Direttive e Scope
Il funzionamento delle direttive consiste nell'esportare comportamenti o informazioni specifiche nello scope lessicale dei loro chiamanti. Così come dichiarare una variabile lessicale rende disponibile in uno scope il nome di un simbolo, usare una direttiva attiva il suo comportamento in un certo scope:
{
# $lessicale non e` visibile; strict non e` attiva
{
use strict;
my $lessicale = 'disponibile qui';
# $lessicale e` visibile; strict e` attiva
...
}
# $lessicale e` di nuovo invisible; strict non e` attiva
}
Inoltre, così come le dichiarazioni lessicali interessano anche gli scope più interni, le dichiarazioni rimangono attive in tali scope:
# scope di file
use strict;
{
# scope interno, dove strict continua a essere attivo
my $interno = 'un altro lessicale';
...
}
Uso delle Direttive
Fate la use di una direttiva come fareste con qualunque altro modulo. Le direttive possono ricevere argomenti, ad esempio un numero minimo di versione accettabile e una lista di argomenti per modificarne il comportamento:
# richiede la dichiarazione delle variabili e proibisce le bareword
use strict qw( subs vars );
Talvolta vi servirà disattivare del tutto o in parte gli effetti di una direttiva in uno scope lessicale annidato. L'istruzione predefinita no annulla un'importazione (Importazione(importing)), il che disattiva gli effetti di una direttiva se essa si conforma alle regole. Per esempio, se volete disattivare la protezione di strict quando dovete manipolare direttamente dei simboli:
use Modern::Perl;
# oppure use strict;
{
no strict 'refs';
# manipolazione della tabella dei simboli
}
Direttive Utili
In Perl 5.10.0 è stata aggiunta la possibilità di scrivere le proprie direttive lessicali in puro codice Perl. perldoc perlpragma vi spiega come fare e la descrizione di $^H in perldoc perlvar vi spiega il suo funzionamento.
Già prima della versione 5.10, Perl 5 includeva svariate direttive predefinite utili:
CPAN ha iniziato a raccogliere delle direttive non predefinite:
Attualmente, questi strumenti non sono molto usati. Gli ultimi due vi aiutano a scrivere codice più corretto, mentre coi primi due vale la pena di fare degli esperimenti all'interno di piccoli progetti. Essi rappresentano delle interessanti potenzialità di Perl 5.
Gestire Programmi Reali
Un libro deve anzitutto insegnarvi a scrivere piccoli programmi per risolvere piccoli problemi d'esempio. In questo modo potete imparare bene la sintassi. Tuttavia, per scrivere programmi reali che risolvono problemi reali dovete anche imparare a gestire il codice scritto nel vostro linguaggio. Come dovete organizzare tale codice? Come potete verificare che funziona? Come potete renderlo robusto rispetto agli errori? Che cosa rende il vostro codice conciso, chiaro e manutenibile?
Il Perl moderno fornisce molti strumenti e tecniche per scrivere programmi reali.
Testing
Il testing è un processo che consiste nella scrittura ed esecuzione di piccoli pezzi di codice al fine di verificare che il vostro software funzioni come vi aspettate. Un testing efficace è essenzialmente l'automatizzazione di un processo che avete sicuramente già eseguito innumerevoli volte a mano: scrivere del codice, eseguirlo e controllare se funziona. L'automazione è il punto più importante. Anziché dover contare su delle persone per ripetere in modo affidabile i controlli a mano, lasciate questo compito al calcolatore.
Perl 5 fornisce degli ottimi strumenti per aiutarvi a scrivere i test nel modo giusto.
Test::More
Per iniziare a fare testing in Perl dovete usare il modulo Test::More distribuito con Perl 5, e in particolare la sua funzione ok(). ok() riceve due parametri, un valore booleano e una stringa che descrive lo scopo del test:
ok( 1, 'il numero uno dovrebbe essere vero' );
ok( 0, '... mentre lo zero non dovrebbe' );
ok( '', 'la stringa vuota dovrebbe avere valore falso' );
ok( '!', '... mentre una stringa non vuota non dovrebbe' );
done_testing();
Qualunque condizione vogliate testare nel vostro programma può essere ridotta a un valore binario. Ogni asserzione di test è una semplice domanda a cui si può rispondere sì o no: questo pezzettino di codice funziona come mi aspetto? Un programma complesso potrebbe avere migliaia di tali condizioni e, in generale, più ne considerate meglio è. Isolare comportamenti specifici in singole asserzioni vi permette di restringere il campo d'azione di bug ed equivoci, specialmente quando modificherete il codice in futuro.
La funzione done_testing() dice a Test::More che il programma ha eseguito con successo tutte le asserzioni di test previste. Se il programma è incappato in una eccezione o è comunque terminato in modo inatteso prima di chiamare done_testing(), il framework di test vi notificherà che qualcosa è andato storto. Senza un meccanismo come done_testing(), come potreste accorgervene? È vero che il codice di questo esempio è troppo semplice per poter fallire, ma è altrettanto vero che il codice troppo semplice per poter fallire fallisce molto più spesso di quanto ci si aspetti.
Test::More vi permette di specificare un piano di test che rappresenta il numero di asserzioni singole che intendete eseguire:
use Test::More tests => 4;
ok( 1, 'il numero uno dovrebbe essere vero' );
ok( 0, '... mentre lo zero non dovrebbe' );
ok( '', 'la stringa vuota dovrebbe avere valore falso' );
ok( '!', '... mentre una stringa non vuota non dovrebbe' );
L'argomento tests di Test::More imposta il piano di test per il programma. In pratica, è una rete di sicurezza. Se sono stati eseguiti meno di quattro test, qualcosa è andato storto. Se sono stati eseguiti più di quattro test, qualcosa è andato storto.
Eseguire Test
Il programma risultante è a tutti gli effetti un programma Perl 5 che produce il seguente output:
ok 1 - il numero uno dovrebbe essere vero
not ok 2 - ... mentre lo zero non dovrebbe
# Failed test '... mentre lo zero non dovrebbe'
# at valori_booleani.t line 4.
not ok 3 - la stringa vuota dovrebbe avere valore falso
# Failed test 'la stringa vuota dovrebbe avere valore falso'
# at valori_booleani.t line 5.
ok 4 - ... mentre una stringa non vuota non dovrebbe
1..4
# Looks like you failed 2 tests of 4.
Questo formato è conforme a uno standard per l'output dei test chiamato TAP, il Test Anything Protocol (http://testanything.org/). Il fallimento di test TAP genera dei messaggi diagnostici per facilitare il debugging.
L'output di un file di test che contiene molte asserzioni (e specialmente molte asserzioni fallite) può essere verboso. Nella maggior parte dei casi, siete interessati soltanto a sapere se tutti i test sono passati e, in caso contrario, i dettagli dei fallimenti. Il modulo Test::Harness distribuito con Perl interpreta il formato TAP e il suo programma associato prove esegue i test e visualizza solo le informazioni più pertinenti:
$ prove valori_booleani.t
valori_booleani.t .. 1/?
# Failed test '... mentre lo zero non dovrebbe'
# at valori_booleani.t line 4.
# Failed test 'la stringa vuota dovrebbe avere valore falso'
# at valori_booleani.t line 5.
# Looks like you failed 2 tests of 4.
valori_booleani.t .. Dubious, test returned 2
(wstat 512, 0x200)
Failed 2/4 subtests
Test Summary Report
-------------------
valori_booleani.t (Wstat: 512 Tests: 4 Failed: 2)
Failed tests: 2-3
È un output molto lungo che riporta qualcosa che dovrebbe essere gíà ovvio: il secondo e il terzo test falliscono perché lo zero e la stringa vuota hanno valore falso. Si possono facilmente evitare tali fallimenti invertendo il senso delle condizioni con l'uso della coercizione booleana (Coercizione Booleana(boolean_coercion)):
ok( ! 0, '... mentre lo zero non dovrebbe' );
ok( ! '', 'la stringa vuota dovrebbe avere valore falso' );
Con queste modifiche, prove visualizza ciò che segue:
$ prove valori_booleani.t
valori_booleani.t .. ok
All tests successful.
Vedete perldoc prove per la descrizione di opzioni utili per il test, come la possibilità di eseguire test in parallelo (-j), di aggiungere automaticamente lib/ al path degli include di Perl (-l), di eseguire ricorsivamente tutti i file di test che si trovano nella directory t/ (-r t) e di eseguire per primi i test più lenti (--state=slow,save).
Il seguente alias proveall della shell bash può essere utile:
alias proveall='prove -j9 --state=slow,save -lr t'
Migliorare i Confronti
Sebbene al cuore di ogni test automatico ci sia la condizione booleana "questo è vero o falso?", ridurre ogni cosa a tale condizione booleana è noioso e offre scarse possibilità di diagnostica. Test::More fornisce svariate altre funzioni di asserzione che possono essere più convenienti.
La funzione is() confronta due valori usando l'operatore eq. Se tali valori sono uguali il test passa. Altrimenti il test fallisce con un messaggio diagnostico:
is( 4, 2 + 2, 'la somma dovrebbe funzionare' );
is( 'frittella', 100, 'le frittelle sono numeri' );
Come vi aspettavate, il primo test passa e il secondo fallisce:
t/test_is.t .. 1/2
# Failed test 'le frittelle sono numeri'
# at t/test_is.t line 8.
# got: 'frittella'
# expected: '100'
# Looks like you failed 1 test of 2.
Mentre ok() fornisce solo il numero di linea del test fallito, is() visualizza il valore atteso e quello ricevuto.
is() applica implicitamente il contesto scalare ai propri valori (Prototipi(prototypes)). Pertanto potete ad esempio controllare il numero di elementi di un array senza valutarlo esplicitamente in contesto scalare:
my @cugini = qw( Riccardo Ale Carlo
Enrico Camilla Maria );
is( @cugini, 6, 'Dovrei avere solo sei cugini' );
... anche se alcuni preferiscono scrivere scalar @cugini per essere più chiari.
La funzione duale isnt() di Test::More confronta due valori usando l'operatore ne e ha successo se essi non sono uguali. Anche in questo caso gli operandi sono valutati in contesto scalare.
Sia is() che isnt() effettuano dei confronti stringa con gli operatori eq e ne di Perl 5. Quasi sempre questa è la cosa giusta da fare, ma per valori complessi come gli oggetti con overloading (Overloading(overloading)) o le variabili doppie (Dualvar(dualvars)) può essere preferibile effettuare il test con un confronto esplicito. La funzione cmp_ok() vi permette di specificare l'operatore di confronto:
cmp_ok( 100, $saldo_corrente, '<=',
'Dovrei avere almeno 100€' );
cmp_ok( $pastore_scozzese, $collie, '==',
'I numeri dei parenti di Lassie dovrebbero essere gli stessi' );
Classi e oggetti hanno dei propri modi peculiari di interagire con i test. Potete verificare se una classe o un oggetto estendono un'altra classe (Ereditarietà (inheritance)) con isa_ok():
my $gorzilla = RobotScimmia->new();
isa_ok( $gorzilla, 'Robot' );
isa_ok( $gorzilla, 'Scimmia' );
isa_ok() fornisce un proprio messaggio diagnostico in caso di fallimento.
can_ok() verifica che una classe o un oggetto possano eseguire uno o più metodi specificati:
can_ok( $gorzilla, 'mangia_banana' );
can_ok( $gorzilla, 'trasformati', 'distruggi_tokyo' );
La funzione is_deeply() confronta due riferimenti per garantire che abbiano contenuti uguali:
use Clone;
my $numeri = [ 4, 8, 15, 16, 23, 42 ];
my $numclonati = Clone::clone( $numeri );
is_deeply( $numeri, $numclonati,
'clone() dovrebbe generare elementi identici' );
Se il confronto fallisce, Test::More fa del suo meglio per fornire una ragionevole diagnostica che indichi la posizione della prima differenza tra le strutture. Vedete i moduli CPAN Test::Differences e Test::Deep per test maggiormente configurabili.
Test::More contiene molte altre funzioni, ma quelle descritte sono le più utili.
Organizzazione dei Test
Le distribuzioni CPAN dovrebbero includere una directory t/ con uno o più file di test i cui nomi terminano col suffisso .t. Per default, quando fate il build di una distribuzione con Module::Build o ExtUtils::MakeMaker, il passo di testing esegue tutti i file t/*.t, sintetizza i loro output e ha successo o fallisce a seconda dei risultati dell'intera suite di test. Non ci sono linee guida specifiche su come gestire i contenuti dei singoli file .t, ma due strategie sono piuttosto comuni:
Un approccio ibrido vi dà più flessibilità; un test potrebbe verificare che tutti i vostri moduli siano compilabili, mentre gli altri test potrebbero verificare che ogni modulo si comporti come previsto. Gestire i test in termini di funzionalità diventa più opportuno man mano che le distribuzioni diventano più grandi; dei file di test troppo grandi sono più difficili da manutenere.
L'uso di file di test separati può anche accelerare lo sviluppo. Se state aggiungendo la capacità di sputare fuoco al vostro RobotScimmia, potreste volere eseguire il solo file di test t/sputa_fuoco.t. Quando la nuova funzionalità si comporta in modo soddisfacente, eseguite l'intera suite di test per verificare che le modifiche locali non abbiano effetti globali indesiderati.
Altri Moduli di Testing
Test::More si basa su un modulo sottostante di nome Test::Builder. Tale modulo gestisce il piano di test e coordina l'output del test col protocollo TAP. Questa separazione permette a diversi moduli di test di condividere lo stesso Test::Builder. Di conseguenza, CPAN contiene centinaia di moduli di test--e tutti possono convivere nello stesso programma.
Vedete il progetto Perl QA (http://qa.perl.org/) per maggiori informazioni sul testing in Perl.
Gestione dei Warning
Anche se c'è più di un modo di scrivere un programma funzionante in Perl 5, alcuni di tali modi possono essere poco eleganti, poco chiari o anche insidiosamente sbagliati. Il sistema di warning opzionali di Perl 5 può aiutarvi a identificare e evitare queste situazioni.
Generare i Warning
Usate la funzione predefinita warn per emettere un warning:
warn 'Qualcosa e` andato storto!';
warn stampa una lista di valori sul filehandle STDERR (Input e Output(filehandle)). Perl aggiunge il nome del file e il numero di linea in cui è stata chiamata warn a meno che l'ultimo elemento della lista termini con un carattere di a capo.
Il modulo Carp distribuito con Perl 5 offre altri meccanismi per generare dei warning. La sua funzione carp() emette un warning dal punto di vista del codice chiamante. Dato il seguente codice di validazione dei parametri di una funzione:
use Carp 'carp';
sub solo_due_argomenti
{
my ($ops, $opd) = @_;
carp( 'Ho ricevuto troppi argomenti' ) if @_ > 2;
...
}
... il warning di arità (Arità (arity)) includerà il nome del file e il numero di linea relativi al codice chiamante, e non quelli relativi alla funzione solo_due_argomenti(). Analogamente, la funzione cluck() di Carp genera una traccia di tutte le chiamate di funzione effettuate fino a quella corrente.
Il modo verbose di Carp aggiunge le tracce delle chiamate a tutti i warning generati da carp() e croak() (Gestione degli Errori(reporting_errors)), dovunque ci si trovi all'interno del programma:
$ perl -MCarp=verbose mio_prog.pl
Quando scrivete dei moduli (Moduli(modules)), usate Carp al posto di warn e die.
Attivare e Disattivare i Warning
Potrebbe capitarvi di incontrare l'argomento da linea di comando -w nel codice poco recente. Tale argomento attiva i warning in tutto il programma, inclusi i moduli esterni scritti e manutenuti da altre persone. Il suo effetto è del tipo "tutto o niente", ma potrebbe esservi utile se intendete eliminare i warning attuali e potenziali nell'intera codebase.
L'approccio moderno è quello di usare la direttiva warnings (footnote: ...o un suo equivalente come use Modern::Perl;.). In questo modo si attivano i warning all'interno di scope lessicali e si indica che gli autori del codice ritengono che esso non debba normalmente produrre dei warning.
Il flag -W attiva i warning nell'intero programma in modo unilaterale, indipendentemente dall'attivazione o disattivazione lessicale con la direttiva warnings. Il flag -X disattiva i warning nell'intero programma in modo unilaterale. Nessuno dei due è usato frequentemente.
I flag -w, -W e -X influenzano il valore della variabile globale $^W. Il codice scritto prima dell'introduzione della direttiva warnings (avvenuta con Perl 5.6.0 nella primavera del 2000) potrebbe localizzare $^W per sopprimere certi warning in un dato scope.
Disabilitare Categorie di Warning
Per disabilitare i warning in uno scope in modo selettivo, usate no warnings; con una lista di argomenti. Omettendo la lista di argomenti disattiverete tutti i warning all'interno dello scope.
perldoc perllexwarn elenca tutte le categorie di warning riconosciute dalla direttiva warnings della vostra versione di Perl 5. La maggior parte di essi rappresenta condizioni interessanti, ma alcuni possono essere peggio che inutili in certe situazioni. Per esempio, il warning recursion viene generato quando Perl si accorge che una funzione ha chiamato se stessa più di cento volte. Se confidate nella vostra capacità di scrittura delle condizioni di terminazione della ricorsione, potete disattivare questo warning nello scope della ricorsione (anche se forse è meglio usare le chiamate di coda; Chiamate di Coda(tail_calls)).
Se state facendo generazione di codice (Generazione di Codice(code_generation)) o state ridefinendo dei simboli localmente, potete disattivare i warning redefine.
Alcuni programmatori Perl esperti disattivano i warning di uninitialized (valori non inizializzati, NdT) nel codice di elaborazione di stringhe che concatena valori con molte origini diverse. Con un'attenta inizializzazione delle variabili potete evitare di disattivare questi warning, ma potreste invece considerarli dei semplici fastidi per questioni di stile e di concisione.
Rendere Fatali i Warning
Se il vostro progetto considera i warning gravi come gli errori, potete renderli fatali. Per promuovere tutti i warning a eccezioni, usate:
use warnings FATAL => 'all';
Potete anche rendere fatali solo alcune categorie specifiche di warning, come l'uso di costrutti deprecati:
use warnings FATAL => 'deprecated';
Con una buona dose di disciplina, questo meccanismo vi consente di scrivere codice molto robusto--ma fate attenzione. Molti warning derivano da condizioni che si verificano durante l'esecuzione. Se la vostra suite di test non tiene conto di tutti i warning in cui potete imbattervi, il vostro programma potrebbe terminare inaspettatamente a causa di un'eccezione non catturata.
Catturare i Warning
Analogamente a come potete catturare le eccezioni, potete anche catturare i warning. La variabile %SIG (footnote: Vedete perldoc perlvar.) contiene i gestori dei segnali fuori banda sollevati da Perl o dal vostro sistema operativo. Per catturare un warning, assegnate un riferimento a funzione a $SIG{__WARN__}:
{
my $warning;
local $SIG{__WARN__} = sub { $warning .= shift };
# fai qualcosa di pericoloso
...
say "Ho catturato il warning:\n$warning" if $warning;
}
Il primo argomento del gestore dei warning è il messaggio di warning. In effetti questa tecnica è meno utile della disattivazione lessicale dei warning--ma può diventare rilevante in moduli di test come il modulo CPAN Test::Warnings, dove il testo del warning è importante.
State attenti al fatto che %SIG è globale. localizzatela nello scope più limitato possibile, ma ricordate che è comunque una variabile globale.
Registrare i Vostri Warning
La direttiva warnings::register vi permette di creare i vostri warning lessicali in modo che gli utenti del vostro codice possano attivarli e disattivarli. Fate la use della direttiva warnings::register in un modulo:
package Scimmia::Pericolosa;
use warnings::register;
Questo codice crea una nuova categoria di warning con il nome del package Scimmia::Pericolosa. Attivate questi warning con use warnings 'Scimmia::Pericolosa' e disattivateli con no warnings 'Scimmia::Pericolosa'.
Usate warnings::enabled() per verificare se una data categoria di warning è attiva nello scope lessicale chiamante. Usate warnings::warnif() per generare un warning solo se i warning sono attivi. Per esempio, per generare un warning della categoria deprecated:
package Scimmia::Pericolosa;
use warnings::register;
sub import
{
warnings::warnif( 'deprecated',
'l'importazione senza parametri da ' . __PACKAGE__ .
' e` deprecata' )
unless @_;
}
Vedete perldoc perllexwarn per maggiori dettagli.
File
La maggior parte dei programmi interagisce in qualche modo con il mondo reale e, in particolare, effettua letture, scritture e manipolazioni in genere sui file. Le origini di Perl come strumento per gli amministratori di sistema ne ha fatto un linguaggio adatto all'elaborazione del testo.
Input e Output
Un filehandle rappresenta lo stato corrente di uno specifico canale di input o di output. Ogni programma Perl 5 ha a disposizione tre filehandle standard, STDIN (l'input del programma), STDOUT (l'output del programma) e STDERR (l'output degli errori del programma). Per default, ogni cosa che stampate con print e say va su STDOUT, mentre gli errori, i warning e tutto ciò che generate con warn() va su STDERR. Questa separazione dell'output vi permette di redirezionare l'output utile e gli errori in due posti diversi--per esempio, un file di output e un log degli errori.
Usate la funzione predefinita open per ottenere un filehandle. Per aprire un file in lettura scrivete:
open my $fh, '<', $nomefile
or die "Non posso leggere da '$nomefile': $!\n";
Il primo operando è un lessicale che conterrà il filehandle risultante. Il secondo operando è la modalità del file, che determina il tipo delle operazioni permesse sul filehandle. L'ultimo operando è il nome del file. Se open fallisce, la clausola die solleva un'eccezione in cui il valore di $! fornisce le ragioni del fallimento.
Potete aprire i file in scrittura, append, lettura e scrittura e in altri modi ancora. Alcune delle modalità di file più importanti sono:
Table: Modalità di File
Simbolo | Spiegazione |
---|---|
< | Apri in lettura |
> | Apri in scrittura, sovrascrivendo il contenuto precedente se il file esiste già e creando il file in caso contrario. |
>> | Apri in scrittura, aggiungendo in coda al contenuto precedente o creando un nuovo file. |
+< | Apri in lettura e scrittura. |
-| | Apri una pipe in lettura verso un processo esterno. |
|- | Apri una pipe in scrittura verso un processo esterno. |
Potete anche creare dei filehandle che leggono e scrivono su un semplice scalare Perl, usando una qualunque modalità di file:
open my $fh_leggi, '<', \$falso_input;
open my $fh_scrivi, '>', \$output_catturato;
fai_qualcosa_di_straordinario( $fh_leggi, $fh_scrivi );
Tutti gli esempi di questa sezione hanno la use autodie; attivata, in modo che si possa evitare senza problemi la gestione degli errori. Se non volete usare autodie, potete farlo--ma ricordatevi di controllare i valori restituiti da tutte le chiamate di sistema per gestire gli errori in modo appropriato.
perldoc perlopentut contiene molti più dettagli sugli usi più esotici di open, inclusa la possibilità di lanciare e controllare altri processi e l'uso di sysopen per un controllo più fine dell'input e dell'output. perldoc perlfaq5 contiene del codice d'esempio per molti casi d'uso comuni dell'IO.
open a Due Argomenti
Il codice meno recente usa la forma di open() a due argomenti, che fonde la modalità e il nome del file:
open my $fh, "> $un_file"
or die "Non posso scrivere su '$un_file': $!\n";
Il fatto che Perl debba estrarre la modalità del file dal nome del file è causa di potenziali problemi. Ogni volta che Perl deve indovinare le vostre intenzioni, correte il rischio che faccia la scelta sbagliata. Peggio ancora, se $un_file proviene da dell'input utente inaffidabile c'è un potenziale problema di sicurezza, dato che la presenza di caratteri indesiderati potrebbe modificare il comportamento del vostro programma.
La open() a tre argomenti è un sostituto più sicuro di questo codice.
Il filehandle di package speciale DATA rappresenta il file corrente. Quando Perl termina la compilazione del file, lascia DATA aperto alla fine dell'unità di compilazione se il file ha una sezione __DATA__ o __END__. Tutto il testo che segue tale token diventa disponibile tramite lettura da DATA. Questo meccanismo è utile per brevi programmi autocontenuti. Vedete perldoc perldata per maggiori dettagli.
Lettura da File
Potete leggere da un filehandle aperto in lettura con la funzione predefinita readline, scritta anche <>. Un idioma molto comune consiste nella lettura di una linea per volta in un ciclo while():
open my $fh, '<', 'un_file';
while (<$fh>)
{
chomp;
say "Ho letto la linea '$_'";
}
In contesto scalare, readline itera su tutte le linee del file finchè non arriva alla fine del file (eof()). Ogni iterazione restituisce la linea successiva. Quando si arriva alla fine del file, viene restituito il valore undef. L'idioma di esempio con while controlla esplicitamente che la variabile di iterazione sia definita, in modo che il ciclo termini soltanto quando si arriva alla fine del file. In altre parole, il codice dell'idioma è una versione abbreviata di:
open my $fh, '<', 'un_file';
while (defined($_ = <$fh>))
{
chomp;
say "Ho letto la linea '$_'";
last if eof $fh;
}
for impone il contesto lista ai suoi operandi. Nel caso di readline, Perl leggerebbe l'intero file prima di elaborarne una parte qualunque. while invece itera leggendo una linea per volta. Quando la quantità di memoria può essere un problema, usate il while.
Ogni linea letta da readline contiene il carattere o i caratteri di marcatura di fine linea. Nella maggior parte dei casi, tale marcatura è una sequenza di caratteri che varia da piattaforma a piattaforma e consiste in un carattere di a capo (\n), un carriage return (\r) o una combinazione dei due (\r\n). Usate chomp per rimuoverla.
Il modo più indicato per leggere un file linea per linea in Perl 5 è il seguente:
open my $fh, '<', $nomefile;
while (my $linea = <$fh>)
{
chomp $linea;
...
}
Per default, Perl accede ai file in modo testo. Se state leggendo dei dati binari, come un file multimediale o un file compresso--usate binmode prima di effettuare qualunque operazione di IO. In questo modo Perl preserverà i dati senza modificarli in alcun modo (footnote: Le possibili modifiche includono invece la conversione di \n nella sequenza di a capo specifica della piattaforma.). Anche se le piattaforme in stile Unix non sempre necessitano di binmode, i programmi portabili dovrebbero usare più cautela (Stringhe e Unicode(unicode)).
Scrittura su File
Potete scrivere su un filehandle aperto in scrittura con print e say:
open my $fh_out, '>', 'file_di_output.txt';
print $fh_out "Ecco una linea di testo\n";
say $fh_out "... ed eccone un'altra";
Notate l'assenza della virgola tra il filehandle e l'operando successivo.
Perl Best Practices di Damian Conway raccommanda l'abitudine di racchiudere il filehandle tra parentesi graffe. Ciò è necessario per disambiguare il parsing di un filehandle contenuto in una variabile aggregata e non fa danni negli altri casi.
Sia print che say ricevono una lista di operandi. Perl 5 usa la variabile globale predefinita $, come separatore tra i valori di una lista e il valore di $\ come ultimo argomento di print e say. Pertanto, queste due linee di codice producono lo stesso risultato:
my @principi = qw( Corwin Eric Random ... );
print @principi;
print join( $,, @principi ) . $\;
Chiusura dei File
Quando avete terminato di manipolare un file, fate una close esplicita sul suo filehandle oppure lasciate semplicemente che il programma esca dal suo scope. In quest'ultimo caso Perl chiuderà il file al vostro posto. Il vantaggio di chiamare esplicitamete close è che potete controllare--e ripristinare--specifiche situazioni di errore, come l'esaurimento dello spazio su un dispositivo di memoria di massa o l'indisponibilità di una connessione di rete.
Come al solito, autodie può gestire questi controlli al vostro posto:
use autodie;
open my $fh, '>', $file;
...
close $fh;
Variabili Speciali di Gestione dei File
Ad ogni linea letta Perl 5 incrementa il valore della variabile $., che serve quindi da contatore di linee.
readline usa il contenuto corrente di $/ come sequenza di fine linea. Per default, il valore di questa variabile è la sequenza di caratteri di fine linea più appropriata per i file di testo sulla vostra piattaforma. A dire il vero, la parola linea è fuorviante. Potete impostare il contenuto di $/ ad una qualunque sequenza di caratteri (footnote: ... anche se, sfortunatamente, non con un'espressione regolare. Perl 5 non supporta questo caso.). Ciò può essere utile per dati fortemente strutturati da cui volete leggere un record alla volta. Dato un file i cui record sono separati da due linee vuote, impostate $/ a \n\n per leggere un record alla volta. Eseguendo chomp su un record letto dal file rimuoverete la sequenza di due caratteri di a capo.
Per default Perl fa il buffering del proprio output, ed effettua operazioni di IO solo quando la quantità di output pendente supera una determinata soglia. In questo modo è possibile aggregare tra loro un certo numero di costose operazioni di IO invece di scrivere separatamente delle piccole quantità di dati. Tuttavia alcune volte potreste desiderare di inviare i dati non appena sono disponibili, estromettendo il buffering--specialmente se state scrivendo un filtro a linea di comando collegato ad altri programmi oppure un servizio di rete che opera su singole linee.
La variabile $| controlla il buffering sul filehandle di output corrente. Quando è impostato a un valore diverso da zero, Perl fa il flush dell'output dopo ciascuna scrittura sul filehandle. Quando è impostato al valore zero, Perl adotta la propria strategia di buffering di default.
Per default i file adottano una strategia con buffering completo. Quando è connesso a un terminale attivo--e non a un altro programma--STDOUT adotta una strategia di buffering a linee, secondo la quale Perl fa il flush di STDOUT ogni volta che incontra un carattere di a capo nell'output.
Su un filehandle lessicale, usate il metodo autoflush() al posto della variabile globale:
open my $fh, '>', 'nocciole.log';
$fh->autoflush( 1 );
...
A partire da Perl 5.14, potete invocare su un filehandle tutti i metodi forniti da IO::File senza dover caricare esplicitamente IO::File. In Perl 5.12 dovete caricare manualmente IO::File e in Perl 5.10 e versioni precedenti dovete caricare invece FileHandle.
I metodi input_line_number() e input_record_separator() di IO::File vi permettono un accesso per singolo filehandle alle funzionalità per cui normalmente dovreste usare le variabili globali predefinite $. e $/. Vedete la documentazione di IO::File, IO::Handle e IO::Seekable per maggiori informazioni.
Directory e Path
Manipolare le directory è simile a manipolare i file, eccetto che non è possibile scrivere sulle directory (footnote: Mentre potete salvare, spostare, rinominare e rimuovere i file.). Per aprire un directory handle usate la funzione nativa opendir:
opendir my $dirh, '/home/scimmiotto/compiti/';
La funzione predefinita readdir legge da una directory. Analogamente a readline, potete iterare sul contenuto delle directory un elemento alla volta o potete assegnarlo a una lista in un'unica operazione:
# iterazione
while (my $file = readdir $dirh)
{
...
}
# appiattimento in una lista
my @listafile = readdir $altrodirh;
In Perl 5.12 è stata aggiunta una funzionalità per cui all'interno di un while readdir imposta $_:
use 5.012;
opendir my $dirh, 'compiti/circo/';
while (readdir $dirh)
{
next if /^\./;
say "Ho trovato il compito $_!";
}
La strana espressione regolare in questo esempio permette di ignorare i cosiddetti file nascosti su sistemi Unix e simili, in cui per default la presenza di un punto iniziale fa sì che essi non compaiano negli elenchi del contenuto delle directory. Vengono ignorati anche i due file speciali . e .., che rappresentano rispettivamente la directory corrente e la sua directory genitore.
I nomi restituiti da readdir sono relativi alla directory stessa. In altre parole, se la directory compiti/ contiene tre file con i nomi mangiare, bere e scimmiottare, readdir restituisce mangiare, bere e scimmiottare e non compiti/mangiare, compiti/bere e compiti/scimmiottare. Invece, un path assoluto è un path completamente qualificato del filesystem.
Per chiudere un directory handle lasciate che il programma esca dal suo scope oppure invocate la funzione predefinita closedir.
Manipolazione di Path
Perl 5 offre una visione "alla Unix" del vostro filesystem e interpreta i path in stile Unix nel modo appropriato al vostro sistema operativo e al vostro filesystem. In altre parole, se siete su Microsoft Windows, potete usare sia il path C:/Miei Documenti/Robot/Bender/ che il path C:\Miei Documenti\Robot\Caprica Sei\.
Sebbene le operazioni di Perl seguano la semantica dei file Unix, l'uso di un modulo può rendere molto più facile la manipolazione multi-piattaforma di file. La famiglia di moduli File::Spec distribuiti con Perl fornisce astrazioni che vi permettono di manipolare path di file in modo sicuro e portabile. È venerando e consolidato, ma è anche piuttosto macchinoso.
La distribuzione CPAN Path::Class fornisce un'interfaccia migliore. Usate la funzione dir() per creare un oggetto che rappresenta una directory e la funzione file() per creare un oggetto che rappresenta un file:
use Path::Class;
my $pasti = dir( 'compiti', 'cucina' );
my $file = file( 'compiti', 'salute', 'robot.txt' );
Potete ottenere oggetti File da directory e viceversa:
my $pranzo = $pasti->file( 'calzone_ai_funghi' );
my $dir_robot = $lista_robot->dir();
Potete anche aprire filehandle di directory e file:
my $fh_dir = $dir->open();
my $fh_robot = $lista_robot->open( 'r' )
or die "Open non riuscita: $!";
Sia Path::Class::Dir che Path::Class::File offrono altre funzionalità interessanti--ma ricordate che se usate un oggetto Path::Class di un qualche tipo con un operatore o una funzione Perl 5 che si aspettano una stringa contenente un path di file, dovete effettuare voi stessi la conversione in stringa dell'oggetto. Si tratta comunque di un piccolo difetto, anche se di lunga data.
my $contenuti = leggi_da_nomefile( "$pranzo" );
Manipolazione di File
Oltre a leggere e scrivere sui file, potete anche manipolarli come fareste direttamente da linea di comando o da un gestore di file. Gli operatori di test di file, chiamati collettivamente "operatori -X" dato che consistono in un trattino seguito da un'unica lettera, esaminano gli attributi di file e directory. Per esempio, per verificare se un file esiste:
say 'Presente!' if -e $nomefile;
L'operatore -e ha un unico operando, ovvero il nome di un file, un filehandle o un directory handle. Se il file esiste, l'espressione valuta a un valore vero. perldoc -f -X elenca tutti gli altri test di file; i più comuni sono:
A partire da Perl 5.10.1, potete accedere alla documentazione di ciascuno di questi operatori, per esempio con perldoc -f -r.
La funzione predefinita rename vi permette di ridenominare un file o di spostarlo da una directory ad un'altra. Riceve due operandi, ovvero il vecchio nome e il nuovo nome del file:
rename 'stella_morta.txt', 'blocco_di_carbonio.txt';
# oppure se volete essere raffinati:
rename 'stella_morta.txt' => 'blocco_di_carbonio.txt';
Non c'è alcuna funzione predefinita per copiare un file, ma il modulo File::Copy distribuito con Perl fornisce sia una funzione copy() che una funzione move(). Usate la funzione predefinita unlink per cancellare uno o più file (la funzione predefinita delete cancella un elemento da un hash, e non un file dal filesystem.). Tutte queste funzioni restituiscono un valore vero in caso di successo e impostano $! in caso di errore.
Path::Class fornisce metodi convenienti per controllare alcuni attributi di file e per rimuovere file in modo portabile.
Perl tiene traccia della directory di esecuzione corrente. Per default, essa è la directory da cui avete lanciato il programma. La funzione cwd() del modulo Cwd distribuito con Perl restituisce il nome della directory di esecuzione corrente. La funzione predefinita chdir tenta di cambiare la directory di esecuzione corrente. Eseguire il codice nella directory corretta è essenziale per manipolare i file con path relativi.
Moduli
Molte persone considerano CPAN (CPAN(cpan)) come il principale punto di forza di Perl 5. Essenzialmente CPAN è un sistema per trovare e installare moduli. Un modulo è un package contenuto in un proprio file caricabile con use o require. Ogni modulo deve contenere codice Perl 5 valido e deve terminare con un'espressione che valuta ad un valore vero, in modo che il parser Perl 5 possa sapere se il caricamento e la compilazione del modulo hanno avuto successo. Non ci sono altri requisiti, ma solo delle convenzioni molto consolidate.
Quando caricate un modulo, Perl spezza il nome del package in componenti separati da doppi due punti (::) e li usa per ricavare un path di file. In pratica, l'istruzione use StranaBestia; fa sì che Perl cerchi un file di nome StranaBestia.pm in ogni directory elencata in @INC, nell'ordine, fino a che lo trova oppure esaurisce la lista.
Analogamente, use StranaBestia::Persistenza; fa sì che Perl cerchi un file di nome Persistenza.pm in ogni directory di nome StranaBestia/ presente in una qualunque directory elencata in @INC. use StranaBestia::UI::SmartPhone; fa sì che Perl cerchi un path relativo di file StranaBestia/UI/SmartPhone.pm in ogni directory in @INC, e così via.
Il file risultante può contenere o meno una dichiarazione di package corrispondente al suo nome--non vi è un requisito tecnico in questo senso--ma ovvie considerazioni di manutenibilità raccomandano di seguire tale convenzione.
perldoc -l Nome::Modulo visualizza il path completo del file .pm corrispondente, purché tale file contenga della documentazione per il modulo. perldoc -lm Nome::Modulo visualizza il path completo del file .pm, indipendentemente dall'esistenza di un file .pod corrispondente. perldoc -m Nome::Modulo visualizza il contenuto del file .pm.
Uso e Importazione
Quando caricate un modulo con use, Perl lo carica dal disco e poi chiama il suo metodo import(), passandogli gli eventuali argomenti che avete specificato. Per convenzione, il metodo import() di un modulo riceve una lista di nomi e esporta funzioni e altri simboli nel namespace chiamante. Si noti che questa è meramente una convenzione; un modulo potrebbe non fornire import() oppure il suo import() potrebbe avere un comportamento diverso. Le direttive (Direttive(pragmas)) come strict usano gli argomenti per modificare il comportamento dello scope lessicale chiamante anziché per esportare simboli:
use strict;
# ... chiama strict->import()
use CGI ':standard';
# ... chiama CGI->import( ':standard' )
use feature qw( say switch );
# ... chiama feature->import( qw( say switch ) )
L'istruzione predefinita no chiama il metodo unimport() di un modulo, se esiste, e gli passa gli eventuali argomenti. Viene usata solitamente con le direttive che modificano qualche comportamento con import():
use strict;
# riferimenti simbolici e bareword non ammessi
# richiesta la dichiarazione delle variabili
{
no strict 'refs';
# riferimenti simbolici ammessi
# strict 'subs' e 'vars' ancora attive
}
Sia use che no hanno effetto durante la compilazione, per cui:
use Nome::Modulo qw( lista di argomenti );
... equivale a:
BEGIN
{
require 'Nome/Modulo.pm';
Nome::Modulo->import( qw( lista di argomenti ) );
}
Analogamente:
no Nome::Modulo qw( lista di argomenti );
... equivale a:
BEGIN
{
require 'Nome/Modulo.pm';
Nome::Modulo->unimport(qw( lista di argomenti ));
}
... inclusa la require del modulo.
Se import() o unimport() non esistono nel modulo, Perl non genera un messaggio di errore. Questi metodi sono realmente opzionali.
Se volete, potete chiamare direttamente import() e unimport(), sebbene abbia poco senso farlo al di fuori di un blocco BEGIN; dopo il termine della compilazione, import() e unimport() potrebbero non avere alcun effetto.
Le istruzioni use e require di Perl 5 distinguono tra maiuscole e minuscole, ma il fatto che Perl sappia distinguere tra strict e Strict non significa che lo stesso valga per la vostra combinazione di sistema operativo e file system. Se scrivete use Strict;, Perl non troverà strict.pm su un filesystem che distingue tra maiuscole e minuscole. Se invece il filesystem non distingue tra maiuscole e minuscole, Perl non avrà obiezioni a caricare Strict.pm, ma quando verrà invocato Strict->import() non accadrà nulla (strict.pm dichiara un package di nome strict).
I programmi portabili sono restrittivi sulla distinzione tra maiuscole e minuscole anche quando non è necessario.
Esportazione
Un modulo può rendere alcuni simboli globali disponibili ad altri package attraverso un processo noto come esportazione--un processo avviato chiamando import() implicitamente o direttamente.
Il modulo Exporter distribuito con Perl fornisce un meccanismo standard per esportare simboli da un modulo. Exporter dipende dalla presenza di variabili globali di package --in particolare @EXPORT_OK e @EXPORT--che contengono una lista di simboli da esportare su richiesta.
Considerate un modulo StranaBestia::Servizi che fornisce svariate funzioni indipendenti le une dalle altre e utilizzabili in tutto il sistema:
package StranaBestia::Servizi;
use Exporter 'import';
our @EXPORT_OK = qw( gira traduci gratta );
...
Ogni pezzo di codice può ora usare questo modulo e, facoltativamente, importare alcune delle tre funzioni esportate (o tutte e tre). Potete anche esportare variabili:
push @EXPORT_OK, qw( $ragno $scimmia $scoiattolo );
I simboli esportati per default vanno elencati in @EXPORT anziché in @EXPORT_OK:
our @EXPORT = qw( danza_bestiale sonno_bestiale );
... in modo che use StranaBestia::Servizi; importi entrambe le funzioni. Ricordate che, se specificate dei simboli da importare, i simboli di default non verranno importati; in tale caso, otterrete solo ciò che chiedete. Per caricare un modulo senza importare alcun simbolo fornite esplicitamente una lista vuota:
# rendi il modulo disponibile, ma non import()are simboli
use StranaBestia::Servizi ();
Indipendentemente dalle liste di importazione, potete sempre chiamare le funzioni in un altro package con i loro nomi qualificati:
StranaBestia::Servizi::gratta();
Il modulo CPAN Sub::Exporter fornisce un'interfaccia migliore per esportare funzioni senza usare variabili globali di package. Esso offre anche opzioni più potenti. Tuttavia, mentre Exporter può esportare variabili, Sub::Exporter può esportare solo funzioni.
Organizzare il Codice con i Moduli
Perl 5 non vi richiede di usare i moduli, né i package, né i namespace. Potete scrivere tutto il vostro codice in un unico file .pl o in svariati file .pl di cui fare la require dove necessario. Avete a disposizione abbastanza flessibilità per gestire il vostro codice nel modo più appropriato a seconda del vostro stile di programmazione, del livello di formalità, rischio e valore del progetto, della vostra esperienza e della vostra dimestichezza con la distribuzione del software in Perl 5.
Tuttavia, un progetto con più di un paio di centinaia di linee di codice può avvantaggiarsi in molti modi di un'organizzazione a moduli:
Anche se non adottate un approccio orientato agli oggetti, modellare ogni entità e responsabilità distinta del vostro sistema con il proprio modulo tiene insieme le parti di codice legate tra di loro e le separa dalle altre.
Distribuzioni
Il modo più semplice per gestire la configurazione, il build, l'impacchettamento, il testing e l'installazione del software è quello di seguire le convenzioni per le distribuzioni CPAN. Una distribuzione è una collezione di metadati e di uno o più moduli (Moduli(modules)) che formano un'unità redistribuibile, testabile e installabile.
Tutte queste linee guida--per l'impacchettamento di una distribuzione, per la risoluzione delle sue dipendenze, per determinare dove installare software, per verificare che funzioni, per visualizzare la documentazione, per gestire un archivio--sono emerse dal generale consenso di migliaia di collaboratori impegnati a lavorare su decine di migliaia di progetti. Una distribuzione che segue gli standard CPAN può essere testata su svariate versioni di Perl 5 e svariate piattaforme hardware entro poche ore da quando viene caricata--il tutto senza necessità dell'intervento umano e con la segnalazione automatica agli autori di eventuali errori.
Anche se decidete di non rilasciare il vostro codice come distribuzione pubblica su CPAN, potete usare gli strumenti e le convenzioni di CPAN anche per gestire il vostro codice privato. La comunità Perl ha creato un'eccellente infrastruttura; perché non dovreste avvantaggiarvene?
Attributi di una Distribuzione
Oltre ad uno o più moduli, una distribuzione contiene svariati altri file e directory:
Una distribuzione ben-formata deve contenere un nome univoco e un singolo numero di versione (spesso derivato dal suo modulo pricipale). Ogni distribuzione pubblica che scaricate da CPAN dovrebbe rispettare questi standard. Il servizio pubblico CPANTS (http://cpants.perl.org/) valuta ogni distribuzione caricata in base alle linee guida di impacchettamento e alle convenzioni, e suggerisce eventuali migliorie. Se il codice segue le linee guida di CPANTS non significa che funzioni, ma che gli strumenti di impacchettamento di CPAN dovrebbero essere in grado di gestire tale distribuzione.
Strumenti CPAN per Gestire le Distribuzioni
Perl 5 vi mette direttamente a disposizione svariati strumenti per installare, sviluppare e gestire le vostre distribuzioni:
Inoltre, potete installare molti moduli CPAN che rendono più facile la vostra vita di sviluppatori:
Progettazione delle Distribuzioni
Il processo di progettazione di una distribuzione meriterebbe un libro a sé (vedete Writing Perl Modules for CPAN di Sam Tregar), ma alcuni principi possono comunque esservi utili. Avviate il vostro progetto con una utility come Module::Starter o Dist::Zilla. Il costo iniziale per apprenderne la configurazione e le regole può sembrare un investimento faticoso, ma i benefici di avere ogni cosa impostata correttamente (e, nel caso di Dist::Zilla, sempre aggiornata) vi solleva da gran parte del lavoro gestionale più noioso.
A questo punto, considerate le seguenti regole:
Il Package UNIVERSAL
Il package UNIVERSAL predefinito in Perl 5 è l'antenato comune di tutti gli altri package--in un senso orientato agli oggetti (Moose(moose)). UNIVERSAL fornisce alcuni metodi che i suoi derivati possono ereditare o sovrascrivere.
Il Metodo isa()
Il metodo isa() riceve una stringa contenente il nome di una classe o di un tipo predefinito. Potete invocarlo come metodo di classe o di istanza su un oggetto. Restituisce un valore vero se il suo invocante è la classe data o una sua derivata, oppure se il suo invocante è un riferimento blessed al tipo dato.
Dato un oggetto $pepe (un riferimento a hash su cui si è fatto il bless alla classe Scimmia, che deriva dalla classe Mammifero):
say $pepe->isa( 'Scimmia' ); # stampa 1
say $pepe->isa( 'Mammifero' ); # stampa 1
say $pepe->isa( 'HASH' ); # stampa 1
say Scimmia->isa( 'Mammifero' ); # stampa 1
say $pepper->isa( 'Delfino' ); # stampa 0
say $pepper->isa( 'ARRAY' ); # stampa 0
say Scimmia->isa( 'HASH' ); # stampa 0
I tipi predefiniti di Perl 5 sono SCALAR, ARRAY, HASH, Regexp, IO e CODE.
Ogni classe può sovrascrivere isa(). Ciò può essere utile quando si usano oggetti emulati (vedete Test::MockObject e Test::MockModule su CPAN) e con il codice che non usa i ruoli (Ruoli(roles)). Tenete presente che ciascuna classe che sovrascrive isa() deve avere in genere una buona ragione per farlo.
Il Metodo can()
Il metodo can() riceve una stringa contenente il nome di un metodo. Restituisce un riferimento alla funzione che implementa tale metodo, se esso esiste. Altrimenti, restituisce un valore falso. Potete chiamarlo su una classe, su un oggetto o sul nome di un package. In quest'ultimo caso, viene restituito un riferimento a una funzione anziché a un metodo (footnote: ... anche se non potete distinguerli, dato tale riferimento.).
Anche se UNIVERSAL::isa() e UNIVERSAL::can() sono metodi (Equivalenza Metodi-Funzioni(method_sub_equivalence)), potete tranquillamente usare quest'ultimo come funzione allo scopo di determinare se una classe esiste in Perl 5. Se UNIVERSAL::can( $nomeclasse, 'can' ) restituisce un valore vero, significa che qualcuno da qualche parte ha definito una classe di nome $nomeclasse. Tale classe potrebbe non essere utilizzabile, ma esiste.
Data una classe di nome ScimmiaRagno con un metodo di nome strilla, potete ottenere un riferimento al metodo con:
if (my $metodo = ScimmiaRagno->can( 'strilla' )) {...}
if (my $metodo = $sm->can( 'strilla' )
{
$sm->$metodo();
}
Usate can() per testare se un package implementa una specifica funzione o metodo:
use Class::Load;
die "Caricamento di $modulo non riuscito!"
unless load_class( $modulo );
if (my $registra = $modulo->can( 'registra' ))
{
$registra->();
}
Mentre il modulo CPAN Class::Load semplifica il caricamento di classi per nome--evitando il balletto dei require--Module::Pluggable vi solleva da gran parte del lavoro necessario per creare e gestire sistemi di plugin. Imparate a usare entrambe le distribuzioni.
Il Metodo VERSION()
Il metodo VERSION() restituisce il valore della variabile $VERSION per il package o la classe appropriata. Se fornite un numero di versione come parametro opzionale, il metodo solleva un'eccezione quando la variabile $VERSION considerata non è uguale o maggiore al parametro.
Dato un modulo ScimmiaUrlatrice con versione 1.23:
say ScimmiaUrlatrice->VERSION(); # stampa 1.23
say $su->VERSION(); # stampa 1.23
say $su->VERSION( 0.0 ); # stampa 1.23
say $su->VERSION( 1.23 ); # stampa 1.23
say $su->VERSION( 2.0 ); # eccezione!
Non ci sono molte ragioni validi per sovrascrivere VERSION().
Il Metodo DOES()
Il metodo DOES() è stato introdotto in Perl 5.10.0. Il suo scopo è quello di supportare l'uso dei ruoli (Ruoli(roles)) nei programmi. Passategli un invocante e il nome di un ruolo, e il metodo restituirà vero se la classe appropriata ha in qualche modo quel ruolo--attraverso l'ereditarietà, la delega, la composizione, l'applicazione di ruolo o un qualunque altro meccanismo.
L'implementazione di default di DOES() si riconduce a isa(), in quanto l'ereditarietà è uno dei meccanismi con cui una classe assume un ruolo. Data una scimmia Cappuccina:
say Cappuccina->DOES( 'Scimmia' ); # stampa 1
say $cappy->DOES( 'Scimmia' ); # stampa 1
say Cappuccina->DOES( 'Invertebrato' ); # stampa 0
Sovrascrivete DOES() se fornite manualmente un ruolo o un altro comportamento allomorfico.
Estendere UNIVERSAL
Potreste essere tentati di memorizzare altri metodi in UNIVERSAL per renderli disponibili a tutte le altre classi e oggetti Perl 5. Resistete a questa tentazione; un comportamento globale può avere effetti collaterali insidiosi proprio perché non è vincolato.
Detto questo, talvolta può essere accettabile abusare di UNIVERSAL a scopo di debugging o per correggere un comportamento di default improprio. Per esempio, la distribuzione UNIVERSAL::ref di Joshua ben Jore rende usabile l'operatore ref() che normalmente è inutile. Le distribuzioni UNIVERSAL::can e UNIVERSAL::isa possono aiutarvi a fare il debug di errori di anti-polimorfismo (Equivalenza Metodi-Funzioni(method_sub_equivalence)). Perl::Critic può rilevare questi ed altri problemi.
Al di fuori di codice molto controllato e di esigenze pragmatiche molto specifiche, non c'è ragione di aggiungere direttamente del codice a UNIVERSAL. Quasi sempre ci sono alternative di progettazione migliori.
Generazione di Codice
I programmatori principianti scrivono più codice del necessario, in parte perché non hanno familiarità col linguaggio, le librerie e gli idiomi, e in parte perché mancano di esperienza. Cominciano scrivendo lunghi pezzi di codice procedurale, poi scoprono le funzioni, poi i parametri, poi gli oggetti e--in qualche caso--le funzioni di ordine superiore e le chiusure.
Man mano che migliorate come programmatori, vi troverete a scrivere sempre meno codice per risolvere uno stesso problema. Userete delle astrazioni migliori e scriverete codice più generico. Potrete riutilizzare il codice--e quando arriverete ad aggiungere funzionalità cancellando del codice potrete considerarvi davvero soddisfatti.
La scrittura di programmi che scrivono programmi al vostro posto--detta metaprogrammazione o generazione di codice--offre ulteriori possibilità di astrazione. Anche se in questo modo potete fare enormi pasticci, potete anche creare cose eccellenti. Per esempio, ciò che rende possibile Moose (Moose(moose)) sono proprio delle tecniche di metaprogrammazione.
L'uso di AUTOLOAD (AUTOLOAD(autoload)) per le funzioni e metodi mancanti è un esempio limitato di questa tecnica; il sistema di dispatch di funzioni e metodi di Perl 5 vi permette di controllare che cosa succede quando essi non vengono trovati normalmente.
eval
La tecnica più semplice di generazione di codice consiste nel definire una stringa contenente un pezzetto di codice Perl valido e compilarla con la forma dell'operatore eval che riceve una stringa. Contrariamente alla forma a blocco dell'operatore eval usata per catturare eccezioni, la eval di una stringa compila il suo contenuto nello scope corrente, che include il package corrente e le proprie istanze dei lessicali.
Un utilizzo comune di questa tecnica è quello di fornire una soluzione di ripiego quando non potete (o non volete) caricare una dipendenza opzionale:
eval { require Scimmia::Rintracciata }
or eval 'sub Scimmia::Rintracciata::log {}';
Se Scimmia::Rintracciata non è disponibile, questo codice fa in modo che la sua funzione log() esista, anche se non fa nulla. Tuttavia questo esempio è ingannevolmente semplice. Usare eval nel modo giusto richiede degli sforzi; dovete gestire i problemi di quotatura per poter includere delle variabili nel vostro codice passato a eval. Interpolare alcune variabili ma non altre è ancora più complesso:
sub genera_accessori
{
my ($nomemet, $nomeattr) = @_;
eval <<"FINE_ACCESSORE";
sub leggi_$nomemet
{
my \$self = shift;
return \$self->{$nomeattr};
}
sub scrivi_$nomemet
{
my (\$self, \$valore) = \@_;
\$self->{$nomeattr} = \$valore;
}
FINE_ACCESSORE
}
Guai a dimenticare un backslash! E buona fortuna a spiegare al vostro evidenziatore di sintassi che cosa state facendo! Peggio ancora, ogni invocazione di eval su stringa genera una nuova struttura dati che rappresenta l'intero codice, e anche compilarla non è certo gratis. Pur con tutti i suoi limiti questa tecnica ha però il vantaggio di essere semplice.
Chiusure Parametriche
Mentre creare semplici accessori e mutatori con eval è banale, le chiusure (Chiusure(closures)) vi permettono di aggiungere parametri al codice generato durante la sua compilazione, senza richiedere ulteriori elaborazioni:
sub genera_accessori
{
my $nomeattr = shift;
my $lettore = sub
{
my $self = shift;
return $self->{$nomeattr};
};
my $scrittore = sub
{
my ($self, $valore) = @_;
$self->{$nomeattr} = $valore;
};
return $lettore, $scrittore;
}
Questo codice evita i noiosi problemi di quotatura e compila ogni chiusura una sola volta. Usa anche meno memoria dato che tutte le istanze della chiusura condividono il codice compilato. L'unica cosa che differisce tra due istanze è il lessicale $nomeattr associato. In un processo con un lungo tempo di vita o con molti accessori questa tecnica può rivelarsi molto utile.
Installare delle chiusure nella tabella dei simboli è piuttosto semplice, anche se poco elegante:
{
my ($leggi, $scrivi) = genera_accessori( 'torta' );
no strict 'refs';
*{ 'leggi_torta' } = $leggi;
*{ 'scrivi_torta' } = $scrivi;
}
La strana sintassi con l'asterisco (footnote: Pensate ad esso come a un sigillo di typeglob, dove un typeglob è un'espressione gergale Perl che sta per "symbol table".) che dereferenzia un hash si riferisce alla tabella dei simboli corrente, che è una porzione del namespace corrente contenente i simboli accessibili globalmente come le variabili globali di package, le funzioni e i metodi. Assegnando un riferimento ad un elemento della tabella dei simboli si effettua l'installazione o la sostituzione di tale elemento. Per promuovere una funzione anonima a metodo memorizzate il riferimento a tale funzione nella tabella dei simboli.
Il modulo CPAN Package::Stash offre un'interfaccia migliore per la manipolazione delle tabelle dei simboli.
L'assegnamento in una tabella dei simboli tramite stringa, anziché tramite un nome letterale di variabile, è detto riferimento simbolico. Durante questa operazione dovete disattivare il controllo dei riferimenti effettuato da strict. Molti programmi contengono un errore insidioso in pezzi di codice simili a quello sopra, quando generano la chiusura e fanno l'assegnamento in un'unica linea:
{
no strict 'refs';
*{ $nomemet } = sub {
# errore insidioso: strict refs e` disattivato anche qui
};
}
Questo esempio disattiva le restrizioni sia nel blocco esterno che nel corpo stesso della funzione. Dato che a violare le restrizioni dei controlli sui riferimenti è soltanto l'assegnamento, è meglio disattivarle soltanto durante tale operazione.
Se il nome del metodo è un letterale stringa nel vostro codice sorgente, anziché il valore contenuto in una variabile, potete fare direttamente l'assegnamento al simbolo rilevante:
{
no warnings 'once';
(*leggi_torta, *scrivi_torta) =
genera_accessori( 'torta' );
}
Assegnare direttamente al glob non viola le restrizioni, ma menzionare ciascun glob una sola volta genera invece un warning di "used only once" (usato una sola volta, NdT) a meno che lo sopprimiate esplicitamente nel relativo scope.
Manipolazioni Durante la Compilazione
Diversamente dal codice scritto esplicitamente, il codice generato con una eval di stringa viene compilato durante l'esecuzione. Mentre potete ragionevolmente assumere che una funzione normale sia disponibile durante l'intera vita del vostro programma, una funzione generata potrebbe non essere disponibile quando vi serve.
Potete forzare Perl ad eseguire il codice che genera altro codice durante la compilazione racchiudendolo in un blocco BEGIN. Quando il parser Perl 5 incontra un blocco etichettato con BEGIN, esegue il parsing dell'intero blocco. Se il blocco non contiene errori di sintassi viene eseguito immediatamente. Al termine dell'esecuzione, il parsing continua come se non fosse stato interrotto.
La differenza tra scrivere:
sub leggi_eta { ... }
sub scrivi_eta { ... }
sub leggi_nome { ... }
sub scrivi_nome { ... }
sub leggi_peso { ... }
sub scrivi_peso { ... }
... e:
sub crea_accessori { ... }
BEGIN
{
for my $accessore (qw( eta nome peso ))
{
my ($leggi, $scrivi) =
crea_accessori( $accessore );
no strict 'refs';
*{ 'leggi_' . $accessore } = $leggi;
*{ 'scrivi_' . $accessore } = $scrivi;
}
}
... è essenzialmente una differenza nel grado di manutenibilità.
In un modulo, ogni pezzo di codice al di fuori delle funzioni viene eseguito quando fate la use del modulo, poiché Perl aggiunge implicitamente un BEGIN che racchiude la require e la import (Importazione(importing)). Il codice del modulo al di fuori delle funzioni viene eseguito prima della chiamata a import(). Se fate soltanto la require del modulo, non viene generato implicitamente un blocco BEGIN. In tal caso, il codice al di fuori delle funzioni viene eseguito alla fine del parsing.
Fate attenzione alle interazioni tra una dichiarazione lessicale (l'associazione di un nome ad uno scope) e un assegnamento lessicale. La prima avviene durante la compilazione, mentre il secondo avviene al momento dell'esecuzione. Il seguente codice contiene un errore insidioso:
# aggiunge un metodo require() a UNIVERSAL
use UNIVERSAL::require;
# errato; non usatelo
my $package_desiderato = 'Scimmia::A::Propulsione';
BEGIN
{
$package_desiderato->require();
$package_desiderato->import();
}
... dato che il blocco BEGIN viene eseguito prima dell'assegnamento del valore stringa a $package_desiderato. Il risultato è un'eccezione relativa al tentativo di invocare il metodo require() su un valore non definito.
Class::MOP
Contrariamente alla possibilità di installare riferimenti a funzione per popolare dei namespace e creare dei metodi, non esiste un modo semplice per creare delle classi da programma in Perl 5. Fortunatamente, Moose ci viene in soccorso con la sua libreria Class::MOP che fornisce il protocollo di meta-oggetti--un meccanismo grazie al quale si possono creare sistemi ad oggetti che manipolano se stessi.
Anziché scrivere il vostro codice in una stringa potenzialmente sbagliata da passare ad eval oppure tentare di accedere manualmente alle tabelle dei simboli, potete manipolare le entità e le astrazioni del vostro programma attraverso oppurtuni oggetti e metodi.
Per creare una classe, scrivete:
use Class::MOP;
my $classe = Class::MOP::Class->create(
'Scimmia::Cacciavite'
);
Potete aggiungere attributi e metodi alla classe quando la create:
my $classe = Class::MOP::Class->create(
'Scimmia::Cacciavite' =>
(
attributes =>
[
Class::MOP::Attribute->new('$materiale'),
Class::MOP::Attribute->new('$colore'),
]
methods =>
{
stringi => sub { ... },
allenta => sub { ... },
}
),
);
... oppure attraverso la metaclasse (l'oggetto che rappresenta la classe) dopo la creazione:
$classe->add_attribute(
esperienza => Class::MOP::Attribute->new('$xp')
);
$classe->add_method( uccidi_zombie => sub { ... } );
... e potete ispezionare la metaclasse:
my @attr = $class->get_all_attributes();
my @met = $class->get_all_methods();
Analogamente, Class::MOP::Attribute e Class::MOP::Method vi consentono di creare, manipolare e fare introspezione su attributi e metodi.
Overloading
Perl 5 non è un linguaggio orientato agli oggetti in modo pervasivo. I suoi tipi di dati predefiniti (scalari, array e hash) non sono oggetti con metodi di cui si può fare l'overload, ma potete controllare il comportamento delle vostre classi e oggetti, specialmente quando ne fate la coercizione o la valutazione contestuale, con l'overloading.
L'utilità dell'overloading è sottile, ma potente. Un esempio interessante è l'overloading del comportamento di un oggetto in contesto booleano, specialmente quando usate una tecnica come il pattern dell'Oggetto Nullo (http://www.c2.com/cgi/wiki?NullObject). In contesto booleano , un oggetto valuta a vero, a meno che facciate l'overload della sua conversione a booleano.
Potete fare l'overload del comportamento dell'oggetto per quasi tutte le operazioni e le coercizioni: conversione a stringa, numero e booeano, iterazione, invocazione, accesso ad array, accesso ad hash, operazioni aritmetiche, operatori di confronto, match intelligente, operazioni bit-a-bit e persino assegnamenti. La conversione a stringa, numero e booeano sono i casi più importanti e di uso più comune.
Overloading di Operazioni Comuni
La direttiva overload vi permette di associare una funzione a un'operazione di cui potete fare l'overload passando delle coppie di argomenti, in cui la chiave è il tipo di overload e il valore è un riferimento alla funzione da chiamare per effettuare l'operazione. Una classe Nullo che fa l'overload della valutazione booleana in modo che restituisca sempre un valore falso potrebbe essere definita come segue:
package Nullo
{
use overload 'bool' => sub { 0 };
...
}
A questo punto è facile aggiungere la conversione a stringa:
package Nullo
{
use overload
'bool' => sub { 0 },
'""' => sub { '(null)' };
}
Fare l'overload della conversione a numero è più complesso, perché gli operatori binari sono generalmente binari (Arità (arity)). Dati due operandi di cui si è fatto l'overload dei metodi per l'addizione, quale dei due ha la precedenza? La risposta deve essere coerente, facile da illustrare e comprensibile anche alle persone che non hanno letto il codice sorgente dell'implementazione.
perldoc overload tenta di spiegare questo aspetto nelle sezioni Calling Conventions for Binary Operations e MAGIC AUTOGENERATION, ma la soluzione più semplice consiste nel fare l'overload della conversione a numero (denotata con '0+') e specificare a overload di usare altri overload come ripiego ogni volta che è possibile:
package Nullo
{
use overload
'bool' => sub { 0 },
'""' => sub { '(null)' },
'0+' => sub { 0 },
fallback => 1;
}
QUando impostate fallback a un valore vero, Perl utilizzerà, se possibile, un qualunque altro overload definito per comporre l'operazione richiesta. Se non è possibile, Perl si comporterà come se non ci fossero degli overload attivi, il che è spesso ciò che desiderate.
Senza fallback, Perl userà soltanto gli overload specifici che avete fornito. Se qualcuno tenta di eseguire un'operazione della quale non avete fatto l'overload viene sollevata un'eccezione.
Overload e Ereditarietà
Le classi derivate ereditano gli overload dalle proprie superclassi. Possono ridefinire questo comportamento in due modi. Se la superclasse usa l'overloading come illustrato sopra, fornendo direttamente dei riferimenti a funzione, la classe derivata deve ridefinire il comportamento di overload della superclasse usando direttamente overload.
Le superclassi possono fornire alle classi discendenti maggiore flessibilità specificando il nome di un metodo da chiamare per implementare l'overloading, anziché condificare direttamente un riferimento a funzione:
package Nullo
{
use overload
'bool' => 'conv_bool',
'""' => 'conv_stringa',
'0+' => 'conv_num',
fallback => 1;
}
In questo caso, ciascuna classe derivata può effettuare in modo diverso le operazioni coinvolte nell'overload sovrascrivendo i metodi con nomi appropriati.
Usi dell'Overloading
Potete essere tentati di usare l'overloading come strumento per definire delle abbreviazioni simboliche per nuove operazioni, ma raramente ci sono buone ragioni per farlo in Perl 5. La distribuzione CPAN IO::All sfrutta questa idea al limite per produrre idee intelligenti di scrittura di codice conciso e componibile. Tuttavia, per ogni API raffinata con un brillante uso dell'overloading, ne vengono create una dozzina che sono invece dei pasticci. Talvolta il miglior codice evita gli eccessi di ingegno in favore della semplicità.
Ridefinire l' addizione, la moltiplicazione anche la concatenazione su una classe Matrice ha senso, perché la notazione esistente per tali operazioni è molto diffusa. Il dominio di un nuovo problema per cui non esista una notazione consolidata non è un buon candidato per l'overloading, come non lo è il dominio di un problema per cui dovete fare salti mortali per travare le corrispondenze tra gli operatori esistenti in Perl e una diversa notazione.
Perl Best Practices di Damian Conway suggerisce un altro uso dell'overloading, per prevenire l'abuso accidentale di oggetti. Per esempio, fare l'overloading della conversione a numero in modo da invocare croak() per gli oggetti che non hanno un'unica rappresentazione numerica ragionevole può aiutarvi a trovare e risolvere bug.
Taint
Perl fornisce strumenti per scrivere programmi sicuri. Tali strumenti non sostituiscono un'attenta analisi e pianificazione, ma possono ricompensare la vostra cautela e comprensione dei problemi aiutandovi ad evitare errori insidiosi.
Uso del Modo Taint
Il modo taint (o taint, marchio NdT) aggiunge metadati a tutti i dati che provengono dall'esterno del vostro programma. Ogni dato derivato da dati marchiati è anch'esso marchiato. Potete usare dati marchiati nel vostro programma, ma se li usate in operazioni che hanno effetti sul mondo esterno--ovvero se li usate in modo non sicuro--Perl solleva un'eccezione fatale.
perldoc perlsec spiega il modo taint in grande dettaglio.
Per attivare il modo taint, lanciate il vostro programma con l'argomento a linea di comando -T. Se passate questo argomento nella linea #! di un programma, dovete poi eseguire direttamente il programma; se invece lo lanciate con perl miaapptainted.pl, senza il flag -T, Perl termina con un'eccezione. Infatti, nel momento in cui Perl incontra il flag nella linea #! è già troppo tardi, ad esempio, per marchiare i dati d'ambiente di %ENV.
Origine dei Dati Marchiati
I dati marchiati hanno due origini: i dati provenienti da un qualsiasi canale di input e l'ambiente operativo del programma. Il primo consiste in qualunque cosa leggiate da un file o riceviate dagli utenti nel caso di programmi web o di rete. Il secondo include gli argomenti a linea di comando, le variabili d'ambiente e i dati restituiti da chiamate di sistema. Anche un'operazione come la lettura da un directory handle produce dati marchiati.
La funzione tainted() del modulo Scalar::Util distribuito con Perl restituisce vero se il suo argomento è marchiato:
die 'Oh no! Dati marchiati!'
if Scalar::Util::tainted( $valore_sospetto );
Rimuovere il Taint dai Dati
Per rimuovere un taint, dovete estrarre delle porzioni sicure dei dati con una cattura in un'espressione regolare. I dati catturati non sono marchiati. Se il vostro input utente consiste di un numero telefonico americano, potete rimuoverne il taint con:
die 'Numero ancora marchiato!'
unless $numero =~ /(\(/d{3}\) \d{3}-\d{4})/;
my $numero_sicuro = $1;
Più è specifico il pattern che specifica ciò che è ammissibile, più sicuro sarà il vostro programma. L'approccio opposto di rifiutare elementi o formati specifici corre il rischio di trascurare dei pericoli. È molto meglio rifiutare qualcosa che era sicuro ma inatteso che ammettere qualcosa di pericoloso che sembra sicuro. Naturalmente, nessuno vi vieta di effettuare la cattura dell'intero contenuto di una variabile--ma in tal caso, perché usare taint?
Rimuovere il Taint dall'Ambiente
La variabile globale predefinita %ENV rappresenta le variabili d'ambiente del sistema. Tali dati sono marchiati poiché i loro valori sono controllati da forze esterne al programma. Ogni variabile d'ambiente che modifica come Perl o la shell trovano file e directory sono potenziali vettori di attacchi. Un programma attento al taint dovrebbe cancellare svariate chiavi da %ENV e impostare $ENV{PATH} a un path specifico e sicuro:
delete @ENV{ qw( IFS CDPATH ENV BASH_ENV ) };
$ENV{PATH} = '/path/ai/file/eseguibili/';
Se non impostate $ENV{PATH} in modo appropriato, riceverete messaggi a proposito della sua scarsa sicurezza. Se tale variabile d'ambiente conteneva la directory d'esecuzione corrente o delle directory relative, oppure se le directory specificate permettono a tutti la scrittura, un attacco intelligente potrebbe dirottare le chiamate di sistema per perpetrare i suoi misfatti.
Per ragioni simili, @INC non contiene la directory di esecuzione corrente quando è attivo il modo taint. Perl ignora anche le variabili d'ambiente PERL5LIB e PERLLIB. Usate la direttiva lib o il flag -I di perl per aggiungere le directory di librerie al programma.
Trappole di Taint
Il modo taint è del tipo "tutto o nulla". O è attivo o non lo è. Talvolta questo induce le persone a usare dei pattern permissivi per rimuovere il taint dai dati, dando un'errata illusione di sicurezza. Considerate attentamente la rimozione del taint.
Sfortunatamente, non tutti i moduli gestiscono i dati marchiati in modo appropriato. Questo è un problema che gli autori CPAN dovrebbero considerare seriamente. Se dovete attivare i controlli di taint in codice preesistente, potete usare il flag -t, che attiva il modo taint ma riduce le violazione di taint da eccezioni a semplici warning. Questo non sostituisce ovviamente il modo taint completo, ma vi permette di mettere in sicurezza programmi esistenti senza la rigidità dell'approccio "tutto o nulla" di -T.
Perl Oltre la Sintassi
Con il Perl scolastico potete arrivare solo fino a un certo livello. Una vera scioltezza nel linguaggio vi permette di usare i suoi pattern e idiomi naturali. I programmatori esperti comprendono come interagiscono e si combinano le funzionalità di Perl.
Preparatevi per una seconda curva di apprendimento di Perl: il modo di pensare "alla Perl". Il risultato sarà del codice conciso, potente e "alla Perl".
Idiomi
Ogni linguaggio--di programmazione o naturale--ha dei pattern comuni di espressione, o idiomi. La Terra gira, ma noi parliamo della nascita e tramonto del Sole. Tutti noi ci vantiamo delle trovate intelligenti e ci vergognamo di quelle stupide quando buttiamo fuori il nostro codice.
Gli idiomi Perl non sono né funzionalità del linguaggio né tecniche di progettazione. Sono piuttosto manierismi e meccanismi che, presi tutti insieme, danno al vostro codice un accento "alla Perl". Non è necessario che li usiate, ma sappiate che sono dei punti di forza di Perl.
L'Oggetto $self
Il sistema a oggetti di Perl 5 (Moose(moose)) tratta l'invocante di un metodo come un semplice parametro. Indipendentemente dal fatto che invochiate un metodo di classe o di istanza, il primo elemento di @_ è sempre l'invocante del metodo. La maggior parte del codice Perl 5 usa per convenzione $class come nome dell'invocante di un metodo di classe e $self come nome dell'oggetto invocante. Questa convenzione è talmente consolidata che alcune utili estensioni come MooseX::Method::Signatures assumono che usiate $self come nome degli oggetti invocanti.
Parametri per Nome
L'elaborazione di liste è una componente fondamentale della valutazione di espressioni in Perl. La possibilità che hanno i programmatori Perl di concatenare espressioni che valutano a liste di lunghezza variabile fornisce innumerevoli opportunità di manipolare efficacemente i dati.
Anche se la semplicità del passaggio di argomenti in Perl (ogni cosa viene appiattita in @_) risulta talvolta persino eccessiva, assegnando da @_ in contesto lista potete spacchettare i parametri per nome come coppie. L'operatore fat comma (Dichiarazione di Hash(declaring_hashes)) trasforma in modo ovvio una normale lista in una lista di coppie di argomenti:
prepara_coppa_di_gelato(
panna_montata => 1,
praline => 1,
banana => 0,
gelato => 'menta e cioccolato',
);
La funzione chiamata può spacchettare questi parametri in un hash e trattare tale hash come un singolo argomento:
sub prepara_coppa_di_gelato
{
my %arg = @_;
my $dessert = compra_gelato( $arg{gelato} );
...
}
Perl Best Practices suggerisce di passare dei riferimenti ad hash anziché degli hash. Questo permette a Perl di effettuare la validazione del riferimento ad hash nel chiamante.
Questa tecnica funziona bene con import() (Importazione(importing)) e altri metodi; elaborate tutti i parametri che desiderate prima di inghiottire tutto ciò che resta in un hash:
sub import
{
my ($class, %arg) = @_;
my $package_chiamante = caller();
...
}
La Trasformata di Schwartz
La trasformata di Schwartz è un'elegante dimostrazione della pervasività della gestione di liste in Perl in quanto è un idioma preso agevolmente a prestito dalla famiglia di linguaggi Lisp.
Supponete di avere un hash che associa i nomi dei vostri colleghi con le loro estensioni telefoniche:
my %estensioni =
(
'001' => 'Anna',
'002' => 'Walter',
'003' => 'Gerardo',
'005' => 'Rudy',
'007' => 'Brenda',
'008' => 'Patrizio',
'011' => 'Luca',
'012' => 'Lidia',
'017' => 'Cristina',
'020' => 'Maurizio',
'023' => 'Marco',
'024' => 'Andrea',
'052' => 'Giorgia',
'088' => 'Nicola',
);
La quotatura delle chiavi di hash con l'operatore fat comma funziona soltanto su entità interpretabili come bareword; le chiavi di questo esempio sono invece numeri--in particolare numeri ottali, poiché iniziano con lo zero. Quasi tutti cascano in questo tranello.
Per ordinare questa lista per nome in ordine alfabetico, dovete ordinare l'hash in base ai suoi valori anziché in base alle sue chiavi. Ordinare i valori è facile:
my @nomi_ordinati = sort values %estensioni;
... ma dovete fare un ulteriore passo per preservare l'associazione tra nomi ed estensioni, e qui entra in gioco la trasformata di Schwartz. Per prima cosa, convertite l'hash in una lista di strutture dati facili da ordinare--in questo caso, array anonimi di due elementi:
my @coppie = map { [ $_, $estensioni{$_} ] }
keys %estensioni;
sort riceve tale lista di array anonimi e confronta i loro secondi elementi (i nomi) come stringhe:
my @coppie_ordinate = sort { $a->[1] cmp $b->[1] }
@coppie;
Il blocco fornito a sort riceve i suoi argomenti in due variabili di package (Scope(scope)) $a e $b (footnote: Vedete perldoc -f sort per una esaustiva discussione delle implicazioni di questa scelta degli scope.). Il blocco di sort prende due argomenti per volta; il primo diventa il contenuto di $a e il secondo il contenuto di $b. Se $a deve precedere $b nel risultato, il blocco restituisce -1. Se i due valori sono equivalenti per quanto riguarda l'ordinamento, il blocco restituisce 0. Infine, se $a deve seguire $b nel risultato, il blocco restituisce 1. Qualunque altro valore di ritorno è considerato un errore.
Invertire un hash sul posto funziona se non avete colleghi con lo stesso nome. I dati del nostro esempio non presentano problemi, ma il vostro codice deve difendervi da questa eventualità.
L'operatore cmp effettua confronti tra stringhe mentre l'operatore <=> effettua confronti numerici.
Una seconda operazione map su @coppie_ordinate converte la struttura dati in una forma più usabile:
my @est_formattate = map { "$_->[1], est. $_->[0]" }
@coppie_ordinate;
... e finalmente potete stampare il risultato:
say for @est_formattate;
La trasformata di Schwartz sfrutta la pervasività della gestione di liste in Perl per evitare l'uso delle variabili temporanee. La combinazione risultante è:
say for
map { " $_->[1], est. $_->[0]" }
sort { $a->[1] cmp $b->[1] }
map { [ $_ => $estensioni{$_} ] }
keys %estensioni;
Leggete l'espressione da destra verso sinistra, nell'ordine di valuazione. Per ciascuna chiave dell'hash delle estensioni, viene creato un array anonimo di due elementi contenente la chiave e il corrispondente valore nell'hash. Tale lista di array anonimi viene ordinata in base ai secondi elementi, ovvero ai valori dell'hash. Infine per ciascuno di questi array ordinati viene generata una stringa di output nel formato opportuno.
La cascata map-sort-map della trasformata di Schwartz converte una struttura dati in una forma più facile da ordinare e quindi in una terza forma.
Mentre l'esempio di ordinamento descritto è semplice, considerate il caso in cui dovete calcolare hash crittografici per grandi file. La trasformata di Schwartz è particolarmente vantaggiosa poiché fa il caching dei risultati delle computazioni costose, eseguendole una sola volta nel map più a destra.
Slurp Semplificato di File
local è fondamentale per gestire le variabili globali predefinite di Perl 5. Dovete aver compreso bene gli scope (Scope(scope)) per usare efficacemente local--ma in tal caso potete usare degli scope brevi e leggeri in modi interessanti. Per esempio, per fare lo slurp di file in uno scalare in un'unica espressione:
my $file = do { local $/ = <$fh> };
# oppure
my $file = do { local $/; <$fh> };
# oppure
my $file; { local $/; $file = <$fh> };
$/ è il separatore dei record di input. Quando viene localizzato, il suo valore viene impostato a undef fino al suo assegnamento. Tale localizzazione ha luogo prima dell'assegnamento e, dato che il valore del separatore non è definito, Perl non ha problemi a leggere l'intero contenuto del filehandle in una volta e ad assegnare il suo valore a $/. Poiché il valore di un blocco do è quello dell'ultima espressione valutata al suo interno, in questo caso il blocco valuta al valore dell'assegnamento, ovvero il contenuto del file. Sebbene al termine del blocco il valore di $/ venga immediatamente ripristinato al suo stato precedente, il valore finale di $file sarà il contenuto del file.
Nel secondo esempio il blocco non contiene assegnamenti e restituisce semplicemente la singola linea letta dal filehandle.
Il terzo esempio evita una seconda copia della stringa che rappresenta il contenuto del file; non è elegante come gli altri, ma usa una quantità minore di memoria.
Questo utile esempio è sicuramente esasperante per le persone che non comprendono a fondo sia local che gli scope. Il modulo CPAN File::Slurp è una buona alternativa e spesso è anche più veloce.
Gestire il Main
Perl non ha una sintassi speciale per la creazione delle chiusure (Chiusure(closures)); può quindi succedere che facciate involontariamente la chiusura di una variabile lessicale. Molti programmi impostano svariate variabili lessicali nello scope del file prima di trasferire l'elaborazione ad altre funzioni. Potreste essere tentati di usare queste variabili direttamente, anziché passare e ricevere i loro valori dalle funzioni, specialmente quando il programma cresce di dimensioni. Sfortunatamente, in questo modo i programmi diventano dipendenti da sottigliezze riguardanti ciò che succede durante il processo di compilazione di Perl 5; una variabile che credevate di inizializzare subito ad uno specifico valore potrebbe essere inizializzata in un momento parecchio successivo.
Per evitare che questo accada, racchiudete il codice principale del vostro programma in una semplice funzione main(). Incapsulate le vostre variabili negli scope appropriati. Quindi, aggiungete una singola linea all'inizio del vostro programma, dopo aver usato tutti i moduli e le direttive di cui avete bisogno:
#!/usr/bin/perl
use Modern::Perl;
...
exit main( @ARGS );
Chiamare main() prima di qualunque altra cosa nel programma vi impone di essere espliciti riguardo all'inizializzazione e all'ordine di compilazione. Chiamare exit con il valore restituito da main() previene l'esecuzione dell'eventuale codice successivo, ma fate attenzione a restituire 0 dal main() quando la sua esecuzione ha successo.
Esecuzione Controllata
La vera differenza tra un programma e un modulo è nell'uso che intendete farne. Gli utenti invocano i programmi direttamente, mentre i programmi caricano i moduli dopo l'inizio dell'esecuzione. Tuttavia, un modulo è fatto di codice Perl allo stesso modo di un programma. Rendere eseguibile un modulo è facile, così come fare in modo che un programma si comporti come un modulo (cosa utile per testare parti di un programma esistente senza creare formalmente un modulo). Tutto ciò che dovete fare è scoprire come Perl ha iniziato a eseguire un pezzo di codice.
L'unico argomento, peraltro opzionale, di caller è il numero di contesti di chiamata (Ricorsione(recursion)) sui quali riportare informazioni. caller(0) riporta informazioni sul contesto di chiamata corrente. Per permettere a un modulo di essere eseguito correttamente sia come programma sia come modulo, mettete tutto il codice eseguibile all'interno di funzioni, aggiungete una funzione main() e scrivete la seguente linea all'inizio del modulo:
main() unless caller(0);
Se non c'è alcun chiamante per il modulo, significa che qualcuno lo ha invocato direttamente come programma (con perl path/del/Modulo.pm invece di use Modulo;).
L'ottavo elemento della lista restituita da caller in contesto lista è un valore vero se il contesto di chiamata rappresenta use o require, altrimenti è undef. Nonostante sia molto accurato, poche persone lo usano.
Validazione Postfissa dei Parametri
CPAN contiene svariati moduli che vi aiutano a verificare i parametri delle vostre funzioni; Params::Validate e MooseX::Params::Validate sono due opzioni valide. Una validazione basilare è facile anche senza tali moduli.
Supponete che la vostra funzione accetti esattamente due argomenti. Potreste scrivere:
use Carp 'croak';
sub sposa_scimmie
{
if (@_ != 2)
{
croak 'Il matrimonio richiede due scimmie!';
}
...
}
... ma, da un punto di vista linguistico, le conseguenze sono più importanti del controllo e meritano di essere all'inizio dell'espressione:
croak 'Il matrimonio richiede due scimmie!' if @_ != 2;
... che potrebbe essere scritto in modo ancora più chiaro come:
croak 'Il matrimonio richiede due scimmie!'
unless @_ == 2;
Queste tecniche di terminazione prematura--specialmente quella con i condizionali postfissi--possono semplificare il resto del vostro codice. Ognuna di tali asserzioni è come una singola riga in una tabella di verità.
Regex En Passant
Molti degli idiomi di Perl 5 si basano sul fatto che le espressioni hanno dei valori:
say my $num_est = my $estensione = 42;
Pur essendo chiaramente poco elegante, questo codice illustra come usare il valore di un'espressione in un'altra espressione. Non è un'idea nuova; avrete quasi sicuramente già usato il valore di ritorno di una funzione in una lista o come argomento di un'altra funzione. Forse però non avete pensato a tutte le sue implicazioni.
Supponete di voler estrarre un nome di battesimo da una combinazione di nome e cognome con un'espressione regolare precompilata in $nome_battesimo_regex:
my ($nome_di_battesimo) = $nome =~ /($nome_battesimo_regex)/;
In contesto lista, un match riuscito di una regex restituisce una lista di tutte le catture (Cattura(regex_captures) e Perl assegna la prima di esse a $nome_di_battesimo.
Per modificare il nome, ad esempio per rimuovere tutti i caratteri non di parola per creare un nome utente da usare per le credenziali di un sistema, potete scrivere:
(my $nome_normalizzato = $nome) =~ tr/A-Za-z//dc;
In Perl 5.14 è stato aggiunto il modificatore non distruttivo /r delle sostituzioni, quindi potete scrivere anche my $nome_normalizzato = $nome =~ tr/A-Za-z//dcr;.
Per prima cosa, assegnate il valore di $nome a $nome_normalizzato, dato che le parentesi modificano la precedenza in modo che l'assegnamento venga eseguito per primo. L'espressione di assegnamento valuta alla variabile $nome_normalizzato, quindi tale variabile diventa il primo operando dell'operatore di traslitterazione.
Questa tecnica è applicabile ad altri operatori di modifica sul posto:
my $eta = 14;
(my $nuova_eta = $eta)++;
say "Il prossimo anno avro` $nuova_eta anni";
Coercizioni Unarie
Il sistema di tipi di Perl 5 fa quasi sempre la cosa giusta quando usate gli operatori corretti. Se usate l'operatore di concatenazione di stringhe, Perl tratterà entrambi gli operando come stringhe. Se usate l'operatore di addizione, Perl tratterà entrambi gli operandi come numeri.
Tuttavia, qualche volta dovete dare a Perl un suggerimento su ciò che intendete fare, usando una coercizione unaria per forzare un modo specifico di valutare un valore.
Per garantire che Perl tratti un valore come numerico, sommategli zero:
my $valore_numerico = 0 + $valore;
Per garantire che Perl tratti un valore come booleano, fatene una doppia negazione:
my $valore_booleano = !! $valore;
Per garantire che Perl tratti un valore come stringa, concatenatelo con la stringa vuota:
my $valore_stringa = '' . $valore;
Sebbene sia estremamente improbabile che dobbiate usare queste coercizioni, è importante riconoscere questi idiomi quando li incontrate. Infatti, anche quando vi sembra una buona idea rimuovere un "inutile" + 0 da un'espressione, ciò potrebbe danneggiare il codice.
Variabili Globali
Perl 5 fornisce svariate variabili sovraglobali (che corrispondono alle variabili globali predefinite) il cui scope è realmente globale e non limitato a un package o a un file. Sfortunatamente, la loro accessibilità globale significa anche che ogni modifica diretta o indiretta dei loro valori può avere effetti su altre parti del programma; inoltre, i loro nomi sono alquanto criptici. Anche i programmatori Perl 5 esperti generalmente ne conoscono a memoria solo alcune, e ben poche persone le conoscono tutte. D'altra parte, solo un piccolo sottoinsieme di queste variabili può davvero esservi utile. La loro lista esaustiva è contenuta in perldoc perlvar.
Gestire le Variabili Sovraglobali
Le nuove versioni di Perl 5 continuano a convertire dei comportamenti globali in comportamenti lessicali; grazie a ciò, è possibile evitare di usare molte di queste variabili sovraglobali. Quando però non potete evitarle, usate local nello scope più ristretto possibile per limitare gli effetti delle vostre modifiche. In questo modo, anche se il vostro codice rimane esposto ai cambiamenti effettuati su tali variabili dal codice che chiamate, evitate almeno possibili sorprese da parte del codice esterno al vostro scope. Come illustrato dall'idioma dello slurp semplificato di file (Slurp Semplificato di File(easy_file_slurping)), local è spesso l'approccio giusto:
my $file; { local $/; $file = <$fh> };
L'effetto della localizzazione di $/ dura soltanto fino alla fine del blocco. E ci sono poche possibilità che la lettura di linee da un filehandle provochi l'esecuzione di codice Perl (footnote: Un filehandle tied (Tie(tie)) è una di queste possibilità.) che modifica il valore di $/ all'interno del blocco do.
Non tutti i casi di utilizzo delle variabili sovraglobali sono così facili da trattare, ma spesso questo approccio funziona.
Talvolta potete aver bisogno di leggere il valore di una variabile sovraglobale, sperando che nel frattempo esso non sia stato modificato da altro codice. Catturare eccezioni con un blocco eval può dar luogo a problemi di concorrenza, dato che i metodi DESTROY() invocati sui lessicali dello scope da cui il programma è appena uscito potrebbero cancellare il valore di $@:
local $@;
eval { ... };
if (my $eccezione = $@) { ... }
Copiate $@ immediatamente dopo aver catturato un'eccezione per preservare il suo contenuto. In alternativa, vedete Try::Tiny (Avvertenze sulle Eccezioni(exception_caveats)).
Nomi in Inglese
Il modulo English distribuito con Perl fornisce dei nomi lunghi (più facili da ricordare) per le variabili sovraglobali al posto di quelli composti di punteggiatura. Potete importarli in un namespace con:
use English '-no_match_vars';
In questo modo potete usare i nomi lunghi documentati in perldoc perlvar all'interno dello scope della direttiva.
Tre variabili sovraglobali relative alle regex ($&, $` e $') impongono una penalità di efficienza che si applica globalmente a tutte le espressioni regolari di un programma. Se dimenticate di importare -no_match_vars, il vostro programma pagherà la penalità anche se non leggete esplicitamente da tali variabili.
I programmi in Perl moderno dovrebbero usare la variabile @- al posto di questo terribile terzetto di variabili.
Variabili Sovraglobali Utili
Molti programmi in Perl 5 moderno possono cavarsela usando solo un paio delle variabili sovraglobali. Probabilmente, incontrerete solo alcune di queste variabili nei programmi reali.
Alternative alle Variabili Sovraglobali
Le maggiori responsabilità di azioni a distanza ricadono sull'IO e sulle condizioni eccezionali. Usare Try::Tiny (Avvertenze sulle Eccezioni(exception_caveats)) vi aiuta a isolarvi dalla complicata semantica della gestione corretta delle eccezioni. localizzare e copiare il valore di $! può aiutarvi a evitare comportamenti strani quando Perl effettua chiamate implicite di sistema. Usare IO::File e i suoi metodi sui filehandle lessicali (Variabili Speciali di Gestione dei File(file_handling_variables)) vi aiuta a prevenire modifiche globali indesiderate al comportamento dell'IO.
Cosa Evitare
Perl 5 non è perfetto. Alcune funzionalità sono difficili da usare correttamente. Altre non hanno mai funzionato bene. Alcune sono bizzarre combinazioni di altre funzionalità con strani casi speciali. Anche se la cosa migliore che potete fare è semplicemente di evitarle, conoscerne il motivo vi aiuterà a trovare soluzioni più efficaci.
Bareword
Perl è un linguaggio malleabile. Potete scrivere programmi nei modi più creativi, manutenibili, criptici e bizzarri che preferite. I buoni programmatori si preoccupano della manutenibilità, ma Perl non ha la presunzione di sapere che cos'è la manutenibilità secondo il vostro punto di vista.
Il parser Perl riconosce le istruzioni native e gli operatori. Usa i sigilli per identificare le variabili e altri simboli di punteggiatura per riconoscere chiamate a funzioni e metodi. Tuttavia, qualche volta il parser deve cercare di indovinare le vostre intenzioni, specialmente quando usate una bareword--un identificatore senza sigilli o altra punteggiatura sintatticamente significativa.
Usi Appropriati delle Bareword
Sebbene la direttiva strict (Direttive(pragmas)) sia rigida nel proibire l'uso di bareword ambigue, accetta invece alcuni usi delle bareword.
Bareword come chiavi di hash
Le chiavi di hash in Perl 5 sono generalmente non ambigue poiché il parser le può identificare come stringhe; in $giochi{flipper}, flipper è ovviamente una stringa.
Occasionalmente, questa interpretazione non è ciò che desiderate, specialmente quando intendete valutare una funzione nativa o definita da voi per produrre la chiave di hash. In tal caso, potete disambiguare fornendo degli argomenti, usando le parentesi degli argomenti di funzione oppure anteponendo un operatore più unario per forzare la valutazione della funzione nativa:
# la chiave e` il letterale 'shift'
my $valore = $elementi{shift};
# la chiave e` il valore prodotto da shift
my $valore = $elementi{shift @_}
# il + unario usa la funzione nativa shift
my $valore = $elementi{+shift};
Bareword come nomi di package
Anche i nomi di package in Perl 5 sono delle bareword. Se vi attenete alla convenzione per cui i nomi di package hanno iniziali maiuscole e le funzioni no, vi imbatterete raramente in collisioni di nomi, ma il parser di Perl 5 deve decidere come interpretare Package->metodo(). Significa "chiama una funzione di nome Package() e poi chiama metodo() sul suo valore di ritorno"? oppure significa "Chiama un metodo di nome metodo() nel namespace Package"? La risposta cambia a seconda del codice che il parser ha già incontrato nel namespace corrente.
Potete forzare il parser a trattare Package come un nome di package appendendo il separatore di package (::) (footnote: Anche tra le persone che capiscono perché questo funziona, pochissime lo usano.):
# probabilmente e` un metodo di classe
Package->metodo();
# sicuramente e` un metodo di classe
Package::->metodo();
Bareword come nomi di blocchi di codice
I blocchi di codice con i nomi speciali AUTOLOAD, BEGIN, CHECK, DESTROY, END, INIT e UNITCHECK sono bareword che dichiarano delle funzioni senza usare l'istruzione nativa sub. Li avete già incontrati in questo libro (Generazione di Codice(code_generation)):
package Scimmia::Maggiordomo;
BEGIN { inizializza_scimmie( __PACKAGE__ ) }
sub AUTOLOAD { ... }
Anche se potete omettere sub nelle dichiarazioni di AUTOLOAD(), poche persone lo fanno.
Bareword come costanti
Le costanti dichiarate con la direttiva constant si usano come bareword:
# non usatelo per una vera autenticazione
use constant NAME => 'Bob';
use constant PASSWORD => '|38pesce!lesso74|';
return unless $nome eq NAME && $pass eq PASSWORD;
Notate che queste costanti non vengono interpolate in una stringa con doppia quotatura.
Le costanti sono un caso speciale di funzioni prototipate (Prototipi(prototypes)). Quando predichiarate una funzione con un prototipo, il parser sa come deve trattare tale funzione e vi avverte in caso ci siano ambiguità di interpretazione. Tutti i problemi dei prototipi si applicano anche alle costanti.
Usi Inappropriati delle Bareword
Indipendentemente da quanta cura mettete nel programmare, le bareword producono codice ambiguo. Nella maggior parte dei casi potete evitare di usarle, ma incontrerete svariati tipi di bareword nel codice preesistente.
Bareword nelle chiamate di funzione
Il code scritto senza usare strict 'subs' può usare delle bareword come nomi di funzione. Aggiungere delle parentesi fa sì che il codice soddisfi le restrizioni. Usate perl -MO=Deparse,-p (vedete perldoc B::Deparse) per scoprire come Perl le interpreta, e quindi aggiungete le parentesi appropriate.
Bareword come valori di hash
Alcuni pezzi di codice poco recente non si preoccupavano di quotare i valori nelle coppie degli hash:
# pessimo stile; da evitare
my %genitori =
(
madre => Anna,
padre => Flavio,
);
Quando non esistono né la funzione Flavio() né la funzione Anna(), Perl interpreta queste bareword come stringhe. strict 'subs', in questo caso, genera invece un errore.
Bareword come filehandle
Prima dell'introduzione dei filehandle lessicali (Riferimenti a Filehandle(lexical_filehandles)), tutti gli handle di file e di directory erano delle bareword. Potete quasi sempre riscrivere questo codice senza problemi in modo che usi dei filehandle lessicali. Fanno eccezione STDIN, STDOUT e STDERR, ma fortunatamente il parser di Perl è in grado di riconoscerli.
Bareword come funzioni per sort
Infine, la funzione nativa sort può ricevere come secondo argomento il nome di una funzione da usare per l'ordinamento. Anche se raramente questo nome risulta ambiguo per il parser, esso può invece confondere le persone che leggono il codice. L'alternativa di fornire un riferimento a funzione in uno scalare è solo un poco migliore:
# stile che usa bareword
my @ordinato = sort confronta_lunghezze @nonordinato;
# riferimento a funzione in scalare
my $confronto = \&confronta_lunghezze;
my @ordinato = sort $confronto @nonordinato;
La seconda opzione evita l'uso di bareword, ma il codice risultante ha una linea in più. Purtropppo, il parser di Perl 5 non accetta la versione con una sola linea a causa delle speciali regole di parsing di sort; non potete usare espressioni arbitrarie (come ottenere un riferimento a una funzione con nome) dove il parser si attende un blocco oppure uno scalare.
# non funziona
my @ordinato = sort \&confronta_lunghezze @nonordinato;
In entrambi i casi, il modo in cui sort invoca la funzione e le passa gli argomenti può essere poco chiaro (vedete perldoc -f sort per i dettagli). Dove possibile, prendete invece in considerazione l'uso della forma a blocco di sort. Se dovete usare una delle forme con funzione, ricordate di aggiungere un commento esplicativo.
Oggetti Indiretti
Perl 5 non ha un operatore new; un costruttore è una qualunque cosa che restituisce un oggetto. Per convenzione, i costruttori sono metodi di classe con nome new(), ma potete scegliere qualunque alternativa vi piaccia. Molti vecchi tutorial sugli oggetti Perl 5 promuovevano l'uso di chiamate di costruttori in stile C++ e Java:
my $q = new CGI; # NON USARE
... invece della più esplicita chiamata di metodo:
my $q = CGI->new();
Queste due sintassi sono equivalenti in tutti i casi, tranne quelli in cui non lo sono.
Invocazioni Indirette con Bareword
Nella forma indiretta (e, più precisamente, nel caso dativo) del primo esempio, il verbo (il metodo) precede il nome a cui si riferisce (l'oggetto). Questo va bene nei linguaggi naturali, ma in Perl 5 introduce delle ambiguità di interpretazione.
Dato che il nome del metodo è una bareword (Bareword(barewords)), il parser deve indovinare qual è l'interpretazione corretta del codice attraverso l'uso di svariate euristiche. Anche se tali euristiche sono ampiamente testate e quasi sempre corrette, nei casi in cui falliscono possono creare confusione. Peggio ancora, esse dipendono dall'ordine di compilazione del codice e dei moduli.
Le difficoltà del parsing aumentano ulteriormente quando il costruttore riceve degli argomenti. In stile indiretto si potrebbe scrivere:
# NON USARE
my $ogg = new Class( arg => $valore );
... facendo assomigliare il nome della classe a una chiamata di funzione. Perl 5 può disambiguare molti di questi casi, ma le sue euristiche dipendono dai nomi di package che il parser ha già incontrato, dalle bareword che ha già interpretato (e da come le ha interpretate) e dai nomi di funzione che sono già stati dichiarati nel package corrente.
Immaginate di imbattervi in una funzione prototipata (Prototipi(prototypes)) il cui nome va in qualche modo in conflitto con il nome di una classe o di un metodo chiamato indirettamente. Anche se è una situazione rara, farne il debug è così problematico che vale la pena evitare le invocazioni indirette.
Limitazioni Scalari della Notazione Indiretta
Un altro problema legato alla sintassi indiretta è che il parser si aspetta una singola espressione scalare come oggetto. Stampare su un filehandle memorizzato in una variabile aggregata sembra facile, ma non lo è:
# NON FUNZIONA COME CI SI ASPETTA
say $config->{output} 'Messaggio diagnostico divertente!';
Perl tenterà di chiamare say sull'oggetto $config.
print, close e say--che sono tutte funzioni native che operano sui filehandle--operano in modo indiretto. Questo comportamento era appropriato quando i filehandle erano globali ai package, ma i filehandle lessicali (Riferimenti a Filehandle(lexical_filehandles)) evidenziano i problemi della sintassi a oggetti indiretta. Per risolvere tali problemi, disambiguate la sottoespressione che produce l'invocante desiderato:
say {$config->{output}} 'Messaggio diagnostico divertente!';
Alternative alla Notazione Indiretta
La notazione diretta di invocazione non ha questi problemi di ambiguità. Per costruire un oggetto, chiamate il metodo costruttore direttamente sul nome della classe:
my $q = CGI->new();
my $ogg = Class->new( arg => $valore );
Questa sintassi contiene ancora un problema di bareword dato che, se esiste una funzione di nome CGI, Perl interpreta la bareword del nome di classe come una chiamata alla funzione, ovvero:
sub CGI;
# avete scritto CGI->new(), ma Perl lo ha letto
my $q = CGI()->new();
Anche se questo problema si presenta raramente, potete disambiguare i nomi di classe appendendo il separatore di package (::) oppure contrassegnando esplicitamente i nomi di classe come letterali stringa:
# separatore di package
my $q = CGI::->new();
# letterale stringa non ambiguo
my $q = 'CGI'->new();
Tuttavia, nella pratica nessuno lo fa.
Per il caso specifico delle operazioni su filehandle, l'uso del dativo è talmente prevalente che potete usare l'approccio di invocazione indiretta se racchiudete l'invocante desiderato tra parentesi graffe. Se usate Perl 5.14 (o se caricate IO::File oppure IO::Handle), potete usare dei metodi sui filehandle lessicali (footnote: Anche se quasi nessuno li usa con print e say.).
Il modulo CPAN Perl::Critic::Policy::Dynamic::NoIndirect (che è un plugin di Perl::Critic) può identificare le invocazioni indirette durante le revisioni del codice. Il modulo CPAN indirect può identificare e proibire il loro uso nei programmi in esecuzione:
# genera un warning in caso di invocazione indiretta
no indirect;
# solleva un'eccezione in caso di invocazione indiretta
no indirect ':fatal';
Prototipi
Un prototipo è un insieme opzionale di metadati associato ad una funzione che cambia il modo in cui il parser interpreta i suoi argomenti. Anche se, a un'analisi superficiale, i prototipi ricordano le segnature di funzione di altri linguaggi, in realtà sono molto diversi.
I prototipi permettono agli utenti di definire delle funzioni che si comportano come quelle native. Considerate la funzione nativa push, che riceve un array e una lista. Mentre normalmente Perl 5 appiattirebbe l'array e la lista in un'unica lista da passare a push, il parser sa che non deve appiattire l'array, in modo che push possa modificarlo sul posto.
I prototipi di funzione sono parte delle dichiarazioni:
sub qui (&@);
sub quo ($$) { ... }
my $qua = sub (&&) { ... };
Un prototipo associato a una dichiarazione anticipata deve corrispondere al prototipo associato alla dichiarazione della funzione; in caso contrario, Perl genera un warning. Curiosamente, potete omettere il prototipo nella dichiarazione anticipata e includerlo nella dichiarazione vera e propria--ma non ci sono ragioni per farlo.
La funzione nativa prototype riceve il nome di una funzione e restituisce una stringa che rappresenta il suo prototipo. Per vedere il prototipo di una funzione nativa, usate la forma CORE:::
$ perl -E "say prototype 'CORE::push';"
\@@
$ perl -E "say prototype 'CORE::keys';"
\%
$ perl -E "say prototype 'CORE::open';"
*;$@
prototype restituisce undef per le funzioni native che non potete emulare:
say prototype 'CORE::system' // 'undef'
# undef; non potete emulare la funzione nativa system
say prototype 'CORE::prototype' // 'undef'
# undef; la funzione nativa prototype non ha un prototipo
Vi ricordate di push?
$ perl -E "say prototype 'CORE::push';"
\@@
Il carattere @ rappresenta una lista. Il backslash impone l'uso di un riferimento all'argomento corrispondente. Quindi, questo prototipo significa che push riceve un riferimento ad array e una lista di valori. Potreste scrivere la seguente funzione miapush:
sub miapush (\@@)
{
my ($array, @resto) = @_;
push @$array, @resto;
}
Altri caratteri usati nei prototipi sono $, per forzare un argomento scalare, % per indicare un hash (usato spesso come riferimento) e & per indicare un blocco di codice. Vedete perldoc perlsub per la documentazione completa.
Il Problema dei Prototipi
I prototipi modificano il modo in cui Perl fa il parsing del vostro codice e possono causare delle conversioni di tipo. Non servono né a documentare il numero o il tipo degli argomenti delle funzioni, né a mappare gli argomenti su parametri per nome.
Le coercizioni di tipo causate dai prototipi hanno comportamenti insidiosi, come quello di forzare il contesto scalare sugli argomenti:
sub uguaglianza_numerica($$)
{
my ($sinistro, $destro) = @_;
return $sinistro == $destro;
}
my @num = 1 .. 10;
say 'Sono uguali, qualunque cosa questo significhi!'
if uguaglianza_numerica @num, 10;
... ma funzionano solo con espressioni semplici:
sub miapush(\@@);
# errore di compilazione: prototipo non corrispondente
# (assegnamento scalare dove ci si aspettava un array)
miapush( my $elems = [], 1 .. 20 );
Per fare il debug di questo errore, gli utenti di miapush devono sapere che esiste un prototipo e conoscere le limitazioni dei prototipi di array. Peggio ancora, errori come questo sono tra i più semplici tra quelli causati dai prototipi.
Usi Appropriati dei Prototipi
Ci sono pochi casi in cui i vantaggi nell'uso dei prototipi superano i problemi, ma tali casi esistono.
Par prima cosa, i prototipi vi permettono di sovrascrivere le funzioni native. Dovete anzitutto verificare se è possibile sovrascrivere una funzione nativa esaminandone il prototipo con un semplice programma di test. Quindi, usate la direttiva subs per comunicare a Perl che intendete sovrascrivere una funzione nativa, e infine dichiarate la vostra sovrascrittura con il prototipo appropriato:
use subs 'push';
sub push (\@@) { ... }
Ricordate che la direttiva subs è attiva per tutto il resto del file, indipendentemente dagli scope lessicali.
La seconda ragione valida per usare i prototipi è per la definizione di costanti durante la compilazione. Quando Perl si imbatte in una funzione dichiarata con un prototipo vuoto (non senza prototipo) e tale funzione valuta a una singola espressione costante, l'ottimizzatore sostituisce tutte le chiamate alla funzione con costanti:
sub PI_GRECO () { 4 * atan2(1, 1) }
Tutto il codice seguente usa il valore calcolato di pi greco al posto della bareword PI_GRECO o di una chiamata a PI_GRECO(), in base allo scope e alla visibilità.
La direttiva nativa constant gestisce questi dettagli al vostro posto. Il modulo CPAN Const::Fast crea delle costanti scalari che potete interpolare all'interno di stringhe.
Un uso ragionevole dei prototipi è quello di estendere la sintassi di Perl per operare su funzioni anonime in forma di blocchi. Il modulo CPAN Test::Exception sfrutta efficacemente questa possibilità per fornire una interessante API con computazione ritardata (footnote: Vedete anche Test::Fatal). La sua funzione throws_ok() riceve tre argomenti: un blocco di codice da eseguire, un'espressione regolare di cui fare il match con la stringa dell'eccezione e una descrizione opzionale del test:
use Test::More tests => 1;
use Test::Exception;
throws_ok
{ my $nonoggetto; $nonoggetto->regala() }
qr/Can't call method "regala" on an undefined/,
'Un metodo chiamato su un invocante non definito deve fallire';
La funzione esportata throws_ok() ha il prototipo &$;$. Il suo primo argomento è un blocco che diventa una funzione anonima. Il secondo argomento è uno scalare. Il terzo argomento è opzionale.
I lettori attenti avranno forse notato che manca la virgola dopo il blocco. Si tratta di una idiosincrasia del parser Perl 5, che si aspetta un carattere di spaziatura dopo un blocco con prototipo invece dell'operatore virgola. È un altro dei problemi della sintassi dei prototipi.
Potete usare throws_ok() senza usare i prototipi:
use Test::More tests => 1;
use Test::Exception;
throws_ok(
sub { my $nonoggetto; $nonoggetto->regala() },
qr/Can't call method "regala" on an undefined/,
'Un metodo chiamato su un invocante non definito deve fallire' );
Un ultimo caso di uso appropriato dei prototipi è quando definite una vostra funzione con nome da usare con sort (footnote: Questo esempio è stato suggerito da Ben Tilly.):
sub ordina_per_lunghezza ($$)
{
my ($sinistro, $destro) = @_;
return length($sinistro) <=> length($destro);
}
my @ordinato = sort ordina_per_lunghezza @nonordinato;
Il prototipo $$ forza Perl a passare le coppie considerate da sort nell'array @_. La documentazione di sort segnala che questo metodo è un po' meno veloce di usare le variabili globali di package $a e $b, ma spesso l'uso di variabili lessicali compensa ampiamente la riduzione di velocità.
Equivalenza Metodi-Funzioni
Il sistema ad oggetti di Perl 5 è deliberatamente minimale (Riferimenti Blessed(blessed_references)). Dato che una classe è un package, Perl non distinge tra una funzione e un metodo memorizzato in un package. La stessa istruzione sub viene usata per dichiarare entrambi. La documentazione può esplicitare le vostre intenzioni, ma Perl non avrà problemi a fare il dispatch ad una funzione invocata come metodo. Allo stesso modo, potete invocare un metodo come se fosse una funzione--qualificata, esportata o in forma di riferimento--se gli passate l'invocante manualmente.
Invocare una cosa sbagliata in modo sbagliato è causa di problemi.
Prospettiva del Chiamante
Considerate una classe con svariati metodi:
package Ordine;
use List::Util 'sum';
...
sub calcola_prezzo
{
my $self = shift;
return sum( 0, $self->articoli_ordinati() );
}
Dato un oggetto Ordine $o, le seguenti chiamate di questo metodo possono sembrare equivalenti:
my $prezzo = $o->calcola_prezzo();
# sbagliato; non usatelo
my $prezzo = Ordine::calcola_prezzo( $o );
Sebbene in questo semplice caso esse producano lo stesso output, la seconda viola l'incapsulamento dell'oggetto aggirando la ricerca del metodo.
Se $o fosse invece una sottoclasse o un allomorfismo (Ruoli(roles)) di Ordine che sovrascrive calcola_prezzo(), aggirare il dispatch del metodo chiamerebbe il metodo sbagliato. Qualunque modifica dell'implementazione di calcola_prezzo(), ad esempio relativa all'ereditarietà o alla delega con AUTOLOAD()--potrebbe far fallire il codice chiamante.
Esiste una situazione in Perl in cui questo comportamento può sembrare necessario. Se forzate la risoluzione senza dispatch di un metodo, come invocate il riferimento a metodo risultante?
my $rif_a_met = $o->can( 'applica_sconto' );
In realtà, avete due possibilità. La prima consiste nello scartare il valore di ritorno del metodo can():
$o->applica_sconto() if $o->can( 'applica_sconto' );
La seconda consiste nell'usare il riferimento stesso insieme alla sintassi di invocazione di metodo:
if (my $rif_a_met = $o->can( 'applica_sconto' ))
{
$o->$rif_a_met();
}
Quando $rif_a_met contiene un riferimento a funzione, Perl fa un'invocazione su tale riferimento passando $o come invocante. Questa sintassi funziona anche quando sono attive delle restrizioni, in modo analogo all'invocazione di un metodo attraverso uno scalare che ne contiene il nome:
my $nome = 'applica_sconto';
$o->$nome();
C'è un piccolo svantaggio nell'invocare un metodo per riferimento; se la struttura del programma cambia tra il momento in cui il riferimento viene memorizzato e il momento in cui viene usato per l'invocazione, tale riferimento potrebbe non corrispondere più al metodo appropriato. Se la classe Ordine fosse cambiata in modo tale che Ordine::applica_sconto non fosse più il metodo giusto da chiamare, il riferimento in $rif_a_met non sarebbe stato aggiornato.
Quando usate questa forma di invocazione, limitate lo scope dei riferimenti.
Prospettiva del Chiamato
Dato che Perl 5 non distingue tra funzioni e metodi nelle dichiarazioni e dato che è possibile (per quanto sconsigliabile) invocare una data funzione sia come funzione che come metodo, si può scrivere una funzione pronta ad essere chiamata in entrambi i modi. Il modulo CGI distribuito con Perl da il cattivo esempio. Le sue funzioni applicano svariate euristiche per determinare se il loro primo argumento è un invocante.
Ciò ha molti svantaggi. Prima di tutto, è difficile predire esattamente quali sono gli invocanti potenzialmente validi per un dato metodo, specialmente quando è possibile avere a che fare con delle sottoclassi. Anche creare una API che non possa essere usata scorrettamente dagli utenti è più difficile, così come scrivere la documentazione. Che cosa succede quando una parte del progetto usa l'interfaccia procedurale e un'altra parte usa l'interfaccia a oggetti?
Se dovete fornire sia un'interfaccia procedurale che una orientata agli oggetti ad una libreria, create due API separate.
Tie
Mentre l'overloading (Overloading(overloading)) vi permette di personalizzare il comportamento di classi e oggetti per tipi specifici di coercizione, un meccanismo detto tying (legatura, NdT) ci permette di personalizzare il comportamento dei tipi nativi (scalari, array, hash e filehandle). Qualunque operazione effettuata su una variabile legata si traduce in una specifica chiamata di metodo.
La funzione nativa tie (lega, NdT) permetteva originariamente di usare spazio su disco come memoria di supporto degli hash, in modo che Perl potesse accedere a file più grandi di quanto potesse essere ospitato in memoria. Il modulo Tie::File distribuito con Perl fornisce un sistema simile e vi permette di trattare i file come se fossero array.
La classe a cui fate il tie di una variabile deve rispettare una interfaccia predefinita per ogni tipo di dato. Vedete perldoc perltie per una panoramica, prima di consultare i moduli Tie::StdScalar, Tie::StdArray e Tie::StdHash della distribuzione Perl per i dettagli più specifici. Iniziate ereditando da una di tali classi, e sovrascrivete i metodi che dovete modificare.
Come se tie non creasse già abbastanza confusione, Tie::Scalar, Tie::Array e Tie::Hash definiscono le interfacce necessarie per legare scalari, array e hash, mentre Tie::StdScalar, Tie::StdArray e Tie::StdHash forniscono le implementazioni di default.
Legare Variabili
per legare una variabile:
use Tie::File;
tie my @file, 'Tie::File', @arg;
Il primo argomento è la variabile da legare, il secondo è il nome della classe a cui legarla e @args è una lista opzionale di argomenti richiesti dalla funzione di legatura. Nel caso di Tie::File, il parametro richiesto è un nome valido di file.
Le funzioni di legatora sono simili a costruttori: TIESCALAR, TIEARRAY(), TIEHASH() e TIEHANDLE() rispettivamente per gli scalari, gli array, gli hash e i filehandle. Ciascuna funzione restituisce un nuovo oggetto che rappresenta la variabile legata. Entrambe le funzioni native tie e tied restituiscono tale oggetto. La maggior parte delle persone, tuttavia, usa tied in contesto booleano.
Implementazione di Variabili Legate
Per implementare la classe di una variabile legata, ereditate anzitutto da uno dei moduli distribuiti con Perl come Tie::StdScalar (footnote: Tie::StdScalar non ha un proprio file .pm associato, perciò dovete usare Tie::Scalar per renderlo disponibile.), quindi sovrascrivete i metodi specifici relativi alle operazioni che volete modificare. Nel caso di uno scalare legato, è probabile che essi includano FETCH e STORE, forse anche TIESCALAR(), ma probabilmente non DESTROY().
Con poco codice, potete creare una classe che tiene traccia di tutte le letture e le scritture su uno scalare:
package Tie::Scalar::Registra
{
use Modern::Perl;
use Tie::Scalar;
use parent -norequire => 'Tie::StdScalar';
sub STORE
{
my ($self, $valore) = @_;
Registro->scrivi("Memorizzo <$valore> (prima era [$$self])", 1);
$$self = $valore;
}
sub FETCH
{
my $self = shift;
Registro->scrivi("Leggo <$$self>", 1);
return $$self;
}
}
1;
Assumete che il metodo scrivi() della classe Registro riceva una stringa e il numero di contesti di chiamata nello stack di cui riportare il numero di linea.
I metodi STORE() e FETCH() trattano $self come uno scalare blessed. Assegnare un valore a tale riferimento a scalare cambia il valore dello scalare e leggere da esso restituisce il valore dello scalare.
Analogamente, i metodi di Tie::StdArray e Tie::StdHash operano rispettivamente su dei riferimenti blessed ad array e hash. La documentazione contenuta in perldoc perltie spiega i numerosi metodi supportati da queste classi, ad esempio per leggere e scrivere valori multipli.
Grazie all'opzione -norequire la direttiva parent evita di tentare di caricare un file per Tie::StdScalar, dato che tale modulo è contenuto in una parte del file Tie/Scalar.pm.
Quando usare le Variabili Legate
Le variabili legate possono sembrare delle divertenti occasioni per mostrare la propria brillantezza, ma spesso producono interfacce confuse. A meno che abbiate delle ragioni molto valide per volere che certi oggetti si comportino come se fossero tipi di dati nativi, evitate di creare le vostre legature. Usare tie è anche molto più lento che usare i tipi nativi a causa di vari aspetti implementativi.
Tra le ragioni valide ci sono quella di facilitare il debugging (ad esempio usare lo scalare tracciato per aiutarvi a capire dove cambia il suo valore) e quella di rendere possibili certe operazioni altrimenti impossibili (ad esempio accedere a file di grandi dimensioni con un uso efficiente della memoria). Le variabili legate sono invece difficilmente utili come interfacce primarie agli oggetti; spesso è troppo difficile e vincolante tentare di uniformare l'intera interfaccia a quella supportata da tie().
Un ultimo avvertimento poco piacevole ma convinente; troppo codice devia dai suoi obiettivi al fine di prevenire l'uso di variabili legate, spesso accidentalmente. Questo è sicuramente un problema, ma la violazione delle aspettative del codice di una libreria rivela normalmente dei bug che non è in vostro potere risolvere.
Cosa Manca
Perl5 non è perfetto, ma è malleabile--in parte perché non esiste un'unica configurazione ideale per ogni programmatore e ogni obiettivo. Alcuni comportamenti utili sono disponibili nelle librerie distribuite con Perl. Altri sono disponibili su CPAN. La vostra efficacia come programmatore Perl dipende da quanto sapete avvantaggiarvi di queste aggiunte.
Default Mancanti
La progettazione di Perl 5 ha tentato di anticipare le nuove direzioni del linguaggio, ma era altrettanto impossibile predire il futuro nel 1994 come lo è nel 2011. Perl 5 ha espanso il linguaggio, ma è rimasto compatibile con Perl 1 del 1987.
Il miglior codice Perl 5 scritto nel 2011 è molto diverso dal miglior codice Perl 5 scritto nel 1994 e dal miglior codice Perl 1 scritto nel 1987.
Sebbene Perl 5 sia distribuito con una ricca libreria, essa non è onnicomprensiva. Molti dei migliori moduli Perl 5 sono distribuiti su CPAN (CPAN(cpan)) anziché insieme a Perl. La meta-distribuzione Task::Kensho include svariate altre distribuzioni che rappresentano il meglio che CPAN abbia da offrire. Dateci un'occhiata come prima cosa ogni volta che dovete risolvere un problema.
Ciò detto, alcune direttive e moduli distribuiti con Perl sono indispensabili ad ogni programmatore Perl serio.
La direttiva strict
La direttiva strict (Direttive(pragmas)) vi permette di disattivare (e ri-attivare) vari costrutti potenti del linguaggio che si prestano ad abusi involontari.
strict proibisce i riferimenti simbolici, richiede che le variabili siano dichiarate (Scope Lessicale(lexical_scope)) e proibisce l'uso di bareword non dichiarate (Bareword(barewords)). Anche se l'uso occasionale dei riferimenti simbolici può essere necessario per manipolare le tabelle dei simboli (Uso e Importazione(import)), l'uso di una variabile come nome di variabile offre il fianco a errori insidiosi di azione a distanza--o, peggio ancora, alla possibilità di manipolazione di strutture dati interne per fini illeciti da parte di input utente scarsamente validato.
Richiedere che le variabili siano dichiarate aiuta a rilevare errori di battitura nei nomi delle variabili e incoraggia la definizione di variabili lessicali negli scope appropriati. È molto più facile capire qual è lo scope che si intende associare a una variabile lessicale se tutte le variabili hanno dichiarazioni my e our nello scope appropriato.
strict ha un effetto lessicale basato sullo scope durante la compilazione del suo uso (Uso e Importazione(import)) e della sua disattivazione (con no). Vedete perldoc strict per maggiori dettagli.
La Direttiva warnings
La direttiva warnings (Gestione dei Warning(handling_warnings)) controlla la generazione delle varie classi di warning in Perl 5, come il tentativo di convertire in stringa il valore undef o l'uso del tipo sbagliato di operatore su certi valori. Genera anche warning relativi all'uso di funzionalità deprecate.
I warning più utili sono quelli che spiegano che Perl ha incontrato difficoltà a interpretare le vostre intenzioni e ha dovuto tentare di indovinare l'interpretazione giusta. Sebbene spesso Perl ci azzecchi, una disambiguazione da parte vostra garantisce che i vostri programmi funzionino correttamente.
La direttiva warnings ha un effetto lessicale sullo scope durante la compilazione del suo uso o disattivazione (con no). Vedete perldoc perllexwarn e perldoc warnings per maggiori dettagli.
Combinate use warnings con use diagnostics per ricevere messaggi diagnostici più estesi per ogni warning presente nei vostri programmi. Questa diagnostica estesa è documentata in perldoc perldiag. Questo comportamento è particolarmente utile quando imparate Perl. Disattivatelo prima di distribuire il vostro programma, perché produce un output verboso che potrebbe riempire i vostri file di log ed esporre troppe informazioni agli utenti.
IO::File e IO::Handle
Prima di Perl 5.14 i filehandle lessicali erano oggetti della classe IO::Handle, ma dovevate caricare esplicitamente IO::Handle prima di poter chiamare dei metodi su di essi. A partire da Perl 5.14, i filehandle lessicali sono istanze del modulo IO::File che Perl carica per voi.
Aggiungete IO::Handle al codice da eseguire in Perl 5.12 o versioni precedenti se volete chiamare dei metodi sui filehandle lessicali.
La direttiva autodie
Perl 5 lascia alla vostra responsabilità gestire (o ignorare) gli errori. Per esempio, se non fate attenzione a controllare il valore di ritorno di ogni chiamata a open() potreste tentare di leggere da un filehandle chiuso--o, peggio ancora, perdere dati che tentate di scriverci. La direttiva autodie modifica questo comportamento di default. Se scrivete:
use autodie;
open my $fh, '>', $file;
... un fallimento della chiamata a open() solleverà un'eccezione. Dato che la reazione più appropriata al fallimento di una chiamata di sistema è quella di sollevare un'eccezione, questa direttiva può rimuovere molto codice ripetitivo dal vostro programma e garantirvi la tranquillità di sapere che non non avete dimenticato di controllare un valore di ritorno.
Questa direttiva viene distribuita con Perl 5 a partire dalla versione 5.10.1. Vedete perldoc autodie per maggiori informazioni.

